
Preface

This guide is complemental to the Natural reference documentation in that it provides basic
information and some longer, in-depth articles on various aspects of programming with Natural.
You should be familiar with this information before you start to write Natural applications.

Describes the differences between the two Natural programming modes:
Reporting Mode and Structured Mode.

Generally, it is recommended to use structured mode exclusively, because it
provides for more clearly structured applications. Therefore, all explanations

Natural Programming
Modes

and examples in this documentation refer to structuredmode. Any peculiarities
of reporting mode will not be taken into consideration.

Within an application, you can use several types of programming objects to
achieve an efficient application structure. This document discusses the various

Object Types

types of Natural programming objects, such as data areas, programs,
subprograms, subroutines, helproutines, maps.

Describes how you define the fields you wish to use in a program.Defining Fields

Explains the benefits of using the Natural programming object “function”,
shows the difference between using function calls and subprogram calls and
describes the methods available for defining and calling a function.

User-Defined Functions

Describes various aspects of usingNatural to access data in anAdabas database.Accessing Data in a
Database

Discusses various aspects of how you can control the format of an output report
created with Natural, that is, the way in which the data are displayed.

Controlling Data Output

Discusses various other aspects of programming with Natural.Further Programming
Aspects

As of Natural 5, generated programs are portable across the platforms UNIX,
OpenVMS and Windows.

Portable Natural
Generated Programs

Provides information on components of Natural which you can use to design
character-based user interfaces for your applications.

Designing Application
User Interfaces

Contains a list of all keywords and words that are reserved in the Natural
programming language.

Natural Reserved
Keywords

The preceding sections of the Programming Guide contain several examples of
Natural programs. In addition, links are provided there to further example

Referenced Example
Programs

programs (mainly for reporting mode) which are contained in this separate
section.

Note:

1. All example programs shown in the Programming Guide are also provided
in source-code form in the Natural library SYSEXPG. The example programs
use data from the files EMPLOYEES and VEHICLES, which are supplied by

xiii



Software AG for demonstration purposes. TheNatural library SYSEXPG also
includes example programs for Natural Functions.

2. Further example programs of using Natural statements are provided in the
Natural library SYSEXSYN and are documented in the section Referenced
Example Programs in the Statements documentation.

3. Please ask your Natural administrator about the availability of the libraries
SYSEXPG and SYSEXSYN at your site.

4. To use any Natural example program to access an Adabas database, the
Adabas nucleus parameter OPTIONSmust be set to TRUNCATION.

Notation vrs or vr

When used in this documentation, the notation vrs or vr represents the relevant product version
(see also Version in the Glossary).

Document Conventions

DescriptionConvention

Identifies elements on a screen.Bold

Identifies service names and locations in the format folder.subfolder.service,
APIs, Java classes, methods, properties.

Monospace font

Identifies:Italic

Variables for which you must supply values specific to your own situation or
environment.
New terms the first time they occur in the text.
References to other documentation sources.

Identifies:Monospace font

Text you must type in.
Messages displayed by the system.
Program code.

Indicates a set of choices from which you must choose one. Type only the information
inside the curly braces. Do not type the { } symbols.

{ }

Separates two mutually exclusive choices in a syntax line. Type one of these choices.
Do not type the | symbol.

|

Indicates one or more options. Type only the information inside the square brackets.
Do not type the [ ] symbols.

[ ]

Indicates that you can type multiple options of the same type. Type only the
information. Do not type the ellipsis (...).

...

Online Information and Support

Software AG Documentation Website

You can find documentation on the Software AG Documentation website at https://documenta-
tion.softwareag.com.

Software AG Empower Product Support Website

If you do not yet have an account for Empower, send an email to empower@softwareag.comwith
your name, company, and company email address and request an account.

Once you have an account, you can open Support Incidents online via the eService section of
Empower at https://empower.softwareag.com/.

About this Documentation

https://documentation.softwareag.com
https://documentation.softwareag.com
https://empower.softwareag.com/


You can find product information on the Software AG Empower Product Support website at ht-
tps://empower.softwareag.com.

To submit feature/enhancement requests, get information about product availability, and download
products, go to Products.

To get information about fixes and to read early warnings, technical papers, and knowledge base
articles, go to the Knowledge Center.

If you have any questions, you can find a local or toll-free number for your country in our Global
Support Contact Directory at https://empower.softwareag.com/public_directory.aspx and give
us a call.

Software AG Tech Community

You can find documentation and other technical information on the SoftwareAGTechCommunity
website at https://techcommunity.softwareag.com. You can:

■ Access product documentation, if you have Tech Community credentials. If you do not, you
will need to register and specify "Documentation" as an area of interest.

■ Access articles, code samples, demos, and tutorials.
■ Use the online discussion forums, moderated by Software AG professionals, to ask questions,
discuss best practices, and learn how other customers are using Software AG technology.

■ Link to external websites that discuss open standards and web technology.

Data Protection

SoftwareAGproducts provide functionalitywith respect to processing of personal data according
to the EU General Data Protection Regulation (GDPR). Where applicable, appropriate steps are
documented in the respective administration documentation.

Purpose of Programming Modes

Natural offers two ways of programming:

■ Reporting Mode
■ Structured Mode

Note: Generally, it is recommended to use structuredmode exclusively, because it provides
for more clearly structured applications.

Reporting Mode

Reporting mode is only useful for the creation of adhoc reports and small programs which do not
involve complex data and/or programming constructs. (If youdecide towrite a program in reporting
mode, be aware that small programs may easily become larger and more complex.)

Please note that certain Natural statements are available only in reporting mode, whereas others
have a specific structure when used in reporting mode. For an overview of the statements that
can be used in reporting mode, see Reporting Mode Statements in the Statements documentation.

Structured Mode

Structured mode is intended for the implementation of complex applications with a clear and
well-defined program structure. The major benefits of structured mode are:

■ The programs have to be written in a more structured way and are therefore easier to read and
consequently easier to maintain.

■ As all fields to be used in a program have to be defined in one central location (instead of being
scattered all over the program, as is possible in reportingmode), overall control of the data used
is much easier.

With structured mode, you also have to make more detail planning before the actual programs
can be coded, thereby avoiding many programming errors and inefficiencies.

For an overview of the statements that can be used in structured mode, see Statements Grouped by
Functions in the Statements documentation.

Programming Guide8

Setting/Changing the Programming Mode

The default programmingmode is set by the Natural administrator with the profile parameter SM.

For further information on the Natural profile and session parameter SM, see SM - Programming in
Structured Mode in the Parameter Reference.

For information on how to change the programming mode, see SM - Programming in Structured
Mode in the Parameter Reference.

Functional Differences

The following major functional differences exist between reporting mode and structured mode:

■ Syntax Related to Closing Loops and Functional Blocks
■ Closing a Processing Loop in Reporting Mode
■ Closing a Processing Loop in Structured Mode
■ Location of Data Elements in a Program
■ Database Reference

Note: For detailed information on functional differences that exist between the twomodes,
see the Statementsdocumentation. It provides separate syntax diagrams and syntax element
descriptions for eachmode-sensitive statement. For a functional overview of the statements
that can be used in reporting mode, see Reporting Mode Statements in the Statements docu-
mentation.

Syntax Related to Closing Loops and Functional Blocks

(CLOSE) LOOP and DO ... DOEND statements are used for this purpose.

END-... statements (except END-DEFINE, END-DECIDE and END-SUBROUTINE) cannot be
used.

Reporting Mode:

Every loop or logical construct must be explicitly closed with a corresponding END-...
statement. Thus, it becomes immediately clear, which loop/logical constructs ends where.

LOOP and DO/DOEND statements cannot be used.

Structured Mode:

The two examples below illustrate the differences between the twomodes in constructingprocessing
loops and logical conditions.

Reporting Mode Example:

The reporting mode example uses the statements DO and DOEND to mark the beginning and end of
the statement block that is based on the AT END OF DATA condition. The END statement closes all
active processing loops.

READ EMPLOYEES BY PERSONNEL-ID
DISPLAY NAME BIRTH
AT END OF DATA

DO
SKIP 2
WRITE / 'LAST SELECTED:' OLD(NAME)

DOEND
END

Structured Mode Example:

The structuredmode example uses an END-ENDDATA statement to close the AT END OF DATA condi-
tion, and an END-READ statement to close the READ loop. The result is a more clearly structured
program in which you can see immediately where each construct begins and ends:

DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH

END-DEFINE
READ MYVIEW BY PERSONNEL-ID

DISPLAY NAME BIRTH
AT END OF DATA

SKIP 2
WRITE / 'LAST SELECTED:' OLD(NAME)

END-ENDDATA
END-READ
END

Closing a Processing Loop in Reporting Mode

The statements END, LOOP (or CLOSE LOOP) or SORTmay be used to close a processing loop.

The LOOP statement can be used to close more than one loop, and the END statement can be used
to close all active loops. These possibilities of closing several loopswith a single statement constitute
a basic difference to structured mode.

A SORT statement closes all processing loops and initiates another processing loop.

Example 1 - LOOP:

FIND ...
FIND ...
...
...
LOOP /* closes inner FIND loop

LOOP /* closes outer FIND loop
...
...

Example 2 - END:

FIND ...
FIND ...
...
...

END /* closes all loops and ends processing

Example 3 - SORT:

FIND ...
FIND ...
...
...

SORT ... /* closes all loops, initiates loop
...
END /* closes SORT loop and ends processing

Closing a Processing Loop in Structured Mode

Structured mode uses a specific loop-closing statement for each processing loop. Also, the END
statement does not close any processing loop. The SORT statementmust be preceded by an END-ALL
statement, and the SORT loop must be closed with an END-SORT statement.

Example 1 - FIND:

FIND ...
FIND ...
...
...
END-FIND /* closes inner FIND loop

END-FIND /* closes outer FIND loop
...

Example 2 - READ:

READ ...
AT END OF DATA
...
END-ENDDATA
...

END-READ /* closes READ loop
...
...
END

Example 3 - SORT:

READ ...
FIND ...
...
...

END-ALL /* closes all loops
SORT /* opens loop
...
...
END-SORT /* closes SORT loop
END

Location of Data Elements in a Program

In reporting mode, you can use database fields without having to define them in a DEFINE DATA
statement; also, you can define user-defined variables anywhere in a program, which means that
they can be scattered all over the program.

In structured mode, all data elements to be used have to be defined in one central location (either
in the DEFINE DATA statement at the beginning of the program, or in a data area outside the pro-
gram).

Database Reference

Reporting Mode:

In reportingmode, database fields and data definitionmodules (DDMs)may be referencedwithout
having been defined in a data area.

Structured Mode:

In structured mode, each database field to be used must be specified in a DEFINE DATA statement
as described in Defining Fields and Accessing Data in an Adabas Database.

II Object Types

This part describes the various types of Natural programming objects that can be used to achieve
an efficient application structure. AllNatural objects are stored inNatural libraries. Natural libraries
are contained in Natural system files.

Using Natural Programming Objects

Data Areas

Programs, Functions, Subprograms and Subroutines

Processing a Rich GUI Page - Adapter

Maps

Helproutines

Multiple Use of Source Code - Copycode

Documenting Natural Objects - Text

Creating Component Based Applications - Class

Using Non-Natural Files - Resource

15



16



3 Using Natural Programming Objects

■ Types of Programming Objects .......................................................................................................... 18
■ Creating and Maintaining Programming Objects .................................................................................... 18

17



Types of Programming Objects

Within a Natural application, you can use the following types of programming objects:

■ Program
■ Class
■ Subprogram
■ Function
■ Adapter
■ Subroutine
■ Copycode
■ Helproutine
■ Text
■ Map
■ Local Data Area
■ Global Data Area
■ Parameter Data Area
■ Resource

Creating and Maintaining Programming Objects

To create and maintain the programming objects, you use the NaturalONE editors.

■ Programs, subprograms, subroutines, functions, copycodes, helproutines, texts, global data
areas, local data areas and parameter data areas are created and maintained with the source
editor.

■ Character-based user interface maps are created and maintained with the map editor.

Complex graphical user interfaces are created and maintained with the Ajax Developer.
■ Data definition modules (DDM) are created and maintained with the DDM editor.

Use of Data Areas

As explained in Defining Fields, all fields that are to be used in a program have to be defined in
a DEFINE DATA statement.

The fields can be defined within the DEFINE DATA statement itself; or they can be defined outside
the program in a separate data area, with the DEFINE DATA statement referencing that data area.

A separate data area is a Natural object that can be used by multiple Natural programs, subpro-
grams, subroutines, helproutines or classes. A data area contains data element definitions, such
as user-defined variables, constants and database fields from a data definition module (DDM).

All data areas are created and edited with the source editor.

Natural supports three types of data area:

■ Local Data Area
■ Global Data Area
■ Parameter Data Area

Local Data Area

Variables defined as local are used only within a single Natural programming object. There are
two options for defining local data:

■ Define local data within a program.
■ Define local data outside a program in a separate Natural programming object, a local data area
(LDA).

Such a local data area is initialized when a program, subprogram or external subroutine that
uses this local data area starts to execute.

For a clear application structure and for easier maintainability, it is usually better to define fields
in data areas outside the programs.

Example 1 - Fields Defined Directly within a DEFINE DATA Statement:

In the following example, the fields are defined directly within the DEFINE DATA statement of the
program.

DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 PERSONNEL-ID

1 #VARI-A (A20)
1 #VARI-B (N3.2)
1 #VARI-C (I4)
END-DEFINE
...

Example 2 - Fields Defined in a Separate Data Area:

In the following example, the same fields are not defined in the DEFINE DATA statement of the
program, but in an LDA, named LDA39, and the DEFINE DATA statement in the program contains
only a reference to that data area.

Program:

DEFINE DATA LOCAL
USING LDA39

END-DEFINE
...

Local Data Area LDA39:

I T L  Name                             F Length     Miscellaneous             
All -- -------------------------------- - ---------- ------------------------->
  V  1 VIEWEMP                                       EMPLOYEES                 
     2 PERSONNEL-ID                     A          8                           
     2 FIRST-NAME                       A         20                           
     2 NAME                             A         20                           
     1 #VARI-A                          A         20                           
     1 #VARI-B                          N        3.2                           
     1 #VARI-C                          I          4                          ↩

Global Data Area

The following topics are covered below:

■ Creating and Referencing a GDA
■ Creating and Deleting GDA Instances

■ Data Blocks

Creating and Referencing a GDA

GDAs are created and modified with the source editor.

A GDA that is referenced by a Natural programming object must be stored in the same Natural
library (or a steplib defined for this library) where the object that references this GDA is stored.

Note: Using a GDA named COMMON for startup:
If a GDAnamed COMMON exists in a library, the programnamed ACOMMON is invoked automat-
ically when you LOGON to that library.

Important: When you build an application where multiple Natural programming objects
reference a GDA, remember that modifications to the data element definitions in the GDA
affect all Natural programming objects that reference that data area. Therefore these objects
must be recompiled (cataloged and stowed)after the GDA has been modified.

To use a GDA, a Natural programming object must reference it with the GLOBAL clause of the
DEFINE DATA statement. Each Natural programming object can reference only one GDA; that is,
a DEFINE DATA statement must not contain more than one GLOBAL clause.

Creating and Deleting GDA Instances

The first instance of aGDA is created and initialized at runtimewhen the firstNatural programming
object that references it starts to execute.

Once a GDA instance has been created, the data values it contains can be shared by all Natural
programming objects that reference thisGDA (DEFINE DATA GLOBAL statement) and that are invoked
by a PERFORM, INPUT or FETCH statement. All objects that share a GDA instance are operating on
the same data elements.

A new GDA instance is created if the following applies:

■ A subprogram that references a GDA (any GDA) is invoked with a CALLNAT statement.
■ A subprogram that does not reference a GDA invokes a programming object that references a
GDA (any GDA).

If a new instance of a GDA is created, the current GDA instance is suspended and the data values
it contains are stacked. The subprogram then references the data values in the newly created GDA
instance. The data values in the suspended GDA instance or instances is inaccessible. A program-
ming object only refers to one GDA instance and cannot access any previous GDA instances. A
GDA data element can only be passed to a subprogram by defining the element as a parameter
in the CALLNAT statement.

When the subprogram returns to the invoking programming object, the GDA instance it references
is deleted and the GDA instance suspended previously is resumed with its data values.

A GDA instance and its contents is deleted if any of the following applies:

■ The next LOGON is performed.
■ Another GDA is referenced on the same level (levels are described later in this section).
■ A RELEASE VARIABLES statement is executed. In this case, the data values in a GDA instance are
reset either when a program at the level 1 finishes executing, or if the program invokes another
program via a FETCH or RUN statement.

The following graphics illustrate how programming objects reference GDAs and share data
elements in GDA instances.

Sharing GDA Instances

The graphic below illustrates that a subprogram referencing a GDA cannot share the data values
in a GDA instance referenced by the invoking program. A subprogram that references the same
GDA as the invoking program creates a new instance of this GDA. The data elements defined in
aGDA that is referenced by a subprogram can, however, be shared by a subroutine or a helproutine
invoked by the subprogram.

The graphic below shows three GDA instances of GDA1 and the final values each GDA instance is

assigned by the data element #GLOB1. The numbers to indicate the hierarchical levels of
the programming objects.

Using FETCH or FETCH RETURN

The graphic below illustrates that programs referencing the same GDA and invoking one another
with the FETCH or FETCH RETURN statement share the data elements defined in this GDA. If any of
these programs does not reference a GDA, the instance of the GDA referenced previously remains
active and the values of the data elements are retained.

The numbers and indicate the hierarchical levels of the programming objects.

Using FETCH with different GDAs

The graphic below illustrates that if a programuses the FETCH statement to invoke another program
that references a different GDA, the current instance of the GDA (here: GDA1) referenced by the
invoking program is deleted. If this GDA is then referenced again by another program, a new in-
stance of this GDA is created where all data elements have their initial values.

You cannot use the FETCH RETURN statement to invoke another program that references a different
GDA.

The number indicates the hierarchical level of the programming objects.

The invoking programs PROG3 and PROG4 affect the GDA instances as follows:

■ The statement GLOBAL USING GDA2 in PROG3 creates an instance of GDA2 and deletes the current
instance of GDA1.

■ The statement GLOBAL USING GDA1 in PROG4 deletes the current instance of GDA2 and creates a
new instance of GDA1. As a result, the WRITE statement displays the value zero (0).

Data Blocks

To save data storage space, you can create a GDA with data blocks.

The following topics are covered below:

■ Example of Data Block Usage
■ Defining Data Blocks
■ Block Hierarchies

Example of Data Block Usage

Data blocks can overlay each other during program execution, thereby saving storage space.

For example, given the following hierarchy, Blocks B and C would be assigned the same storage
area. Thus it would not be possible for Blocks B and C to be in use at the same time. Modifying
Block B would result in destroying the contents of Block C.

Defining Data Blocks

You define data blocks in the source editor. You establish the block hierarchy by specifying which
block is subordinate to which: you do this by entering the name of the “parent” block in the com-
ment field of the block definition.

In the following example, SUB-BLOCKB and SUB-BLOCKC are subordinate to MASTER-BLOCKA;
SUB-BLOCKD is subordinate to SUB-BLOCKB.

The maximum number of block levels is 8 (including the master block).

Example:

Global Data Area G-BLOCK:

I T L Name F Leng Index/Init/EM/Name/Comment
- - - -------------------------------- - ---- ---------------------------------

B MASTER-BLOCKA
1 MB-DATA01 A 10

B SUB-BLOCKB MASTER-BLOCKA
1 SBB-DATA01 A 20

B SUB-BLOCKC MASTER-BLOCKA
1 SBC-DATA01 A 40

B SUB-BLOCKD SUB-BLOCKB

1 SBD-DATA01 A 40

To make the specific blocks available to a program, you use the following syntax in the DEFINE
DATA statement:

Program 1:

DEFINE DATA GLOBAL
USING G-BLOCK
WITH MASTER-BLOCKA

END-DEFINE

Program 2:

DEFINE DATA GLOBAL
USING G-BLOCK
WITH MASTER-BLOCKA.SUB-BLOCKB

END-DEFINE

Program 3:

DEFINE DATA GLOBAL
USING G-BLOCK
WITH MASTER-BLOCKA.SUB-BLOCKC

END-DEFINE

Program 4:

DEFINE DATA GLOBAL
USING G-BLOCK
WITH MASTER-BLOCKA.SUB-BLOCKB.SUB-BLOCKD

END-DEFINE

With this structure, Program 1 can share the data in MASTER-BLOCKAwith Program 2, Program 3
or Program 4. However, Programs 2 and 3 cannot share the data areas of SUB-BLOCKB and
SUB-BLOCKC because these data blocks are defined at the same level of the structure and thus occupy
the same storage area.

Block Hierarchies

Care needs to be taken when using data block hierarchies. Let us assume the following scenario
with three programs using a data block hierarchy:

Program 1:

DEFINE DATA GLOBAL
USING G-BLOCK
WITH MASTER-BLOCKA.SUB-BLOCKB

END-DEFINE
*
MOVE 1234 TO SBB-DATA01
FETCH 'PROGRAM2'
END

Program 2:

DEFINE DATA GLOBAL
USING G-BLOCK
WITH MASTER-BLOCKA

END-DEFINE
*
FETCH 'PROGRAM3'
END

Program 3:

DEFINE DATA GLOBAL
USING G-BLOCK
WITH MASTER-BLOCKA.SUB-BLOCKB

END-DEFINE
*
WRITE SBB-DATA01
END

Explanation:

■ Program1uses the global data area G-BLOCKwith MASTER-BLOCKA and SUB-BLOCKB. The program
modifies a field in SUB-BLOCKB and fetches Program 2 which specifies only MASTER-BLOCKA in
its data definition.

■ Program 2 resets (deletes the contents of) SUB-BLOCKB. The reason is that a program on Level 1
(for example, a program calledwith a FETCH statement) resets any data blocks that are subordinate
to the blocks it defines in its own data definition.

■ Program 2 now fetches Program 3 which is to display the field modified in Program 1, but it
returns an empty screen.

For details on program levels, seeMultiple Levels of Invoked Objects.

Parameter Data Area

A subprogram is invoked with a CALLNAT statement. With the CALLNAT statement, parameters can
be passed from the invoking object to the subprogram.

These parameters must be defined with a DEFINE DATA PARAMETER statement in the subprogram:

■ they can be defined in the PARAMETER clause of the DEFINE DATA statement itself; or
■ they can be defined in a separate parameter data area,with the DEFINE DATA PARAMETER statement
referencing that PDA.

The following topics are covered below:

■ Parameters Defined within DEFINE DATA PARAMETER Statement

■ Parameters Defined in Parameter Data Area

Parameters Defined within DEFINE DATA PARAMETER Statement

Parameters Defined in Parameter Data Area

In the same way, parameters that are passed to an external subroutine via a PERFORM statement
must be defined with a DEFINE DATA PARAMETER statement in the external subroutine.

In the invoking object, the parameter variables passed to the subprogram/subroutine need not be
defined in a PDA; in the illustrations above, they are defined in the LDA used by the invoking
object (but they could also be defined in a GDA).

The sequence, format and length of the parameters specified with the CALLNAT/PERFORM statement
in the invoking object must exactly match the sequence, format and length of the fields specified
in the DEFINE DATA PARAMETER statement of the invoked subprogram/subroutine. However, the
names of the variables in the invoking object and the invoked subprogram/subroutine need not
be the same (as the parameter data are transferred by address, not by name).

To guarantee that the data element definitions used in the invoking program are identical to the
data element definitions used in the subprogram or external subroutine, you can specify a PDA

in a DEFINE DATA LOCAL USING statement. By using a PDAas an LDAyou can avoid the extra effort
of creating an LDA that has the same structure as the PDA.

This document discusses those object typeswhich can be invoked as routines; that is, as subordinate
programs.

Helproutines and maps, although they are also invoked from other objects, are strictly speaking
not routines as such, and are therefore discussed in separate documents; see Helproutines and
Maps.

A Modular Application Structure

Typically, a Natural application does not consist of a single huge program, but is split into several
modules. Each of these modules will be a functional unit of manageable size, and each module is
connected to the other modules of the application in a clearly defined way. This provides for a
well structured application, which makes its development and subsequent maintenance a lot
easier and faster.

During the execution of amain program, other programs, subprograms, subroutines, helproutines
andmaps can be invoked. These objects can in turn invoke other objects (for example, a subroutine
can itself invoke another subroutine). Thus, the modular structure of an application can become
quite complex and extend over several levels.

Multiple Levels of Invoked Objects

Each invoked object is one level below the level of the object from which it was invoked; that is,
with each invocation of a subordinate object, the level number is incremented by 1.

Anyprogram that is directly executed is at Level 1; any subprogram, subroutine,map or helproutine
directly invoked by themain program is at Level 2; when such a subroutine in turn invokes another
subroutine, the latter is at Level 3.

A program invoked with a FETCH statement from within another object is classified as a main
program, operating from Level 1. A program that is invoked with FETCH RETURN, however, is
classified as a subordinate program and is assigned a level one below that of the invoking object.

The following illustration is an example of multiple levels of invoked objects and also shows how
these levels are counted:

If you wish to ascertain the level number of the object that is currently being executed, you can
use the system variable *LEVEL (which is described in the System Variables documentation).

This document discusses the following Natural object types, which can be invoked as routines
(that is, subordinate programs):

■ program
■ function
■ subroutine
■ subprogram

Helproutines and maps, although they are also invoked from other objects, are strictly speaking
not routines as such, and are therefore discussed in separate documents; see Helproutines and
Maps.

Basically, programs, subprograms and subroutines differ from one another in the way data can
be passed between them and in their possibilities of sharing each other's data areas. Therefore the
decision which object type to use for which purpose depends very much on the data structure of
your application.

Program

A program can be executed - and thus tested - by itself.

A program can also be invoked from another object with a FETCH or FETCH RETURN statement. The
invoking object can be another program, a subprogram, function, subroutine or helproutine.

■ When a program is invoked with FETCH RETURN, the execution of the invoking object will be
suspended - not terminated - and the fetched programwill be activated as a subordinate program.
When the execution of the FETCHed program is terminated, the invoking objectwill be re-activated
and its execution continued with the statement following the FETCH RETURN statement.

■ When a program is invoked with FETCH, the execution of the invoking object will be terminated
and the FETCHed program will be activated as a main program. The invoking object will not be
re-activated upon termination of the fetched program.

The following topics are covered below:

■ Program Invoked with FETCH RETURN

■ Program Invoked with FETCH

Program Invoked with FETCH RETURN

Aprogram invokedwith FETCH RETURN can access the global data area used by the invoking object.

In addition, every program can have its own local data area, in which the fields that are to be used
only within the program are defined.

However, a program invoked with FETCH RETURN cannot have its own global data area.

Program Invoked with FETCH

A program invoked with FETCH as a main program usually establishes its own global data area
(as shown in the illustration above). However, it could also use the same global data area as estab-
lished by the invoking object.

Note: A source program can also be invoked with a RUN statement; see the RUN statement
in the Statements documentation.

Function

The Natural object of type function contains one DEFINE FUNCTION statement for the definition of
a single function and the END statement. The function itself can be invoked using the function call
syntax.

The basic structure of the DEFINE FUNCTION statement is like this:

DEFINE FUNCTION function-name
RETURNS ...
DEFINE DATA
...
END-DEFINE

statements ...

END-FUNCTION

And the function call has the structure:

function-name (< ... >)

The DEFINE FUNCTION statement offers the keyword RETURNS for the result value of a function.
With the DEFINE DATA statement, the parameters for the function call aswell as local and independ-
ent variables for the function logic can be defined. A global data area (for example, GDA1) cannot
be referenced inside the function definition.

The block of statements after the RETURNS keyword and the DEFINE DATA statement must contain
all those statements which are to be executed when the function is called.

Parameter data areas (for example, PDA1) may be used to access parameters for function calls and
function definitions in order to minimize the maintainance effort when changing parameters.

The function call can be used either as an operand within a Natural statement or stand-alone in
place of a Natural statement. The arguments of the function call must be specified using a special
bracket notation: (<...>).

The DEFINE PROTOTYPE statement may be used to define the result and parameter layouts for a
certain function. This may be considered if the function is not yet available or if a variable function
call is to be used. At compilation time, the type of the result variable and the parameters will be
checked.

The DEFINE PROTOTYPE statement may be included in a copycode object if the function call, and
therefore the prototype definition can be used in several programming objects.

For further information, see the section User-Defined Functions.

Subroutine

The statements that make up a subroutine must be defined within a DEFINE SUBROUTINE ...
END-SUBROUTINE statement block.

A subroutine is invoked with a PERFORM statement.

A subroutine may be an inline subroutine or an external subroutine:

■ Inline Subroutine
An inline subroutine is defined within the object which contains the PERFORM statement that in-
vokes it.

■ External Subroutine
An external subroutine is defined in a separate object - of type subroutine - outside the object
which invokes it.

If you have a block of code which is to be executed several times within an object, it is useful to
use an inline subroutine. You then only have to code this block once within a DEFINE SUBROUTINE
statement block and invoke it with several PERFORM statements.

The following topics are covered below:

■ Inline Subroutine
■ Data Available to an Inline Subroutine
■ External Subroutine

■ Data Available to an External Subroutine

Inline Subroutine

An inline subroutine can be contained within a programming object of type program, function,
subprogram, subroutine or helproutine.

If an inline subroutine is so large that it impairs the readability of the object inwhich it is contained,
you may consider putting it into an external subroutine, so as to enhance the readability of your
application.

Data Available to an Inline Subroutine

An inline subroutine has access to the local data area and the global data area used by the object
in which it is contained.

External Subroutine

An external subroutine - that is, an object of type subroutine - cannot be executed by itself. It must
be invoked from another object. The invoking object can be a program, function, subprogram,
subroutine or helproutine.

Data Available to an External Subroutine

An external subroutine can access the global data area used by the invoking object.

Moreover, parameters can be passed with the PERFORM statement from the invoking object to the
external subroutine. These parameters must be defined either in the DEFINE DATA PARAMETER
statement of the subroutine, or in a parameter data area used by the subroutine.

In addition, an external subroutine can have its local data area, in which the fields that are to be
used only within the subroutine are defined.

However, an external subroutine cannot have its own global data area.

Subprogram

Typically, a subprogram would contain a generally available standard function that is used by
various objects in an application.

A subprogram cannot be executed by itself. It must be invoked from another object. The invoking
object can be a program, function, subprogram, subroutine or helproutine.

A subprogram is invoked with a CALLNAT statement.

When the CALLNAT statement is executed, the execution of the invoking object will be suspended
and the subprogram executed. After the subprogram has been executed, the execution of the in-
voking object will be continued with the statement following the CALLNAT statement.

Data Available to a Subprogram

With the CALLNAT statement, parameters can be passed from the invoking object to the subprogram.
These parameters are the only data available to the subprogram from the invoking object. They
must be defined either in the DEFINE DATA PARAMETER statement of the subprogram, or in a para-
meter data area used by the subprogram.

In addition, a subprogram can have its own local data area, in which the fields to be used within
the subprogram are defined.

If a subprogram in turn invokes a subroutine or helproutine, it can also establish its own global
data area to be shared with the subroutine/helproutine.

Processing Flow when Invoking a Routine

When the CALLNAT, PERFORM or FETCH RETURN statement that invokes a routine - a subprogram, an
external subroutine, or a program respectively - is executed, the execution of the invoking object
is suspended and the execution of the routine begins.

The execution of the routine continues until either its END statement is reached or processing of
the routine is stopped by an ESCAPE ROUTINE statement being executed.

In either case, processing of the invoking object will then continue with the statement following
the CALLNAT, PERFORM or FETCH RETURN statement used to invoke the routine.

Example:

6 Processing a Rich GUI Page - Adapter

TheNatural object of type “adapter” is used to represent a rich GUI page in a Natural application.
This object type plays a similar role for the processing of a rich GUI page as the object type map
plays for terminal I/O processing. But it is different from a map in that it does not contain layout
information.

An object of type adapter is generated from an external page layout. It serves as an interface that
enables a Natural application to send data to an external I/O system for presentation andmodific-
ation, using an externally defined and stored page layout. The adapter contains the Natural code
necessary to perform this task.

An application program refers to an adapter in the PROCESS PAGE USING statement.

For information on the object type “adapter”, see the Natural for Ajax documentation.

7 Maps

As an alternative to specifying screen layouts dynamically, the INPUT statement offers the possib-
ility to use predefined map layouts which makes use of the Natural object type “map”.

Benefits of Using Maps

Using predefined map layouts rather than dynamic screen-layout specifications offers various
advantages such as:

■ Clearly structured applications as a result of a consequent separation of program logic and
display logic.

■ Map layout modifications possible without making changes to the main programs.
■ The language of an applications's user interface can be easily adapted for internationalization
or localization.

The benefit of using programming objects such as maps will become obvious when it comes to
maintaining existing Natural applications.

Types of Maps

Maps (screen layouts) are those parts of an application which the users see on their screens.

The following types of maps exist:

■ Input Map
The dialog with the user is carried out via input maps.

■ Output Map
If an application produces any output report, this report can be displayed on the screen by using
an output map.

■ Help Map
Helpmaps are, in principle, like any othermaps, but when they are assigned as help, additional
checks are performed to ensure their usability for help purpose.

The object type “map” comprises

■ the map body which defines the screen layout and
■ an associated parameter data area (PDA) which, as a sort of interface, contains data definitions
such as name, format, length of each field presented on a specific map.

Related Topics:

■ For information on selection boxes that can be attached to input fields, see SB - Selection Box in
the INPUT statement documentation and SB - Selection Box in the Parameter Reference.

■ For information on split screen maps where the upper portion may be used as an output map
and the lower portion as an inputmap, see Split-Screen Feature in the INPUT statement document-
ation.

Creating Maps

Maps and help map layouts are created and edited in the map editor.

The map editor is used to create character-based user interfaces. If you want to create complex
graphical user interfaces, youwill use Ajax Developer instead. Ajax Developer includes a number
of tools. Its central tool is the Layout Painter which is used to define layouts for HTML pages. The
appropriate local data area (LDA) is created and maintained in the source editor.

■ For information on input processing using screen layouts specified dynamically, see Syntax 1 -
Dynamic Screen Layout Specification in the INPUT statement documentation.

■ For information on input processing using amap layout createdwith themap editor, see Syntax
2 - Using Predefined Map Layout in the INPUT statement documentation.

Starting/Stopping Map Processing

An input map is invoked with an INPUT USING MAP statement.

An output map is invoked with a WRITE USING MAP statement.

Processing of a map can be stopped with an ESCAPE ROUTINE statement in a processing rule.

Helproutines have specific characteristics to facilitate the processing of help requests. They may
be used to implement complex and interactive help systems. They are created with the source
editor.

Invoking Help

A Natural user can invoke a Natural helproutine either by entering the help character in a field,
or by pressing the help key (usually PF1). The default help character is a question mark (?).

■ The help character must be entered only once.
■ The help character must be the only character modified in the input string.
■ The help character must be the first character in the input string.

If a helproutine is specified for a numeric field, Natural will allow a question mark to be entered
for the purpose of invoking the helproutine for that field. Natural will still check that valid numeric
data are provided as field input.

If not already specified, the help key may be specified with the SET KEY statement:

SET KEY PF1=HELP

A helproutine can only be invoked by a user if it has been specified in the program ormap from
which it is to be invoked.

Specifying Helproutines

A helproutine may be specified:

■ in a program: at statement level and at field level;
■ in a map: at map level and at field level.

If a user requests help for a field for which no help has been specified, or if a user requests help
without a field being referenced, the helproutine specified at the statement ormap level is invoked.

Ahelproutinemay also be invoked byusing a REINPUT USING HELP statement (either in the program
itself or in a processing rule). If the REINPUT USING HELP statement contains a MARK option, the
helproutine assigned to the marked field is invoked. If no field-specific helproutine is assigned,
the map helproutine is invoked.

A REINPUT statement in a helproutine may only apply to INPUT statements within the same hel-
proutine.

The name of a helproutine may be specified either with the session parameter HE of an INPUT
statement:

INPUT (HE='HELP2112')

or by using the map editor.

The name of a helproutine may be specified as an alphanumeric constant or as an alphanumeric
variable containing the name. If it is a constant, the name of the helproutine must be specified
within apostrophes.

Programming Considerations for Helproutines

Processing of a helproutine can be stopped with an ESCAPE ROUTINE statement.

Be careful when using END OF TRANSACTION or BACKOUT TRANSACTION statements in a helproutine,
because this will affect the transaction logic of the main program.

Passing Parameters to Helproutines

A helproutine can access the currently active global data area (but it cannot have its own global
data area). In addition, it can have its own local data area.

Data may also be passed from/to a helproutine via parameters. A helproutine may have up to 20
explicit parameters and one implicit parameter. The explicit parameters are specified with the HE
operand after the helproutine name:

HE='MYHELP','001'

The implicit parameter is the field for which the helproutine was invoked:

INPUT #A (A5) (HE='YOURHELP','001')

where 001 is an explicit parameter and #A is the implicit parameter/the field.

This is specified within the DEFINE DATA PARAMETER statement of the helproutine as:

DEFINE DATA PARAMETER
1 #PARM1 (A3) /* explicit parameter
1 #PARM2 (A5) /* implicit parameter
END-DEFINE

Please note that the implicit parameter (#PARM2 in the above example)may be omitted. The implicit
parameter is used to access the field for which help was requested, and to return data from the
helproutine to the field. For example, youmight implement a calculator program as a helproutine
and have the result of the calculations returned to the field.

When help is called, the helproutine is called before the data are passed from the screen to the
program data areas. This means that helproutines cannot access data entered within the same
screen transaction.

Once help processing is complete, the screen data will be refreshed: any fields which have been
modified by the helproutine will be updated - excluding fields which had been modified by the
user before the helproutine was invoked, but including the field for which help was requested.
Exception: If the field forwhich helpwas requested is split into several parts by dynamic attributes
(DY session parameter), and the part in which the question mark is entered is after a part modified
by the user, the field content will not be modified by the helproutine.

Attribute control variables are not evaluated again after the processing of the helproutine, even
if they have been modified within the helproutine.

Equal Sign Option

The equal sign (=) may be specified as an explicit parameter:

INPUT PERSONNEL-NUMBER (HE='HELPROUT',=)

This parameter is processed as an internal field (format/lengthA65)which contains the field name
(or map name if specified at map level). The corresponding helproutine starts with:

DEFINE DATA PARAMETER
1 FNAME (A65) /* contains 'PERSONNEL-NUMBER'
1 FVALUE (N8) /* value of field (optional)
END-DEFINE

This option may be used to access one common helproutine which reads the field name and
provides field-specific help by accessing the application online documentation or the Predict data
dictionary.

Array Indices

If the field selected by the help character or the help key is an array element, its indices are supplied
as implicit parameters (1 - 3 depending on rank, regardless of the explicit parameters).

The format/length of these parameters is I2.

INPUT A(*,*) (HE='HELPROUT',=)

The corresponding helproutine starts with:

DEFINE DATA PARAMETER
1 FNAME (A65) /* contains 'A'
1 FVALUE (N8) /* value of selected element
1 FINDEX1 (I2) /* 1st dimension index
1 FINDEX2 (I2) /* 2nd dimension index
END-DEFINE
...

Help as a Window

The size of a help to be displayedmay be smaller than the screen size. In this case, the help appears
on the screen as a window, enclosed by a frame, for example:

*******************************************************************************
PERSONNEL INFORMATION

PLEASE ENTER NAME: ?_________________
PLEASE ENTER CITY: __________________

+---------------------------+
! !
! Type in the name of an !
! employee in the first !
! field and press ENTER. !
! You will then receive !
! a list of all employees !
! of that name. !
! !
! For a list of employees !
! of a certain name who !
! live in a certain city, !
! type in a name in the !
! first field and a city !
! in the second field !
! and press ENTER. !

*******************! !*******************************
+---------------------------+

Within a helproutine, the size of the window may be specified as follows:

■ by a FORMAT statement (for example, to specify the page size and line size: FORMAT PS=15 LS=30);
■ by an INPUT USING MAP statement; in this case, the size defined for the map (in its map settings)
is used;

■ by a DEFINE WINDOW statement; this statement allows you to either explicitly define a window
size or leave it to Natural to automatically determine the size of the window depending on its
contents.

The position of a help window is computed automatically from the position of the field for which
help was requested. Natural places the window as close as possible to the corresponding field
without overlaying the field. With the DEFINE WINDOW statement, you may bypass the automatic
positioning and determine the window position yourself.

For further information on window processing, please refer to the DEFINE WINDOW statement in
the Statements documentation and the terminal command %W in the Terminal Commands document-
ation.

9 Multiple Use of Source Code - Copycode

This chapter describes the advantages and the use of copycode.

Use of Copycode

Copycode is a portion of source code which can be included in another object via an INCLUDE
statement.

So, if you have a statement block which is to appear in identical form in several objects, you may
use copycode instead of coding the statement block several times. This reduces the coding effort
and also ensures that the blocks are really identical.

Processing of Copycode

The copycode is included at compilation; that is, the source-code lines from the copycode are not
physically inserted into the object that contains the INCLUDE statement, but they will be included
in the compilation process and are thus part of the resulting object module.

Consequently, when you modify the source code of copycode, you also have to newly compile
(STOW) all objects which use that copycode.

Attention:

■ Copycode cannot be executed on its own. It cannot be stowed, but only saved.
■ An END statement must not be placed within a copycode.

For further information, refer to the description of the INCLUDE statement (in the Statements docu-
mentation).

10 Documenting Natural Objects - Text

The Natural object type “text” is used to write text rather than programs.

Use of Text Objects

You can use this type of object to documentNatural objects inmore detail than you can, for example,
within the source code of a program.

Text objects may also be useful at sites where Predict is not available for program documentation
purposes.

Writing Text

You write the text using the source editor.

The only difference in handling as opposed to writing programs is that there is no lower to upper
case translation, that is, the text you write stays as it is.

You can write any text you wish (there is no syntax check).

Text objects can only be saved, they cannot be stowed. They cannot be executed, but only displayed
in the editor.

11 Creating Component Based Applications - Class

Classes are used to apply an object based programming style.

12 Using Non-Natural Files - Resource

This section describes the Natural object of type resource.

Non-Natural file types, such as HTML files, XML style sheets, etc., supported by Natural for
OpenVMS are located in the different libraries in the RES directory. Natural delivers some with
the product, but you can also save your own resources for your applications in RES.

Use of Resources

Natural distinguishes two kinds of resources:

■ Shared Resources
A shared resource is any non-Natural file that is used in aNatural application and ismaintained
in the Natural library system.

■ Private Resources
A private resource is a file that is assigned to one and only one Natural object and is considered
to be part of that object. An object can have at most one private resource file. At the moment,
only Natural dialogs have private resources.

Both shared and private resources belonging to a Natural library are maintained in a subdirectory
named ..\RES in the directory that represents the Natural library in the file system.

Shared Resources

A shared resource is any non-Natural file that is used in a Natural application and is maintained
in the Natural library system. A non-Natural file that is to be used as a shared resource must be
contained in the subdirectory named ..\RES of a Natural library.

Example of Using a Shared Resource

The bitmapMYPICTURE.BMP is to be displayed in a bitmap control in a dialog MYDLG, contained
in a library MYLIB. First the bitmap is put into the Natural library MYLIB by moving it into the dir-
ectory ..\MYLIB\RES. The following code snippet from the dialog MYDLG shows how it is then
assigned to the bitmap control:

DEFINE DATA LOCAL
01 #BM-1 HANDLE OF BITMAP 
... 
END-DEFINE 
* (Creation of the Bitmap control omitted.) 
... 
#BM-1.BITMAP-FILE-NAME := "MYPICTURE.BMP" ... ↩

The advantages of using the bitmap as a shared resource are:

■ The file name can be specified in the Natural dialog without a path name.
■ The file can be kept in a Natural library together with the Natural object that uses it.

Note: In previous Natural versions non-Natural files were usually kept in a directory that
was defined with the environment variable NATGUI_BMP. Existing applications that use this
approachwill work in the sameway as before, becauseNatural always searches for a shared
resource file in this directory, if it was not found in the current library.

Private Resources

Private resources are used internally byNatural to store binary data that is part of Natural objects.
These files are recognized by the file name extension NR*, where * is a character that depends on
the type of theNatural object. Naturalmaintains private resource files and their contents automat-
ically. A Natural object can have a maximum of one private resource file.

Currently, only Natural dialogs have a private resource file. This file is used to store the configur-
ation of ActiveX controls that are defined in a dialog and are configured with their own property
pages.

Example of Private Resources

The name of the private resource file of the dialog MYDLG isMYDLG.NR3.

Natural creates, modifies and deletes this file automatically as needed, when the dialog is created,
modified, deleted, etc.

The private resource file is used to store binary data related to the dialog MYDLG.

API for Processing Resources

In the library SYSEXT, the following application programming interface (API) exists, which gives
user applications access to resources' unique user exit routines:

PurposeAPI

Write, read, delete a resource by using short or long name.USR4208N

III Defining Fields

This part describes how you define the fields you wish to use in a program. These fields can be
database fields and user-defined fields.

Use and Structure of DEFINE DATA Statement

User-Defined Variables

Function Call

Introduction to Dynamic Variables and Fields

Using Dynamic and Large Variables

User-Defined Constants

Initial Values (and the RESET Statement)

Redefining Fields

Arrays

X-Arrays

Please note that only the major options of the DEFINE DATA statement are discussed here. Further
options are described in the Statements documentation.

The particulars of database fields are described in Accessing Data in an Adabas Database. On
principle, the features and examples described there for Adabas also apply to other database
management systems. Differences, if any, are described in the relevant database interface docu-
mentation and in the Statements documentation or Parameter Reference.

13 Use and Structure of DEFINE DATA Statement

■ Field Definitions in DEFINE DATA Statement ........................................................................................ 78
■ Defining Fields within a DEFINE DATA Statement .................................................................................. 78
■ Defining Fields in a Separate Data Area ............................................................................................... 79
■ Structuring a DEFINE DATA Statement Using Level Numbers .................................................................. 79

77



The first statement in a Natural program written in structured modemust always be a DEFINE
DATA statement which is used to define fields for use in a program.

Field Definitions in DEFINE DATA Statement

In the DEFINE DATA statement, you define all the fields - database fields as well as user-defined
variables - that are to be used in the program.

There are two ways to define the fields:

■ The fields can be defined within the DEFINE DATA statement itself (see below).
■ The fields can be defined outside the program in a local or global data area, with the DEFINE
DATA statement referencing that data area (see below).

If fields are used bymultiple programs/routines, they should be defined in a data area outside the
programs.

For a clear application structure, it is usually better to define fields in data areas outside the pro-
grams.

Data areas are created and maintained with the source editor.

In thefirst example below, the fields are definedwithin the DEFINE DATA statement of the program.
In the second example, the same fields are defined in a local data area (LDA), and the DEFINE
DATA statement only contains a reference to that data area.

Defining Fields within a DEFINE DATA Statement

The following example illustrates howfields can be definedwithin the DEFINE DATA statement itself:

DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 PERSONNEL-ID

1 #VARI-A (A20)
1 #VARI-B (N3.2)
1 #VARI-C (I4)
END-DEFINE
...

Defining Fields in a Separate Data Area

The following example illustrates how fields can be defined in a local data area (LDA):

Program:

DEFINE DATA LOCAL
USING LDA39 

END-DEFINE 
... ↩

Local Data Area LDA39:

I T L Name                             F Leng Index/Init/EM/Name/Comment       
- - - -------------------------------- - ---- ---------------------------------
  V 1 VIEWEMP                                 EMPLOYEES    
    2 NAME                             A   20                           
    2 FIRST-NAME                       A   20
    2 PERSONNEL-ID                     A    8                   
    1 #VARI-A                          A   20
    1 #VARI-B                          N  3.2
    1 #VARI-C                          I    4
 ↩

Structuring a DEFINE DATA Statement Using Level Numbers

The following topics are covered:

■ Structuring and Grouping Your Definitions
■ Level Numbers in View Definitions
■ Level Numbers in Field Groups
■ Level Numbers in Redefinitions

Structuring and Grouping Your Definitions

Level numbers are used within the DEFINE DATA statement to indicate the structure and grouping
of the definitions. This is relevant with:

■ view definitions
■ field groups
■ redefinitions

Level numbers are 1- or 2-digit numbers in the range from 01 to 99 (the leading zero is optional).

Generally, variable definitions are on Level 1.

The level numbering in view definitions, redefinitions and groups must be sequential; no level
numbers may be skipped.

Level Numbers in View Definitions

If you define a view, the specification of the view namemust be on Level 1, and the fields the view
is comprised of must be on Level 2. (For details on view definitions, see Database Access.)

Example of Level Numbers in View Definition:

DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 BIRTH

...
END-DEFINE

Level Numbers in Field Groups

The definition of groups provides a convenient way of referencing a series of consecutive fields.
If you define several fields under a common group name, you can reference the fields later in the
program by specifying only the group name instead of the names of the individual fields.

The group name must be specified on Level 1, and the fields contained in the group must be one
level lower.

For group names, the same naming conventions apply as for user-defined variables.

Example of Level Numbers in Group:

DEFINE DATA LOCAL
1 #FIELDA (N2.2)
1 #FIELDB (I4)
1 #GROUPA

2 #FIELDC (A20)
2 #FIELDD (A10)
2 #FIELDE (N3.2)

1 #FIELDF (A2) 
... 
END-DEFINE ↩

In this example, the fields #FIELDC, #FIELDD and #FIELDE are defined under the common group
name #GROUPA. The other three fields are not part of the group. Note that #GROUPA only serves as
a group name and is not a field in its own right (and therefore does not have a format/length
definition).

Level Numbers in Redefinitions

If you redefine a field, the REDEFINE option must be on the same level as the original field, and the
fields resulting from the redefinition must be one level lower. For details on redefinitions, see
Redefining Fields.

Example of Level Numbers in Redefinition:

DEFINE DATA LOCAL
1 VIEWEMP VIEW OF STAFFDDM

2 BIRTH
2 REDEFINE BIRTH

3 #YEAR-OF-BIRTH (N4)
3 #MONTH-OF-BIRTH (N2)
3 #DAY-OF-BIRTH (N2)

1 #FIELDA (A20)
1 REDEFINE #FIELDA

2 #SUBFIELD1 (N5)
2 #SUBFIELD2 (A10)
2 #SUBFIELD3 (N5) 

...  
END-DEFINE ↩

In this example, the database field BIRTH is redefined as three user-defined variables, and the user-
defined variable #FIELDA is redefined as three other user-defined variables.

User-defined variables are fields which you define yourself in a program. They are used to store
values or intermediate results obtained at some point in program processing for additional pro-
cessing or display.

See also Naming Conventions for User-Defined Variables in Using Natural.

Definition of Variables

You define a user-defined variable by specifying its name and its format/length in the DEFINE
DATA statement.

You define the characteristics of a variable with the following notation:

(r,format-length/index)

This notation follows the variable name, optionally separated by one or more blanks.

No blanks are allowed between the individual elements of the notation.

The individual elements may be specified selectively as required, but when used together, they
must be separated by the characters as indicated above.

Example:

In this example, a user-defined variable of alphanumeric format and a length of 10 positions is
defined with the name #FIELD1.

DEFINE DATA LOCAL
1 #FIELD1 (A10)
...
END-DEFINE

Notes:

1. If operating in structuredmode or if a program contains a DEFINE DATA LOCAL clause, variables
cannot be defined dynamically in a statement.

2. This does not apply to application-independent variables (AIVs); see also Defining Application-
Independent Variables

Referencing of Database Fields Using (r) Notation

A statement label or the source-code line number can be used to refer to a previous Natural
statement. This can be used to override Natural's default referencing (as described for each state-
ment, where applicable), or for documentation purposes. See also Loop Processing, Referencing
Statements within a Program.

The following topics are covered below:

■ Default Referencing of Database Fields
■ Referencing with Statement Labels
■ Referencing with Source-Code Line Numbers

Default Referencing of Database Fields

Generally, the following applies if you specify no statement reference notation:

■ By default, the innermost active database loop (FIND, READ or HISTOGRAM) in which the database
field in question has been read is referenced.

■ If the field is not read in any active database loop, the last previous GET statement (in reporting
mode also FIND FIRST or FIND UNIQUE statement) is referenced which is not contained in an
already closed loop and which has read the field.

Referencing with Statement Labels

AnyNatural statement which causes a processing loop to be initiated and/or causes data elements
to be accessed in the database may be marked with a symbolic label for subsequent referencing.

A label may be specified either in the form label. before the referencing object or in parentheses
(label.) after the referencing object (but not both simultaneously).

The naming conventions for labels are identical to those for variables. The period after the label
name serves to identify the entry as a label.

Example:

...
RD. READ PERSON-VIEW BY NAME STARTING FROM 'JONES'

FD. FIND AUTO-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (FD.)
DISPLAY NAME (RD.) FIRST-NAME (RD.) MAKE (FD.)

END-FIND
END-READ
...

85Programming Guide

User-Defined Variables



Referencing with Source-Code Line Numbers

A statement may also be referenced by using the number of the source-code line in which the
statement is located.

All four digits of the line number must be specified (leading zeros must not be omitted).

Example:

...
0110 FIND EMPLOYEES-VIEW WITH NAME = 'SMITH'
0120 FIND VEHICLES-VIEW WITH MODEL = 'FORD'
0130 DISPLAY NAME (0110) MODEL (0120)
0140 END-FIND
0150 END-FIND
...

Renumbering of Source-Code Line Number References

Numeric four-digit source-code line numbers that reference a statement (see Referencing of
Database Fields Using (r) Notation and also Referencing Statements within a Program) are also
renumbered if the Natural source program is renumbered. For the user's convenience and to aid
in readability and debugging, all source code line number references that occur in a statement, an
alphanumeric constant or a comment are renumbered. The position of the source code line number
reference in the statement or alphanumeric constant (start, middle, end) does not matter.

The following patterns are recognized as being a valid source code line number reference and are
renumbered (nnnn is a four-digit number):

Sample StatementPattern

ESCAPE BOTTOM (0150)(nnnn)

DISPLAY ADDRESS-LINE(0010/1:5)(nnnn/

DISPLAY NAME(0010,A10/1:5)(nnnn,

If the left parenthesis or the four-digit number nnnn is followed by a blank, or the four-digit
number nnnn is followed by a period, the pattern is not considered to be a valid source code line
number reference.

To avoid that a four-digit number that is contained in an alphanumeric constant is unintentionally
renumbered, the constant should be split up and the different parts should be concatenated to
form a single value by use of a hyphen.

Example:

Z := 'XXXX (1234,00) YYYY'

should be replaced by

Z := 'XXXX (1234' - ',00) YYYY'

Format and Length of User-Defined Variables

Format and length of a user-defined variable are specified in parentheses after the variable name.

Fixed-length variables can be defined with the following formats and corresponding lengths.

For the definition of Format and Length in dynamic variables, seeDefinition ofDynamic Variables.

Internal Length (in Bytes)Definable LengthExplanationFormat

1 - 10737418241 - 1073741824 (1GB)AlphanumericA

1 - 10737418241 - 1073741824 (1GB)BinaryB

2-Attribute ControlC

4-DateD

4 or 84 or 8Floating PointF

1, 2 or 41 , 2 or 4IntegerI

1-LogicalL

1 - 291 - 29Numeric (unpacked)N

1 - 151 - 29Packed numericP

7-TimeT

2 - 10737418241 - 536870912 (0.5 GB)Unicode (UTF-16)U

Length can only be specified if format is specified. With some formats, the length need not be ex-
plicitly specified (as shown in the table above).

For fields defined with format N or P, you can use decimal position notation in the form nn.m,
where nn represents the number of positions before the decimal point, and m represents the number
of positions after the decimal point. The sum of the values of nn and mmust not exceed 29, and
the value of mmust not exceed 7.

Notes:

1. When a user-defined variable of format P is output with a DISPLAY, WRITE, or INPUT statement,
Natural internally converts the format to N for the output.

2. In reportingmode, if format and length are not specified for a user-defined variable, the default
format/length N7 will be used, unless this default assignment has been disabled by the pro-
file/session parameter FS.

For a database field, the format/length as defined for the field in the data definitionmodule (DDM)
apply. (In reporting mode, it is also possible to define in a program a different format/length for
a database field.)

In structured mode, format and length may only be specified in a data area definition or with a
DEFINE DATA statement.

Example of Format/Length Definition - Structured Mode:

DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME

1 #NEW-SALARY (N6.2)
END-DEFINE
...
FIND EMPLOY-VIEW ...
...
COMPUTE #NEW-SALARY = ...
...

In reporting mode, format/length may be defined within the body of the program, if no DEFINE
DATA statement is used.

Example of Format/Length Definition - Reporting Mode:

...

...
FIND EMPLOYEES

... ... COMPUTE #NEW-SALARY(N6.2) = ...

...

Special Formats

In addition to the standard alphanumeric (A) and numeric (B, F, I, N, P) formats, Natural supports
the following special formats:

■ Format C - Attribute Control
■ Formats D - Date, and T - Time
■ Format L - Logical

Programming Guide88

User-Defined Variables



■ Format: Handle

Format C - Attribute Control

A variable defined with format C may be used to assign attributes dynamically to a field used in
a DISPLAY, INPUT, PRINT, PROCESS PAGE or WRITE statement.

For a variable of format C, no length can be specified. The variable is always assigned a length of
2 bytes by Natural.

Example:

DEFINE DATA LOCAL
1 #ATTR (C)
1 #A (N5)
END-DEFINE
...
MOVE (AD=I CD=RE) TO #ATTR
INPUT #A (CV=#ATTR)
...

For further information, see the session parameter CV.

Formats D - Date, and T - Time

Variables defined with formats D and T can be used for date and time arithmetic and display.
Format D can contain date information only. Format T can contain date and time information; in
otherwords, date information is a subset of time information. Time is counted in tenths of seconds.

For variables of formats D and T, no length can be specified. A variable with format D is always
assigned a length of 4 bytes (P6) and a variable of format T is always assigned a length of 7 bytes
(P12) byNatural. If the profile parameter MAXYEAR is set to 9999, a variable with format D is always
assigned a length of 4 bytes (P7) and a variable of format T is always assigned a length of 7 bytes
(P13) by Natural.

Example:

DEFINE DATA LOCAL
1 #DAT1 (D)
END-DEFINE
*
MOVE *DATX TO #DAT1
ADD 7 TO #DAT1
WRITE '=' #DAT1
END

For further information, see the session parameter EM and the system variables *DATX and *TIMX.

The value in a date field must be in the range from 1st January 1582 to 31st December 2699.

Format L - Logical

A variable defined of format L may be used as a logical condition criterion. It can take the value
TRUE or FALSE.

For a variable of format L, no length can be specified. A variable of format L is always assigned a
length of 1 byte by Natural.

Example:

DEFINE DATA LOCAL
1 #SWITCH(L)
END-DEFINE
MOVE TRUE TO #SWITCH
...
IF #SWITCH

...
MOVE FALSE TO #SWITCH

ELSE
...
MOVE TRUE TO #SWITCH

END-IF

For further information on logical value presentation, see the session parameter EM.

Format: Handle

A variable defined as HANDLE OF OBJECT can be used as an object handle.

Index Notation

An index notation is used for fields that represent an array.

An integer numeric constant or user-defined variable may be used in index notations. A user-
defined variable can be specified using one of the following formats: N (numeric), P (packed), I
(integer) or B (binary), where format B may be used only with a length of less than or equal to 4.

A system variable, system function or qualified variable cannot be used in index notations.

Array Definition - Examples:

1. #ARRAY (3)
Defines a one-dimensional array with three occurrences.

2. FIELD ( label.,A20/5) orlabel.FIELD(A20/5)
Defines an array from a database field referencing the statementmarked by label.with format
alphanumeric, length 20 and 5 occurrences.

3. #ARRAY (N7.2/1:5,10:12,1:4)
Defines an array with format/length N7.2 and three array dimensions with 5 occurrences in the
first, 3 occurrences in the second and 4 occurrences in the third dimension.

4. FIELD ( label./i:i + 5) orlabel.FIELD(i:i + 5)
Defines an array from a database field referencing the statement marked by label..

FIELD represents a multiple-value field or a field from a periodic group where i specifies the
offset indexwithin the database occurrence. The size of the arraywithin the program is defined
as 6 occurrences (i:i + 5). The database offset index is specified as a variable to allow for the
positioning of the program array within the occurrences of the multiple-value field or periodic
group. For any repositioning of i, a new access must be made to the database using a GET or
GET SAME statement.

Natural allows for the definition of arrays where the index does not have to begin with 1. At
runtime, Natural checks that index values specified in the reference do not exceed the maximum
size of dimensions as specified in the definition.

Notes:

1. For compatibilitywith earlierNatural versions, an array rangemay be specified using a hyphen
(-) instead of a colon (:).

2. A mix of both notations, however, is not permitted.

3. The hyphen notation is only allowed in reporting mode (but not in a DEFINE DATA statement).

The maximum index value is 1,073,741,824. The maximum size of a data area per programming
object is 1,073,741,824 bytes (1 GB).

Simple arithmetic expressions using the plus (+) and minus (-) operators may be used in index
references. When arithmetic expressions are used as indices, these operators must be preceded
and followed by a blank.

Arrays in group structures are resolved by Natural field by field, not group occurrence by group
occurrence.

Example of Group Array Resolution:

DEFINE DATA LOCAL
1 #GROUP (1:2)
2 #FIELDA (A5/1:2)
2 #FIELDB (A5)

END-DEFINE
...

If the group defined above were output in a WRITE statement:

WRITE #GROUP (*)

the occurrences would be output in the following order:

#FIELDA(1,1) #FIELDA(1,2) #FIELDA(2,1) #FIELDA(2,2) #FIELDB(1) #FIELDB(2)

and not:

#FIELDA(1,1) #FIELDA(1,2) #FIELDB(1) #FIELDA(2,1) #FIELDA(2,2) #FIELDB(2)

Array Referencing - Examples:

1. #ARRAY (1)
References the first occurrence of a one-dimensional array.

2. #ARRAY (7:12)
References the seventh to twelfth occurrence of a one-dimensional array.

3. #ARRAY (i + 5)
References the i+fifth occurrence of a one-dimensional array.

4. #ARRAY (5,3:7,1:4)
Reference is made within a three dimensional array to occurrence 5 in the first dimension, oc-
currences 3 to 7 (5 occurrences) in the second dimension and 1 to 4 (4 occurrences) in the third
dimension.

5. An asterisk may be used to reference all occurrences within a dimension:

DEFINE DATA LOCAL 
1 #ARRAY1 (N5/1:4,1:4) 
1 #ARRAY2 (N5/1:4,1:4) 
END-DEFINE 
... 
ADD #ARRAY1 (2,*) TO #ARRAY2 (4,*) 
... ↩

Using a Slash before an Array Occurrence

If a variable name is followed by a 4-digit number enclosed in parentheses, Natural interprets this
number as a line-number reference to a statement. Therefore a 4-digit array occurrence must be
preceded by a slash (/) to indicate that it is an array occurrence; for example:

#ARRAY(/1000)

not:

#ARRAY(1000)

because the latter would be interpreted as a reference to source code line 1000.

If an index variable name could bemisinterpreted as a format/length specification, a slash (/) must
be used to indicate that an index is being specified. If, for example, the occurrence of an array is
defined by the value of the variable N7, the occurrence must be specified as:

#ARRAY (/N7)

not:

#ARRAY (N7)

because the latter would be misinterpreted as the definition of a 7-byte numeric field.

Referencing a Database Array

The following topics are covered below:

■ Referencing Multiple-Value Fields and Periodic-Group Fields
■ Referencing Arrays Defined with Constants
■ Referencing Arrays Defined with Variables
■ Referencing Multiple-Defined Arrays

Note: Before executing the following example programs, please run the program INDEXTST

in the library SYSEXPG to create an example record that uses 10 different language codes.

Referencing Multiple-Value Fields and Periodic-Group Fields

Amultiple-value field or periodic-group fieldwithin a view/DDMmay be defined and referenced
using various index notations.

For example, the first to tenth values and the Ith to Ith+10 values of the same multiple-value
field/periodic-group field of a database record:

DEFINE DATA LOCAL
1 I (I2)
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 LANG (1:10)
2 LANG (I:I+10)

END-DEFINE

or:

RESET I (I2)
...
READ EMPLOYEES
OBTAIN LANG(1:10) LANG(I:I+10)

Notes:

1. The same lower bound index may only be used once per array (this applies to constant indexes
as well as variable indexes).

2. For an array definition using a variable index, the lower bound must be specified using the
variable by itself, and the upper boundmust be specified using the same variable plus a constant.

Referencing Arrays Defined with Constants

An array definedwith constants may be referenced using either constants or variables. The upper
bound of the array cannot be exceeded. The upper boundwill be checked byNatural at compilation
time if a constant is used.

Reporting Mode Example:

** Example 'INDEX1R': Array definition with constants (reporting mode)
***********************************************************************
*
READ (1) EMPLOYEES WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

OBTAIN LANG (1:10)
/*
WRITE 'LANG(1:10):' LANG (1:10) //
WRITE 'LANG(1) :' LANG (1) / 'LANG(5:9) :' LANG (5:9)

LOOP
*
END

Structured Mode Example:

** Example 'INDEX1S': Array definition with constants (structured mode)
***********************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY
2 LANG (1:10)

END-DEFINE
*
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

WRITE 'LANG(1:10):' LANG (1:10) //
WRITE 'LANG(1) :' LANG (1) / 'LANG(5:9) :' LANG (5:9)

END-READ
END

If a multiple-value field or periodic-group field is defined several times using constants and is to
be referenced using variables, the following syntax is used.

Reporting Mode Example:

** Example 'INDEX2R': Array definition with constants (reporting mode)
** (multiple definition of same database field)
***********************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 LANG (1:5)
2 LANG (4:8)

END-DEFINE
*
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

DISPLAY 'NAME' NAME
'LANGUAGE/1:3' LANG (1.1:3)
'LANGUAGE/6:8' LANG (4.3:5)

LOOP
*
END

Structured Mode Example:

** Example 'INDEX2S': Array definition with constants (structured mode)
** (multiple definition of same database field)
***********************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 LANG (1:5)
2 LANG (4:8)

END-DEFINE
*
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

DISPLAY 'NAME' NAME
'LANGUAGE/1:3' LANG (1.1:3)
'LANGUAGE/6:8' LANG (4.3:5)

END-READ
*
END

Referencing Arrays Defined with Variables

Multiple-value fields or periodic-group fields in arrays definedwith variables must be referenced
using the same variable.

Reporting Mode Example:

** Example 'INDEX3R': Array definition with variables (reporting mode)
***********************************************************************
RESET I (I2)
*
I := 1
READ (1) EMPLOYEES WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

OBTAIN LANG (I:I+10)
/*
WRITE 'LANG(I) :' LANG (I) /

'LANG(I+5:I+7):' LANG (I+5:I+7)
LOOP
*
END

Structured Mode Example:

** Example 'INDEX3S': Array definition with variables (structured mode)
***********************************************************************
DEFINE DATA LOCAL
1 I (I2)
*
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 LANG (I:I+10)

END-DEFINE
*
I := 1
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

WRITE 'LANG(I) :' LANG (I) /
'LANG(I+5:I+7):' LANG (I+5:I+7)

END-READ
END

If a different index is to be used, an unambiguous reference to the first encountered definition of
the array with variable index must be made. This is done by qualifying the index expression as
shown below.

Reporting Mode Example:

** Example 'INDEX4R': Array definition with variables (reporting mode)
***********************************************************************
RESET I (I2) J (I2)
*
I := 2
J := 3
*
READ (1) EMPLOYEES WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

OBTAIN LANG (I:I+10)
/*
WRITE 'LANG(I.J) :' LANG (I.J) /

'LANG(I.1:5):' LANG (I.1:5)
LOOP
*
END

Structured Mode Example:

** Example 'INDEX4S': Array definition with variables (structured mode)
***********************************************************************
DEFINE DATA LOCAL
1 I (I2)
1 J (I2)
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 LANG (I:I+10)

END-DEFINE
*
I := 2
J := 3
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

WRITE 'LANG(I.J) :' LANG (I.J) /
'LANG(I.1:5):' LANG (I.1:5)

END-READ
END

The expression I. is used to create an unambiguous reference to the array definition and “positions”
to the first value within the read array range (LANG(I.1:5)).

The current content of I at the time of the database access determines the starting occurrence of
the database array.

Referencing Multiple-Defined Arrays

For multiple-defined arrays, a reference with qualification of the index expression is usually ne-
cessary to ensure an unambiguous reference to the desired array range.

Reporting Mode Example:

** Example 'INDEX5R': Array definition with constants (reporting mode)
** (multiple definition of same database field)
***********************************************************************
DEFINE DATA LOCAL /* For reporting mode programs
1 EMPLOY-VIEW VIEW OF EMPLOYEES /* DEFINE DATA is recommended

2 NAME /* to use multiple definitions
2 CITY /* of same database field
2 LANG (1:10)
2 LANG (5:10)

*
1 I (I2)
1 J (I2)
END-DEFINE
*
I := 1
J := 2

*
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

WRITE 'LANG(1.1:10) :' LANG (1.1:10) /
'LANG(1.I:I+2):' LANG (1.I:I+2) //

WRITE 'LANG(5.1:5) :' LANG (5.1:5) /
'LANG(5.J) :' LANG (5.J)

LOOP
END

Structured Mode Example:

** Example 'INDEX5S': Array definition with constants (structured mode)
** (multiple definition of same database field)
***********************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 LANG (1:10)
2 LANG (5:10)

*
1 I (I2)
1 J (I2)
END-DEFINE
*
*
I := 1
J := 2
*
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'

WRITE 'LANG(1.1:10) :' LANG (1.1:10) /
'LANG(1.I:I+2):' LANG (1.I:I+2) //

WRITE 'LANG(5.1:5) :' LANG (5.1:5) /
'LANG(5.J) :' LANG (5.J)

END-READ
END

A similar syntax is also used if multiple-value fields or periodic-group fields are defined using
index variables.

Reporting Mode Example:

** Example 'INDEX6R': Array definition with variables (reporting mode)
** (multiple definition of same database field)
***********************************************************************
DEFINE DATA LOCAL
1 I (I2) INIT <1>
1 J (I2) INIT <2>
1 N (I2) INIT <1>
1 EMPLOY-VIEW VIEW OF EMPLOYEES /* For reporting mode programs

2 NAME /* DEFINE DATA is recommended

2 CITY /* to use multiple definitions
2 LANG (I:I+10) /* of same database field
2 LANG (J:J+5)
2 LANG (4:5)

*
END-DEFINE
*
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'
*

WRITE 'LANG(I.I) :' LANG (I.I) /
'LANG(1.I:I+2):' LANG (I.I:I+10) //

*
WRITE 'LANG(J.N) :' LANG (J.N) /

'LANG(J.2:4) :' LANG (J.2:4) //
*

WRITE 'LANG(4.N) :' LANG (4.N) /
'LANG(4.N:N+1):' LANG (4.N:N+1) /

LOOP
END

Structured Mode Example:

** Example 'INDEX6S': Array definition with variables (structured mode)
** (multiple definition of same database field)
***********************************************************************
DEFINE DATA LOCAL
1 I (I2) INIT <1>
1 J (I2) INIT <2>
1 N (I2) INIT <1>
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 LANG (I:I+10)
2 LANG (J:J+5)
2 LANG (4:5)

*
END-DEFINE
*
READ (1) EMPLOY-VIEW WITH NAME = 'WINTER' WHERE CITY = 'LONDON'
*

WRITE 'LANG(I.I) :' LANG (I.I) /
'LANG(1.I:I+2):' LANG (I.I:I+10) //

*
WRITE 'LANG(J.N) :' LANG (J.N) /

'LANG(J.2:4) :' LANG (J.2:4) //
*

WRITE 'LANG(4.N) :' LANG (4.N) /
'LANG(4.N:N+1):' LANG (4.N:N+1) /

END-READ
END

Referencing the Internal Count for a Database Array (C* Notation)

It is sometimes necessary to reference a multiple-value field and/or a periodic group without
knowing how many values/occurrences exist in a given record. Adabas maintains an internal
count of the number of values of each multiple-value field and the number of occurrences of each
periodic group. This countmay be referenced by specifying C* immediately before the field name.

Note concerning databases other than Adabas:

The explicit format and length permitted to declare a C* field is either

■ integer (I) with a length of 2 bytes (I2) or 4 bytes (I4),
■ numeric (N) or packed (P) with only integer (but no precision) digits; for example (N3).

If no explicit format and length is supplied, format/length (N3) is assumed as default.

Examples:

Returns the count of the number of values for the multiple-value field LANG.C*LANG

Returns the count of the number of occurrences for the periodic group INCOME.C*INCOME

Returns the count of the number of values for the multiple-value field BONUS in periodic
group occurrence 1 (assuming that BONUS is amultiple-value fieldwithin a periodic group.)

C*BONUS(1)

Example Program Using the C* Variable:

** Example 'CNOTX01': C* Notation
************************************************************************
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 C*INCOME
2 INCOME

3 SALARY (1:5)
3 C*BONUS (1:2)
3 BONUS (1:2,1:2)

2 C*LANG
2 LANG (1:2)

*
1 #I (N1)
END-DEFINE
*
LIMIT 2
READ EMPL-VIEW BY CITY

/*
WRITE NOTITLE 'NAME:' NAME /

'NUMBER OF LANGUAGES SPOKEN:' C*LANG 5X
'LANGUAGE 1:' LANG (1) 5X
'LANGUAGE 2:' LANG (2)

/*
WRITE 'SALARY DATA:'
FOR #I FROM 1 TO C*INCOME

WRITE 'SALARY' #I SALARY (1.#I)
END-FOR
/*
WRITE 'THIS YEAR BONUS:' C*BONUS(1) BONUS (1,1) BONUS (1,2)

/ 'LAST YEAR BONUS:' C*BONUS(2) BONUS (2,1) BONUS (2,2)
SKIP 1

END-READ
END

Output of Program CNOTX01:

NAME: SENKO
NUMBER OF LANGUAGES SPOKEN: 1 LANGUAGE 1: ENG LANGUAGE 2:
SALARY DATA:
SALARY 1 36225
SALARY 2 29900
SALARY 3 28100
SALARY 4 26600
SALARY 5 25200
THIS YEAR BONUS: 0 0 0
LAST YEAR BONUS: 0 0 0

NAME: CANALE
NUMBER OF LANGUAGES SPOKEN: 2 LANGUAGE 1: FRE LANGUAGE 2: ENG
SALARY DATA:
SALARY 1 202285
THIS YEAR BONUS: 1 23000 0
LAST YEAR BONUS: 0 0 0

C* for Multiple-Value Fields Within Periodic Groups

For a multiple-value field within a periodic group, you can also define a C* variable with an index
range specification.

The following examples use the multiple-value field BONUS, which is part of the periodic group
INCOME. All three examples yield the same result.

Example 1 - Reporting Mode:

** Example 'CNOTX02': C* Notation (multiple-value fields)
************************************************************************
*
LIMIT 2
READ EMPLOYEES BY CITY

OBTAIN C*BONUS (1:3)
BONUS (1:3,1:3)

/*
DISPLAY NAME C*BONUS (1:3) BONUS (1:3,1:3)

LOOP
*
END

Example 2 - Structured Mode:

** Example 'CNOTX03': C* Notation (multiple-value fields)
************************************************************************
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 INCOME (1:3)

3 C*BONUS
3 BONUS (1:3)

END-DEFINE
*
LIMIT 2
READ EMPL-VIEW BY CITY

/*
DISPLAY NAME C*BONUS (1:3) BONUS (1:3,1:3)

END-READ
*
END

Example 3 - Structured Mode:

** Example 'CNOTX04': C* Notation (multiple-value fields)
************************************************************************
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 C*BONUS (1:3)
2 INCOME (1:3)

3 BONUS (1:3)
END-DEFINE
*
LIMIT 2
READ EMPL-VIEW BY CITY

/*
DISPLAY NAME C*BONUS (*) BONUS (*,*)

END-READ
*
END

Caution: As the Adabas format buffer does not permit ranges for count fields, they are
generated as individual fields; therefore a C* index range for a large array may cause an
Adabas format buffer overflow.

Qualifying Data Structures

To identify a field when referencing it, you may qualify the field; that is, before the field name,
you specify the name of the level-1 data element in which the field is located and a period.

If a field cannot be identified uniquely by its name (for example, if the same field name is used in
multiple groups/views), you must qualify the field when you reference it.

The combination of level-1 data element and field name must be unique.

Example:

DEFINE DATA LOCAL
1 FULL-NAME

2 LAST-NAME (A20)
2 FIRST-NAME (A15)

1 OUTPUT-NAME
2 LAST-NAME (A20)
2 FIRST-NAME (A15)

END-DEFINE
...
MOVE FULL-NAME.LAST-NAME TO OUTPUT-NAME.LAST-NAME
...

The qualifier must be a level-1 data element.

Example:

DEFINE DATA LOCAL
1 GROUP1

2 SUB-GROUP
3 FIELD1 (A15)
3 FIELD2 (A15)

END-DEFINE
...
MOVE 'ABC' TO GROUP1.FIELD1
...

Qualifying a Database Field:

If you use the same name for a user-defined variable and a database field (which you should not
do anyway), you must qualify the database field when you want to reference it

Caution: If you do not qualify the database field when you want to reference it, the user-
defined variable will be referenced instead.

Examples of User-Defined Variables

DEFINE DATA LOCAL 
1 #A1 (A10)       /* Alphanumeric, 10 positions. 
1 #A2 (B4)        /* Binary, 4 positions. 
1 #A3 (P4)        /* Packed numeric, 4 positions and 1 sign position. 
1 #A4 (N7.2)      /* Unpacked numeric,  
                  /* 7 positions before and 2 after decimal point. 
1 #A5 (N7.)       /* Invalid definition!!! 
1 #A6 (P7.2)      /* Packed numeric, 7 positions before and 2 after decimal point  
                  /* and 1 sign position. 
1 #INT1 (I1)      /* Integer, 1 byte. 
1 #INT2 (I2)      /* Integer, 2 bytes. 
1 #INT3 (I3)      /* Invalid definition!!! 
1 #INT4 (I4)      /* Integer, 4 bytes. 
1 #INT5 (I5)      /* Invalid definition!!! 
1 #FLT4 (F4)      /* Floating point, 4 bytes. 
1 #FLT8 (F8)      /* Floating point, 8 bytes. 
1 #FLT2 (F2)      /* Invalid definition!!! 
1 #DATE (D)       /* Date (internal format/length P6). 
1 #TIME (T)       /* Time (internal format/length P12). 
1 #SWITCH (L)     /* Logical, 1 byte (TRUE or FALSE). 
                  /*
END-DEFINE ↩

call-name

(< [([prototype-cast] [intermediate-result-definition])] [parameter] [,[parameter]] ... >)

[array-index-expression]

Related topics:

■ Object type Function
■ User-defined Functions
■ DEFINE FUNCTION statement
■ DEFINE PROTOTYPE statement

Calling User-Defined Functions

A function call performs an invocation of a user-defined function, a special kind of a subroutine
which is implemented in a separate programming object of type function.

Usually, a function call is provided with parameters and returns a result. It may be used within
aNatural statement instead of a read-only operand. In this case, the function has to return a result,
which is then processed by the statement like a field containing the same value.

It is also possible to use a function call stand-alone in place of a Natural statement. Then the
function call need not return a result value.

There are different ways to call a function:

■ Symbolic Function Call
■ Variable Function Call

Symbolic Function Call

When the specified call-name represents the name of the function to be executed, the function
call is denoted as a symbolic function call. Be aware, the function name is defined in the DEFINE
FUNCTION statement and does not necessarily have to match the name of the module in which the
function is defined. The function interface result and parameter layouts used to resolve a function
call are either specified by a DEFINE PROTOTYPE statement with the same name or loaded automat-
ically if the cataloged version of the called function exists.

Variable Function Call

A function can also be called in an indirect form, denoted as a variable function call. In this case,
the specified call-name is an alphanumeric variable, which contains the name of the called function
at execution time. This variable has to be referenced in a prototype definition (DEFINE PROTOTYPE
statement) in conjunction with the keyword VARIABLE. If this prototype does not contain the right
result layout or parameter layout, another prototype can be assigned with the (PT=) option.

Function Result

According to the function definition, a function call may return a single result field. This can be a
scalar value or an array field, which is processed like a temporary field in the statement where
the function call is embedded. If the result is an array, the function call must be immediately fol-
lowed by an array-index-expression addressing the required occurrences.

For example, to access the first occurrence of the array returned:

#FCT(<#A,#B>)(1)

Evaluation Sequence

If a single or multiple function calls are used within a statement, the evaluation of all functions is
performed in a separate step before the statement execution will be started. The function calls are
executed in the same order in which they appear in the statement.

Restrictions

Function calls are not allowed in the following situations:

■ in positions where the operand value is changed by the Natural statement (for example, MOVE
1 TO #FCT(<..>) );

■ in a DEFINE DATA statement;
■ in a database access statement (READ, FIND, SELECT, UPDATE, STORE, etc.);
■ in an AT BREAK or IF BREAK statement;
■ as an argument of Natural system functions (AVER, SUM, *TRIM, etc.);
■ as an array index notation;
■ as a parameter of a function call.

If a function call is used in an INPUT statement, the return value will be treated like a constant
value. This leads to an automatic assignment of attribute (AD=O) to make this field write-protected
(for output only).

Syntax Description

A function call may consist of the following syntax elements:

■ call-name
■ prototype-cast
■ intermediate-result-definition
■ Parameter(s)
■ array-index-expression

call-name

function-name

variable-name

Operand Definition Table:

Dynamic DefinitionReferencing PermittedPossible FormatsPossible StructureOperand

noyesUAASvariable-name

Syntax Element Description:

DescriptionSyntax Element

Function Name:function-name

function-name is the name of the function to be called. Be aware, the function name
is defined in the DEFINE FUNCTION statement and does not necessarily have to match
the name of the module in which the function is defined. If a prototype with the same
name has already been defined before, this prototype is used to pick up the function
interface; that is, the result layout definition and the required parameters.

Variable Function Name:variable-name

variable-name is the name of the variable containing the real name of the function to
be called at runtime. In order to declare the call name as a variable name, a prototype
with the same name, which is classified with keyword VARIABLE, must be defined
before.

prototype-cast

prototype-name
PT=

prototype-variable-name

For every function call, Natural tries to get information on the function result and the calling
parameters. This leads to a search for a prototype with the same name as the call name. If such a
prototype is not available or if this prototype does not contain the right result layout or parameter
layout, another prototype can be linked to the function call with a (PT=) clause. In this case, the
referenced prototype steps in place and is used to define the function result layout and the para-
meter layout. The calling mode (symbolic or variable) declared in the referenced prototype is ig-
nored.

Syntax Element Description:

DescriptionSyntax Element

Prototype Name:prototype-name

prototype-name is the identifier of the prototype whose result and
parameters layouts are to be used.

Prototype Variable Name:prototype-variable-name

prototype-variable-name is the name of an alphanumeric field used as
function name in a function call. At execution time it has to contain the name
of the function to be called.

The name has to follow the same rules which apply for a variable reference,
including field qualification, but without array index references.

intermediate-result-definition

format-length [/array-definition]

IR=

[(array-definition)] HANDLE OF OBJECT

[/array-definition]) DYNAMIC(
A
U
B

This clause can be used to specify the format-length/array-definition of the return value for
a function call without using an explicit or implicit prototype definition, that is, it enables the ex-
plicit specification of the function result layout. If a prototype is available for this function call or
if the cataloged version of the called function exists, the result format given in the (IR=) clause is
checked for move compatibility.

Syntax Element Description:

DescriptionSyntax Element

Format/Length Definition:format-length

The format and length of the field.

For information on format/length definition of user-defined variables, see Format
and Length of User-Defined Variables.

Array Dimension Definition:array-definition

With an array-definition, you define the lower and upper bounds of the
dimensions in an array definition.

See Array Dimension Definition in the Statements documentation.

Handle of Object:HANDLE OF OBJECT

Used in conjunction with NaturalX.

Data Format:A, B or U

Possible formats are alphanumeric, binary or Unicode for dynamic variables.

Dynamic Variable:DYNAMIC

A field may be defined as DYNAMIC.

For further information onprocessingdynamic variables, see Introduction toDynamic
Variables and Fields.

Parameter(s)

Parameters are the data values passed to the function. They can be provided as a constant value
or a variable, depending onwhat is defined in the DEFINE DATA PARAMETER section of the function
definition. The semantic and syntactic rules which apply to the function parameters are the same
as described in the parameters section of subprograms; see Parameters in the description of the
CALLNAT statement.

nX

M
O )(AD=operand

A

Operand Definition Table:

Dynamic DefinitionReferencing
Permitted

Possible FormatsPossible StructureOperand

noyesOGCLTDBFIPNAGASCoperand

Syntax Element Description:

DescriptionSyntax
Element

Parameters to be Skipped:nX

With the notation nX you can specify that the next n parameters are to be skipped (for example,
1X to skip the next parameter, or 3X to skip the next three parameters); this means that for the next
n parameters no values are passed to the function.

A parameter that is to be skipped must be defined with the keyword OPTIONAL in the function's
DEFINE DATA PARAMETER statement. OPTIONALmeans that a value can - but need not - be passed
from the invoking object to such a parameter.

Attribute Definition:AD=

If operand is a variable, you can mark it in one of the following ways:

Non-modifiable:AD=O

See session parameter AD=O.

Note: Internally, AD=O is processed in the same way as
BY VALUE (see the section
parameter-data-definition in the description of
the DEFINE DATA statement).

Modifiable:AD=M

See session parameter AD=M.

This is the default setting.

Input only:AD=A

See session parameter AD=A.

Note: If operand is a constant, the attribute definition AD cannot be explicitly specified. For
constants, AD=O always applies.

array-index-expression

If the result returned by the function call is an array, an index notationmust be provided to address
the demanded array occurrences.

For details, refer to Index Notation in User-Defined Variables in the Programming Guide.

Example

The example FUNCEX01 uses the functions F#ADDITION, F#CHAR, F#EVEN and F#TEXT, which are
defined in the sample functions FUNCEX02, FUNCEX03, FUNCEX04 and FUNCEX05.

** Example 'FUNCEX01': Function call (Program)
************************************************************************
DEFINE DATA LOCAL

1 #NUM (I2) INIT <5>
1 #A (I2) INIT <1>
1 #B (I2) INIT <2>
1 #C (I2) INIT <3>
1 #CHAR (A1) INIT <'A'>

END-DEFINE
*
IF #NUM = F#ADDITION(<#A,#B,#C>) /* Function with three parameters.

WRITE 'Sum of #A,#B,#C' #NUM
ELSE

IF #NUM = F#ADDITION(<1X,#B,#C>) /* Function with optional parameters.
WRITE 'Sum of #B,#C' #NUM

END-IF
END-IF
*
DECIDE ON FIRST #CHAR

VALUE F#CHAR (<>)(1) /* Function with result array.
WRITE 'Character A found'

VALUE F#CHAR (<>)(2)
WRITE 'Character B found'

NONE
IGNORE

END-DECIDE
*
IF F#EVEN(<#B>) /* Function with logical result value.

WRITE #B 'is an even number'
END-IF
*
F#TEXT(<'Hello', '*'>) /* Function used as a statement.
*
WRITE F#TEXT(<(IR=A12) 'Good'>) /* Function with intermediate result.
*
END

The function F#ADDITION is defined in the sample function FUNCEX02 in library SYSEXPG.

** Example 'FUNCEX02': Function call (Function)
************************************************************************
DEFINE FUNCTION F#ADDITION

RETURNS (I2)
DEFINE DATA PARAMETER

1 #PARM1 (I2) OPTIONAL
1 #PARM2 (I2) OPTIONAL
1 #PARM3 (I2) OPTIONAL

END-DEFINE                                                            
  /*                                                                    
  RESET F#ADDITION
  IF #PARM1 SPECIFIED                                                   
    F#ADDITION := F#ADDITION + #PARM1                                             ↩
  
  END-IF                                                                
  IF #PARM2 SPECIFIED                                                   
    F#ADDITION := F#ADDITION + #PARM2                                             ↩
  
  END-IF                                                                
  IF #PARM3 SPECIFIED                                                   
    F#ADDITION := F#ADDITION + #PARM3                                             ↩
  
  END-IF                                                                
  /*                                                                    
END-FUNCTION                                                            
*                                                                       
END ↩

The function F#CHAR is defined in the example function FUNCEX03 in library SYSEXPG.

** Example 'FUNCEX03': Function call (Function)
************************************************************************
DEFINE FUNCTION F#CHAR

RETURNS (A1/1:2)
/*
F#CHAR(1) := 'A'
F#CHAR(2) := 'B'
/*

END-FUNCTION                                                            
*                                                                       
END  ↩

The function F#EVEN is defined in the example function FUNCEX04 in library SYSEXPG.

** Example 'FUNCEX04': Function call (Function)
************************************************************************
DEFINE FUNCTION F#EVEN

RETURNS (L)
DEFINE DATA
PARAMETER

1 #NUM (N4) BY VALUE
LOCAL

1 #REST (I2)
END-DEFINE
/*
DIVIDE 2 INTO #NUM REMAINDER #REST
/*
IF #REST = 0

F#EVEN := TRUE
ELSE

F#EVEN := FALSE
END-IF
/*

END-FUNCTION                                                            
*                                                                       
END ↩

The function F#TEXT is defined in the sample function FUNCEX05 in library SYSEXPG.

** Example 'FUNCEX05': Function call (Function)
************************************************************************
DEFINE FUNCTION F#TEXT

RETURNS (A20) BY VALUE
DEFINE DATA
PARAMETER

1 #TEXT1 (A5) BY VALUE
1 #TEXT2 (A1) BY VALUE OPTIONAL

LOCAL
1 #FRAME (A3)

END-DEFINE
/*
IF #TEXT2 SPECIFIED

MOVE ALL #TEXT2 TO #FRAME
/*
COMPRESS #FRAME #TEXT1 'world' #FRAME INTO F#TEXT
/*
WRITE F#TEXT

ELSE
COMPRESS #TEXT1 'morning' INTO F#TEXT
/*

END-IF
/*

END-FUNCTION                                                            
*                                                                       
END ↩

Output of Program FUNCEX01

Sum of #B,#C 5
Character A found

2 is an even number
*** Hello world ***
Good morning

Purpose of Dynamic Variables

In that the maximum size of large data structures (for example, pictures, sounds, videos) may not
exactly be known at application development time, Natural additionally provides for the definition
of alphanumeric and binary variables with the attribute DYNAMIC. The value space of variables
which are defined with this attribute will be extended dynamically at execution time when it be-
comes necessary (for example, during an assignment operation: #picture1 := #picture2). This
means that large binary and alphanumeric data structures may be processed in Natural without
the need to define a limit at development time. The execution-time allocation of dynamic variables
is of course subject to available memory restrictions. If the allocation of dynamic variables results
in an insufficent memory condition being returned by the underlying operating system, the ON
ERROR statement can be used to intercept this error condition; otherwise, an error message will be
returned by Natural.

The Natural system variable *LENGTH can be used obtain the length (in terms of code units) of the
value space which is currently used for a given dynamic variable. For A and B formats, the size
of one code unit is 1 byte. For U format, the size of one code unit is 2 bytes (UTF-16). Natural
automatically sets *LENGTH to the length of the source operand during assignments in which the
dynamic variable is involved. *LENGTH(field) therefore returns the length (in terms of code units)
currently used for a dynamic Natural field or variable.

If the dynamic variable space is no longer needed, the REDUCE or RESIZE statements can be used
to reduce the space used for the dynamic variable to zero (or any other desired size). If the upper
limit of memory usage is known for a specific dynamic variable, the EXPAND statement can be used
to set the space used for the dynamic variable to this specific size.

If a dynamic variable is to be initialized, the MOVE ALL UNTIL statement should be used for this
purpose.

Definition of Dynamic Variables

Because the actual size of large alphanumeric and binary data structuresmay not be exactly known
at application development time, the definition of dynamic variables of format A, B or U can be
used to manage these structures. The dynamic allocation and extension (reallocation) of large
variables is transparent to the application programming logic. Dynamic variables are defined
without any length.Memorywill be allocated either implicitly at execution time,when the dynamic
variable is used as a target operand, or explicitly with an EXPAND or RESIZE statement.

Dynamic variables can only be defined in a DEFINE DATA statement using the following syntax:

level variable-name ( A ) DYNAMIC

level variable-name ( B ) DYNAMIC

level variable-name ( U ) DYNAMIC

Restrictions:

The following restrictions apply to a dynamic variable:

■ A redefinition of a dynamic variable is not allowed.
■ A dynamic variable may not be contained in a REDEFINE clause.

Value Space Currently Used for a Dynamic Variable

The length (in terms of code units) of the currently used value space of a dynamic variable can be
obtained from the systemvariable *LENGTH. *LENGTH is set to the (used) length of the source operand
during assignments automatically.

Caution: Due to performance considerations, the storage area that is allocated to hold the
value of the dynamic variable may be larger than the value of *LENGTH (used size available
to the programmer). You should not rely on the storage that is allocated beyond the used
length as indicated by *LENGTH: it may be released at any time, even if the respective dynamic
variable is not accessed. It is not possible for theNatural programmer to obtain information
about the currently allocated size. This is an internal value.

*LENGTH(field) returns the used length (in terms of code units) of a dynamic Natural field or
variable. For A and B formats, the size of one code unit is 1 byte. For U format, the size of one code
unit is 2 bytes (UTF-16). *LENGTHmay be used only to get the currently used length for dynamic
variables.

Size Limitation Check

Profile Parameter USIZE

For dynamic variables, a size limitation check at compile time is not possible because no length is
defined for dynamic variables. The size of user buffer area (USIZE) indicates the size of the user
buffer in virtual memory. The user buffer contains all data dynamically allocated by Natural. If a
dynamic variable is allocated or extended at execution time and the USIZE limitation is exceeded,
an error message will be returned.

Allocating/Freeing Memory Space for a Dynamic Variable

The statements EXPAND, REDUCE and RESIZE are used to explicitly allocate and free memory space
for a dynamic variable.

Syntax:

EXPAND [SIZE OF] DYNAMIC [VARIABLE] operand1 TO operand2

REDUCE [SIZE OF] DYNAMIC [VARIABLE] operand1 TO operand2

RESIZE [SIZE OF] DYNAMIC [VARIABLE] operand1 TO operand2

- where operand1 is a dynamic variable and operand2 is a non-negative numeric size value.

EXPAND

Function

The EXPAND statement is used to increase the allocated length of the dynamic variable (operand1)
to the specified length (operand2).

Changing the Specified Size

The length currently used (as indicated by the Natural system variable *LENGTH, see above) for
the dynamic variable is not modified.

If the specified length (operand2) is less than the allocated length of the dynamic variable, the
statement will be ignored.

REDUCE

Function

The REDUCE statement is used to reduce the allocated length of the dynamic variable (operand1)
to the specified length (operand2).

The storage allocated for the dynamic variable (operand1) beyond the specified length (operand2)
may be released at any time, when the statement is executed or at a later time.

Changing the Specified Length

If the length currently used (as indicated by the Natural system variable *LENGTH, see above) for
the dynamic variable is greater than the specified length (operand2), the system variable *LENGTH
of this dynamic variable is set to the specified length. The content of the variable is truncated, but
not modified.

If the given length is larger than the currently allocated storage of the dynamic variable, the
statement will be ignored.

RESIZE

Function

The RESIZE statement adjusts the currently allocated length of the dynamic variable (operand1)
to the specified length (operand2).

Changing the Specified Length

If the specified length is smaller then the used length (as indicated by the Natural system variable
*LENGTH, see above) of the dynamic variable, the used length is reduced accordingly.

If the specified length is larger than the currently allocated length of the dynamic variable, the al-
located length of the dynamic variable is increased. The currently used length (as indicated by the
system variable *LENGTH) of the dynamic variable is not affected and remains unchanged.

If the specified length is the same as the currently allocated length of the dynamic variable, the
execution of the RESIZE statement has no effect.

General Remarks

Generally, the following rules apply:

■ A dynamic alphanumeric field may be used wherever an alphanumeric field is allowed.
■ A dynamic binary field may be used wherever a binary field is allowed.
■ A dynamic Unicode field may be used wherever a Unicode field is allowed.

Exception:

Dynamic variables are not allowed within the SORT statement. To use dynamic variables in a
DISPLAY, WRITE, PRINT, REINPUT or INPUT statement, you must use either the session parameter AL
or EM to define the length of the variable.

The used length (as indicated by the Natural system variable *LENGTH, see Value Space Currently
Used for a Dynamic Variable) and the size of the allocated storage of dynamic variables are equal
to zero until the variable is accessed as a target operand for the first time. Due to assignments or
othermanipulation operations, dynamic variablesmay be firstly allocated or extended (reallocated)
to the exact size of the source operand.

The size of a dynamic variablemay be extended if it is used as amodifiable operand (target operand)
in the following statements:

operand1 (destination operand in an assignment).ASSIGN

See Parameter Transfer with Dynamic Variables (except if AD=O, or if BY VALUE exists in
the corresponding parameter data area).

CALLNAT

operand2, see Processing.COMPRESS

operand1 in the DELETE REPLACE clause.EXAMINE

operand2 (destination operand), see Function.MOVE

(except if AD=O, or if BY VALUE exists in the corresponding parameter data area).PERFORM

operand1 and operand2, see Handling of Large and Dynamic Variables.READ WORK FILE

operand4.SEPARATE

parameter in the INTO Clause.SELECT (SQL)

operand3 (except if AD=O).SEND METHOD

Currently, there is the following limit concerning the usage of large variables:

Parameter size less than 64 KB per parameter (no limit for CALLwith INTERFACE4 option).CALL

In the following sections, the use of dynamic variables is discussed in more detail on the basis of
examples.

Assignments with Dynamic Variables

Generally, an assignment is done in the current used length (as indicated by the Natural system
variable *LENGTH) of the source operand. If the destination operand is a dynamic variable, its
current allocated size is possibly extended in order tomove the source operandwithout truncation.

Example:

#MYDYNTEXT1 := OPERAND
MOVE OPERAND TO #MYDYNTEXT1
/* #MYDYNTEXT1 IS AUTOMATICALLY EXTENDED UNTIL THE SOURCE OPERAND CAN BE COPIED ↩

MOVE ALL, MOVE ALL UNTILwith dynamic target operands are defined as follows:

■ MOVE ALLmoves the source operand repeatedly to the target operand until the used length
(*LENGTH) of the target operand is reached. The system variable *LENGTH is not modified. If
*LENGTH is zero, the statement will be ignored.

■ MOVE ALL operand1 TO operand2 UNTIL operand3moves operand1 repeatedly to operand2
until the length specified in operand3 is reached. If operand3 is greater than *LENGTH(operand2),
operand2 is extended and *LENGTH(operand2) is set to operand3. If operand3 is less than
*LENGTH(operand2), the used length is reduced to operand3. If operand3 equals
*LENGTH(operand2), the behavior is equivalent to MOVE ALL.

Example:

#MYDYNTEXT1 := 'ABCDEFGHIJKLMNO'          /* *LENGTH(#MYDYNTEXT1) = 15
MOVE ALL 'AB' TO #MYDYNTEXT1              /* CONTENT OF #MYDYNTEXT1 = ↩
'ABABABABABABABA';  
                                          /* *LENGTH IS STILL 15
MOVE ALL 'CD' TO #MYDYNTEXT1 UNTIL 6      /* CONTENT OF #MYDYNTEXT1 = 'CDCDCD';
                                          /* *LENGTH = 6
MOVE ALL 'EF' TO #MYDYNTEXT1 UNTIL 10     /* CONTENT OF #MYDYNTEXT1 = 'EFEFEFEFEF'; 
                                          /* *LENGTH = 10

MOVE JUSTIFIED is rejected at compile time if the target operand is a dynamic variable.

MOVE SUBSTR and MOVE TO SUBSTR are allowed. MOVE SUBSTRwill lead to a runtime error if a sub-
string behind the used length of a dynamic variable (*LENGTH) is referenced. MOVE TO SUBSTRwill
lead to a runtime error if a sub-string position behind *LENGTH + 1 is referenced, because this
would lead to an undefined gap in the content of the dynamic variable. If the target operand

should be extended by MOVE TO SUBSTR (for example if the second operand is set to *LENGTH+1),
the third operand is mandatory.

Valid syntax:

#OP2 := *LENGTH(#MYDYNTEXT1)
MOVE SUBSTR (#MYDYNTEXT1, #OP2) TO OPERAND                /* MOVE LAST CHARACTER ↩
TO OPERAND
#OP2 := *LENGTH(#MYDYNTEXT1) + 1
MOVE OPERAND TO SUBSTR(#MYDYNTEXT1, #OP2, #lEN_OPERAND)   /* CONCATENATE OPERAND ↩
TO #MYDYNTEXT1          ↩

Invalid syntax:

#OP2 := *LENGTH(#MYDYNTEXT1) + 1
MOVE SUBSTR (#MYDYNTEXT1, #OP2, 10) TO OPERAND            /* LEADS TO RUNTIME ERROR; ↩
UNDEFINED SUB-STRING  
#OP2 := *LENGTH(#MYDYNTEXT1 + 10) 
MOVE OPERAND TO SUBSTR(#MYDYNTEXT1, #OP2, #EN_OPERAND)    /* LEADS TO RUNTIME ERROR; ↩
UNDEFINED GAP
#OP2 := *LENGTH(#MYDYNTEXT1) + 1
MOVE OPERAND TO SUBSTR(#MYDYNTEXT1, #OP2)                 /* LEADS TO RUNTIME ERROR; ↩
UNDEFINED LENGTH

Assignment Compatibility

Example:

#MYDYNTEXT1   := #MYSTATICVAR1
#MYSTATICVAR1 := #MYDYNTEXT2   ↩

If the source operand is a static variable, the used length of the dynamic destination operand
(*LENGTH(#MYDYNTEXT1)) is set to the format length of the static variable and the source value is
copied in this length including trailing blanks (alphanumeric and Unicode fields) or binary zeros
(for binary fields).

If the destination operand is static and the source operand is dynamic, the dynamic variable is
copied in its currently used length. If this length is less than the format length of the static variable,
the remainder is filled with blanks (for alphanumeric and Unicode fields) or binary zeros (for
binary fields). Otherwise, the value will be truncated. If the currently used length of the dynamic
variable is 0, the static target operand is filled with blanks (for alphanumeric and Unicode fields)
or binary zeros (for binary fields).

Initialization of Dynamic Variables

Dynamic variables can be initialized with blanks (alphanumeric and Unicode fields) or zeros
(binary fields) up to the currently used length (= *LENGTH) using the RESET statement. The system
variable *LENGTH is not modified.

Example:

DEFINE DATA LOCAL
1 #MYDYNTEXT1   (A)  DYNAMIC
END-DEFINE
#MYDYNTEXT1 := 'SHORT TEXT'
WRITE *LENGTH(#MYDYNTEXT1)          /* USED LENGTH = 10
RESET #MYDYNTEXT1                   /* USED LENGTH = 10, VALUE = 10 BLANKS  ↩

To initialize a dynamic variable with a specified value in a specified size, the MOVE ALL UNTIL
statement may be used.

Example:

MOVE ALL 'Y' TO #MYDYNTEXT1 UNTIL 15          /* #MYDYNTEXT1 CONTAINS 15 'Y'S, USED ↩
LENGTH = 15 ↩

String Manipulation with Dynamic Alphanumeric Variables

If a modifiable operand is a dynamic variable, its current allocated size is possibly extended in
order to perform the operation without truncation or an error message. This is valid for the con-
catenation (COMPRESS) and separation of dynamic alphanumeric variables (SEPARATE).

Example:

** Example 'DYNAMX01': Dynamic variables (with COMPRESS and SEPARATE)
************************************************************************
DEFINE DATA LOCAL
1 #MYDYNTEXT1 (A) DYNAMIC
1 #TEXT (A20)
1 #DYN1 (A) DYNAMIC
1 #DYN2 (A) DYNAMIC
1 #DYN3 (A) DYNAMIC
END-DEFINE
*
MOVE ' HELLO WORLD ' TO #MYDYNTEXT1
WRITE #MYDYNTEXT1 (AL=25) 'with length' *LENGTH (#MYDYNTEXT1)
/* dynamic variable with leading and trailing blanks
*

MOVE ' HELLO WORLD ' TO #TEXT
*
MOVE #TEXT TO #MYDYNTEXT1
WRITE #MYDYNTEXT1 (AL=25) 'with length' *LENGTH (#MYDYNTEXT1)
/* dynamic variable with whole variable length of #TEXT
*
COMPRESS #TEXT INTO #MYDYNTEXT1
WRITE #MYDYNTEXT1 (AL=25) 'with length' *LENGTH (#MYDYNTEXT1)
/* dynamic variable with leading blanks of #TEXT
*
*
#MYDYNTEXT1 := 'HERE COMES THE SUN'
SEPARATE #MYDYNTEXT1 INTO #DYN1 #DYN2 #DYN3 IGNORE
*
WRITE / #MYDYNTEXT1 (AL=25) 'with length' *LENGTH (#MYDYNTEXT1)
WRITE #DYN1 (AL=25) 'with length' *LENGTH (#DYN1)
WRITE #DYN2 (AL=25) 'with length' *LENGTH (#DYN2)
WRITE #DYN3 (AL=25) 'with length' *LENGTH (#DYN3)
/* #DYN1, #DYN2, #DYN3 are automatically extended or reduced
*
EXAMINE #MYDYNTEXT1 FOR 'SUN' REPLACE 'MOON'
WRITE / #MYDYNTEXT1 (AL=25) 'with length' *LENGTH (#MYDYNTEXT1)
/* #MYDYNTEXT1 is automatically extended or reduced
*
END

Note: In case of non-dynamic variables, an error message may be returned.

Logical Condition Criterion (LCC) with Dynamic Variables

Generally, a read-only operation (such as a comparison) with a dynamic variable is done with its
currently used length. Dynamic variables are processed like static variables if they are used in a
read-only (non-modifiable) context.

Example:

IF #MYDYNTEXT1 = #MYDYNTEXT2 OR #MYDYNTEXT1 = "**" THEN ...
IF #MYDYNTEXT1 < #MYDYNTEXT2 OR #MYDYNTEXT1 < "**" THEN ...
IF #MYDYNTEXT1 > #MYDYNTEXT2 OR #MYDYNTEXT1 > "**" THEN ...

Trailing blanks for alphanumeric andUnicode variables or leading binary zeros for binary variables
are processed in the same way for static and dynamic variables. For example, alphanumeric vari-
ables containing the values AA and AA followed by a blank will be considered being equal, and
binary variables containing the values H’0000031’ and H’3031’will be considered being equal.
If a comparison result should only be TRUE in case of an exact copy, the used lengths of the dynamic
variables have to be compared in addition. If one variable is an exact copy of the other, their used
lengths are also equal.

Programming Guide130

Using Dynamic and Large Variables



Example:

#MYDYNTEXT1 := 'HELLO' /* USED LENGTH IS 5
#MYDYNTEXT2 := 'HELLO ' /* USED LENGTH IS 10
IF #MYDYNTEXT1 = #MYDYNTEXT2 THEN ... /* TRUE
IF #MYDYNTEXT1 = #MYDYNTEXT2 AND

*LENGTH(#MYDYNTEXT1) = *LENGTH(#MYDYNTEXT2) THEN ... /* FALSE

Two dynamic variables are compared position by position (from left to right for alphanumeric
variables, and right to left for binary variables) up to the minimum of their used lengths. The first
position where the variables are not equal determines if the first or the second variable is greater
than, less than or equal to the other. The variables are equal if they are equal up to the minimum
of their used lengths and the remainder of the longer variable contains only blanks for alphanu-
meric dynamic variables or binary zeros for binary dynamic variables. To compare two Unicode
dynamic variables, trailing blanks are removed fromboth values before the ICU collation algorithm
is used to compare the two resulting values. See also Logical Condition Criteria in the Unicode and
Code Page Support documentation.

Example:

#MYDYNTEXT1 := 'HELLO1' /* USED LENGTH IS 6
#MYDYNTEXT2 := 'HELLO2' /* USED LENGTH IS 10
IF #MYDYNTEXT1 < #MYDYNTEXT2 THEN ... /* TRUE
#MYDYNTEXT2 := 'HALLO'
IF #MYDYNTEXT1 > #MYDYNTEXT2 THEN ... /* TRUE

Comparison Compatibility

Comparisons between dynamic and static variables are equivalent to comparisons between dy-
namic variables. The format length of the static variable is interpreted as its used length.

Example:

#MYSTATTEXT1 :=  'HELLO'                      /* FORMAT LENGTH OF MYSTATTEXT1 IS ↩
A20  
#MYDYNTEXT1  :=  'HELLO'                      /* USED LENGTH IS 5
IF  #MYSTATTEXT1  =  #MYDYNTEXT1  THEN  ...   /* TRUE
IF  #MYSTATTEXT1  >  #MYDYNTEXT1  THEN  ...   /* FALSE

AT/IF-BREAK of Dynamic Control Fields

The comparison of the break control fieldwith its old value is performed position by position from
left to right. If the old and the new value of the dynamic variable are of different length, then for
comparison, the value with shorter length is padded to the right (with blanks for alphanumeric
and Unicode dynamic values or binary zeros for binary values).

In case of an alphanumeric or Unicode break control field, trailing blanks are not significant for
the comparison, that is, trailing blanks do not mean a change of the value and no break occurs.

In case of a binary break control field, trailing binary zeros are not significant for the comparison,
that is, trailing binary zeros do not mean a change of the value and no break occurs.

Parameter Transfer with Dynamic Variables

Dynamic variables may be passed as parameters to a called program object (CALLNAT, PERFORM).
A call-by-reference is possible because the value space of a dynamic variable is contiguous. A call-
by-value causes an assignment with the variable definition of the caller as the source operand and
the parameter definition as the destination operand. A call-by-value result causes in addition the
movement in the opposite direction.

For a call-by-reference, both definitionsmust be DYNAMIC. If only one of them is DYNAMIC, a runtime
error is raised. In the case of a call-by-value (result), all combinations are possible. The following
table illustrates the valid combinations:

Call By Reference

ParameterCaller
DynamicStatic

NoYesStatic

YesNoDynamic

The formats of dynamic variables A or B must match.

Call by Value (Result)

ParameterCaller
DynamicStatic

YesYesStatic

YesYesDynamic

Note: In the case of static/dynamic or dynamic/static definitions, a value truncation may
occur according to the data transfer rules of the appropriate assignments.

Example 1:

** Example 'DYNAMX02': Dynamic variables (as parameters)
************************************************************************
DEFINE DATA LOCAL
1 #MYTEXT (A) DYNAMIC
END-DEFINE
*
#MYTEXT := '123456' /* extended to 6 bytes, *LENGTH(#MYTEXT) = 6
*
CALLNAT 'DYNAMX03' USING #MYTEXT                                        
*                                                                       
WRITE *LENGTH(#MYTEXT)          /* *LENGTH(#MYTEXT) = 8                 
*                                                                       
END                                                                  ↩

Subprogram DYNAMX03:

** Example 'DYNAMX03': Dynamic variables (as parameters)                
************************************************************************
DEFINE DATA PARAMETER                                                   
1 #MYPARM (A) DYNAMIC  BY VALUE RESULT                                  
END-DEFINE                                                              
*                                                                       
WRITE *LENGTH(#MYPARM)                  /* *LENGTH(#MYPARM) = 6         
#MYPARM := '1234567'                    /* *LENGTH(#MYPARM) = 7         
#MYPARM := '12345678'                   /* *LENGTH(#MYPARM) = 8         
EXPAND DYNAMIC VARIABLE #MYPARM TO 10   /* 10 bytes are allocated       
*                                                                       
WRITE *LENGTH(#MYPARM)                  /* *LENGTH(#MYPARM) = 8         
*                                                                       
/* content of #MYPARM is moved back to #MYTEXT                          
/* used length of #MYTEXT = 8                                           
*                                                                       
END                                                                    ↩

Example 2:

** Example 'DYNAMX04': Dynamic variables (as parameters)
************************************************************************
DEFINE DATA LOCAL
1 #MYTEXT (A) DYNAMIC
END-DEFINE
*
#MYTEXT := '123456' /* extended to 6 bytes, *LENGTH(#MYTEXT) = 6
*
CALLNAT 'DYNAMX05' USING #MYTEXT                                        
*                                                                       
WRITE *LENGTH(#MYTEXT)          /* *LENGTH(#MYTEXT) = 8                 
                                /* at least 10 bytes are                
                                /* allocated (extended in DYNAMX05)     
*                                                                       
END                                                               ↩

Subprogram DYNAMX05:

** Example 'DYNAMX05': Dynamic variables (as parameters)                
************************************************************************
DEFINE DATA PARAMETER                                                   
1 #MYPARM (A) DYNAMIC                                                   
END-DEFINE                                                              
*                                                                       
WRITE *LENGTH(#MYPARM)                  /* *LENGTH(#MYPARM) = 6         
#MYPARM := '1234567'                    /* *LENGTH(#MYPARM) = 7         
#MYPARM := '12345678'                   /* *LENGTH(#MYPARM) = 8         
EXPAND DYNAMIC VARIABLE #MYPARM TO 10   /* 10 bytes are allocated       
*                                                                       
WRITE *LENGTH(#MYPARM)                  /* *LENGTH(#MYPARM) = 8         
*                                                                       
END                                                                    ↩

CALL 3GL Program

Dynamic and large variables can sensibly be used with the CALL statement when the option
INTERFACE4 is used. Using this option leads to an interface to the 3GL program with a different
parameter structure.

This usage requires someminor changes in the 3GLprogram, but provides the following significant
benefits as compared with the older FINFO structure.

■ No limitation on the number of passed parameters (former limit 40).
■ No limitation on the parameter's data size (former limit 64 KB per parameter).

■ Full parameter information can be passed to the 3GL program including array information.
Exported functions are providedwhich allow secure access to the parameter data (formerly you
had to take care not to overwrite memory inside of Natural)

For further information on the FINFO structure, see the CALL INTERFACE4 statement.

Before calling a 3GLprogramwith dynamic parameters, it is important to ensure that the necessary
buffer size is allocated. This can be done explicitly with the EXPAND statement.

If an initialized buffer is required, the dynamic variable can be set to the initial value and to the
necessary size by using the MOVE ALL UNTIL statement. Natural provides a set of functions that
allow the 3GL program to obtain information about the dynamic parameter and to modify the
length when parameter data is passed back.

Example:

MOVE ALL ' ' TO #MYDYNTEXT1 UNTIL 10000
/* a buffer of length 10000 is allocated
/* #MYDYNTEXT1 is initialized with blanks
/* and *LENGTH(#MYDYNTEXT1) = 10000

CALL INTERFACE4 'MYPROG' USING #MYDYNTEXT1
WRITE *LENGTH(#MYDYNTEXT1)

/* *LENGTH(#MYDYNTEXT1) may have changed in the 3GL program

For a more detailed description, refer to the CALL statement in the Statements documentation.

Work File Access with Large and Dynamic Variables

The following topics are covered below:

■ PORTABLE and UNFORMATTED
■ ASCII, ASCII-COMPRESSED and SAG
■ Special Conditions for TRANSFER and ENTIRE CONNECTION

PORTABLE and UNFORMATTED

Large and dynamic variables can be written into work files or read from work files using the two
work file types PORTABLE and UNFORMATTED. For these types, there is no size restriction for dynamic
variables. However, large variablesmay not exceed amaximumfield/record length of 32766 bytes.

For the work file type PORTABLE, the field information is stored within the work file. The dynamic
variables are resized during READ if the field size in the record is different from the current size.

The work file type UNFORMATTED can be used, for example, to read a video from a database and
store it in a file directly playable by other utilities. In the WRITE WORK statement, the fields are
written to the file with their byte length. All data types (DYNAMIC or not) are treated the same. No

structural information is inserted.Note thatNatural uses a bufferingmechanism, so you can expect
the data to be completely written only after a CLOSE WORK. This is especially important if the file
is to be processed with another utility while Natural is running.

With the READ WORK statement, fields of fixed length are read with their whole length. If the end-
of-file is reached, the remainder of the current field is filled with blanks. The following fields are
unchanged. In the case of DYNAMIC data types, all the remainder of the file is read unless it exceeds
1073741824 bytes. If the end of file is reached, the remaining fields (variables) are kept unchanged
(normal Natural behavior).

ASCII, ASCII-COMPRESSED and SAG

Theworkfile typesASCII,ASCII-COMPRESSEDandSAG (binary) cannot handledynamic variables
and will produce an error. Large variables for these work file types pose no problem unless the
maximum field/record length of 32766 bytes is exceeded.

Special Conditions for TRANSFER and ENTIRE CONNECTION

In conjunctionwith the READ WORK FILE statement, thework file type TRANSFER can handle dynamic
variables. There is no size limit for dynamic variables. The work file type ENTIRE CONNECTION
cannot handle dynamic variables. They can both, however, handle large variableswith amaximum
field/record length of 1073741824 bytes.

In conjunction with the WRITE WORK FILE statement, the work file type TRANSFER can handle dy-
namic variables with a maximum field/record length of 32766 bytes. The work file type ENTIRE
CONNECTION cannot handle dynamic variables. They can both, however, handle large variables
with a maximum field/record length of 1073741824 bytes.

DDM Generation and Editing for Varying Length Columns

Depending on the data types, the related database format A or format B is generated. For the
databases' data type VARCHAR the Natural length of the column is set to the maximum length of
the data type as defined in theDBMS. If a data type is very large, the keyword DYNAMIC is generated
at the length field position.

For all varying length columns, an LINDICATOR field L@<column-name>will be generated. For the
databases' data type VARCHAR, an LINDICATOR field with format/length I2 will be generated. For
large data types (see list below) the format/length will be I4.

In the context of database access, the LINDICATOR handling offers the chance to get the length of
the field to be read or to set the length of the field to be written independent of a defined buffer
length (or independent of *LENGTH). Usually, after a retrieval function, *LENGTHwill be set to the
corresponding length indicator value.

Example DDM:

  T  L  Name                      F   Leng        S   D   Remark
     :
     1  L@PICTURE1                I   4                                     /* ↩
length indicator
     1  PICTURE1                  B   DYNAMIC             IMAGE
     1  N@PICTURE1                I   2                                     /* NULL ↩
indicator
     1  L@TEXT1                   I   4                                     /* ↩
length indicator
     1  TEXT1                     A   DYNAMIC             TEXT
     1  N@TEXT1                   I   2                                     /* NULL ↩
indicator
     1  L@DESCRIPTION             I   2                                     /* ↩
length indicator
     1  DESCRIPTION               A   1000                VARCHAR(1000) 
     :
     :
  ~~~~~~~~~~~~~~~~~~~~~~Extended Attributes~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~/* ↩
concerning PICTURE1
   Header               :    ---
   Edit Mask            :    ---
   Remarks              :   IMAGE

The generated formats are varying length formats. The Natural programmer has the chance to
change the definition from DYNAMIC to a fixed length definition (extended field editing) and can
change, for example, the corresponding DDMfield definition for VARCHAR data types to amultiple
value field (old generation).

Example:

  T  L  Name                  F   Leng          S   D   Remark
     :
     1  L@PICTURE1            I   4                                         /* ↩
length indicator
     1  PICTURE1              B   1000000000            IMAGE
     1  N@PICTURE1            I   2                                         /* NULL ↩
indicator
     1  L@TEXT1               I   4                                         /* ↩
length indicator
     1  TEXT1                 A   5000                  TEXT
     1  N@TEXT1               I   2                                         /* NULL ↩
indicator
     1  L@DESCRIPTION         I   2                                         /* ↩
length indicator
  M  1  DESCRIPTION           A   100                   VARCHAR(1000) 
       :
       :
  ~~~~~~~~~~~~~~~~~~~~Extended Attributes~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~/* ↩
concerning PICTURE1

   Header             :    ---
   Edit Mask          :    ---
   Remarks            :   IMAGE

Accessing Large Database Objects

To access a database with large objects (CLOBs or BLOBs), a DDMwith corresponding large al-
phanumeric, Unicode or binary fields is required. If a fixed length is defined and if the database
large object does not fit into this field, the large object is truncated. If the programmer does not
know the definitive length of the database object, it will make sense to work with dynamic fields.
As many reallocations as necessary are done to hold the object. No truncation is performed.

Example Program:

DEFINE DATA LOCAL

1 person VIEW OF xyz-person
  2 last_name
  2 first_name_1
  2 L@PICTURE1                   /* I4 length indicator for PICTURE1
  2 PICTURE1                     /* defined as dynamic in the DDM
  2 TEXT1                        /* defined as non-dynamic in the DDM

END-DEFINE

SELECT * INTO VIEW person FROM xyz-person                   /* PICTURE1 will be ↩
read completely
                             WHERE last_name = 'SMITH'      /* TEXT1 will be ↩
truncated to fixed length 5000  

   WRITE  'length of PICTURE1: '  L@PICTURE1                /* the L-INDICATOR will ↩
contain the length 
                                                            /* of PICTURE1 (= ↩
*LENGTH(PICTURE1)
   /* do something with PICTURE1 and TEXT1

   L@PICTURE1  :=  100000       
   INSERT INTO xyz-person  (*) VALUES (VIEW person)         /* only the first 100000 ↩
Bytes of PICTURE1
                                                            /* are inserted
END-SELECT ↩

If a format-length definition is omitted in the view, this is taken from theDDM. In reportingmode,
it is now possible to specify any length, if the corresponding DDMfield is defined as DYNAMIC. The
dynamic field will be mapped to a field with a fixed buffer length. The other way round is not
possible.

VIEW format / length definitionDDM format/length definition

valid-(An)

valid(An)

only valid in reporting mode(Am)

invalid(A) DYNAMIC

valid-(A) DYNAMIC

valid(A) DYNAMIC

only valid in reporting mode(An)

only valid in reporting mode(Am / i : j)

(equivalent for Format B variables)

Parameter with LINDICATOR Clause in SQL Statements

If the LINDICATOR field is defined as an I2 field, the SQL data type VARCHAR is used for sending or
receiving the corresponding column. If the LINDICATOR host variable is specified as I4, a large object
data type (CLOB/BLOB) is used.

If the field is defined as DYNAMIC, the column is read in an internal loop up to its real length. The
LINDICATOR field and the system variable *LENGTH are set to this length. In the case of a fixed-length
field, the column is read up to the defined length. In both cases, the field is written up to the value
defined in the LINDICATOR field.

Performance Aspects with Dynamic Variables

If a dynamic variable is to be expanded in small quantitiesmultiple times (for example, byte-wise),
use the EXPAND statement before the iterations if the upper limit of required storage is (approxim-
ately) known. This avoids additional overhead to adjust the storage needed.

Use the REDUCE or RESIZE statement if the dynamic variable will no longer be needed, especially
for variables with a high value of the system variable *LENGTH. This enables Natural you to release
or reuse the storage. Thus, the overall performance may be improved.

The amount of the allocated memory of a dynamic variable may be reduced using the REDUCE
DYNAMIC VARIABLE statement. In order to (re)allocate a variable to a specified length, the EXPAND
statement can be used. (If the variable should be initialized, use the MOVE ALL UNTIL statement.)

Example:

** Example 'DYNAMX06': Dynamic variables (allocated memory)
************************************************************************
DEFINE DATA LOCAL
1 #MYDYNTEXT1 (A) DYNAMIC
1 #LEN (I4)
END-DEFINE
*
#MYDYNTEXT1 := 'a' /* used length is 1, value is 'a'

/* allocated size is still 1
WRITE *LENGTH(#MYDYNTEXT1)
*
EXPAND DYNAMIC VARIABLE #MYDYNTEXT1 TO 100

/* used length is still 1, value is 'a'
/* allocated size is 100

*
CALLNAT 'DYNAMX05' USING #MYDYNTEXT1
WRITE *LENGTH(#MYDYNTEXT1)

/* used length and allocated size
/* may have changed in the subprogram

*
#LEN := *LENGTH(#MYDYNTEXT1)
REDUCE DYNAMIC VARIABLE #MYDYNTEXT1 TO #LEN

/* if allocated size is greater than used length,
/* the unused memory is released

*
REDUCE DYNAMIC VARIABLE #MYDYNTEXT1 TO 0
WRITE *LENGTH(#MYDYNTEXT1)

/* free allocated memory for dynamic variable
END

Rules:

■ Use dynamic operands where it makes sense.
■ Use the EXPAND statement if upper limit of memory usage is known.
■ Use the REDUCE statement if the dynamic operand will no longer be needed.

Outputting Dynamic Variables

Dynamic variables may be used inside output statements such as the following:

NotesStatement

With these statements, you must set the format of the output or input of dynamic variables
using the AL (Alphanumeric Length for Output) or EM (Edit Mask) session parameters.

DISPLAY

WRITE

INPUT

--REINPUT

Because the output of the PRINT statement is unformatted, the output of dynamic variables in
the PRINT statement need not be set using AL and EM parameters. In other words, these
parameters may be omitted.

PRINT

Dynamic X-Arrays

A dynamic X-array may be allocated by first specifying the number of occurrences and then ex-
panding the length of the previously allocated array occurrences.

Example:

DEFINE DATA LOCAL
1 #X-ARRAY(A/1:*) DYNAMIC
END-DEFINE
*
EXPAND ARRAY #X-ARRAY TO (1:10) /* Current boundaries (1:10)
#X-ARRAY(*) := 'ABC'
EXPAND ARRAY #X-ARRAY TO (1:20) /* Current boundaries (1:20)
#X-ARRAY(11:20) := 'DEF'

Constants can be used throughout Natural programs. This document discusses the types of con-
stants that are supported and how they are used.

Numeric Constants

The following topics are covered below:

■ Numeric Constants
■ Validation of Numeric Constants

Numeric Constants

A numeric constant may contain 1 to 29 numeric digits, a special character as decimal separator
(period or comma) and a sign.

Examples:

1234 +1234 -1234

12.34 +12.34 -12.34

MOVE 3 TO #XYZ
COMPUTE #PRICE = 23.34
COMPUTE #XYZ = -103
COMPUTE #A = #B * 6074

Note: Internally, numeric constants without decimal digits are represented in integer form
(format I), while numeric constantswith decimal digits, aswell as numeric constantswithout
decimal digits that are too large to fit into format I, are represented in packed form (format
P).

Example:

LengthFormatNumeric Constant
ToFrom

>=10P<= -2147483649

4I-32769-2147483648

2I32767-32768

4I214748364732768

>=10P>= 2147483648

Validation of Numeric Constants

When numeric constants are used within one of the statements COMPUTE, MOVE, or DEFINE DATA
with INIT option, Natural checks at compilation timewhether a constant value fits into the corres-
ponding field. This avoids runtime errors in situations where such an error condition can already
be detected during compilation.

Alphanumeric Constants

The following topics are covered below:

■ Alphanumeric Constants
■ Apostrophes Within Alphanumeric Constants
■ Concatenation of Alphanumeric Constants

Alphanumeric Constants

An alphanumeric constantmay contain 1 to 1 1073741824 bytes (1 GB) of alphanumeric characters.

An alphanumeric constant must be enclosed in either apostrophes (')

'text'

or quotation marks (")

"text"

Examples:

MOVE 'ABC' TO #FIELDX
MOVE '% INCREASE' TO #TITLE
DISPLAY "LAST-NAME" NAME

Note: An alphanumeric constant that is used to assign a value to a user-defined variable
must not be split between statement lines.

Apostrophes Within Alphanumeric Constants

If you want an apostrophe to be part of an alphanumeric constant that is enclosed in apostrophes,
you must write this as two apostrophes or as a single quotation mark.

If you want an apostrophe to be part of an alphanumeric constant that is enclosed in quotation
marks, you write this as a single apostrophe.

Example:

If you want the following to be output:

HE SAID, 'HELLO'

you can use any of the following notations:

WRITE 'HE SAID, ''HELLO'''
WRITE 'HE SAID, "HELLO"'
WRITE "HE SAID, ""HELLO"""
WRITE "HE SAID, 'HELLO'"

Note: If quotation marks are not converted to apostrophes as shown above, this is due to
the setting of profile parameter TQMARK (Translate Quotation Marks); ask your Natural ad-
ministrator for details.

Concatenation of Alphanumeric Constants

Alphanumeric constants may be concatenated to form a single value by use of a hyphen.

Examples:

MOVE 'XXXXXX' - 'YYYYYY' TO #FIELD
MOVE "ABC" - 'DEF' TO #FIELD

In this way, alphanumeric constants can also be concatenated with hexadecimal constants.

Unicode Constants

The following topics are covered below:

■ Unicode Text Constants
■ Apostrophes Within Unicode Text Constants
■ Unicode Hexadecimal Constants

■ Concatenation of Unicode Constants

Unicode Text Constants

A Unicode text constant must be preceded by the character U and enclosed in either apostrophes
(')

U'text'

or quotation marks (")

U"text"

Example:

U'HELLO'

The compiler stores this text constant in the generated program in Unicode format (UTF-16).

Apostrophes Within Unicode Text Constants

If you want an apostrophe to be part of a Unicode text constant that is enclosed in apostrophes,
you must write this as two apostrophes or as a single quotation mark.

If youwant an apostrophe to be part of aUnicode text constant that is enclosed in quotationmarks,
you write this as a single apostrophe.

Example:

If you want the following to be output:

HE SAID, 'HELLO'

you can use any of the following notations:

WRITE U'HE SAID, ''HELLO'''
WRITE U'HE SAID, "HELLO"'
WRITE U"HE SAID, ""HELLO"""
WRITE U"HE SAID, 'HELLO'"

Note: If quotation marks are not converted to apostrophes as shown above, this is due to
the setting of the profile parameter TQ (Translate Quotation Marks); ask your Natural ad-
ministrator for details.

Unicode Hexadecimal Constants

The following syntax is used to supply a Unicode character or a Unicode string by its hexadecimal
notation:

UH'hhhh...'

where h represents a hexadecimal digit (0-9, A-F). Since a UTF-16 Unicode character consists of a
double-byte, the number of hexadecimal characters supplied has to be a multiple of four.

Example:

This example defines the string 45.

UH'00340035'

Concatenation of Unicode Constants

Concatenation of Unicode text constants (U) andUnicode hexadecimal constants (UH) is allowed.

Valid Example:

MOVE U'XXXXXX' - UH'00340035' TO #FIELD

Unicode text constants or Unicode hexadecimal constants cannot be concatenated with code page
alphanumeric constants or H constants.

Invalid Example:

MOVE U'ABC' - 'DEF' TO #FIELD
MOVE UH'00340035' - H'414243' TO #FIELD

Further Valid Example:

DEFINE DATA LOCAL
1 #U10 (U10)               /* Unicode variable with 10 (UTF-16) characters, total ↩
byte length = 20
1 #UD  (U)  DYNAMIC        /* Unicode variable with dynamic length
END-DEFINE
*
#U10 := U'ABC'             /* Constant is created as X'004100420043' in the object, ↩
the UTF-16 representation for string 'ABC'.

#U10 := UH'004100420043'   /* Constant supplied in hexadecimal format only, ↩
corresponds to U'ABC'

#U10 := U'A'-UH'0042'-U'C' /* Constant supplied in mixed formats, corresponds to ↩
U'ABC'.
END

Programming Guide148

User-Defined Constants



Date and Time Constants

The following topics are covered below:

■ Date Constant
■ Time Constant
■ Extended Time Constant

Date Constant

A date constant may be used in conjunction with a format D variable.

Date constants may have the following formats:

International date formatD'yyyy-mm-dd'

German date formatD'dd.mm.yyyy'

European date formatD'dd/mm/yyyy'

US date formatD'mm/dd/yyyy'

where dd represents the number of the day, mm the number of the month and yyyy the year.

Example:

DEFINE DATA LOCAL
1 #DATE (D)
END-DEFINE
...
MOVE D'2004-03-08' TO #DATE
...

The default date format is controlled by the profile parameter DTFORM (Date Format) as set by the
Natural administrator.

Time Constant

A time constant may be used in conjunction with a format T variable.

A time constant has the following format:

T'hh:ii:ss'

where hh represents hours, iiminutes and ss seconds.

Example:

DEFINE DATA LOCAL
1 #TIME (T)
END-DEFINE
...
MOVE T'11:33:00' TO #TIME
...

Extended Time Constant

A time variable (format T) can contain date and time information, date information being a subset
of time information; however, with a “normal” time constant (prefix T) only the time information
of a time variable can be handled:

T'hh:ii:ss'

With an extended time constant (prefix E), it is possible to handle the full content of a time variable,
including the date information:

E'yyyy-mm-dd hh:ii:ss'

Apart from that, the use of an extended time constant in conjunction with a time variable is the
same as for a normal time constant.

Note: The format in which the date information has to be specified in an extended time
constant depends on the setting of the profile parameter DTFORM. The extended time constant
shown above assumes DTFORM=I (international date format).

Hexadecimal Constants

The following topics are covered below:

■ Hexadecimal Constants

■ Concatenation of Hexadecimal Constants

Hexadecimal Constants

A hexadecimal constant may be used to enter a value which cannot be entered as a standard key-
board character.

A hexadecimal constant may contain 1 to 1073741824 bytes (1 GB) of alphanumeric characters.

A hexadecimal constant is prefixed with an H. The constant itself must be enclosed in apostrophes
andmay consist of the hexadecimal characters 0 - 9, A - F. Two hexadecimal characters are required
to represent one byte of data.

The hexadecimal representation of a character varies, depending on whether your computer uses
anASCII or EBCDIC character set.When you transfer hexadecimal constants to another computer,
you may therefore have to convert the characters.

ASCII examples:

H'313233' (equivalent to the alphanumeric constant '123')
H'414243' (equivalent to the alphanumeric constant 'ABC')

EBCDIC examples:

H'F1F2F3' (equivalent to the alphanumeric constant '123')
H'C1C2C3' (equivalent to the alphanumeric constant 'ABC')

When a hexadecimal constant is transferred to another field, it will be treated as an alphanumeric
value (format A).

The data transfer of an alphanumeric value (format A) to a field which is defined with a format
other than A,U or B is not allowed. Therefore, a hexadecimal constant used as initial value in a
DEFINE DATA statement is rejected with the syntax error NAT0094 if the corresponding variable
is not of format A, U or B.

Example:

DEFINE DATA LOCAL
1 #I(I2) INIT <H'000F'>      /* causes a NAT0094 syntax error 
END-DEFINE                         ↩

Note: If a hexadecimal constant is output that contains any characters from the ranges H'00'
to H'1F' or H'80' to H'A0', these characterswill not be output, as theywould be interpreted
as terminal control characters. As of Version 2.2 these hex constants are not suppressed.

Concatenation of Hexadecimal Constants

Hexadecimal constants may be concatenated by using a hyphen between the constants.

ASCII example:

H'414243' - H'444546' (equivalent to 'ABCDEF')

EBCDIC example:

H'C1C2C3' - H'C4C5C6' (equivalent to 'ABCDEF')

In this way, hexadecimal constants can also be concatenated with alphanumeric constants.

Logical Constants

The logical constants TRUE and FALSEmay be used to assign a logical value to a field defined with
Format L.

Example:

DEFINE DATA LOCAL
1 #FLAG (L)
END-DEFINE
...
MOVE TRUE TO #FLAG
...
IF #FLAG ...

statement ...
MOVE FALSE TO #FLAG

END-IF
...

Floating Point Constants

Floating point constants can be used with variables defined with format F.

Example:

Programming Guide152

User-Defined Constants



DEFINE DATA LOCAL
1 #FLT1 (F4)
END-DEFINE
...
COMPUTE #FLT1 = -5.34E+2
...

Attribute Constants

Attribute constants can be usedwith variables definedwith format C (control variables). This type
of constant must be enclosed within parentheses.

The following attributes may be used:

DescriptionAttribute

defaultAD=D

blinkingAD=B

intensifiedAD=I

non-displayAD=N

reverse videoAD=V

underlinedAD=U

cursive/italicAD=C

dynamic attributeAD=Y

protectedAD=P

blueCD=BL

greenCD=GR

neutralCD=NE

pinkCD=PI

redCD=RE

turquoiseCD=TU

yellowCD=YE

See also session parameters AD and CD.

Example:

DEFINE DATA LOCAL
1 #ATTR (C)
1 #FIELD (A10)
END-DEFINE
...
MOVE (AD=I CD=BL) TO #ATTR
...
INPUT #FIELD (CV=#ATTR)
...

Handle Constants

The handle constant NULL-HANDLE can be used with object handles.

Defining Named Constants

If you need to use the same constant value several times in a program, you can reduce the main-
tenance effort by defining a named constant:

■ Define a field in the DEFINE DATA statement,
■ assign a constant value to it, and
■ use the field name in the program instead of the constant value.

Thus, when the value has to be changed, you only have to change it once in the DEFINE DATA
statement and not everywhere in the program where it occurs.

You specify the constant value in angle brackets with the keyword CONSTANT after the field defin-
ition in the DEFINE DATA statement.

■ If the value is alphanumeric, it must be enclosed in apostrophes.
■ If the value is text inUnicode format, it must be preceded by the character U andmust be enclosed
in apostrophes.

■ If the value is in hexadecimal Unicode format, it must be preceded by the characters UH and
must be enclosed in apostrophes.

Example:

DEFINE DATA LOCAL
1 #FIELDA (N3) CONSTANT <100>
1 #FIELDB (A5) CONSTANT <'ABCDE'>
1 #FIELDC (U5) CONSTANT <U'ABCDE'>
1 #FIELDD (U5) CONSTANT <UH'00410042004300440045'>
END-DEFINE
...

During the execution of the program, the value of such a named constant cannot be modified.

This chapter describes the default initial values of user-defined variables, explains how you can
assign an initial value to a user-defined variable and how you can use the RESET statement to reset
the field value to its default initial value or the initial value defined for that variable in the DEFINE
DATA statement.

Default Initial Value of a User-Defined Variable/Array

If you specify no initial value for a field, the field will be initialized with a default initial value
depending on its format:

Default Initial ValueFormat

0B, F, I, N, P

blankA, U

F(ALSE)L

D' 'D

T'00:00:00'T

(AD=D)C

NULL-HANDLEObject Handle

Assigning an Initial Value to a User-Defined Variable/Array

In the DEFINE DATA statement, you can assign an initial value to a user-defined variable. If the
initial value is alphanumeric, it must be enclosed in apostrophes.

■ Assigning a Modifiable Initial Value
■ Assigning a Constant Initial Value
■ Assigning a Natural System Variable as Initial Value
■ Assigning Characters as Initial Value for Alphanumeric Variables

Assigning a Modifiable Initial Value

If the variable/array is to be assigned a modifiable initial value, you specify the initial value in
angle brackets with the keyword INIT after the variable definition in the DEFINE DATA statement.
The value(s) assignedwill be used each time the variable/array is referenced. The value(s) assigned
can be modified during program execution.

Example:

Programming Guide158

Initial Values (and the RESET Statement)



DEFINE DATA LOCAL
1 #FIELDA (N3) INIT <100>
1 #FIELDB (A20) INIT <'ABC'>
END-DEFINE
...

Assigning a Constant Initial Value

If the variable/array is to be treated as a named constant, you specify the initial value in angle
brackets with the keyword CONSTANT after the variable definition in the DEFINE DATA statement.
The constant value(s) assignedwill be used each time the variable/array is referenced. The value(s)
assigned cannot be modified during program execution.

Example:

DEFINE DATA LOCAL
1 #FIELDA (N3) CONST <100>
1 #FIELDB (A20) CONST <'ABC'>
END-DEFINE
...

Assigning a Natural System Variable as Initial Value

The initial value for a field may also be the value of a Natural system variable.

Example:

In this example, the system variable *DATX is used to provide the initial value.

DEFINE DATA LOCAL
1 #MYDATE (D) INIT <*DATX>
END-DEFINE
...

Assigning Characters as Initial Value for Alphanumeric Variables

As initial value, a variable can also be filled, entirely or partially, with a specific single character
or string of characters (only possible for alphanumeric variables).

■ Filling an entire field:
With the option FULL LENGTH <character(s)>, the entire field is filled with the specified char-
acter(s).

In this example, the entire field will be filled with asterisks.

DEFINE DATA LOCAL
1 #FIELD (A25) INIT FULL LENGTH <'*'>
END-DEFINE
...

■ Filling the firstn positions of a field:
With the option LENGTH n <character(s)>, the first n positions of the field are filled with the
specified character(s).

In this example, the first 4 positions of the field will be filled with exclamation marks.

DEFINE DATA LOCAL
1 #FIELD (A25) INIT LENGTH 4 <'!'>
END-DEFINE
...

Resetting a User-Defined Variable to its Initial Value

The RESET statement is used to reset the value of a field. Two options are available:

■ Reset to Default Initial Value
■ Reset to Initial Value Defined in DEFINE DATA

Notes:

1. A field declared with a CONSTANT clause in the DEFINE DATA statement may not be referenced
in a RESET statement, since its content cannot be changed.

2. In reporting mode, the RESET statement may also be used to define a variable, provided that
the program contains no DEFINE DATA LOCAL statement.

Reset to Default Initial Value

RESET (without INITIAL) sets the content of each specifiedfield to itsdefault initial valuedepending
on its format.

Example:

DEFINE DATA LOCAL 
1 #FIELDA (N3)  INIT <100> 
1 #FIELDB (A20) INIT <'ABC'> 
1 #FIELDC (I4)  INIT <5> 
END-DEFINE 
... 
... 
RESET #FIELDA                    /* resets field value to default initial value 
... ↩

Reset to Initial Value Defined in DEFINE DATA

RESET INITIAL sets each specified field to the initial value as defined for the field in the DEFINE
DATA statement.

For a field declared without INIT clause in the DEFINE DATA statement, RESET INITIAL has the
same effect as RESET (without INITIAL).

Example:

DEFINE DATA LOCAL 
1 #FIELDA (N3)  INIT <100> 
1 #FIELDB (A20) INIT <'ABC'> 
1 #FIELDC (I4)  INIT <5> 
END-DEFINE 
... 
RESET INITIAL #FIELDA #FIELDB #FIELDC /* resets field values to initial values as ↩
defined in DEFINE DATA
...

Redefinition is used to change the format of a field, or to divide a single field into segments.

Using the REDEFINE Option of DEFINE DATA

The REDEFINE option of the DEFINE DATA statement can be used to redefine a single field - either
a user-defined variable or a database field - as one or more new fields. A group can also be re-
defined.

Important: Dynamic variables are not allowed in a redefinition.

The REDEFINE option redefines byte positions of a field from left to right, regardless of the format.
Byte positions must match between original field and redefined field(s).

The redefinition must be specified immediately after the definition of the original field.

Example 1:

In the following example, the database field BIRTH is redefined as three newuser-defined variables:

DEFINE DATA LOCAL
01 EMPLOY-VIEW VIEW OF STAFFDDM

02 NAME
02 BIRTH
02 REDEFINE BIRTH

03 #BIRTH-YEAR (N4)
03 #BIRTH-MONTH (N2)
03 #BIRTH-DAY (N2)

END-DEFINE
...

Example 2:

In the following example, the group #VAR2, which consists of two user-defined variables of format
N and P respectively, is redefined as a variable of format A:

DEFINE DATA LOCAL
01 #VAR1 (A15)
01 #VAR2

02 #VAR2A (N4.1)
02 #VAR2B (P6.2)

01 REDEFINE #VAR2
02 #VAR2RD (A10)

END-DEFINE
...

With the notation FILLER nX you can define n filler bytes - that is, segments which are not to be
used - in the field that is being redefined. (The definition of trailing filler bytes is optional.)

Example 3:

In the following example, the user-defined variable #FIELD is redefined as three new user-defined
variables, each of format/length A2. The FILLER notations indicate that the 3rd and 4th and 7th to
10th bytes of the original field are not be used.

DEFINE DATA LOCAL
1 #FIELD (A12)
1 REDEFINE #FIELD

2 #RFIELD1 (A2)
2 FILLER 2X
2 #RFIELD2 (A2)
2 FILLER 4X
2 #RFIELD3 (A2)

END-DEFINE
...

Example Program Illustrating the Use of a Redefinition

The following program illustrates the use of a redefinition:

** Example 'DDATAX01': DEFINE DATA
************************************************************************
DEFINE DATA LOCAL
01 VIEWEMP VIEW OF EMPLOYEES

02 NAME
02 FIRST-NAME
02 SALARY (1:1)

*
01 #PAY (N9)
01 REDEFINE #PAY

02 FILLER 3X
02 #USD (N3)
02 #OOO (N3)

END-DEFINE
*
READ (3) VIEWEMP BY NAME STARTING FROM 'JONES'

MOVE SALARY (1) TO #PAY
DISPLAY NAME FIRST-NAME #PAY #USD #OOO

END-READ
END

Output of Program DDATAX01:

Note how #PAY and the fields resulting from its definition are displayed:

Page      1                                                  04-11-11  14:15:54
 
        NAME              FIRST-NAME         #PAY    #USD #OOO
-------------------- -------------------- ---------- ---- ----
 
JONES                VIRGINIA                  46000   46    0
JONES                MARSHA                    50000   50    0
JONES                ROBERT                    31000   31    0              
            ↩

Natural supports the processing of arrays. Arrays are multi-dimensional tables, that is, two or
more logically related elements identified under a single name. Arrays can consist of single data
elements ofmultiple dimensions or hierarchical data structureswhich contain repetitive structures
or individual elements.

Defining Arrays

In Natural, an array can be one-, two- or three-dimensional. It can be an independent variable,
part of a larger data structure or part of a database view.

Important: Dynamic variables are not allowed in an array definition.

To define a one-dimensional array

■ After the format and length, specify a slash followed by a so-called “index notation”, that is,
the number of occurrences of the array.

For example, the following one-dimensional array has three occurrences, each occurrence
being of format/length A10:

DEFINE DATA LOCAL
1 #ARRAY (A10/1:3)
END-DEFINE
...

To define a two-dimensional array

■ Specify an index notation for both dimensions:

DEFINE DATA LOCAL
1 #ARRAY (A10/1:3,1:4)
END-DEFINE
...

A two-dimensional array can be visualized as a table. The array defined in the example above
would be a table that consists of 3 “rows” and 4 “columns”:

Initial Values for Arrays

To assign initial values to one or more occurrences of an array, you use an INIT specification,
similar to that for “ordinary” variables, as shown in the following examples.

Assigning Initial Values to One-Dimensional Arrays

The following examples illustrate how initial values are assigned to a one-dimensional array.

■ To assign an initial value to one occurrence, you specify:

1 #ARRAY (A1/1:3) INIT (2) <'A'>

A is assigned to the second occurrence.
■ To assign the same initial value to all occurrences, you specify:

1 #ARRAY (A1/1:3) INIT ALL <'A'>

A is assigned to every occurrence. Alternatively, you could specify:

1 #ARRAY (A1/1:3) INIT (*) <'A'>

■ To assign the same initial value to a range of several occurrences, you specify:

1 #ARRAY (A1/1:3) INIT (2:3) <'A'>

A is assigned to the second to third occurrence.
■ To assign a different initial value to every occurrence, you specify:

1 #ARRAY (A1/1:3) INIT <'A','B','C'>

A is assigned to the first occurrence, B to the second, and C to the third.
■ To assign different initial values to some (but not all) occurrences, you specify:

1 #ARRAY (A1/1:3) INIT (1) <'A'> (3) <'C'>

A is assigned to the first occurrence, and C to the third; no value is assigned to the second occur-
rence.

Alternatively, you could specify:

1 #ARRAY (A1/1:3) INIT <'A',,'C'>

■ If fewer initial values are specified than there are occurrences, the last occurrences remain empty:

1 #ARRAY (A1/1:3) INIT <'A','B'>

A is assigned to the first occurrence, and B to the second; no value is assigned to the third occur-
rence.

Assigning Initial Values to Two-Dimensional Arrays

This section illustrates how initial values are assigned to a two-dimensional array. The following
topics are covered:

■ Preliminary Information
■ Assigning the Same Value
■ Assigning Different Values

Preliminary Information

For the examples shown in this section, let us assume a two-dimensional array with three occur-
rences in the first dimension (“rows”) and four occurrences in the second dimension (“columns”):

1 #ARRAY (A1/1:3,1:4)

Vertical: First Dimension (1:3), Horizontal: Second Dimension (1:4):

(1,4)(1,3)(1,2)(1,1)

(2,4)(2,3)(2,2)(2,1)

(3,4)(3,3)(3,2)(3,1)

The first set of examples illustrates how the same initial value is assigned to occurrences of a two-
dimensional array; the second set of examples illustrates how different initial values are assigned.

In the examples, please note in particular the usage of the notations * and V. Both notations refer
to all occurrences of the dimension concerned: * indicates that all occurrences in that dimension
are initialized with the same value, while V indicates that all occurrences in that dimension are
initialized with different values.

Assigning the Same Value

■ To assign an initial value to one occurrence, you specify:

1 #ARRAY (A1/1:3,1:4) INIT (2,3) <'A'>

A

■ To assign the same initial value to one occurrence in the second dimension - in all occurrences
of the first dimension - you specify:

1 #ARRAY (A1/1:3,1:4) INIT (*,3) <'A'>

A

A

A

■ To assign the same initial value to a range of occurrences in the first dimension - in all occurrences
of the second dimension - you specify:

1 #ARRAY (A1/1:3,1:4) INIT (2:3,*) <'A'>

AAAA

AAAA

■ To assign the same initial value to a range of occurrences in each dimension, you specify:

1 #ARRAY (A1/1:3,1:4) INIT (2:3,1:2) <'A'>

AA

AA

■ To assign the same initial value to all occurrences (in both dimensions), you specify:

1 #ARRAY (A1/1:3,1:4) INIT ALL <'A'>

AAAA

AAAA

AAAA

Alternatively, you could specify:

1 #ARRAY (A1/1:3,1:4) INIT (*,*) <'A'>

Assigning Different Values

■ 1 #ARRAY (A1/1:3,1:4) INIT (V,2) <'A','B','C'>

A

B

C

■ 1 #ARRAY (A1/1:3,1:4) INIT (V,2:3) <'A','B','C'>

AA

BB

CC

■ 1 #ARRAY (A1/1:3,1:4) INIT (V,*) <'A','B','C'>

AAAA

BBBB

CCCC

■ 1 #ARRAY (A1/1:3,1:4) INIT (V,*) <'A',,'C'>

AAAA

CCCC

■ 1 #ARRAY (A1/1:3,1:4) INIT (V,*) <'A','B'>

AAAA

BBBB

■ 1 #ARRAY (A1/1:3,1:4) INIT (V,1) <'A','B','C'> (V,3) <'D','E','F'>

DA

EB

FC

■ 1 #ARRAY (A1/1:3,1:4) INIT (3,V) <'A','B','C','D'>

DCBA

■ 1 #ARRAY (A1/1:3,1:4) INIT (*,V) <'A','B','C','D'>

DCBA

DCBA

DCBA

■ 1 #ARRAY (A1/1:3,1:4) INIT (2,1) <'A'> (*,2) <'B'> (3,3) <'C'> (3,4) <'D'>

B

BA

DCB

■ 1 #ARRAY (A1/1:3,1:4) INIT (2,1) <'A'> (V,2) <'B','C','D'> (3,3) <'E'> (3,4) <'F'>

B

CA

FED

A Three-Dimensional Array

A three-dimensional array could be visualized as follows:

The array illustrated here would be defined as follows (at the same time assigning an initial value
to the highlighted field in Row 1, Column 2, Plane 2):

DEFINE DATA LOCAL
1 #ARRAY2

2 #ROW (1:4)
3 #COLUMN (1:3)
4 #PLANE (1:3)

5 #FIELD2 (P3) INIT (1,2,2) <100>
END-DEFINE
...

If defined as a local data area in the source editor, the same array would look as follows:

I T L Name                             F Leng Index/Init/EM/Name/Comment       
- - - -------------------------------- - ---- ---------------------------------
    1 #ARRAY2                                                                  
    2 #ROW                                    (1:4)                            
    3 #COLUMN                                 (1:3)                            
    4 #PLANE                                  (1:3)                            
I   5 #FIELD2                          P    3                    ↩

Arrays as Part of a Larger Data Structure

Themultiple dimensions of an arraymake it possible to define data structures analogous toCOBOL
or PL1 structures.

Example:

DEFINE DATA LOCAL
1 #AREA

2 #FIELD1 (A10)
2 #GROUP1 (1:10)

3 #FIELD2 (P2)
3 #FIELD3 (N1/1:4)

END-DEFINE
...

In this example, the data area #AREA has a total size of:

10 + (10 * (2 + (1 * 4))) bytes = 70 bytes

#FIELD1 is alphanumeric and 10 bytes long. #GROUP1 is the name of a sub-areawithin #AREA, which
consists of 2 fields and has 10 occurrences. #FIELD2 is packed numeric, length 2. #FIELD3 is the
second field of #GROUP1with four occurrences, and is numeric, length 1.

To reference a particular occurrence of #FIELD3, two indices are required: first, the occurrence of
#GROUP1must be specified, and second, the particular occurrence of #FIELD3must also be specified.
For example, in an ADD statement later in the sameprogram, #FIELD3would be referenced as follows:

ADD 2 TO #FIELD3 (3,2)

Database Arrays

Adabas supports array structures within the database in the form ofmultiple-value fields and
periodic groups. These are described under Database Arrays.

The following example shows a DEFINE DATA view containing a multiple-value field:

DEFINE DATA LOCAL
1 EMPLOYEES-VIEW VIEW OF EMPLOYEES

2 NAME
2 ADDRESS-LINE (1:10) /* <--- MULTIPLE-VALUE FIELD

END-DEFINE
...

The same view in a local data area would look as follows:

I T L Name F Leng Index/Init/EM/Name/Comment
- - - -------------------------------- - ---- ---------------------------------

V 1 EMPLOYEES-VIEW EMPLOYEES
2 NAME A 20

M 2 ADDRESS-LINE A 20 (1:10) /* MU-FIELD

Using Arithmetic Expressions in Index Notation

A simple arithmetic expression may also be used to express a range of occurrences in an array.

Examples:

Values of the field MA are referenced, beginningwith value I and endingwith value I+5.MA (I:I+5)

Values of the field MA are referenced, beginning with value I+2 and ending with value
J-3.

MA (I+2:J-3)

Only the arithmetic operators plus (+) and minus (-) may be used in index expressions.

Arithmetic Support for Arrays

Arithmetic support for arrays include operations at array level, at row/column level, and at indi-
vidual element level.

Only simple arithmetic expressions are permitted with array variables, with only one or two op-
erands and an optional third variable as the receiving field.

Only the arithmetic operators plus (+) and minus (-) are allowed for expressions defining index
ranges.

Examples of Array Arithmetics

The following examples assume the following field definitions:

DEFINE DATA LOCAL
01 #A (N5/1:10,1:10)
01 #B (N5/1:10,1:10)
01 #C (N5)
END-DEFINE
...

1. ADD #A(*,*) TO #B(*,*)

The result operand, array #B, contains the addition, element by element, of the array #A and the
original value of array #B.

2. ADD 4 TO #A(*,2)

The second column of the array #A is replaced by its original value plus 4.

3. ADD 2 TO #A(2,*)

The second row of the array #A is replaced by its original value plus 2.

4. ADD #A(2,*) TO #B(4,*)

The value of the second row of array #A is added to the fourth row of array #B.

5. ADD #A(2,*) TO #B(*,2)

This is an illegal operation and will result in a syntax error. Rows may only be added to rows
and columns to columns.

Programming Guide178

Arrays



6. ADD #A(2,*) TO #C

All values in the second row of the array #A are added to the scalar value #C.

7. ADD #A(2,5:7) TO #C

The fifth, sixth, and seventh column values of the second row of array #A are added to the
scalar value #C.

When an ordinary array field is defined, you have to specify the index bounds exactly, hence the
number of occurrences for each dimension. At runtime, the complete array field is existent by
default; each of its defined occurrences can be accessed without performing additional allocation
operations. The size layout cannot be changed anymore; you may neither add nor remove field
occurrences.

However, if the number of occurrences needed is unknown at development time, but you want
to flexibly increase or decrease the number of the array fields at runtime, you should use what is
called an X-array (eXtensible array).

An X-array can be resized at runtime and can help you manage memory more efficiently. For ex-
ample, you can use a large number of array occurrences for a short time and then reduce memory
when the application is no longer using the array.

Definition

An X-array is an array of which the number of occurrences is undefined at compile time. It is
defined in a DEFINE DATA statement by specifying an asterisk (*) for at least one index bound of
at least one array dimension. An asterisk (*) character in the index definition represents a variable
index boundwhich can be assigned to a definite value during program execution. Only one bound
- either upper or lower - may be defined as variable, but not both.

An X-array can be defined whenever a (fixed) array can be defined, i.e. at any level or even as an
indexed group. It cannot be used to access MU-/PE-fields of a database view. Amultidimensional
array may have a mixture of constant and variable bounds.

Example:

DEFINE DATA LOCAL
1 #X-ARR1 (A5/1:*)          /* lower bound is fixed at 1, upper bound is variable
1 #X-ARR2 (A5/*)            /* shortcut for (A5/1:*)
1 #X-ARR3 (A5/*:100)        /* lower bound is variable, upper bound is fixed at 100
1 #X-ARR4 (A5/1:10,1:*)     /* 1st dimension has a fixed index range with (1:10)
END-DEFINE                  /* 2nd dimension has fixed lower bound 1 and variable ↩
upper bound

Storage Management of X-Arrays

Occurrences of an X-array must be allocated explicitly before they can be accessed. To increase or
decrease the number of occurrences of a dimension, the statements EXPAND, RESIZE and REDUCE
may be used.

However, the number of dimensions of the X-array (1, 2 or 3 dimensions) cannot be changed.

Example:

DEFINE DATA LOCAL
1 #X-ARR(I4/10:*)
END-DEFINE
EXPAND ARRAY #X-ARR TO (10:10000)
/* #X-ARR(10) to #X-ARR(10000) are accessible
WRITE *LBOUND(#X-ARR) /* is 10

*UBOUND(#X-ARR) /* is 10000
*OCCURRENCE(#X-ARR) /* is 9991

#X-ARR(*) := 4711 /* same as #X-ARR(10:10000) := 4711
/* resize array from current lower bound=10 to upper bound =1000
RESIZE ARRAY #X-ARR TO (*:1000)
/* #X-ARR(10) to #X-ARR(1000) are accessible
/* #X-ARR(1001) to #X-ARR(10000) are released
WRITE *LBOUND(#X-ARR) /* is 10

*UBOUND(#X-ARR) /* is 1000
*OCCURRENCE(#X-ARR) /* is 991

/* release all occurrences
REDUCE ARRAY #X-ARR TO 0
WRITE *OCCURRENCE(#X-ARR) /* is 0

Storage Management of X-Group Arrays

If you want to increase or decrease occurrences of X-group arrays, you must distinguish between
independent and dependent dimensions.

A dimension which is specified directly (not inherited) for an X-(group) array is independent.

A dimension which is not specified directly, but inherited for an array is dependent.

Only independent dimensions of an X-array can be changed in the statements EXPAND, RESIZE and
REDUCE; dependent dimensions must be changed using the name of the corresponding X-group
array which owns this dimension as independent dimension.

Example - Independent/Dependent Dimensions:

DEFINE DATA LOCAL
1 #X-GROUP-ARR1(1:*) /* (1:*)

2 #X-ARR1 (I4) /* (1:*)
2 #X-ARR2 (I4/2:*) /* (1:*,2:*)
2 #X-GROUP-ARR2 /* (1:*)

3 #X-ARR3 (I4) /* (1:*)
3 #X-ARR4 (I4/3:*) /* (1:*,3:*)
3 #X-ARR5 (I4/4:*, 5:*) /* (1:*,4:*,5:*)

END-DEFINE

The following table shows whether the dimensions in the above program are independent or de-
pendent.

Independent DimensionDependent DimensionName

(1:*)#X-GROUP-ARR1

(1:*)#X-ARR1

(2:*)(1:*)#X-ARR2

(1:*)#X-GROUP-ARR2

(1:*)#X-ARR3

(3:*)(1:*)#X-ARR4

(4:*,5:*)(1:*)#X-ARR5

The only index notation permitted for a dependent dimension is either a single asterisk (*), a range
defined with asterisks (*:*) or the index bounds defined.

This is to indicate that the bounds of the dependent dimensionmust be kept as they are and cannot
be changed.

The occurrences of the dependent dimensions can only be changed by manipulating the corres-
ponding array groups.

EXPAND ARRAY #X-GROUP-ARR1 TO (1:11) /* #X-ARR1(1:11) are allocated
/* #X-ARR3(1:11) are allocated

EXPAND ARRAY #X-ARR2 TO (*:*, 2:12) /* #X-ARR2(1:11, 2:12) are allocated
EXPAND ARRAY #X-ARR2 TO (1:*, 2:12) /* same as before
EXPAND ARRAY #X-ARR2 TO (* , 2:12) /* same as before
EXPAND ARRAY #X-ARR4 TO (*:*, 3:13) /* #X-ARR4(1:11, 3:13) are allocated
EXPAND ARRAY #X-ARR5 TO (*:*, 4:14, 5:15) /* #X-ARR5(1:11, 4:14, 5:15) are allocated

The EXPAND statements may be coded in an arbitrary order.

The following use of the EXPAND statement is not allowed, since the arrays only have dependent
dimensions.

EXPAND ARRAY #X-ARR1 TO ...
EXPAND ARRAY #X-GROUP-ARR2 TO ...
EXPAND ARRAY #X-ARR3 TO ...

Referencing an X-Array

The occurrences of an X-array must be allocated by an EXPAND or RESIZE statement before they
can be accessed.

As a general rule, an attempt to address a non existent X-array occurrence leads to a runtime error.
In some statements, however, the access to a nonmaterialized X-array field does not cause an error
situation if all occurrences of an X-array are referenced using the complete range notation, for ex-
ample: #X-ARR(*). This applies to

■ parameters used in a CALL statement,
■ parameters used in the statements CALLNAT, PERFORM or OPEN DIALOG, if defined as optional
parameters,

■ source fields used in a COMPRESS statement,
■ output fields supplied in a PRINT statement,
■ fields referenced in a RESET statement.

If individual occurrences of a non materialized X-array are referenced in one of these statements,
a corresponding error message is issued.

Example:

DEFINE DATA LOCAL 
1 #X-ARR (A10/1:*) /* X-array only defined, but not allocated
END-DEFINE        
RESET #X-ARR(*)    /* no error, because complete field referenced with (*)   
RESET #X-ARR(1:3)  /* runtime error, because individual occurrences (1:3) are ↩
referenced
END  ↩

The asterisk (*) notation in an array reference stands for the complete range of a dimension. If the
array is an X-array, the asterisk is the index range of the currently allocated lower and upper bound
values, which are determined by the system variables *LBOUND and *UBOUND.

Parameter Transfer with X-Arrays

X-arrays that are used as parameters are treated in the same way as constant arrays with regard
to the verification of the following:

■ format,
■ length,
■ dimension or
■ number of occurrences.

In addition, X-array parameters can also change the number of occurrences using the statement
RESIZE, REDUCE or EXPAND. The question if a resize of an X-array parameter is permitted depends
on three factors:

■ the type of parameter transfer used, that is by reference or by value,
■ the definition of the caller or parameter X-array, and
■ the type of X-array range being passed on (complete range or subrange).

The following tables demonstrate when an EXPAND, RESIZE or REDUCE statement can be applied to
an X-array parameter.

Example with Call By Value

ParameterCaller

X-ArrayVariable (1:V)Static

yesnonoStatic

yesnonoX-array subrange, for example:

CALLNAT...#XA(1:5)

yesnonoX-array complete range, for example:

CALLNAT...#XA(*)

Call By Reference/Call By Value Result

ParameterCaller

X-Array with a fixed
upper bound, e.g.

X-Array with a fixed
lower bound, e.g.

Variable
(1:V)

Static

DEFINE DATA ↩
PARAMETER
1 #PX (A10/*:1)

DEFINE DATA ↩
PARAMETER
1 #PX (A10/1:*)

nonononoStatic

nonononoX-array subrange, for example:

CALLNAT...#XA(1:5)

noyesnonoX-Array with a fixed lower bound,
complete range, for example:

DEFINE DATA LOCAL
1 #XA(A10/1:*)
...
CALLNAT...#XA(*)

yesnononoX-Array with a fixed upper bound,
complete range, for example:

DEFINE DATA LOCAL
1 #XA(A10/*:1)
...
CALLNAT...#XA(*)

Parameter Transfer with X-Group Arrays

The declaration of an X-group array implies that each element of the group will have the same
values for upper boundary and lower boundary. Therefore, the number of occurrences of dependent
dimensions of fields of an X-group array can only be changed when the group name of the X-
group array is given with a RESIZE, REDUCE or EXPAND statement (see Storage Management of X-
Group Arrays above).

Members of X-group arrays may be transferred as parameters to X-group arrays defined in a
parameter data area. The group structures of the caller and the callee need not necessarily be

187Programming Guide

X-Arrays



identical. A RESIZE, REDUCE or EXPAND done by the callee is only possible as far as the X-group array
of the caller stays consistent.

Example - Elements of X-Group Array Passed as Parameters:

Program:

DEFINE DATA LOCAL
1 #X-GROUP-ARR1(1:*) /* (1:*)

2 #X-ARR1 (I4) /* (1:*)
2 #X-ARR2 (I4) /* (1:*)

1 #X-GROUP-ARR2(1:*) /* (1:*)
2 #X-ARR3 (I4) /* (1:*)
2 #X-ARR4 (I4) /* (1:*)

END-DEFINE
...
CALLNAT ... #X-ARR1(*) #X-ARR4(*)
...
END

Subprogram:

DEFINE DATA PARAMETER
1 #X-GROUP-ARR(1:*) /* (1:*)

2 #X-PAR1 (I4) /* (1:*)
2 #X-PAR2 (I4) /* (1:*)

END-DEFINE
...
RESIZE ARRAY #X-GROUP-ARR to (1:5)
...
END

The RESIZE statement in the subprogram is not possible. It would result in an inconsistent number
of occurrences of the fields defined in the X-group arrays of the program.

X-Array of Dynamic Variables

An X-array of dynamic variables may be allocated by first specifying the number of occurrences
using the EXPAND statement and then assigning a value to the previously allocated array occurrences.

Example:

DEFINE DATA LOCAL
1 #X-ARRAY(A/1:*) DYNAMIC
END-DEFINE
EXPAND ARRAY #X-ARRAY TO (1:10)

/* allocate #X-ARRAY(1) to #X-ARRAY(10) with zero length.
/* *LENGTH(#X-ARRAY(1:10)) is zero

#X-ARRAY(*) := 'abc'
/* #X-ARRAY(1:10) contains 'abc',
/* *LENGTH(#X-ARRAY(1:10)) is 3

EXPAND ARRAY #X-ARRAY TO (1:20)
/* allocate #X-ARRAY(11) to #X-ARRAY(20) with zero length
/* *LENGTH(#X-ARRAY(11:20)) is zero

#X-ARRAY(11:20) := 'def'
/* #X-ARRAY(11:20) contains 'def'
/* *LENGTH(#X-ARRAY(11:20)) is 3

Lower and Upper Bound of an Array

The system variables *LBOUND and *UBOUND contain the current lower and upper bound of an array
for the specified dimension(s): (1,2 or 3).

If no occurrences of an X-array have been allocated, the access to *LBOUND or *UBOUND is undefined
for the variable index bounds, that is, for the boundaries that are represented by an asterisk (*)
character in the index definition, and leads to a runtime error. In order to avoid a runtime error,
the system variable *OCCURRENCEmay be used to check against zero occurrences before *LBOUND
or *UBOUND is evaluated:

Example:

IF *OCCURRENCE (#A) NE 0 AND *UBOUND(#A) < 100 THEN ...

Related topics:

■ Natural object type function
■ Function Call
■ Natural statements DEFINE FUNCTION, DEFINE PROTOTYPE

Introduction to User-Defined Functions

A user-defined function is a programming object of type function, containing Natural statements
which implement a specific functional task. The invocation of a user-defined function (“function
call”) usually has a number of parameters and returns a result.

The syntactical representation of a function call is the function name (call-name) followed by a
special bracket notation containing the parameters, which is, for example: FCTNAME(<...>).

A function can be called from any place within a Natural statement where an operand is expected
which is only read, but notmodified. The result returned by the function is processed by the statement
at the place where the function call is embedded, like a field containing the same value.

A function can also be called in a stand-alonemode, outside of anyNatural statement. In this case,
the function result is not processed.

Usually, the result value returned by the function depends on the parameters provided with the
function call. As well as with a Natural subprogram, a parameter can be definedwithin a function
as a “by reference”, “by value” or “by value result” field. This makes it possible to provide para-
meter values in a function call which are only transfer compatible towhat is defined in the function
definition. Moreover, it allows you to exchange data between the calling object and the function
not only via the function result, but also via parameters. The correctness of the parameter list and
the compatibility of the result value is checked at compilation, either by means of an existing
cataloged function object or by means of a result and parameter layout definition described with
a DEFINE PROTOTYPE statement.

All function calls used in aNatural statement are evaluated in a separate step before the statement
execution is started. They are evaluated in the same order in which they appear in the statement.

Restrictions

At some places in a programming object, function calls cannot be used. This includes

■ positions where the operand value is changed by the Natural statement;
■ all kinds of database calls (for example, FIND, READ);
■ DEFINE DATA statement;
■ IF BREAK statement;
■ AT BREAK statement;
■ array index expressions;
■ Natural system functions (for example, AVER, SUM);
■ parameters of a function call.

Function Call versus Subprogram Call

The following is a comparison of the characteristics of a function call with those of a subprogram
call.

■ What is similar?
■ What is different?
■ Example of a Function Call
■ Example of a Subprogram Call

What is similar?

The following similarities exist between a function call and a subprogram call:

■ The programming code forming the routine logic is coded inside a separate object, either in a
function or a subprogram.

■ Parameters are defined in the object using a DEFINE DATA PARAMETER statement, with various
communication modes (for example, “by value”).

What is different?

The following differences exist between a function call and a subprogram call:

■ A function call can be used at any position in a Natural statement where a read-only operand
is possible (for exceptions, see Restrictions), whereas subprograms can be invoked only via the
CALLNAT statement.

■ A function returns a result valuewhich can instantly be processed by the statement that includes
the function call. The use of a temporary variable is not required. A CALLNAT statement can only
return data via its parameters. To process such a value with another statement, it needs to be
declared as an explicit variable.

■ Parameters and the result of a function call are always verified if the called function already
exists at compilation time. Subprogram calls are checked only if the compiler option PCHECK is
set to ON.

■ The name of a function, which is used to call the function, is definedwithin the DEFINE FUNCTION
statement and may differ from the name of the object containing the function. Similar to sub-
routines, the name of a function may have up to 32 characters. A subprogram is called by the
name of the subprogram object. Therefore, themaximumname length is limited to 8 characters.

Example of a Function Call

The following example shows a program calling a function, and the function definition created
with a DEFINE FUNCTION statement.

Program Calling the Function

** Example 'FUNCAX01': Calling a function (Program)
************************************************************************
*
WRITE 'Function call' F#ADD(< 2,3 >) /* Function call.

/* No temporary variables needed.
*
END

Definition of Function F#ADD

** Example 'FUNCAX02': Calling a function (Function)
************************************************************************
DEFINE FUNCTION F#ADD

RETURNS #RESULT (I4)
DEFINE DATA PARAMETER

1 #SUMMAND1 (I4) BY VALUE
1 #SUMMAND2 (I4) BY VALUE

END-DEFINE
/*
#RESULT := #SUMMAND1 + #SUMMAND2
/*

END-FUNCTION                                                            
*                                                                       
END ↩

Example of a Subprogram Call

To implement the same functionality as shown in the example of a function call by using a subpro-
gram call instead, you need to specify temporary variables.

Program Calling the Subprogram

The following example shows a program calling a subprogram, involving the use of a temporary
variable.

** Example 'FUNCAX03': Calling a subprogram (Program)
************************************************************************
DEFINE DATA LOCAL

1 #RESULT (I4) INIT <0>
END-DEFINE
*
CALLNAT 'FUNCAX04' #RESULT 2 3 /* Result is stored in #RESULT.
*
WRITE '=' #RESULT /* Print out the result of the

/* subprogram.
*
END

Called Subprogram FUNCAX04

** Example 'FUNCAX04': Calling a subprogram (Subprogram)
************************************************************************
DEFINE DATA PARAMETER

1 #RESULT (I4) BY VALUE RESULT
1 #SUMMAND1 (I4) BY VALUE
1 #SUMMAND2 (I4) BY VALUE

END-DEFINE                                                              
*                                                                       
#RESULT := #SUMMAND1 + #SUMMAND2                                        
*                                                                       
END ↩

Function Definition (DEFINE FUNCTION)

A function is created in a separate Natural object of type function. It contains a single DEFINE
FUNCTION statement, which defines the parameter interface, the result value and the program code
forming the operation logic.

Symbolic and Variable Function Call

There are two modes to call a function, either in a direct form or indirect form. In the direct form
(denoted as “symbolic” function call), the function name specified in the call is exactly the name
of the function itself. In the indirect form (denoted as “variable” function call), the name specified
in the function call is an alphanumeric variable with any name, which contains the name of the
called function at runtime.

To define a variable function call, it is always necessary to use a DEFINE PROTOTYPE VARIABLE
statement. Otherwise, the function call is assumed to be a symbolic function call; this means, the
name itself is regarded as the function name.

See Function Call for more details about this topic.

Function Result and Parameters

Usually, function calls are used within Natural statements instead of variables or constant values.
Since the compiler strictly requires the layout of the operands involved, it is essential to get the
format, length and array structure of a function result. Moreover, if the parameter structure of a
function is known, the parameters supplied in a function call can be checked for correctness at
compilation time.

There are three options to provide this information

■ with a DEFINE PROTOTYPE statement;
■ implicit by the existent object (cataloged version) of the called function, which is automatically
loaded by the compiler if no DEFINE PROTOTYPE statement was found before;

■ with an explicit (IR=) clause specified in the function call.

The simplest way is certainly to use the second option, as it always takes the information from the
object which is called at runtime. However, this is only possible if the cataloged object of the called
function is available. If a function is called with a variable function call, via a variable containing
the function name, the name of the function object is unknown and as a consequence cannot be

identified and loaded. In this case a DEFINE PROTOTYPE statement can be used to describe the
function interface. The first two options comprise a full validation of the result and the parameters.

If neither a DEFINE PROTOTYPE statement nor the existent object of a function call is available, two
casting options can be used in a function call. These are

■ (IR=) to specify the function result format/length/array structure. This option does not incorporate
any parameter checks.

■ (PT=) to use a previously defined prototype with a name other than the function name.

Explicit Prototype Definition (DEFINE PROTOTYPE)

To specify the interface of a certain function, a DEFINE PROTOTYPE statement can be used. This
statement defines the layout of the parameters, which are to be passed in a function call, and the
format/length of the result field returned by the function. Furthermore, it indicates whether a
function call is a “symbolic” or a “variable” function call.

When a function call was found, Natural tries to locate a prototype with the same name as the
used function name. If such a prototype is found, the function result and parameter layouts of this
prototype are used to resolve the function call. Especially, the decision on the function call mode
(“symbolic” or “variable”) is made at this place.

Implicit (Automatic) Prototype Definition

If a function call is resolved in a program, the compiler searches for a DEFINE PROTOTYPE statement
with the function name, which has been defined before. If such a statement cannot be found,
Natural tries to load the called function into the buffer pool. If successful, the function layout of
the result and and the parameters is extracted from the object data and kept as if it was provided
by an explicit DEFINE PROTOTYPE statement for this function. This manner of use is denoted as
automatic prototype definition. It assures great conformity between the interface definition (at compile
time) and the passing accuracy at runtime.

Prototype Cast (PT Clause)

In order to get the interface layout of a called function, Natural tries to locate a DEFINE PROTOTYPE
statement with the same name as the function identifier. If such a statement is not available, and
the called function object cannot be loaded (see Implicit (Automatic) Prototype Definition ), a
(PT=) clause can be specified in the function call. If such a clause is applied, the DEFINE PROTOTYPE
statementwith the referenced name (which is different from the function name) is used to describe
the function result and to validate the parameters.

Example:

#I := #MULT(<(PT=#ADD) 2 , 3>)

In this example, the function #MULT is called, but the result and parameter layouts are used from
a prototype whose name is #ADD.

Intermediate Result Definition (IR Clause)

Usually, the function result is specified by a DEFINE PROTOTYPE statement, which is either coded
explicitly or will be created automatically with the function object (see Implicit (Automatic) Pro-
totype Definition ). If such a definition is not available, the result layout can be specified by using
the (IR=) clause in the function call. If this clause is used, it determines which format/length the
compiler should use for the result field in the statement generation. This clause can also be specified
if the prototype definition is available for a function call. In this case, the result layout in the pro-
totype is overruled by the (IR=) clause specification; the parameter checks, however, are performed
according to the prototype definition.

Combinations of Possible Prototype Definitions

In order to resolve a function call, the compiler needs information on

■ the function call mode (symbolic or variable);
■ the layout (format/length) of the function result;
■ the layout (format/length) of the function parameters.

Different options allow you to provide this data, which are the explicit prototype definition, the
implicit prototype definition, the (PT=) option, and the (IR=) option. But which one has an effect
if multiple of these clauses are used?

A function call is used as a variable function call if there is a related prototype with the same
name,which contains a VARIABLE clause. In all other cases, the function call is treated as a symbolic
function call.

The result is determined in the following order:

■ the definition provided in (IR=), if this clause is specified;
■ the RETURNS definition in the prototype referenced in (PT=), if this clause is specified;
■ the explicit prototype definition (DEFINE PROTOTYPE) with the same name as used in the function
call, if it exists;

■ the implicit prototype definition,which is loaded automatically from the existing function object.

If none of these options applies, a syntax error is raised.

The parameter checks are performed according to the definition in:

■ the prototype definition referenced in (PT=), if this clause is specified;
■ the explicit prototype definition (DEFINE PROTOTYPE) with the same name as used in the function
call, if it exists;

■ the implicit prototype definition,which is loaded automatically from the existing function object.

If none of these options applies, the parameter validation is not performed. This allows you to
supply any number and layout of parameters in the function call, without receiving a syntax error.

Example with Multiple Definitions in a Function Call

Program:

** Example 'FUNCBX01': Declare result value and parameters (Program)
************************************************************************
*
DEFINE DATA LOCAL

1 #PROTO-NAME (A20)
1 #PARM1 (I4)
1 #PARM2 (I4)

END-DEFINE
*
DEFINE PROTOTYPE VARIABLE #PROTO-NAME

RETURNS (I4)
DEFINE DATA PARAMETER

1 #P1 (I4) BY VALUE OPTIONAL
1 #P2 (I4) BY VALUE

END-DEFINE
END-PROTOTYPE
*
#PROTO-NAME := 'F#MULTI'
#PARM1 := 3

#PARM2 := 5
*
WRITE #PROTO-NAME(<#PARM1, #PARM2>)
WRITE #PROTO-NAME(<1X ,5>)
*
WRITE F#MULTI(<(PT=#PROTO-NAME) #PARM1,#PARM2>)
*
WRITE F#MULTI(<(IR=N20) #PARM1, #PARM2>)
*
END

Function F#MULTI:

** Example 'FUNCBX02': Declare result value and parameters (Function)
************************************************************************
DEFINE FUNCTION F#MULTI

RETURNS #RESULT (I4) BY VALUE
DEFINE DATA PARAMETER

1 #FACTOR1 (I4) BY VALUE OPTIONAL
1 #FACTOR2 (I4) BY VALUE

END-DEFINE
/*
IF #FACTOR1 SPECIFIED

#RESULT := #FACTOR1 * #FACTOR2
ELSE

#RESULT := #FACTOR2 * 10
END-IF
/*

END-FUNCTION                                                            
*                                                                       
END ↩

Evaluation Sequence of Functions in Statements

Instead of operands, function calls can be used directly in statements. However, this is only allowed
with operands which are only read, but not modified by the statement.

All function calls are evaluated before the statement execution starts. The returned result values
are stored in temporary fields and passed to the statement. The functions are executed in the same
order in which they appear in the statement. If a function call has parameters which are modified
by the function execution, you should consider that this can influence the statement result. This
may apply if the same parameter is used at another place in the same statement.

Example:

Before the COMPUTE statement is started, variable #I has the value 1. In a first step, function F#RETURN
is executed. This changes #I to value 2 and returns the same value as the function result. After

Programming Guide202

User-Defined Functions



this, the COMPUTE operation starts and adds the incremented #I (2) and the temporary field (2)
to a sum of 4.

Program:

** Example 'FUNCCX01': Parameter changed within function (Program)
************************************************************************
DEFINE DATA LOCAL

1 #I (I2) INIT <1>
1 #RESULT (I2)

END-DEFINE
*
COMPUTE #RESULT := #I + F#RETURN(<#I>) /* First evaluate function call,

/* then execute the addition.
*
WRITE '#I :' #I /

'#RESULT:' #RESULT
*
END

Function:

** Example 'FUNCCX02': Parameter changed within function (Function)
************************************************************************
DEFINE FUNCTION F#RETURN

RETURNS #RESULT (I2) BY VALUE
DEFINE DATA PARAMETER

1 #PARM1 (I2) BY VALUE RESULT
END-DEFINE
/*
#PARM1 := #PARM1 + 1 /* Increment parameter.
#RESULT := #PARM1 /* Set result value.
/*

END-FUNCTION                                                            
*                                                                       
END ↩

Output of Program FUNCCX01:

#I     :      2    
#RESULT:      4 ↩

Using a Function as a Statement

A function can also be called stand-alone, without being embedded in other statements. In this
case, the function return value is completely ignored.

If such an execution mode is desired, only the function call is coded, which then stands for a
statement. In order to prevent an unwanted link to the previous statement in the source code, a
semicolon must be used to explicitly separate the function call from this statement.

Example:

Program:

** Example 'FUNCDX01': Using a function as a statement (Program)
************************************************************************
DEFINE DATA LOCAL

1 #A (I4) INIT <1>
1 #B (I4) INIT <2>

END-DEFINE
*
*
WRITE 'Write:' #A #B
F#PRINT-ADD(< 2,3 >) /* Function call belongs to operand list

/* immediately preceding it.
*
WRITE // '*************************' //
*
WRITE 'Write:' #A #B; /* Semicolon separates operands and function.
F#PRINT-ADD(< 2,3 >) /* Function call does not belong to the

/* operand list.
*
END

Function:

** Example 'FUNCDX02': Using a function as a statement (Function)
************************************************************************
DEFINE FUNCTION F#PRINT-ADD

RETURNS (I4)
DEFINE DATA PARAMETER

1 #SUMMAND1 (I4) BY VALUE
1 #SUMMAND2 (I4) BY VALUE

END-DEFINE
/*
F#PRINT-ADD := #SUMMAND1 + #SUMMAND2 /* Result of function call.
WRITE 'Function call:' F#PRINT-ADD
/*

END-FUNCTION                                                            

*                                                                       
END ↩

Output of Program FUNCDX01:

Function call:           5                       
Write:           1           2           5       

*************************                        

Write:           1           2                   
Function call:           5 ↩

V Accessing Data in a Database

This part describes various aspects of accessing data in a database with Natural.

Natural and Database Access

Accessing Data in an Adabas Database

This chapter gives an overview of the facilities that Natural provides for accessing different types
of database management systems.

Database Management Systems Supported by Natural

Natural offers specific database interfaces for the following types of databasemanagement systems
(DBMS):

■ Nested-relational DBMS (Adabas)

The following topics are covered below:

■ Adabas

Adabas

Via its integratedAdabas interface, Natural can access Adabas databases either on a localmachine
or on remote computers. For remote access, an additional routing and communication software
such as Entire Net-Work is necessary. In any case, the type of host machine running the Adabas
database is transparent for the Natural user.

Profile Parameters Influencing Database Access

There are variousNatural profile parameters to define howNatural handles the access to databases.

For a detailed parameter description, refer to the corresponding section in the Parameter Reference.

Access through Data Definition Modules

To enable convenient and transparent access to the different database management systems, a
special object, the “data definition module” (DDM), is used in Natural. This DDM establishes the
connection between the Natural data structures and the data structures in the database system to
be used. Such a database structuremight be a table in an SQLdatabase, a file in anAdabas database.
Hence, the DDM hides the real structure of the database accessed from the Natural application.
DDMs are created using the Natural DDM editor.

Natural is capable of accessing multiple types of databases from within a single application by
using references toDDMs that represent the specific data structures in the specific database system.
The diagram below shows an application that connects to different types of database.

Natural's Data Manipulation Language

Natural has a built-in datamanipulation language (DML) that allowsNatural applications to access
all database systems supported byNatural using the same language statements such as FIND, READ,
STORE or DELETE. These statements can be used in a Natural application without knowing the type
of database that is going to be accessed.

Natural determines the real type of database system from its configuration files and translates the
DML statements into database-specific commands; that is, it generates direct commands for
Adabas, SQL statement strings and host variable structures for SQL databases.

Because some of the Natural DML statements provide functionality that cannot be supported for
all database types, the use of this functionality is restricted to specific database systems. Please,
note the corresponding database-specific considerations in the statements documentation.

Natural's Special SQL Statements

In addition to the “normal” Natural DML statements, Natural provides a set of SQL statements
for a more specific use in conjunction with SQL database systems; see SQL Statements Overview
(in the Statements documentation).

Flexible SQL and facilities forworkingwith stored procedures complete the set of SQL commands.
These statements can be used for SQL database access only and are not valid for Adabas or other
non-SQL-databases.

This chapter describes various aspects of accessing data in an Adabas database with Natural.

Adabas Database Management Interfaces ADA and ADA2

Natural'sAdabas databasemanagement interfacesADAandADA2are considereddistinct database
interfaces like, for example, ADA and SQL.

Database type ADA is Natural's default interface to Adabas databases. It is the appropriate choice
if new Adabas functionality as introduced with Adabas Version 6 on Open Systems and Adabas
Version 8 on mainframes is not concerned.

Database type ADA2 is provided as an extended interface to Adabas databases as of Version 6 on
Open Systems and Adabas as of Version 8 on mainframes. In particular, it supports Adabas LA
fields, Adabas large object fields and extended Adabas buffer lengths. The support of Adabas LA
and large object fields implies the use of Natural format (A) DYNAMIC in a view definition, the
support of extended Adabas buffer lengths enables the definition of view sizes that exceed 64 KB.
For further information, refer to Defining a Database View.

Database type ADA2 does not supportmulti-fetch processing. Corresponding global and local
definitions are ignored at runtime.

SoftwareAGproductswhich have their own systemfiles require a corresponding physical database
of database type ADA.

Natural objects that were compiled with database type ADA can be executed in an environment
where the corresponding Adabas database is defined as database type ADA2.

Data Definition Modules - DDMs

For Natural to be able to access a database file, a logical definition of the physical database file is
required. Such a logical file definition is called a data definition module (DDM).

This section covers the following topics:

Programming Guide214

Accessing Data in an Adabas Database



■ Use of Data Definition Modules

Use of Data Definition Modules

The data definitionmodule contains information about the individual fields of the file - information
which is relevant for the use of these fields in a Natural program. A DDM constitutes a logical
view of a physical database file.

For each physical file of a database, one or more DDMs can be defined. And for each DDM one
or more data views can be defined as described View Definition in the DEFINE DATA statement
documentation and explained in the section Defining a Database View.

DDMs are defined by the Natural administrator with Predict (or, if Predict is not available, with
the corresponding Natural function).

Database Arrays

Adabas supports array structures within the database in the form of multiple-value fields and
periodic groups.

This section covers the following topics:

■ Multiple-Value Fields
■ Periodic Groups
■ Referencing Multiple-Value Fields and Periodic Groups
■ Multiple-Value Fields within Periodic Groups
■ Referencing Multiple-Value Fields within Periodic Groups
■ Referencing the Internal Count of a Database Array

Multiple-Value Fields

A multiple-value field is a field which can have more than one value (up to 65534, depending on
the Adabas version and definition of the FDT) within a given record.

Example:

Assuming that the above is a record in an employees file, the first field (Name) is an elementary
field, which can contain only one value, namely the name of the person; whereas the second field
(Languages), which contains the languages spoken by the person, is a multiple-value field, as a
person can speak more than one language.

Periodic Groups

Aperiodic group is a group of fields (whichmay be elementary fields and/ormultiple-value fields)
that may have more than one occurrence (up to 65534, depending on the Adabas version and
definition of the field definition table (FDT)) within a given record.

The different values of a multiple-value field are usually called “occurrences”; that is, the number
of occurrences is the number of values which the field contains, and a specific occurrence means
a specific value. Similarly, in the case of periodic groups, occurrences refer to a group of values.

Example:

Assuming that the above is a record in a vehicles file, the first field (Name) is an elementary field
which contains the name of a person; Cars is a periodic group which contains the automobiles
owned by that person. The periodic group consists of three fields which contain the registration
number, make andmodel of each automobile. Each occurrence of Cars contains the values for one
automobile.

Referencing Multiple-Value Fields and Periodic Groups

To reference one or more occurrences of a multiple-value field or a periodic group, you specify
an “index notation” after the field name.

Examples:

The following examples use the multiple-value field LANGUAGES and the periodic group CARS from
the previous examples.

The various values of the multiple-value field LANGUAGES can be referenced as follows.

ExplanationExample

References the first value (SPANISH).LANGUAGES (1)

The value of the variable X determines the value to be referenced.LANGUAGES (X)

References the first three values (SPANISH, CATALAN and FRENCH).LANGUAGES (1:3)

References the sixth to tenth values.LANGUAGES (6:10)

The values of the variables X and Y determine the values to be referenced.LANGUAGES (X:Y)

The various occurrences of the periodic group CARS can be referenced in the same manner:

ExplanationExample

References the first occurrence (B-123ABC/SEAT/IBIZA).CARS (1)

The value of the variable X determines the occurrence to be referenced.CARS (X)

References the first two occurrences (B-123ABC/SEAT/IBIZA and B-999XYZ/VW/GOLF).CARS (1:2)

References the fourth to seventh occurrences.CARS (4:7)

The values of the variables X and Y determine the occurrences to be referenced.CARS (X:Y)

Multiple-Value Fields within Periodic Groups

An Adabas array can have up to two dimensions: a multiple-value field within a periodic group.

Example:

Assuming that the above is a record in a vehicles file, the first field (Name) is an elementary field
which contains the name of a person; Cars is a periodic group, which contains the automobiles
owned by that person. This periodic group consists of three fields which contain the registration
number, servicing dates and make of each automobile. Within the periodic group Cars, the field
Servicing is a multiple-value field, containing the different servicing dates for each automobile.

Referencing Multiple-Value Fields within Periodic Groups

To reference one ormore occurrences of amultiple-value fieldwithin a periodic group, you specify
a “two-dimensional” index notation after the field name.

Examples:

The following examples use the multiple-value field SERVICINGwithin the periodic group CARS
from the example above. The various values of themultiple-value field can be referenced as follows:

ExplanationExample

References the first value of SERVICING in the first occurrence of CARS
(31-05-97).

SERVICING (1,1)

References the first value of SERVICING in the first five occurrences of CARS.SERVICING (1:5,1)

References the first ten values of SERVICING in the first five occurrences of
CARS.

SERVICING (1:5,1:10)

Referencing the Internal Count of a Database Array

It is sometimes necessary to reference a multiple-value field or a periodic groupwithout knowing
how many values/occurrences exist in a given record. Adabas maintains an internal count of the
number of values in each multiple-value field and the number of occurrences of each periodic
group. This count may be read in a READ statement by specifying C* immediately before the field
name.

The count is returned in format/length N3. See Referencing the Internal Count for a Database
Array for further details.

ExplanationExample

Returns the number of values of the multiple-value field LANGUAGES.C*LANGUAGES

Returns the number of occurrences of the periodic group CARS.C*CARS

Returns the number of values of the multiple-value field SERVICING in the first
occurrence of a periodic group (assuming that SERVICING is a multiple-value field
within a periodic group.)

C*SERVICING (1)

Defining a Database View

To be able to use database fields in a Natural program, you must specify the fields in a database
view.

In the view, you specify the name of the data definition module (see Data Definition Modules -
DDMs) from which the fields are to be taken, and the names of the database fields themselves
(that is, their long names, not their database-internal short names).

The view may comprise an entire DDM or only a subset of it. The order of the fields in the view
need not be the same as in the underlying DDM.

As described in the section Statements forDatabaseAccess, the viewname is used in the statements
READ, FIND, HISTOGRAM to determine which database is to be accessed.

For further information on the complete syntax of the viewdefinition option or on the definition/re-
definition of a group of fields, see View Definition in the description of the DEFINE DATA statement
in the Statements documentation.

Basically, you have the following options to define a database view:

■ Inside the Program
You can define a database view inside the program, that is, directly within the DEFINE DATA
statement of the program.

■ Outside the Program
You can define a database view outside the program, that is, in a separate programming object:
either a local data area (LDA) or a global data area (GDA), with the DEFINE DATA statement of
the program referencing that data area.

To define a database view inside the program

1 At Level 1, specify the view name as follows:

1 view-name VIEW OF ddm-name

where view-name is the name you choose for the view, ddm-name is the name of the DDM
from which the fields specified in the view are taken.

2 At Level 2, specify the names of the database fields from the DDM.

In the illustration below, the name of the view is ABC, and it comprises the fields NAME,
FIRST-NAME and PERSONNEL-ID from the DDM XYZ.

In the view, the format and length of a database field need not be specified, as these are already
defined in the underlying DDM.

Sample Program:

In this example, the view-name is VIEWEMP, and the ddm-name is EMPLOYEES, and the names of
the database fields taken from the DDM are NAME, FIRST-NAME and PERSONNEL-ID.

DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 PERSONNEL-ID

1 #VARI-A (A20)
1 #VARI-B (N3.2)
1 #VARI-C (I4)
END-DEFINE
...

To define a database view outside the program

1 In the program, specify:

DEFINE DATA LOCAL
USING <data-area-name>

END-DEFINE
...

where data-area-name is the name you choose for the local or global data area, for example,
LDA39.

2 In the data area to be referenced:

1. At Level 1 in the Name column, specify the name you choose for the view, and in the
Miscellaneous column, the name of the DDM from which the fields specified in the view
are taken.

2. At Level 2, specify the names of the database fields from the DDM.

Example LDA39:

In this example, the view name is VIEWEMP, the DDM name is EMPLOYEES, and the names
of the database fields taken from the DDM are PERSONNEL-ID, FIRST-NAME and NAME.

I T L  Name                             F Length     Miscellaneous          ↩
   
All -- -------------------------------- - ---------- ------------------------->
  V  1 VIEWEMP                                       EMPLOYEES              ↩
   
     2 PERSONNEL-ID                     A          8                        ↩
   
     2 FIRST-NAME                       A         20                        ↩
   
     2 NAME                             A         20                        ↩

     1 #VARI-A                          A         20                        ↩
   
     1 #VARI-B                          N        3.2                        ↩
   
     1 #VARI-C                          I          4                        ↩
  ↩

Considerations Concerning Databases of Type ADA2

With databases of type ADA2, the following applies:

■ If large alphanumeric (LA) or large object (LOB) fields (Adabas LA/LB option) are to be used,
these fields can be specifiedwithin the viewdefinitionwith both fixed format/length, for example,
A20 or U20, and dynamic format/length, for example, (A)DYNAMIC or U(DYNAMIC).

■ Length indicator fields L@... can also be specifiedwithin views if they are related to LA or LOB
fields.

Statements for Database Access

To read data from a database, the following statements are available:

MeaningStatement

Select a range of records from a database in a specified sequence.READ

Select from a database those records which meet a specified search criterion.FIND

Read only the values of one database field, or determine the number of records which meet
a specified search criterion.

HISTOGRAM

READ Statement

The following topics are covered:

■ Use of READ Statement
■ Basic Syntax of READ Statement
■ Example of READ Statement
■ Limiting the Number of Records to be Read
■ STARTING/ENDING Clauses
■ WHERE Clause

■ Further Example of READ Statement

Use of READ Statement

The READ statement is used to read records from a database. The records can be retrieved from the
database

■ in the order in which they are physically stored in the database (READ IN PHYSICAL SEQUENCE),
or

■ in the order of Adabas Internal Sequence Numbers (READ BY ISN), or
■ in the order of the values of a descriptor field (READ IN LOGICAL SEQUENCE).

In this document, only READ IN LOGICAL SEQUENCE is discussed, as it is the most frequently used
form of the READ statement.

For information on the other two options, please refer to the description of the READ statement in
the Statements documentation.

Basic Syntax of READ Statement

The basic syntax of the READ statement is:

READ view IN LOGICAL SEQUENCE BY descriptor

or shorter:

READ view LOGICAL BY descriptor

- where

is the name of a view defined in the DEFINE DATA statement and as explained in Defining
a Database View.

view

is the name of a database field defined in that view. The values of this field determine the
order in which the records are read from the database.

descriptor

If you specify a descriptor, you need not specify the keyword LOGICAL:

READ view BY descriptor

If you do not specify a descriptor, the recordswill be read in the order of values of the field defined
as default descriptor (under Default Sequence) in theDDM. However, if you specify no descriptor,
you must specify the keyword LOGICAL:

READ view LOGICAL

Example of READ Statement

** Example 'READX01': READ
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 PERSONNEL-ID
2 JOB-TITLE

END-DEFINE
*
READ (6) MYVIEW BY NAME

DISPLAY NAME PERSONNEL-ID JOB-TITLE
END-READ
END

Output of Program READX01:

With the READ statement in this example, records from the EMPLOYEES file are read in alphabetical
order of their last names.

The program will produce the following output, displaying the information of each employee in
alphabetical order of the employees' last names.

Page 1 04-11-11 14:15:54

NAME PERSONNEL CURRENT
ID POSITION

-------------------- --------- -------------------------

ABELLAN 60008339 MAQUINISTA
ACHIESON 30000231 DATA BASE ADMINISTRATOR
ADAM 50005800 CHEF DE SERVICE
ADKINSON 20008800 PROGRAMMER
ADKINSON 20009800 DBA
ADKINSON 2001100

If you wanted to read the records to create a report with the employees listed in sequential order
by date of birth, the appropriate READ statement would be:

READ MYVIEW BY BIRTH

You can only specify a field which is defined as a “descriptor” in the underlyingDDM (it can also
be a subdescriptor, superdescriptor, hyperdescriptor or phonetic descriptor or a non-descriptor).

Limiting the Number of Records to be Read

As shown in the previous example program, you can limit the number of records to be read by
specifying a number in parentheses after the keyword READ:

READ (6) MYVIEW BY NAME

In that example, the READ statement would read no more than 6 records.

Without the limit notation, the above READ statement would read all records from the EMPLOYEES
file in the order of last names from A to Z.

STARTING/ENDING Clauses

The READ statement also allows you to qualify the selection of records based on the value of a
descriptor field. With an EQUAL TO/STARTING FROM option in the BY clause, you can specify the
value at which reading should begin. (Instead of using the keyword BY, you may specify the
keyword WITH, which would have the same effect). By adding a THRU/ENDING AT option, you can
also specify the value in the logical sequence at which reading should end.

For example, if youwanted a list of those employees in the order of job titles startingwith TRAINEE
and continuing on to Z, you would use one of the following statements:

READ MYVIEW WITH JOB-TITLE = 'TRAINEE'
READ MYVIEW WITH JOB-TITLE STARTING FROM 'TRAINEE'
READ MYVIEW BY JOB-TITLE = 'TRAINEE'
READ MYVIEW BY JOB-TITLE STARTING FROM 'TRAINEE'

Note that the value to the right of the equal sign (=) or STARTING FROM option must be enclosed in
apostrophes. If the value is numeric, this text notation is not required.

The sequence of records to be read can be evenmore closely specified by adding an end limit with
a THRU/ENDING AT clause.

To read just the records with the job title TRAINEE, you would specify:

READ MYVIEW BY JOB-TITLE STARTING FROM 'TRAINEE' THRU 'TRAINEE'
READ MYVIEW WITH JOB-TITLE EQUAL TO 'TRAINEE'

ENDING AT 'TRAINEE'

To read just the records with job titles that begin with A or B, you would specify:

READ MYVIEW BY JOB-TITLE = 'A' THRU 'C'
READ MYVIEW WITH JOB-TITLE STARTING FROM 'A' ENDING AT 'C'

The values are read up to and including the value specified after THRU/ENDING AT. In the two ex-
amples above, all records with job titles that begin with A or B are read; if there were a job title C,
this would also be read, but not the next higher value CA.

WHERE Clause

The WHERE clause may be used to further qualify which records are to be read.

For instance, if you wanted only those employees with job titles starting from TRAINEEwho are
paid in US currency, you would specify:

READ MYVIEW WITH JOB-TITLE = 'TRAINEE'
WHERE CURR-CODE = 'USD'

The WHERE clause can also be used with the BY clause as follows:

READ MYVIEW BY NAME
WHERE SALARY = 20000

The WHERE clause differs from the BY clause in two respects:

■ The field specified in the WHERE clause need not be a descriptor.
■ The expression following the WHERE option is a logical condition.

The following logical operators are possible in a WHERE clause:

=EQEQUAL

¬=NENOT EQUAL TO

<LTLESS THAN

<=LELESS THAN OR EQUAL TO

>GTGREATER THAN

>=GEGREATER THAN OR EQUAL TO

The following program illustrates the use of the STARTING FROM, ENDING AT and WHERE clauses:

** Example 'READX02': READ (with STARTING, ENDING and WHERE clause)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 JOB-TITLE
2 INCOME (1:2)

3 CURR-CODE
3 SALARY
3 BONUS (1:1)

END-DEFINE
*
READ (3) MYVIEW WITH JOB-TITLE
STARTING FROM 'TRAINEE' ENDING AT 'TRAINEE'

WHERE CURR-CODE (*) = 'USD'
DISPLAY NOTITLE NAME / JOB-TITLE 5X INCOME (1:2)
SKIP 1

END-READ
END

Output of Program READX02:

NAME INCOME
CURRENT
POSITION CURRENCY ANNUAL BONUS

CODE SALARY
------------------------- -------- ---------- ----------

SENKO USD 23000 0
TRAINEE USD 21800 0

BANGART USD 25000 0
TRAINEE USD 23000 0

LINCOLN USD 24000 0
TRAINEE USD 22000 0

Further Example of READ Statement

See the following example program:

■ READX03 - READ statement

FIND Statement

The following topics are covered:

■ Use of FIND Statement
■ Basic Syntax of FIND Statement
■ Limiting the Number of Records to be Processed
■ WHERE Clause
■ Example of FIND Statement with WHERE Clause
■ IF NO RECORDS FOUND Condition
■ Further Examples of FIND Statement

Use of FIND Statement

The FIND statement is used to select from a database those records which meet a specified search
criterion.

Basic Syntax of FIND Statement

The basic syntax of the FIND statement is:

FIND RECORDS IN view WITH field = value

or shorter:

FIND view WITH field = value

- where

is the name of a view as defined in the DEFINE DATA statement and as explained in Defining a
Database View.

view

is the name of a database field as defined in that view.field

You can only specify a fieldwhich is defined as a “descriptor” in the underlyingDDM (it can
also be a subdescriptor, superdescriptor, hyperdescriptor or phonetic descriptor).

For the complete syntax, refer to the FIND statement documentation.

Limiting the Number of Records to be Processed

In the same way as with the READ statement described above, you can limit the number of records
to be processed by specifying a number in parentheses after the keyword FIND:

FIND (6) RECORDS IN MYVIEW WITH NAME = 'CLEGG'

In the above example, only the first 6 records that meet the search criterion would be processed.

Without the limit notation, all records that meet the search criterion would be processed.

Note: If the FIND statement contains a WHERE clause (see below), records which are rejected
as a result of the WHERE clause are not counted against the limit.

WHERE Clause

With the WHERE clause of the FIND statement, you can specify an additional selection criterionwhich
is evaluated after a record (selected with the WITH clause) has been read and before any processing
is performed on the record.

Example of FIND Statement with WHERE Clause

** Example 'FINDX01': FIND (with WHERE)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 JOB-TITLE
2 CITY

END-DEFINE
*
FIND MYVIEW WITH CITY = 'PARIS'

WHERE JOB-TITLE = 'INGENIEUR COMMERCIAL'
DISPLAY NOTITLE CITY JOB-TITLE PERSONNEL-ID NAME

END-FIND
END

Note: In this example only those records which meet the criteria of the WITH clause and the
WHERE clause are processed in the DISPLAY statement.

Output of Program FINDX01:

CITY CURRENT PERSONNEL NAME
POSITION ID

-------------------- ------------------------- --------- --------------------

PARIS INGENIEUR COMMERCIAL 50007300 CAHN
PARIS INGENIEUR COMMERCIAL 50006500 MAZUY
PARIS INGENIEUR COMMERCIAL 50004700 FAURIE
PARIS INGENIEUR COMMERCIAL 50004400 VALLY
PARIS INGENIEUR COMMERCIAL 50002800 BRETON
PARIS INGENIEUR COMMERCIAL 50001000 GIGLEUX
PARIS INGENIEUR COMMERCIAL 50000400 KORAB-BRZOZOWSKI

IF NO RECORDS FOUND Condition

If no records are found that meet the search criteria specified in the WITH and WHERE clauses, the
statementswithin the FIND processing loop are not executed (for the previous example, thiswould
mean that the DISPLAY statement would not be executed and consequently no employee data
would be displayed).

However, the FIND statement also provides an IF NO RECORDS FOUND clause, which allows you to
specify processing you wish to be performed in the case that no records meet the search criteria.

Example:

** Example 'FINDX02': FIND (with IF NO RECORDS FOUND)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME

END-DEFINE
*
FIND MYVIEW WITH NAME = 'BLACKSMITH'

IF NO RECORDS FOUND
WRITE 'NO PERSON FOUND.'

END-NOREC
DISPLAY NAME FIRST-NAME

END-FIND
END

The above program selects all records in which the field NAME contains the value BLACKSMITH. For
each selected record, the name andfirst name are displayed. If no recordwith NAME = 'BLACKSMITH'
is found on the file, the WRITE statement within the IF NO RECORDS FOUND clause is executed.

Output of Program FINDX02:

Page 1 04-11-11 14:15:54

NAME FIRST-NAME
-------------------- --------------------

NO PERSON FOUND.

Further Examples of FIND Statement

See the following example programs:

■ FINDX07 - FIND (with several clauses)
■ FINDX08 - FIND (with LIMIT)
■ FINDX09 - FIND (using *NUMBER, *COUNTER, *ISN)
■ FINDX10 - FIND (combined with READ)
■ FINDX11 - FIND NUMBER (with *NUMBER)

HISTOGRAM Statement

The following topics are covered:

■ Use of HISTOGRAM Statement
■ Syntax of HISTOGRAM Statement
■ Limiting the Number of Values to be Read
■ STARTING/ENDING Clauses
■ WHERE Clause
■ Example of HISTOGRAM Statement

Use of HISTOGRAM Statement

The HISTOGRAM statement is used to either read only the values of one database field, or determine
the number of records which meet a specified search criterion.

The HISTOGRAM statement does not provide access to any database fields other than the one specified
in the HISTOGRAM statement.

Syntax of HISTOGRAM Statement

The basic syntax of the HISTOGRAM statement is:

HISTOGRAM VALUE IN view FOR field

or shorter:

HISTOGRAM view FOR field

- where

is the name of a view as defined in the DEFINE DATA statement and as explained in Defining a
Database View.

view

is the name of a database field as defined in that view.field

For the complete syntax, refer to the HISTOGRAM statement documentation.

Limiting the Number of Values to be Read

In the same way as with the READ statement, you can limit the number of values to be read by
specifying a number in parentheses after the keyword HISTOGRAM:

HISTOGRAM (6) MYVIEW FOR NAME

In the above example, only the first 6 values of the field NAMEwould be read.

Without the limit notation, all values would be read.

STARTING/ENDING Clauses

Like the READ statement, the HISTOGRAM statement also provides a STARTING FROM clause and an
ENDING AT (or THRU) clause to narrow down the range of values to be read by specifying a
starting value and ending value.

Examples:

HISTOGRAM MYVIEW FOR NAME STARTING from 'BOUCHARD'
HISTOGRAM MYVIEW FOR NAME STARTING from 'BOUCHARD' ENDING AT 'LANIER'
HISTOGRAM MYVIEW FOR NAME from 'BLOOM' THRU 'ROESER'

WHERE Clause

The HISTOGRAM statement also provides a WHERE clausewhichmay be used to specify an additional
selection criterion that is evaluated after a value has been read and before any processing is performed
on the value. The field specified in the WHERE clause must be the same as in the main clause of the
HISTOGRAM statement.

Example of HISTOGRAM Statement

** Example 'HISTOX01': HISTOGRAM
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 CITY
END-DEFINE
*
LIMIT 8
HISTOGRAM MYVIEW CITY STARTING FROM 'M'

DISPLAY NOTITLE CITY 'NUMBER OF/PERSONS' *NUMBER *COUNTER
END-HISTOGRAM
END

In this program, the system variables *NUMBER and *COUNTER are also evaluated by the HISTOGRAM
statement, and outputwith the DISPLAY statement. *NUMBER contains the number of database records
that contain the last value read; *COUNTER contains the total number of values which have been
read.

Output of Program HISTOX01:

CITY NUMBER OF CNT
PERSONS

-------------------- ----------- -----------

MADISON 3 1
MADRID 41 2
MAILLY LE CAMP 1 3
MAMERS 1 4
MANSFIELD 4 5
MARSEILLE 2 6
MATLOCK 1 7
MELBOURNE 2 8

Multi-Fetch Clause

This section covers the multi-fetch record retrieval functionality for Adabas databases.

The multi-fetch functionality described in this section is only supported for databases of type ADA.
With database type ADA2, the multi-fetch clause is not supported.

The following topics are covered:

■ Purpose of Multi-Fetch Feature
■ Statements Supported
■ Considerations for Multi-Fetch Usage

Purpose of Multi-Fetch Feature

In standard mode, Natural does not read multiple records with a single database call; it always
operates in a one-record-per-fetch mode. This kind of operation is solid and stable, but can take
some time if a large number of database records are being processed. To improve the performance
of those programs, you can use multi-fetch processing.

By default, Natural uses single-fetch to retrieve data from Adabas databases. This default can be
configured using the Natural profile parameter MFSET.

Values ON (multi-fetch) and OFF (single-fetch) define the default behavior. If MFSET is set to NEVER,
Natural always uses single-fetch mode and ignores any settings at statement level.

The default processing mode can also be overridden at statement level.

Statements Supported

Multi-fetch processing is supported for the following statements that do not involve database
modification:

■ FIND

■ READ

■ HISTOGRAM

For more information on the syntax, see the description of the MULTI-FETCH clause of the FIND,
READ or HISTOGRAM statements.

Considerations for Multi-Fetch Usage

If nested database loops that refer to the same Adabas file contain UPDATE statements in one of the
inner loops, Natural continues processing the outer loops with the updated values. This implies
in multi-fetch mode, that an outer logical READ loop has to be repositioned if an inner database
loop updates the value of the descriptor that is used for sequence control in the outer loop. If this
attempt leads to a conflict for the current descriptor, an error is returned. To avoid this situation,
we recommend that you disable multi-fetch in the outer database loops.

In general, multi-fetch mode improves performance when accessing Adabas databases. In some
cases, however, it might be advantageous to use single-fetch to enhance performance, especially
if database modifications are involved.

Database Processing Loops

This section discusses processing loops required to process data that have been selected from a
database as a result of a FIND, READ or HISTOGRAM statement.

The following topics are covered:

■ Creation of Database Processing Loops
■ Hierarchies of Processing Loops
■ Example of Nested FIND Loops Accessing the Same File
■ Further Examples of Nested READ and FIND Statements

Creation of Database Processing Loops

Natural automatically creates the necessary processing loops which are required to process data
that have been selected from a database as a result of a FIND, READ or HISTOGRAM statement.

Example:

In the following exampe, the FIND loop selects all records from the EMPLOYEES file in which the
field NAME contains the value ADKINSON and processes the selected records. In this example, the
processing consists of displaying certain fields from each record selected.

** Example 'FINDX03': FIND
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY

END-DEFINE
*

FIND MYVIEW WITH NAME = 'ADKINSON'
DISPLAY NAME FIRST-NAME CITY

END-FIND
END

If the FIND statement contained a WHERE clause in addition to the WITH clause, only those records
that were selected as a result of the WITH clause andmet the WHERE criteria would be processed.

The following diagram illustrates the flow logic of a database processing loop:

Hierarchies of Processing Loops

The use ofmultiple FIND and/or READ statements creates a hierarchy of processing loops, as shown
in the following example:

Example of Processing Loop Hierarchy

** Example 'FINDX04': FIND (two FIND statements nested)
************************************************************************
DEFINE DATA LOCAL
1 PERSONVIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME

1 AUTOVIEW VIEW OF VEHICLES
2 PERSONNEL-ID
2 MAKE
2 MODEL

END-DEFINE
*
EMP. FIND PERSONVIEW WITH NAME = 'ADKINSON'

VEH. FIND AUTOVIEW WITH PERSONNEL-ID = PERSONNEL-ID (EMP.)
DISPLAY NAME MAKE MODEL

END-FIND
END-FIND
END

The above program selects from the EMPLOYEES file all peoplewith the name ADKINSON. Each record
(person) selected is then processed as follows:

1. The second FIND statement is executed to select the automobiles from the VEHICLES file, using
as selection criterion the PERSONNEL-IDs from the records selected from the EMPLOYEES file with
the first FIND statement.

2. The NAME of each person selected is displayed; this information is obtained from the EMPLOYEES
file. The MAKE and MODEL of each automobile owned by that person is also displayed; this inform-
ation is obtained from the VEHICLES file.

The second FIND statement creates an inner processing loop within the outer processing loop of
the first FIND statement, as shown in the following diagram.

The diagram illustrates the flow logic of the hierarchy of processing loops in the previous example
program:

Example of Nested FIND Loops Accessing the Same File

It is also possible to construct a processing loop hierarchy in which the same file is used at both
levels of the hierarchy:

** Example 'FINDX05': FIND (two FIND statements on same file nested)
************************************************************************
DEFINE DATA LOCAL
1 PERSONVIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY

1 #NAME (A40)
END-DEFINE
*
WRITE TITLE LEFT JUSTIFIED

'PEOPLE IN SAME CITY AS:' #NAME / 'CITY:' CITY SKIP 1
*
FIND PERSONVIEW WITH NAME = 'JONES'

WHERE FIRST-NAME = 'LAUREL'
COMPRESS NAME FIRST-NAME INTO #NAME
/*
FIND PERSONVIEW WITH CITY = CITY

DISPLAY NAME FIRST-NAME CITY
END-FIND

END-FIND
END

The above program first selects all people with name JONES and first name LAUREL from the
EMPLOYEES file. Then all who live in the same city are selected from the EMPLOYEES file and a list
of these people is created. All field values displayed by the DISPLAY statement are taken from the
second FIND statement.

Output of Program FINDX05:

PEOPLE IN SAME CITY AS: JONES LAUREL
CITY: BALTIMORE

NAME FIRST-NAME CITY
-------------------- -------------------- --------------------

JENSON MARTHA BALTIMORE
LAWLER EDDIE BALTIMORE
FORREST CLARA BALTIMORE
ALEXANDER GIL BALTIMORE
NEEDHAM SUNNY BALTIMORE
ZINN CARLOS BALTIMORE
JONES LAUREL BALTIMORE

Further Examples of Nested READ and FIND Statements

See the following example programs:

■ READX04 - READ statement (in combination with FIND and the system variables *NUMBER
and *COUNTER)

■ LIMITX01 - LIMIT statement (for READ, FIND loop processing)

Database Update - Transaction Processing

This section describes howNatural performs database updating operations based on transactions.

The following topics are covered:

■ Logical Transaction
■ Record Hold Logic
■ Backing Out a Transaction
■ Restarting a Transaction
■ Example of Using Transaction Data to Restart a Transaction

Logical Transaction

Natural performs database updating operations based on transactions, which means that all
database update requests are processed in logical transaction units. A logical transaction is the
smallest unit of work (as defined by you) which must be performed in its entirety to ensure that
the information contained in the database is logically consistent.

A logical transaction may consist of one or more update statements (DELETE, STORE, UPDATE) in-
volving one ormore database files. A logical transactionmay also spanmultipleNatural programs.

A logical transaction beginswhen a record is put on “hold”; Natural does this automaticallywhen
the record is read for updating, for example, if a FIND loop contains an UPDATE or DELETE statement.

The end of a logical transaction is determined by an END TRANSACTION statement in the program.
This statement ensures that all updates within the transaction have been successfully applied, and
releases all records that were put on “hold” during the transaction.

Example:

DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
END-DEFINE
FIND MYVIEW WITH NAME = 'SMITH'

DELETE
END TRANSACTION

END-FIND
END

Each record selected would be put on “hold”, deleted, and then - when the END TRANSACTION
statement is executed - released from “hold”.

Note: TheNatural profile parameter ETEOP, as set by theNatural administrator, determines
whether or not Natural will generate an END TRANSACTION statement at the end of each
Natural program. Ask your Natural administrator for details.

Example of STORE Statement:

The following example program adds new records to the EMPLOYEES file.

** Example 'STOREX01': STORE (Add new records to EMPLOYEES file)
*
** CAUTION: Executing this example will modify the database records!
************************************************************************
DEFINE DATA LOCAL
1 EMPLOYEE-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID(A8)
2 NAME (A20)
2 FIRST-NAME (A20)
2 MIDDLE-I (A1)
2 SALARY (P9/2)
2 MAR-STAT (A1)
2 BIRTH (D)
2 CITY (A20)
2 COUNTRY (A3)

*
1 #PERSONNEL-ID (A8)
1 #NAME (A20)
1 #FIRST-NAME (A20)
1 #INITIAL (A1)
1 #MAR-STAT (A1)
1 #SALARY (N9)
1 #BIRTH (A8)
1 #CITY (A20)
1 #COUNTRY (A3)
1 #CONF (A1) INIT <'Y'>
END-DEFINE

*
REPEAT

INPUT 'ENTER A PERSONNEL ID AND NAME (OR ''END'' TO END)' //
'PERSONNEL-ID : ' #PERSONNEL-ID //
'NAME : ' #NAME /
'FIRST-NAME : ' #FIRST-NAME

/*********************************************************************
/* validate entered data
/*********************************************************************
IF #PERSONNEL-ID = 'END' OR #NAME = 'END'

STOP
END-IF
IF #NAME = ' '

REINPUT WITH TEXT 'ENTER A LAST-NAME'
MARK 2 AND SOUND ALARM

END-IF
IF #FIRST-NAME = ' '

REINPUT WITH TEXT 'ENTER A FIRST-NAME'
MARK 3 AND SOUND ALARM

END-IF
/*********************************************************************
/* ensure person is not already on file
/*********************************************************************
FIP2. FIND NUMBER EMPLOYEE-VIEW WITH PERSONNEL-ID = #PERSONNEL-ID
/*
IF *NUMBER (FIP2.) > 0

REINPUT 'PERSON WITH SAME PERSONNEL-ID ALREADY EXISTS'
MARK 1 AND SOUND ALARM

END-IF
/*********************************************************************
/* get further information
/*********************************************************************
INPUT

'ENTER EMPLOYEE DATA' ////
'PERSONNEL-ID :' #PERSONNEL-ID (AD=IO) /
'NAME :' #NAME (AD=IO) /
'FIRST-NAME :' #FIRST-NAME (AD=IO) ///
'INITIAL :' #INITIAL /
'ANNUAL SALARY :' #SALARY /
'MARITAL STATUS :' #MAR-STAT /
'DATE OF BIRTH (YYYYMMDD) :' #BIRTH /
'CITY :' #CITY /
'COUNTRY (3 CHARS) :' #COUNTRY //
'ADD THIS RECORD (Y/N) :' #CONF (AD=M)

/*********************************************************************
/* ENSURE REQUIRED FIELDS CONTAIN VALID DATA
/*********************************************************************
IF #SALARY < 10000

REINPUT TEXT 'ENTER A PROPER ANNUAL SALARY' MARK 2
END-IF
IF NOT (#MAR-STAT = 'S' OR = 'M' OR = 'D' OR = 'W')

REINPUT TEXT 'ENTER VALID MARITAL STATUS S=SINGLE ' -

'M=MARRIED D=DIVORCED W=WIDOWED' MARK 3
END-IF
IF NOT(#BIRTH = MASK(YYYYMMDD) AND #BIRTH = MASK(1582-2699))

REINPUT TEXT 'ENTER CORRECT DATE' MARK 4
END-IF
IF #CITY = ' '

REINPUT TEXT 'ENTER A CITY NAME' MARK 5
END-IF
IF #COUNTRY = ' '

REINPUT TEXT 'ENTER A COUNTRY CODE' MARK 6
END-IF
IF NOT (#CONF = 'N' OR= 'Y')

REINPUT TEXT 'ENTER Y (YES) OR N (NO)' MARK 7
END-IF
IF #CONF = 'N'

ESCAPE TOP
END-IF
/*********************************************************************
/* add the record with STORE
/*********************************************************************
MOVE #PERSONNEL-ID TO EMPLOYEE-VIEW.PERSONNEL-ID
MOVE #NAME TO EMPLOYEE-VIEW.NAME
MOVE #FIRST-NAME TO EMPLOYEE-VIEW.FIRST-NAME
MOVE #INITIAL TO EMPLOYEE-VIEW.MIDDLE-I
MOVE #SALARY TO EMPLOYEE-VIEW.SALARY (1)
MOVE #MAR-STAT TO EMPLOYEE-VIEW.MAR-STAT
MOVE EDITED #BIRTH TO EMPLOYEE-VIEW.BIRTH (EM=YYYYMMDD)
MOVE #CITY TO EMPLOYEE-VIEW.CITY
MOVE #COUNTRY TO EMPLOYEE-VIEW.COUNTRY
/*
STP3. STORE RECORD IN FILE EMPLOYEE-VIEW
/*
/*********************************************************************
/* mark end of logical transaction
/*********************************************************************
END OF TRANSACTION
RESET INITIAL #CONF

END-REPEAT
END

Output of Program STOREX01:

ENTER A PERSONNEL ID AND NAME (OR 'END' TO END)

PERSONNEL ID :

NAME :
FIRST NAME :

Record Hold Logic

If Natural is used with Adabas, any record which is to be updated will be placed in “hold” status
until an END TRANSACTION or BACKOUT TRANSACTION statement is issued or the transaction time
limit is exceeded.

When a record is placed in “hold” status for one user, the record is not available for update by
another user. Another user who wishes to update the same record will be placed in “wait” status
until the record is released from “hold” when the first user ends or backs out his/her transaction.

To prevent users from being placed in wait status, the session parameter WH (Wait for Record in
Hold Status) can be used (see the Parameter Reference).

When you use update logic in a program, you should consider the following:

■ The maximum time that a record can be in hold status is determined by the Adabas transaction
time limit (Adabas parameter TT). If this time limit is exceeded, youwill receive an errormessage
and all database modifications done since the last END TRANSACTIONwill be made undone.

■ The number of records on hold and the transaction time limit are affected by the size of a
transaction, that is, by the placement of the END TRANSACTION statement in the program. Restart
facilities should be consideredwhen decidingwhere to issue an END TRANSACTION. For example,
if a majority of records being processed are not to be updated, the GET statement is an efficient
way of controlling the “holding” of records. This avoids issuing multiple END TRANSACTION
statements and reduces the number of ISNs on hold. When you process large files, you should
bear in mind that the GET statement requires an additional Adabas call. An example of a GET
statement is shown below.

Example of Hold Logic:

** Example 'GETX01': GET (put single record in hold with UPDATE stmt)
**
** CAUTION: Executing this example will modify the database records!
***********************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 SALARY (1)

END-DEFINE
*
RD. READ EMPLOY-VIEW BY NAME

DISPLAY EMPLOY-VIEW
IF SALARY (1) > 1500000

/*
GE. GET EMPLOY-VIEW *ISN (RD.)
/*
WRITE '=' (50) 'RECORD IN HOLD:' *ISN(RD.)
COMPUTE SALARY (1) = SALARY (1) * 1.15
UPDATE (GE.)

END TRANSACTION
END-IF

END-READ
END

Backing Out a Transaction

During an active logical transaction, that is, before the END TRANSACTION statement is issued, you
can cancel the transaction by using a BACKOUT TRANSACTION statement. The execution of this
statement removes all updates that have been applied (including all records that have been added
or deleted) and releases all records held by the transaction.

Restarting a Transaction

With the END TRANSACTION statement, you can also store transaction-related information. If pro-
cessing of the transaction terminates abnormally, you can read this information with a GET
TRANSACTION DATA statement to ascertain where to resume processing when you restart the
transaction.

Example of Using Transaction Data to Restart a Transaction

The following program updates the EMPLOYEES and VEHICLES files. After a restart operation, the
user is informedof the last EMPLOYEES record successfully processed. The user can resumeprocessing
from that EMPLOYEES record. It would also be possible to set up the restart transaction message to
include the last VEHICLES record successfully updated before the restart operation.

** Example 'GETTRX01': GET TRANSACTION
*
** CAUTION: Executing this example will modify the database records!
************************************************************************
DEFINE DATA LOCAL
01 PERSON VIEW OF EMPLOYEES

02 PERSONNEL-ID (A8)
02 NAME (A20)
02 FIRST-NAME (A20)
02 MIDDLE-I (A1)
02 CITY (A20)

01 AUTO VIEW OF VEHICLES
02 PERSONNEL-ID (A8)
02 MAKE (A20)
02 MODEL (A20)

*
01 ET-DATA

02 #APPL-ID (A8) INIT <' '>
02 #USER-ID (A8)
02 #PROGRAM (A8)
02 #DATE (A10)
02 #TIME (A8)

02 #PERSONNEL-NUMBER (A8)
END-DEFINE
*
GET TRANSACTION DATA #APPL-ID #USER-ID #PROGRAM

#DATE #TIME #PERSONNEL-NUMBER
*
IF #APPL-ID NOT = 'NORMAL' /* if last execution ended abnormally
AND #APPL-ID NOT = ' '

INPUT (AD=OIL)
// 20T '*** LAST SUCCESSFUL TRANSACTION ***' (I)
/ 20T '***********************************'
/// 25T 'APPLICATION:' #APPL-ID
/ 32T 'USER:' #USER-ID
/ 29T 'PROGRAM:' #PROGRAM
/ 24T 'COMPLETED ON:' #DATE 'AT' #TIME
/ 20T 'PERSONNEL NUMBER:' #PERSONNEL-NUMBER

END-IF
*
REPEAT

/*
INPUT (AD=MIL) // 20T 'ENTER PERSONNEL NUMBER:' #PERSONNEL-NUMBER
/*
IF #PERSONNEL-NUMBER = '99999999'

ESCAPE BOTTOM
END-IF
/*
FIND1. FIND PERSON WITH PERSONNEL-ID = #PERSONNEL-NUMBER

IF NO RECORDS FOUND
REINPUT 'SPECIFIED NUMBER DOES NOT EXIST; ENTER ANOTHER ONE.'

END-NOREC
FIND2. FIND AUTO WITH PERSONNEL-ID = #PERSONNEL-NUMBER
IF NO RECORDS FOUND

WRITE 'PERSON DOES NOT OWN ANY CARS'
ESCAPE BOTTOM

END-NOREC
IF *COUNTER (FIND2.) = 1 /* first pass through the loop

INPUT (AD=M)
/ 20T 'EMPLOYEES/AUTOMOBILE DETAILS' (I)
/ 20T '----------------------------'
/// 20T 'NUMBER:' PERSONNEL-ID (AD=O)
/ 22T 'NAME:' NAME ' ' FIRST-NAME ' ' MIDDLE-I
/ 22T 'CITY:' CITY
/ 22T 'MAKE:' MAKE
/ 21T 'MODEL:' MODEL

UPDATE (FIND1.) /* update the EMPLOYEES file
ELSE /* subsequent passes through the loop

INPUT NO ERASE (AD=M IP=OFF) //////// 28T MAKE / 28T MODEL
END-IF
/*
UPDATE (FIND2.) /* update the VEHICLES file
/*
MOVE *APPLIC-ID TO #APPL-ID

MOVE *INIT-USER TO #USER-ID
MOVE *PROGRAM TO #PROGRAM
MOVE *DAT4E TO #DATE
MOVE *TIME TO #TIME
/*
END TRANSACTION #APPL-ID #USER-ID #PROGRAM

#DATE #TIME #PERSONNEL-NUMBER
/*

END-FIND /* for VEHICLES (FIND2.)
END-FIND /* for EMPLOYEES (FIND1.)

END-REPEAT /* for REPEAT
*
STOP /* Simulate abnormal transaction end
END TRANSACTION 'NORMAL '
END

Selecting Records Using ACCEPT/REJECT

This section discusses the statements ACCEPT and REJECTwhich are used to select records based
on user-specified logical criteria.

The following topics are covered:

■ Statements Usable with ACCEPT and REJECT
■ Example of ACCEPT Statement
■ Logical Condition Criteria in ACCEPT/REJECT Statements
■ Example of ACCEPT Statement with AND Operator
■ Example of REJECT Statement with OR Operator
■ Further Examples of ACCEPT and REJECT Statements

Statements Usable with ACCEPT and REJECT

The statements ACCEPT and REJECT can be used in conjunctionwith the database access statements:

■ READ

■ FIND

■ HISTOGRAM

Example of ACCEPT Statement

** Example 'ACCEPX01': ACCEPT IF
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 JOB-TITLE
2 CURR-CODE (1:1)
2 SALARY (1:1)

END-DEFINE
*
READ (20) MYVIEW BY NAME WHERE CURR-CODE (1) = 'USD'
ACCEPT IF SALARY (1) >= 40000

DISPLAY NAME JOB-TITLE SALARY (1)
END-READ
END

Output of Program ACCEPX01:

   Page      1                                                  04-11-11  11:11:11
                                                                                 
          NAME                  CURRENT            ANNUAL                  
                               POSITION            SALARY                  
  -------------------- ------------------------- ----------                
                                                                                 
  ADKINSON             DBA                            46700                
  ADKINSON             MANAGER                        47000                
  ADKINSON             MANAGER                        47000                
  AFANASSIEV           DBA                            42800                
  ALEXANDER            DIRECTOR                       48000                
  ANDERSON             MANAGER                        50000                
  ATHERTON             ANALYST                        43000                
  ATHERTON             MANAGER                        40000               ↩

Logical Condition Criteria in ACCEPT/REJECT Statements

The statements ACCEPT and REJECT allow you to specify logical conditions in addition to those that
were specified in WITH and WHERE clauses of the READ statement.

The logical condition criteria in the IF clause of an ACCEPT / REJECT statement are evaluated after
the record has been selected and read.

Logical condition operators include the following (seeLogical Condition Criteria formore detailed
information):

:=EQEQUAL

¬=NENOT EQUAL TO

<LTLESS THAN

<=LELESS EQUAL

>GTGREATER THAN

>=GEGREATER EQUAL

Logical condition criteria in ACCEPT / REJECT statements may also be connected with the Boolean
operators AND, OR, and NOT. Moreover, parentheses may be used to indicate logical grouping; see
the following examples.

Example of ACCEPT Statement with AND Operator

The following program illustrates the use of the Boolean operator AND in an ACCEPT statement.

** Example 'ACCEPX02': ACCEPT IF ... AND ...
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 JOB-TITLE
2 CURR-CODE (1:1)
2 SALARY (1:1)

END-DEFINE
*
READ (20) MYVIEW BY NAME WHERE CURR-CODE (1) = 'USD'
ACCEPT IF SALARY (1) >= 40000

AND SALARY (1) <= 45000
DISPLAY NAME JOB-TITLE SALARY (1)

END-READ
END

Output of Program ACCEPX02:

Page      1                                                  04-12-14  12:22:01
                                                                               
        NAME                  CURRENT            ANNUAL                        
                             POSITION            SALARY                        
-------------------- ------------------------- ----------                      
                                                                               
AFANASSIEV           DBA                            42800                      
ATHERTON             ANALYST                        43000                      
ATHERTON             MANAGER                        40000                     ↩

Example of REJECT Statement with OR Operator

The following program, which uses the Boolean operator OR in a REJECT statement, produces the
same output as the ACCEPT statement in the example above, as the logical operators are reversed.

** Example 'ACCEPX03': REJECT IF ... OR ...
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 JOB-TITLE
2 CURR-CODE (1:1)
2 SALARY (1:1)

END-DEFINE
*
READ (20) MYVIEW BY NAME WHERE CURR-CODE (1) = 'USD'
REJECT IF SALARY (1) < 40000

OR SALARY (1) > 45000
DISPLAY NAME JOB-TITLE SALARY (1)

END-READ
END

Output of Program ACCEPX03:

Page      1                                                  04-12-14  12:26:27
                                                                               
        NAME                  CURRENT            ANNUAL                        
                             POSITION            SALARY                        
-------------------- ------------------------- ----------                      
                                                                               
AFANASSIEV           DBA                            42800                      
ATHERTON             ANALYST                        43000                      
ATHERTON             MANAGER                        40000                     ↩

Further Examples of ACCEPT and REJECT Statements

See the following example programs:

■ ACCEPX04 - ACCEPT IF ... LESS THAN ...
■ ACCEPX05 - ACCEPT IF ... AND ...
■ ACCEPX06 - REJECT IF ... OR ...

AT START/END OF DATA Statements

This section discusses the use of the statements AT START OF DATA and AT END OF DATA.

The following topics are covered:

■ AT START OF DATA Statement
■ AT END OF DATA Statement
■ Example of AT START OF DATA and AT END OF DATA Statements
■ Further Examples of AT START OF DATA and AT END OF DATA

AT START OF DATA Statement

The AT START OF DATA statement is used to specify any processing that is to be performed after
the first of a set of records has been read in a database processing loop.

The AT START OF DATA statement must be placed within the processing loop.

If the AT START OF DATA processing produces any output, this will be output before the first field
value. By default, this output is displayed left-justified on the page.

AT END OF DATA Statement

The AT END OF DATA statement is used to specify processing that is to be performed after all records
for a database processing loop have been processed.

The AT END OF DATA statement must be placed within the processing loop.

If the AT END OF DATA processing produces any output, this will be output after the last field value.
By default, this output is displayed left-justified on the page.

Example of AT START OF DATA and AT END OF DATA Statements

The following example program illustrates the use of the statements AT START OF DATA and AT
END OF DATA.

The Natural system variable *TIME has been incorporated into the AT START OF DATA statement
to display the time of day.

The Natural system function OLD has been incorporated into the AT END OF DATA statement to
display the name of the last person selected.

** Example 'ATSTAX01': AT START OF DATA
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 JOB-TITLE
2 INCOME (1:1)

3 CURR-CODE
3 SALARY
3 BONUS (1:1)

END-DEFINE
*
WRITE TITLE 'XYZ EMPLOYEE ANNUAL SALARY AND BONUS REPORT' /
READ (3) MYVIEW BY CITY STARTING FROM 'E'

DISPLAY GIVE SYSTEM FUNCTIONS
NAME (AL=15) JOB-TITLE (AL=15) INCOME (1)

/*
AT START OF DATA

WRITE 'RUN TIME:' *TIME /
END-START
AT END OF DATA

WRITE / 'LAST PERSON SELECTED:' OLD (NAME) /
END-ENDDATA

END-READ
*
AT END OF PAGE

WRITE / 'AVERAGE SALARY:' AVER (SALARY(1))
END-ENDPAGE
END

The program produces the following output:

                  XYZ EMPLOYEE ANNUAL SALARY AND BONUS REPORT 
                                                              
     NAME           CURRENT                 INCOME            
                   POSITION                                   
                                CURRENCY   ANNUAL     BONUS   
                                  CODE     SALARY             
--------------- --------------- -------- ---------- ----------
                                                              
RUN TIME: 12:43:19.1                                          
                                                              
DUYVERMAN       PROGRAMMER      USD           34000          0
PRATT           SALES PERSON    USD           38000       9000
MARKUSH         TRAINEE         USD           22000          0
                                                              
LAST PERSON SELECTED: MARKUSH                                 
                                                              
                                                              
AVERAGE SALARY:      31333                                   ↩

Further Examples of AT START OF DATA and AT END OF DATA

See the following example programs:

■ ATENDX01 - AT END OF DATA
■ ATSTAX02 - AT START OF DATA
■ WRITEX09 - WRITE (in combination with AT END OF DATA )

Unicode Data

Natural enables users to access wide-character fields (format W) in an Adabas database.

The following topics are covered:

■ Data Definition Module
■ Access Configuration
■ Restrictions

Data Definition Module

Adabas wide-character fields (W) are mapped to Natural format U (Unicode).

The length definition for a Natural field of format U corresponds to half the size of the Adabas
field of formatW. AnAdabaswide-character field of length 200 is, for example, mapped to (U100)
in Natural.

Access Configuration

Natural receives data fromAdabas and sends data to Adabas using UTF-16 as common encoding.

This encoding is specified with the OPRB parameter and sent to Adabas with the open request. It
is used for wide-character fields and applies to the entire Adabas user session.

Restrictions

Wide-character fields (W) of variable length are not supported.

Collating descriptors are not supported.

For further information on Adabas and Unicode support refer to the specific Adabas product
documentation.

VI Controlling Data Output

This part describes how to proceed if a Natural program is to produce multiple reports. Further-
more, it discusses various aspects of how you can control the format of an output report created
with Natural, that is, the way in which the data are displayed.

Report Specification - (rep) Notation

Layout of an Output Page

Statements DISPLAY and WRITE

Index Notation for Multiple-Value Fields and Periodic Groups

Page Titles, Page Breaks, Blank Lines

Column Headers

Parameters to Influence the Output of Fields

Edit Masks - EM Parameter

Unicode Edit Masks - EMU Parameter

Vertical Displays

(rep) is the output report identifier for which a statement is applicable.

Use of Report Specifications

If a Natural program is to produce multiple reports, the notation (rep)must be specified with
each output statement (see Statements Concerned, below) which is to be used to create output for
any report other than the first report (Report 0).

A value of 0 - 31may be specified.

The value for (rep)may also be a logical namewhich has been assigned using the DEFINE PRINTER
statement, see Example 2 below.

Statements Concerned

The notation (rep) can be used with the following output statements:

AT END OF PAGE | AT TOP OF PAGE | DISPLAY | EJECT | FORMAT | NEWPAGE | PRINT | SKIP | SUSPEND
IDENTICAL SUPPRESS | WRITE | WRITE TITLE | WRITE TRAILER

Examples of Report Specification

Example 1 - Multiple Reports

DISPLAY (1) NAME ...
WRITE (4) NAME ...

Example 2 - Using Logical Names

DEFINE PRINTER (LIST=5) OUTPUT 'LPT1'
WRITE (LIST) NAME ...

The following topics are covered:

Statements Influencing a Report Layout

The following statements have an impact on the layout of the report:

FunctionStatement

With this statement, you can specify a page title, that is, text to be output at the top
of a page. By default, page titles are centered and not underlined.

WRITE TITLE

With this statement, you can specify a page trailer, that is, text to be output at the
bottom of a page. By default, the trailer lines are centered and not underlined.

WRITE TRAILER

With this statement, you can specify any processing that is to be performedwhenever
a new page of the report is started. Any output from this processing will be output
below the page title.

AT TOP OF PAGE

With this statement, you can specify any processing that is to be performedwhenever
an end-of-page condition occurs. Any output from this processing will be output
below any page trailer (as specified with the WRITE TRAILER statement).

AT END OF PAGE

With this statement, you specify processing that is to be performed after the first
record has been read in a database processing loop. Any output from this processing
will be output before the first field value. See note below.

AT START OF DATA

With this statement, you specify processing that is to be performed after all records
for a processing loop have been processed. Any output from this processing will be
output immediately after the last field value. See note below.

AT END OF DATA

With these statements, you control the format in which the field values that have
been read are to be output. See section Statements DISPLAY and WRITE.

DISPLAY / WRITE

Note: The relevance of the statements AT START OF DATA and AT END OF DATA for the output
of data is described under Database Access, AT START/END OF DATA Statements. The
other statements listed above are discussed in other sections of the part Controlling Data
Output.

General Layout Example

The following example program illustrates the general layout of an output page:

** Example 'OUTPUX01': Several sections of output
************************************************************************
DEFINE DATA LOCAL
1 EMP-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 BIRTH

END-DEFINE
*
WRITE TITLE '********** Page Title **********'
WRITE TRAILER '********** Page Trailer **********'
*
AT TOP OF PAGE

WRITE '===== Top of Page ====='
END-TOPPAGE
AT END OF PAGE

WRITE '===== End of Page ====='
END-ENDPAGE
*
READ (10) EMP-VIEW BY NAME

/*
DISPLAY NAME FIRST-NAME BIRTH (EM=YYYY-MM-DD)
/*
AT START OF DATA

WRITE '>>>>> Start of Data >>>>>'
END-START
AT END OF DATA

WRITE '<<<<< End of Data <<<<<'
END-ENDDATA

END-READ
END

Output of Program OUTPUX01:

********** Page Title **********
===== Top of Page =====

NAME FIRST-NAME DATE
OF

BIRTH
-------------------- -------------------- ----------

>>>>> Start of Data >>>>>
ABELLAN KEPA 1961-04-08
ACHIESON ROBERT 1963-12-24
ADAM SIMONE 1952-01-30
ADKINSON JEFF 1951-06-15
ADKINSON PHYLLIS 1956-09-17
ADKINSON HAZEL 1954-03-19
ADKINSON DAVID 1946-10-12
ADKINSON CHARLIE 1950-03-02
ADKINSON MARTHA 1970-01-01
ADKINSON TIMMIE 1970-03-03

<<<<< End of Data <<<<<
********** Page Trailer **********

===== End of Page =====

The following topics are covered:

DISPLAY Statement

The DISPLAY statement produces output in column format; that is, the values for one field are
output in a column underneath one another. If multiple fields are output, that is, if multiple
columns are produced, these columns are output next to one another horizontally.

The order in which fields are displayed is determined by the sequence in which you specify the
field names in the DISPLAY statement.

The DISPLAY statement in the following program displays for each person first the personnel
number, then the name and then the job title:

** Example 'DISPLX01': DISPLAY
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH
2 JOB-TITLE

END-DEFINE
*
READ (3) VIEWEMP BY BIRTH

DISPLAY PERSONNEL-ID NAME JOB-TITLE
END-READ
END

Output of Program DISPLX01:

Page 1 04-11-11 14:15:54

PERSONNEL NAME CURRENT
ID POSITION

--------- -------------------- -------------------------

30020013 GARRET TYPIST
30016112 TAILOR WAREHOUSEMAN
20017600 PIETSCH SECRETARY

To change the order of the columns that appear in the output report, simply reorder the field
names in the DISPLAY statement. For example, if you prefer to list employee names first, then job
titles followed by personnel numbers, the appropriate DISPLAY statement would be:

** Example 'DISPLX02': DISPLAY
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH
2 JOB-TITLE

END-DEFINE
*
READ (3) VIEWEMP BY BIRTH

DISPLAY NAME JOB-TITLE PERSONNEL-ID
END-READ
END

Output of Program DISPLX02:

Page 1 04-11-11 14:15:54

NAME CURRENT PERSONNEL
POSITION ID

-------------------- ------------------------- ---------

GARRET TYPIST 30020013
TAILOR WAREHOUSEMAN 30016112
PIETSCH SECRETARY 20017600

A header is output above each column. Various ways to influence this header are described in the
document Column Headers.

WRITE Statement

The WRITE statement is used to produce output in free format (that is, not in columns). In contrast
to the DISPLAY statement, the following applies to the WRITE statement:

■ If necessary, it automatically creates a line advance; that is, a field or text element that does not
fit onto the current output line, is automatically output in the next line.

■ It does not produce any headers.
■ The values of a multiple-value field are output next to one another horizontally, and not under-
neath one another.

The two example programs shown below illustrate the basic differences between the DISPLAY
statement and the WRITE statement.

You can also use the two statements in combination with one another, as described later in the
document Vertical Displays, Combining DISPLAY and WRITE .

Example of DISPLAY Statement

** Example 'DISPLX03': DISPLAY
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 SALARY (1:3)

END-DEFINE
*
READ (2) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY NAME FIRST-NAME SALARY (1:3)
END-READ
END

Output of Program DISPLX03:

Page 1 04-11-11 14:15:54

NAME FIRST-NAME ANNUAL
SALARY

-------------------- -------------------- ----------

JONES VIRGINIA 46000
42300
39300

JONES MARSHA 50000
46000
42700

Example of WRITE Statement

** Example 'WRITEX01': WRITE
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 SALARY (1:3)

END-DEFINE
*
READ (2) VIEWEMP BY NAME STARTING FROM 'JONES'

WRITE NAME FIRST-NAME SALARY (1:3)
END-READ
END

Output of Program WRITEX01:

Page 1 04-11-11 14:15:55

JONES VIRGINIA 46000 42300 39300
JONES MARSHA 50000 46000 42700

Column Spacing - SF Parameter and nX Notation

By default, the columns output with a DISPLAY statement are separated from one another by one
space.

With the session parameter SF, you can specify the default number of spaces to be inserted between
columns output with a DISPLAY statement. You can set the number of spaces to any value from 1
to 30.

The parameter can be specified with a FORMAT statement to apply to the whole report, or with a
DISPLAY statement at statement level, but not at element level.

With the nX notation in the DISPLAY statement, you can specify the number of spaces (n) to be in-
serted between two columns. An nX notation overrides the specification made with the SF para-
meter.

** Example 'DISPLX04': DISPLAY (with nX)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH
2 JOB-TITLE

END-DEFINE
*
FORMAT SF=3
READ (3) VIEWEMP BY BIRTH

DISPLAY PERSONNEL-ID NAME 5X JOB-TITLE
END-READ
END

Output of Program DISPLX04:

The above example program produces the following output, where the first two columns are
separated by 3 spaces due to the SF parameter in the FORMAT statement, while the second and third
columns are separated by 5 spaces due to the notation 5X in the DISPLAY statement:

Page 1 04-11-11 14:15:54

PERSONNEL NAME CURRENT
ID POSITION

--------- -------------------- -------------------------

30020013 GARRET TYPIST
30016112 TAILOR WAREHOUSEMAN
20017600 PIETSCH SECRETARY

The nX notation is also available with the WRITE statement to insert spaces between individual
output elements:

WRITE PERSONNEL-ID 5X NAME 3X JOB-TITLE

With the above statement, 5 spaces will be inserted between the fields PERSONNEL-ID and NAME,
and 3 spaces between NAME and JOB-TITLE.

Tab Setting - nT Notation

With the nT notation, which is availablewith the DISPLAY and the WRITE statement, you can specify
the position where an output element is to be output.

** Example 'DISPLX05': DISPLAY (with nT)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME

END-DEFINE
*
READ (3) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY 5T NAME 30T FIRST-NAME
END-READ
END

Output of Program DISPLX05:

The above program produces the following output, where the field NAME is output starting in the
5th position (counted from the left margin of the page), and the field FIRST-NAME starting in the
30th position:

Page 1 04-11-11 14:15:54

NAME FIRST-NAME
-------------------- --------------------

JONES VIRGINIA
JONES MARSHA
JONES ROBERT

Line Advance - Slash Notation

With a slash (/) in a DISPLAY or WRITE statement, you cause a line advance.

■ In a DISPLAY statement, a slash causes a line advance between fields and within text.
■ In a WRITE statement, a slash causes a line advance only when placed between fields; within text,
it is treated like an ordinary text character.

When placed between fields, the slash must have a blank on either side.

For multiple line advances, you specify multiple slashes.

Example 1 - Line Advance in DISPLAY Statement:

** Example 'DISPLX06': DISPLAY (with slash '/')
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 DEPARTMENT

END-DEFINE
*
READ (3) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY NAME / FIRST-NAME 'DEPART-/MENT' DEPARTMENT
END-READ
END

Output of Program DISPLX06:

The above DISPLAY statement produces a line advance after each value of the field NAME and
within the text DEPART-MENT:

Page 1 04-11-11 14:15:54

NAME DEPART-
FIRST-NAME MENT

-------------------- -------

JONES SALE
VIRGINIA
JONES MGMT
MARSHA
JONES TECH
ROBERT

Example 2 - Line Advance in WRITE Statement:

** Example 'WRITEX02': WRITE (with line advance)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 DEPARTMENT

END-DEFINE
*
READ (3) VIEWEMP BY NAME STARTING FROM 'JONES'

WRITE NAME / FIRST-NAME 'DEPART-/MENT' DEPARTMENT //
END-READ
END

Output of Program WRITEX02:

The above WRITE statement produces a line advance after each value of the field NAME, and a double
line advance after each value of the field DEPARTMENT, but none within the text DEPART-/MENT:

Page 1 04-11-11 14:15:55

JONES
VIRGINIA DEPART-/MENT SALE

JONES
MARSHA DEPART-/MENT MGMT

JONES
ROBERT DEPART-/MENT TECH

Example 3 - Line Advance in DISPLAY and WRITE Statements:

** Example 'DISPLX21': DISPLAY (usage of slash '/' in DISPLAY and WRITE)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 FIRST-NAME
2 ADDRESS-LINE (1)

END-DEFINE
*
WRITE TITLE LEFT JUSTIFIED UNDERLINED

*TIME
5X 'PEOPLE LIVING IN SALT LAKE CITY'
21X 'PAGE:' *PAGE-NUMBER /
15X 'AS OF' *DAT4E //

*
WRITE TRAILER UNDERLINED 'REGISTER OF' / 'SALT LAKE CITY'
*
READ (2) EMPLOY-VIEW WITH CITY = 'SALT LAKE CITY'

DISPLAY NAME /
FIRST-NAME
'HOME/CITY' CITY
'STREET/OR BOX NO.' ADDRESS-LINE (1)

SKIP 1
END-READ
END

Output of Program DISPLX21:

14:15:54.6 PEOPLE LIVING IN SALT LAKE CITY PAGE: 1
AS OF 11/11/2004

-------------------------------------------------------------------------------
NAME HOME STREET

FIRST-NAME CITY OR BOX NO.
-------------------- -------------------- --------------------

ANDERSON SALT LAKE CITY 3701 S. GEORGE MASON
JENNY

SAMUELSON SALT LAKE CITY 7610 W. 86TH STREET
MARTIN

REGISTER OF
SALT LAKE CITY

-------------------------------------------------------------------------------

Further Examples of DISPLAY and WRITE Statements

See the following example programs:

■ DISPLX13 - DISPLAY (compare with WRITEX08 using WRITE)
■ WRITEX08 - WRITE (compare with DISPLX13 using DISPLAY)
■ DISPLX14 - DISPLAY (with AL, SF and nX)
■ WRITEX09 - WRITE (in combination with AT END OF DATA)

This chapter describes how you can use the index notation (n:n) to specify how many values of
a multiple-value field or how many occurrences of a periodic group are to be output.

Use of Index Notation

With the index notation (n:n) you can specify howmany values of a multiple-value field or how
many occurrences of a periodic group are to be output.

For example, the field INCOME in the DDM EMPLOYEES is a periodic group which keeps a record of
the annual incomes of an employee for each year he/she has been with the company.

These annual incomes are maintained in chronological order. The income of the most recent year
is in occurrence 1.

If you wanted to have the annual incomes of an employee for the last three years displayed - that
is, occurrences 1 to 3 - you would specify the notation (1:3) after the field name in a DISPLAY or
WRITE statement (as shown in the following example program).

Example of Index Notation in DISPLAY Statement

** Example 'DISPLX07': DISPLAY (with index notation)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH
2 INCOME (1:3)

3 CURR-CODE
3 SALARY
3 BONUS (1:1)

END-DEFINE
*
READ (3) VIEWEMP BY BIRTH

DISPLAY PERSONNEL-ID NAME INCOME (1:3)
SKIP 1

END-READ
END

Output of Program DISPLX07:

Note that a DISPLAY statement outputs multiple values of a multiple-value field underneath one
another:

Page 1 04-11-11 14:15:54

PERSONNEL NAME INCOME
ID

CURRENCY ANNUAL BONUS
CODE SALARY

--------- -------------------- -------- ---------- ----------

30020013 GARRET UKL 4200 0
UKL 4150 0

0 0

30016112 TAILOR UKL 7450 0
UKL 7350 0
UKL 6700 0

20017600 PIETSCH USD 22000 0
USD 20200 0
USD 18700 0

As a WRITE statement displays multiple values horizontally instead of vertically, this may cause
a line overflow and a - possibly undesired - line advance.

If you use only a single field within a periodic group (for example, SALARY) instead of the entire
periodic group, and if you also insert a slash (/) to cause a line advance (as shown in the following
example between NAME and JOB-TITLE), the report format becomes manageable.

Example of Index Notation in WRITE Statement

** Example 'WRITEX03': WRITE (with index notation)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH
2 JOB-TITLE
2 SALARY (1:3)

END-DEFINE
*
READ (3) VIEWEMP BY BIRTH

WRITE PERSONNEL-ID NAME / JOB-TITLE SALARY (1:3)
SKIP 1

END-READ
END

Output of Program WRITEX03:

277Programming Guide

Index Notation for Multiple-Value Fields and Periodic Groups



Page 1 04-11-11 14:15:55

30020013 GARRET
TYPIST 4200 4150 0

30016112 TAILOR
WAREHOUSEMAN 7450 7350 6700

20017600 PIETSCH
SECRETARY 22000 20200 18700

This chapter describes various ways of controlling page breaks in a report, the output of page
titles at the top of each report page and the generation of empty lines in an output report.

Default Page Title

For each page output via a DISPLAY or WRITE statement, Natural automatically generates a single
default title line. This title line contains the page number, the date and the time of day.

Example:

WRITE 'HELLO'
END

The above program produces the following output with default page title:

Page      1                                                  04-12-14  13:19:33
                                                                               
HELLO                                                                         ↩

Suppress Page Title - NOTITLE Option

If you wish your report to be output without page titles, you add the keyword NOTITLE to the
statement DISPLAY or WRITE.

Example - DISPLAY with NOTITLE:

** Example 'DISPLX20': DISPLAY (with NOTITLE)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 FIRST-NAME

END-DEFINE
*
READ (5) EMPLOY-VIEW BY CITY FROM 'BOSTON'

DISPLAY NOTITLE NAME FIRST-NAME CITY
END-READ
END

Output of Program DISPLX20:

NAME FIRST-NAME CITY
-------------------- -------------------- --------------------

SHAW LESLIE BOSTON
STANWOOD VERNON BOSTON
CREMER WALT BOSTON
PERREAULT BRENDA BOSTON
COHEN JOHN BOSTON

Example - WRITE with NOTITLE:

WRITE NOTITLE 'HELLO'
END

The above program produces the following output without page title:

HELLO                                                                          ↩

Define Your Own Page Title - WRITE TITLE Statement

If you wish a page title of your own to be output instead of the Natural default page title, you use
the statement WRITE TITLE.

The following topics are covered below:

■ Specifying Text for Your Title
■ Specifying Empty Lines after the Title
■ Title Justification and/or Underlining
■ Title with Page Number

Specifying Text for Your Title

With the statement WRITE TITLE, you specify the text for your title (in apostrophes).

WRITE TITLE 'THIS IS MY PAGE TITLE'
WRITE 'HELLO'
END

The above program produces the following output:

                             THIS IS MY PAGE TITLE                             
HELLO                                                                         ↩

Specifying Empty Lines after the Title

With the SKIP option of the WRITE TITLE statement, you can specify the number of empty lines
to be output immediately below the title line. After the keyword SKIP, you specify the number of
empty lines to be inserted.

WRITE TITLE 'THIS IS MY PAGE TITLE' SKIP 2
WRITE 'HELLO'
END

The above program produces the following output:

                             THIS IS MY PAGE TITLE                             
                                                                               
                                                                               
HELLO                                                                         ↩

SKIP is not only available as part of the WRITE TITLE statement, but also as a stand-alone statement.

Title Justification and/or Underlining

By default, the page title is centered on the page and not underlined.

The WRITE TITLE statement provides the following options which can be used independent of
each other:

EffectOption

Causes the page trailer to be displayed left-justified.LEFT JUSTIFIED

Causes the title to be displayed underlined. The underlining runs the width of the line
size (see alsoNatural profile and session parameter LS). By default, titles are underlined

UNDERLINED

with a hyphen (-). However, with the UC session parameter you can specify another
character to be used as underlining character (seeUnderlining Character for Titles and
Headers).

The following example shows the effect of the LEFT JUSTIFIED and UNDERLINED options:

WRITE TITLE LEFT JUSTIFIED UNDERLINED 'THIS IS MY PAGE TITLE'
SKIP 2
WRITE 'HELLO'
END

The above program produces the following output:

THIS IS MY PAGE TITLE                                                          
-------------------------------------------------------------------------------
                                                                               
                                                                               
HELLO                                                                         ↩

The WRITE TITLE statement is executed whenever a new page is initiated for the report.

Title with Page Number

In the following examples, the system variable *PAGE-NUMBER is used in conjunctionwith the WRITE
TITLE statement to output the page number in the title line.

** Example 'WTITLX01': WRITE TITLE (with *PAGE-NUMBER)
************************************************************************
DEFINE DATA LOCAL
1 VEHIC-VIEW VIEW OF VEHICLES

2 MAKE
2 YEAR
2 MAINT-COST (1)

END-DEFINE
*
LIMIT 5
*
READ VEHIC-VIEW
END-ALL
SORT BY YEAR USING MAKE MAINT-COST (1)

DISPLAY NOTITLE YEAR MAKE MAINT-COST (1)
AT BREAK OF YEAR

MOVE 1 TO *PAGE-NUMBER
NEWPAGE

END-BREAK
/*
WRITE TITLE LEFT JUSTIFIED

'YEAR:' YEAR 15X 'PAGE' *PAGE-NUMBER
END-SORT
END

Output of Program WTITLX01:

YEAR: 1980 PAGE 1
YEAR MAKE MAINT-COST
----- -------------------- ----------

1980 RENAULT 20000
1980 RENAULT 20000
1980 PEUGEOT 20000

Logical Page and Physical Page

A logical page is the output produced by a Natural program. A physical page is your terminal screen
on which the output is displayed; or it may be the piece of paper on which the output is printed.

The size of the logical page is determined by the number of lines output by the Natural program.

If more lines are output than fit onto one screen, the logical page will exceed the physical screen,
and the remaining lines will be displayed on the next screen.

Note: If information you wish to appear at the bottom of the screen (for example, output
created by a WRITE TRAILER or AT END OF PAGE statement) is output on the next screen in-
stead, reduce the logical page size accordingly (with the session parameter PS, which is
discussed below).

Page Size - PS Parameter

With the parameter PS (Page Size for Natural Reports), you determine the maximum number of
lines per (logical) page for a report.

When the number of lines specified with the PS parameter is reached, a page advance occurs
(unless page advance is controlledwith a NEWPAGE or EJECT statement; seePageAdvance Controlled
by EJ Parameter below).

The PS parameter can be set either at session level, or within a program with the following state-
ments:

At report level:

■ FORMAT PS=nn

At statement level:

■ DISPLAY (PS=nn)

■ WRITE (PS=nn)

■ WRITE TITLE (PS=nn)

■ WRITE TRAILER (PS=nn)

■ INPUT (PS=nn)

Page Advance

A page advance can be triggered by one of the following methods:

■ Page Advance Controlled by EJ Parameter
■ Page Advance Controlled by EJECT or NEWPAGE Statements
■ Eject/New Page when less than n Lines Left

These methods are discussed below.

Page Advance Controlled by EJ Parameter

With the session parameter EJ (Page Eject), you determinewhether page ejects are to be performed
or not. By default, EJ=ON applies, which means that page ejects will be performed as specified.

If you specify EJ=OFF, page break information will be ignored. This may be useful to save paper
during test runs where page ejects are not needed.

The EJ parameter can be set at session level; for example:

GLOBALS EJ=OFF

The EJ parameter setting is overridden by the EJECT statement.

Page Advance Controlled by EJECT or NEWPAGE Statements

The following topics are covered below:

■ Page Advance without Title/Header on Next Page
■ Page Advance with End/Top-of-Page Processing

Page Advance without Title/Header on Next Page

The EJECT statement causes a page advance without a title or header line being generated on the
next page. A newphysical page is startedwithout any top-of-page or end-of-page processing being
performed (for example, no WRITE TRAILER or AT END OF PAGE, WRITE TITLE, AT TOP OF PAGE
or *PAGE-NUMBER processing).

The EJECT statement overrides the EJ parameter setting.

Page Advance with End/Top-of-Page Processing

The NEWPAGE statement causes a page advance with associated end-of-page and top-of-page pro-
cessing. A trailer line will be displayed, if specified. A title line, either default or user-specified,
will be displayed on the new page, unless the NOTITLE option has been specified in a DISPLAY or
WRITE statement (as described above).

If the NEWPAGE statement is not used, page advance is automatically controlled by the setting of
the PS parameter; see Page Size - PS Parameter above).

Eject/New Page when less than n Lines Left

Both the NEWPAGE statement and the EJECT statement provide a WHEN LESS THAN n LINES LEFT
option.With this option, you specify a number of lines (n). The NEWPAGE/EJECT statement will then
be executed if - at the time the statement is processed - less than n lines are available on the current
page.

Example 1:

FORMAT PS=55
...
NEWPAGE WHEN LESS THAN 7 LINES LEFT
...

In this example, the page size is set to 55 lines.

If only 6 or less lines are left on the current page at the time when the NEWPAGE statement is pro-
cessed, the NEWPAGE statement is executed and a page advance occurs. If 7 or more lines are left,
the NEWPAGE statement is not executed and no page advance occurs; the page advance then occurs
depending on the session parameter PS (Page Size for Natural Reports), that is, after 55 lines.

Example 2:

** Example 'NEWPAX02': NEWPAGE (in combination with EJECT and
** parameter PS)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 JOB-TITLE

END-DEFINE
*
FORMAT PS=15
*
READ (9) EMPLOY-VIEW BY CITY STARTING FROM 'BOSTON'

AT START OF DATA
EJECT
WRITE /// 20T '%' (29) /

20T '%%' 47T '%%' /
20T '%%' 3X 'REPORT OF EMPLOYEES' 47T '%%' /
20T '%%' 3X ' SORTED BY CITY ' 47T '%%' /
20T '%%' 47T '%%' /
20T '%' (29) /

NEWPAGE
END-START
AT BREAK OF CITY

NEWPAGE WHEN LESS 3 LINES LEFT
END-BREAK
DISPLAY CITY (IS=ON) NAME JOB-TITLE

END-READ
END

New Page with Title

The NEWPAGE statement also provides a WITH TITLE option. If this option is not used, a default title
will appear at the top of the new page or a WRITE TITLE statement or NOTITLE clause will be ex-
ecuted.

The WITH TITLE option of the NEWPAGE statement allows you to override these with a title of your
own choice. The syntax of the WITH TITLE option is the same as for the WRITE TITLE statement.

Example:

NEWPAGE WITH TITLE LEFT JUSTIFIED 'PEOPLE LIVING IN BOSTON:'

The followingprogram illustrates the use of the session parameter PS (Page Size forNatural Reports)
and the NEWPAGE statement. Moreover, the system variable *PAGE-NUMBER is used to display the
current page number.

** Example 'NEWPAX01': NEWPAGE
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 CITY
2 DEPT

END-DEFINE
*
FORMAT PS=20
READ (5) VIEWEMP BY CITY STARTING FROM 'M'

DISPLAY NAME 'DEPT' DEPT 'LOCATION' CITY
AT BREAK OF CITY

NEWPAGE WITH TITLE LEFT JUSTIFIED
'EMPLOYEES BY CITY - PAGE:' *PAGE-NUMBER

END-BREAK
END-READ
END

Output of Program NEWPAX01:

Note the position of the page breaks and the title line:

Page 1 04-11-11 14:15:54

NAME DEPT LOCATION
-------------------- ------ --------------------

FICKEN TECH10 MADISON
KELLOGG TECH10 MADISON
ALEXANDER SALE20 MADISON

Page 2:

EMPLOYEES BY CITY - PAGE: 2
NAME DEPT LOCATION

-------------------- ------ --------------------

DE JUAN SALE03 MADRID
DE LA MADRID PROD01 MADRID

Page 3:

EMPLOYEES BY CITY - PAGE: 3

Page Trailer - WRITE TRAILER Statement

The following topics are covered below:

■ Specifying a Page Trailer
■ Considering Logical Page Size
■ Page Trailer Justification and/or Underlining

Specifying a Page Trailer

The WRITE TRAILER statement is used to output text (in apostrophes) at the bottom of a page.

WRITE TRAILER 'THIS IS THE END OF THE PAGE'

The statement is executed when an end-of-page condition is detected, or as a result of a SKIP or
NEWPAGE statement.

Considering Logical Page Size

As the end-of-page condition is checked only after an entire DISPLAY or WRITE statement has been
processed, it may occur that the logical page size (that is, the number of lines output by a DISPLAY
or WRITE statement) causes the physical size of the output page to be exceeded before the WRITE
TRAILER statement is executed.

To ensure that a page trailer actually appears at the bottom of a physical page, you should set the
logical page size (with the PS session parameter) to a value less than the physical page size.

Page Trailer Justification and/or Underlining

By default, the page trailer is displayed centered on the page and not underlined.

The WRITE TRAILER statement provides the following options which can be used independent of
each other:

EffectOption

Causes the page trailer to be displayed left justified.LEFT JUSTIFIED

The underlining runs the width of the line size (see also Natural profile and session
parameter LS). By default, titles are underlined with a hyphen (-). However, with the

UNDERLINED

UC session parameter you can specify another character to be used as underlining
character (see Underlining Character for Titles and Headers).

The following examples show the use of the LEFT JUSTIFIED and UNDERLINED options of the WRITE
TRAILER statement:

Example 1:

WRITE TRAILER LEFT JUSTIFIED UNDERLINED 'THIS IS THE END OF THE PAGE'

Example 2:

** Example 'WTITLX02': WRITE TITLE AND WRITE TRAILER
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 FIRST-NAME
2 ADDRESS-LINE (1)

END-DEFINE
*
WRITE TITLE LEFT JUSTIFIED UNDERLINED

*TIME
5X 'PEOPLE LIVING IN SALT LAKE CITY'
21X 'PAGE:' *PAGE-NUMBER /

15X 'AS OF' *DAT4E //
*
WRITE TRAILER UNDERLINED 'REGISTER OF' / 'SALT LAKE CITY'
*
READ (2) EMPLOY-VIEW WITH CITY = 'SALT LAKE CITY'

DISPLAY NAME /
FIRST-NAME
'HOME/CITY' CITY
'STREET/OR BOX NO.' ADDRESS-LINE (1)

SKIP 1
END-READ
END

Generating Blank Lines - SKIP Statement

The SKIP statement is used to generate one or more blank lines in an output report.

Example 1 - SKIP in conjunction with WRITE and DISPLAY:

** Example 'SKIPX01': SKIP (in conjunction with WRITE and DISPLAY)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 FIRST-NAME
2 ADDRESS-LINE (1)

END-DEFINE
*
WRITE TITLE LEFT JUSTIFIED UNDERLINED

'PEOPLE LIVING IN SALT LAKE CITY AS OF' *DAT4E 7X
'PAGE:' *PAGE-NUMBER

SKIP 3
*
READ (2) EMPLOY-VIEW WITH CITY = 'SALT LAKE CITY'

DISPLAY NAME / FIRST-NAME CITY ADDRESS-LINE (1)
SKIP 1

END-READ
END

Example 2 - SKIP in conjunction with DISPLAY VERT:

** Example 'SKIPX02': SKIP (in conjunction with DISPLAY VERT)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY
2 JOB-TITLE

END-DEFINE
*
READ (2) EMPLOY-VIEW WITH JOB-TITLE = 'SECRETARY'

DISPLAY NOTITLE VERT
NAME FIRST-NAME / CITY

SKIP 3
END-READ
*
NEWPAGE
*
READ (2) EMPLOY-VIEW WITH JOB-TITLE = 'SECRETARY'

DISPLAY NOTITLE
NAME FIRST-NAME / CITY

SKIP 3
END-READ
END

AT TOP OF PAGE Statement

The AT TOP OF PAGE statement is used to specify any processing that is to be performedwhenever
a new page of the report is started.

If the AT TOP OF PAGE processing produces any output, this will be output below the page title
(with a skipped line in between).

By default, this output is displayed left-justified on the page.

Example:

** Example 'ATTOPX01': AT TOP OF PAGE
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 MAR-STAT
2 BIRTH
2 CITY

2 JOB-TITLE
2 DEPT

END-DEFINE
*
LIMIT 10
READ EMPLOY-VIEW BY PERSONNEL-ID FROM '20017000'

DISPLAY NOTITLE (AL=10)
NAME DEPT JOB-TITLE CITY 5X
MAR-STAT 'DATE OF/BIRTH' BIRTH (EM=YY-MM-DD)

/*
AT TOP OF PAGE

WRITE / '-BUSINESS INFORMATION-'
26X '-PRIVATE INFORMATION-'

END-TOPPAGE
END-READ
END

Output of Program ATTOPX01:

-BUSINESS INFORMATION- -PRIVATE INFORMATION-
NAME DEPARTMENT CURRENT CITY MARITAL DATE OF

CODE POSITION STATUS BIRTH
---------- ---------- ---------- ---------- ---------- --------

CREMER TECH10 ANALYST GREENVILLE S 70-01-01
MARKUSH SALE00 TRAINEE LOS ANGELE D 79-03-14
GEE TECH05 MANAGER CHAPEL HIL M 41-02-04
KUNEY TECH10 DBA DETROIT S 40-02-13
NEEDHAM TECH10 PROGRAMMER CHATTANOOG S 55-08-05
JACKSON TECH10 PROGRAMMER ST LOUIS D 70-01-01
PIETSCH MGMT10 SECRETARY VISTA M 40-01-09
PAUL MGMT10 SECRETARY NORFOLK S 43-07-07
HERZOG TECH05 MANAGER CHATTANOOG S 52-09-16
DEKKER TECH10 DBA MOBILE W 40-03-03

AT END OF PAGE Statement

The AT END OF PAGE statement is used to specify any processing that is to be performedwhenever
an end-of-page condition occurs.

If the AT END OF PAGE processing produces any output, this will be output after any page trailer
(as specified with the WRITE TRAILER statement).

By default, this output is displayed left-justified on the page.

The same considerations described above for page trailers regarding physical and logical page
sizes and the number of lines output by a DISPLAY or WRITE statement also apply to AT END OF
PAGE output.

Example:

** Example 'ATENPX01': AT END OF PAGE (with system function available
** via GIVE SYSTEM FUNCTIONS in DISPLAY)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 JOB-TITLE
2 SALARY (1)

END-DEFINE
*
READ (10) EMPLOY-VIEW BY PERSONNEL-ID = '20017000'

DISPLAY NOTITLE GIVE SYSTEM FUNCTIONS
NAME JOB-TITLE 'SALARY' SALARY(1)

/*
AT END OF PAGE

WRITE / 24T 'AVERAGE SALARY: ...' AVER(SALARY(1))
END-ENDPAGE

END-READ
END

Output of Program ATENPX01:

NAME CURRENT SALARY
POSITION

-------------------- ------------------------- ----------

CREMER ANALYST 34000
MARKUSH TRAINEE 22000
GEE MANAGER 39500
KUNEY DBA 40200
NEEDHAM PROGRAMMER 32500
JACKSON PROGRAMMER 33000
PIETSCH SECRETARY 22000
PAUL SECRETARY 23000
HERZOG MANAGER 48500
DEKKER DBA 48000

AVERAGE SALARY: ... 34270            
            ↩

Further Example

See the following example program:

■ DISPLX21 - DISPLAY (with slash '/' and compare with WRITE)

The following topics are covered:

Default Column Headers

By default, each database field outputwith a DISPLAY statement is displayedwith a default column
header (which is defined for the field in the DDM).

** Example 'DISPLX01': DISPLAY
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH
2 JOB-TITLE

END-DEFINE
*
READ (3) VIEWEMP BY BIRTH

DISPLAY PERSONNEL-ID NAME JOB-TITLE
END-READ
END

Output of Program DISPLX01:

The above example program uses default headers and produces the following output.

Page 1 04-11-11 14:15:54

PERSONNEL NAME CURRENT
ID POSITION

--------- -------------------- -------------------------

30020013 GARRET TYPIST
30016112 TAILOR WAREHOUSEMAN
20017600 PIETSCH SECRETARY

Suppress Default Column Headers - NOHDR Option

If you wish your report to be output without column headers, add the keyword NOHDR to the
DISPLAY statement.

DISPLAY NOHDR PERSONNEL-ID NAME JOB-TITLE

Define Your Own Column Headers

If you wish column headers of your own to be output instead of the default headers, you specify
'text' (in apostrophes) immediately before a field, text being the header to be used for the field.

** Example 'DISPLX08': DISPLAY (with column title in 'text')
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH
2 JOB-TITLE

END-DEFINE
*
READ (3) VIEWEMP BY BIRTH

DISPLAY PERSONNEL-ID
'EMPLOYEE' NAME
'POSITION' JOB-TITLE

END-READ
END

Output of Program DISPLX08:

The above program contains the header EMPLOYEE for the field NAME, and the header POSITION for
the field JOB-TITLE; for the field PERSONNEL-ID, the default header is used. The program produces
the following output:

Page 1 04-11-11 14:15:54

PERSONNEL EMPLOYEE POSITION
ID

--------- -------------------- -------------------------

30020013 GARRET TYPIST
30016112 TAILOR WAREHOUSEMAN
20017600 PIETSCH SECRETARY

Combining NOTITLE and NOHDR

To create a report that has neither page title nor column headers, you specify the NOTITLE and
NOHDR options together in the following order:

DISPLAY NOTITLE NOHDR PERSONNEL-ID NAME JOB-TITLE

Centering of Column Headers - HC Parameter

By default, column headers are centered above the columns.With the HC parameter, you can influ-
ence the placement of column headers.

If you specify

headers will be left-justified.HC=L

headers will be right-justified.HC=R

headers will be centered.HC=C

The HC parameter can be used in a FORMAT statement to apply to the whole report, or it can be used
in a DISPLAY statement at both statement level and element level, for example:

DISPLAY (HC=L) PERSONNEL-ID NAME JOB-TITLE

Width of Column Headers - HW Parameter

With the HW parameter, you determine the width of a column output with a DISPLAY statement.

If you specify

thewidth of a DISPLAY column is determined by either the length of the header text or the length
of the field, whichever is longer. This also applies by default.

HW=ON

the width of a DISPLAY column is determined only by the length of the field. However, HW=OFF
only applies to DISPLAY statements which do not create headers; that is, either a first DISPLAY
statement with NOHDR option or a subsequent DISPLAY statement.

HW=OFF

The HW parameter can be used in a FORMAT statement to apply to the entire report, or it can be used
in a DISPLAY statement at both statement level and element (field) level.

Filler Characters for Headers - Parameters FC and GC

With the FC parameter, you specify the filler characterwhich will appear on either side of a header
produced by a DISPLAY statement across the full columnwidth if the columnwidth is determined
by the field length and not by the header (see HW parameter above); otherwise FCwill be ignored.

When a group of fields or a periodic group is output via a DISPLAY statement, a group header is
displayed across all field columns that belong to that group above the headers for the individual
fieldswithin the group.With the GC parameter, you can specify the filler characterwhichwill appear
on either side of such a group header.

While the FC parameter applies to the headers of individual fields, the GC parameter applies to the
headers for groups of fields.

The parameters FC and GC can be specified in a FORMAT statement to apply to the whole report, or
they can be specified in a DISPLAY statement at both statement level and element (field) level.

** Example 'FORMAX01': FORMAT (with parameters FC, GC)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 INCOME (1:1)

3 CURR-CODE
3 SALARY
3 BONUS (1:1)

END-DEFINE
*
FORMAT FC=* GC=$
*
READ (3) VIEWEMP BY NAME

DISPLAY NAME (FC==) INCOME (1)
END-READ
END

Output of Program FORMAX01:

Page 1 04-11-11 14:15:54

========NAME======== $$$$$$$$$$$$INCOME$$$$$$$$$$$$

CURRENCY **ANNUAL** **BONUS***
CODE SALARY

-------------------- -------- ---------- ----------

ABELLAN PTA 1450000 0
ACHIESON UKL 10500 0
ADAM FRA 159980 23000

Underlining Character for Titles and Headers - UC Parameter

By default, titles and headers are underlined with a hyphen (-).

With the UC parameter, you can specify another character to be used as underlining character.

The UC parameter can be specified in a FORMAT statement to apply to the whole report, or it can be
specified in a DISPLAY statement at both statement level and element (field) level.

** Example 'FORMAX02': FORMAT (with parameter UC)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 BIRTH
2 JOB-TITLE

END-DEFINE
*
FORMAT UC==
*
WRITE TITLE LEFT JUSTIFIED UNDERLINED 'EMPLOYEES REPORT'
SKIP 1
READ (3) VIEWEMP BY BIRTH

DISPLAY PERSONNEL-ID (UC=*) NAME JOB-TITLE
END-READ
END

In the above program, the UC parameter is specified at program level and at element (field) level:
the underlining character specified with the FORMAT statement (=) applies for the whole report -
except for the field PERSONNEL-ID, for which a different underlining character (*) is specified.

Output of Program FORMAX02:

EMPLOYEES REPORT
===============================================================================

PERSONNEL NAME CURRENT
ID POSITION

********* ==================== =========================

30020013 GARRET TYPIST
30016112 TAILOR WAREHOUSEMAN
20017600 PIETSCH SECRETARY

Suppressing Column Headers - Slash Notation

With the notation apostrophe-slash-apostrophe ('/'), you can suppress default column headers for
individual fields displayed with a DISPLAY statement. While the NOHDR option suppresses the
headers of all columns, the notation '/' can be used to suppress the header for an individual
column.

The apostrophe-slash-apostrophe ('/') notation is specified in the DISPLAY statement immediately
before the name of the field for which the column header is to be suppressed.

Compare the following two examples:

Example 1:

DISPLAY NAME PERSONNEL-ID JOB-TITLE

In this case, the default column headers of all three fields will be displayed:

Page      1                                                  04-11-11  14:15:54
                                                                               
        NAME         PERSONNEL          CURRENT                                
                          ID             POSITION                          
-------------------- --------- -------------------------                 
                                                                               
ABELLAN              60008339  MAQUINISTA               
ACHIESON             30000231  DATA BASE ADMINISTRATOR                   
ADAM                 50005800  CHEF DE SERVICE                           
ADKINSON             20008800  PROGRAMMER                                
ADKINSON             20009800  DBA                                       
ADKINSON             20011000  SALES PERSON          ↩

Example 2:

DISPLAY '/' NAME PERSONNEL-ID JOB-TITLE

In this case, the notation '/' causes the column header for the field NAME to be suppressed:

Page      1                                                  04-11-11  14:15:54
                                                                               
                     PERSONNEL          CURRENT                          
                        ID             POSITION                          
                     --------- -------------------------                 
                                                                               
ABELLAN              60008339  MAQUINISTA                                
ACHIESON             30000231  DATA BASE ADMINISTRATOR                   
ADAM                 50005800  CHEF DE SERVICE                           
ADKINSON             20008800  PROGRAMMER                                

ADKINSON             20009800  DBA                                       
ADKINSON             20011000  SALES PERSON     ↩

Further Examples of Column Headers

See the following example programs:

■ DISPLX15 - DISPLAY (with FC, UC)
■ DISPLX16 - DISPLAY (with '/', 'text', 'text/text')

This chapter discusses the use of those Natural profile and/or session parameters which you can
use to control the output format of fields.

Overview of Field-Output-Relevant Parameters

Natural provides several profile and/or session parameters you can use to control the format in
which fields are output:

FunctionParameter

With these session parameters, you can specify characters that are to be displayed before
or after a field or before a field value.

LC, IC and TC

With these session parameters, you can specify characters in Unicode format that are
to be displayed before or after a field or before a field value.

LCU, ICU and TCU

With these session parameters, you can increase or reduce the output length of fields.AL and NL

With this session parameter, you can specify the default output length for an
alphanumeric map field of format U.

DL

With this session parameter, you can determine whether negative values are to be
displayed with or without a minus sign.

SG

With this session parameter, you can suppress the display of subsequent identical field
values.

IS

With this profile and session parameter, you can determine whether field values of 0
are to be displayed or not.

ZP

With this session parameter, you can suppress the display of empty lines generated by
a DISPLAY or WRITE statement.

ES

These parameters are discussed in the following sections.

Leading Characters - LC Parameter

With the session parameter LC, you can specify leading characters that are to be displayed imme-
diately before a field that is output with a DISPLAY statement. The width of the output column is
enlarged accordingly. You can specify 1 to 10 characters.

By default, values are displayed left-justified in alphanumeric fields and right-justified in numeric
fields. (These defaults can be changed with the AD parameter; see the Parameter Reference). When
a leading character is specified for an alphanumeric field, the character is therefore displayed im-
mediately before the field value; for a numeric field, a number of spaces may occur between the
leading character and the field value.

The LC parameter can be used with the following statements:

■ FORMAT

■ DISPLAY

The LC parameter can be set at statement level and at element level.

Unicode Leading Characters - LCU Parameter

The session parameter LCU is identical to the session parameter LC. The difference is that the
leading characters are always stored in Unicode format.

This allows you to specify leading characters with mixed characters from different code pages,
and assures that always the correct character is displayed independent of the installed system
code page.

For further information, see Unicode and Code Page Support in the Natural Programming Language,
Session Parameters, section EMU, ICU, LCU, TCU versus EM, IC, LC, TC.

The parameters LCU and ICU cannot both be applied to one field.

Insertion Characters - IC Parameter

With the session parameter IC, you specify the characters to be inserted in the column immediately
preceding the value of a field that is output with a DISPLAY statement. You can specify 1 to 10 charac-
ters.

For a numeric field, the insertion characters will be placed immediately before the first significant
digit that is output, with no intervening spaces between the specified character and the field value.
For alphanumeric fields, the effect of the IC parameter is the same as that of the LC parameter.

The parameters LC and IC cannot both be applied to one field.

The IC parameter can be used with the following statements:

■ FORMAT

■ DISPLAY

The IC parameter can be set at statement level and at element level.

Unicode Insertion Characters - ICU Parameter

The session parameter ICU is identical to the session parameter IC. The difference is that the insertion
characters are always stored in Unicode format.

This allows you to specify insertion characters with mixed characters from different code pages,
and assures that always the correct character is displayed independent of the installed system
code page.

For further information, see Unicode and Code Page Support in the Natural Programming Language,
Session Parameters, section EMU, ICU, LCU, TCU versus EM, IC, LC, TC.

The parameters LCU and ICU cannot both be applied to one field.

Trailing Characters - TC Parameter

With the session parameter TC, you can specify trailing characters that are to be displayed imme-
diately to the right of a field that is outputwith a DISPLAY statement. Thewidth of the output column
is enlarged accordingly. You can specify 1 to 10 characters.

The TC parameter can be used with the following statements:

■ FORMAT

■ DISPLAY

The TC parameter can be set at statement level and at element level.

Unicode Trailing Characters - TCU Parameter

The session parameter TCU is identical to the session parameter TC. The difference is that the trailing
characters are always stored in Unicode format.

This allows you to specify trailing characters with mixed characters from different code pages,
and assures that always the correct character is displayed independent of the installed system
code page.

For further information, see Unicode and Code Page Support in the Natural Programming Language,
Session Parameters, section EMU, ICU, LCU, TCU versus EM, IC, LC, TC.

Output Length - AL and NL Parameters

With the session parameter AL, you can specify the output length for an alphanumeric field; with the
NL parameter, you can specify the output length for a numeric field. This determines the length of a
field as it will be output, which may be shorter or longer than the actual length of the field (as
defined in the DDM for a database field, or in the DEFINE DATA statement for a user-defined vari-
able).

Both parameters can be used with the following statements:

■ FORMAT

■ DISPLAY

■ WRITE

■ PRINT

■ INPUT

Both parameters can be set at statement level and at element level.

Note: If an edit mask is specified, it overrides an NL or AL specification. Edit masks are de-
scribed in Edit Masks - EM Parameter.

Display Length for Output - DL Parameter

Note: You should use the Web I/O Interface to make use of the full functionality of the DL
parameter. When using the terminal emulation, it is not possible, for example, to scroll in
a field when the value defined with DL is smaller than the field length.

With the session parameter DL, you can specify the display length for a field of format A or U, since
the display width of a Unicode string can be twice the length of the string, and the user must be
able to display the whole string. The default will be the length, for example, for a format/length
U10, the display length can be 10 to 20, whereas the default length (when DL is not specified) is
10.

The session parameter DL can be used with the following statements:

■ FORMAT

■ DISPLAY

■ WRITE

■ PRINT

■ INPUT

The session parameter DL can be set at statement level and at element level.

The difference between the session parameters AL and DL is that AL defines the data length of a
field whereas DL defines the number of columns which are used on the screen for displaying the
field. The user can scroll in input fields to view the entire content of a field if the value specified
with the DL session parameter is less than the length of the field data.

Using the DL parameter with a length that is smaller than the length of the field is only recommen-
dedwith theWeb I/O Interface.When runningNatural in a terminal emulation, scrolling in a field
is not possible and so the effect is the same as using the AL parameter. Moreover, when changing
the field contents, all characters which are beyond the display length will be lost.

Note: DL is allowed for A-format fields as well. In conjunction with the Web I/O Interface,
this would allow making the edit control size smaller than the content of a field.

Example:

DEFINE DATA LOCAL
1 #U1 (U10)
1 #U2 (U10)
END-DEFINE
*
#U1 := U'latintxt00'
#U2 := U' '
*
INPUT (AD=M) #U1 #U2
END

The above programproduces the following outputwhere the content of the field #U2 is incomplete:

#U1 latintxt00 #U2

When the session parameter DL is used with the field #U2 and is specified accordingly, the content
of this field will be displayed correctly:

DEFINE DATA LOCAL
1 #U1 (U10)
1 #U2 (U10)
END-DEFINE
*
#U1 := U'latintxt00'
#U2 := U' '
*
INPUT (AD=M) #U1 #U2 (DL=20)
END

Result:

#U1 latintxt00 #U2

Sign Position - SG Parameter

With the session parameter SG, you can determine whether or not a sign position is to be allocated
for numeric fields.

■ By default, SG=ON applies, which means that a sign position is allocated for numeric fields.
■ If you specify SG=OFF, negative values in numeric fields will be output without a minus sign (-
).

The SG parameter can be used with the following statements:

■ FORMAT

■ DISPLAY

■ PRINT

■ WRITE

■ INPUT

The SG parameter can be set at both statement level and element level.

Note: If an edit mask is specified, it overrides an SG specification. Edit masks are described
in Edit Masks - EM Parameter.

Example Program without Parameters

** Example 'FORMAX03': FORMAT (without FORMAT and compare with FORMAX04)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 SALARY (1:1)
2 BONUS (1:1,1:1)

END-DEFINE
*
READ (5) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY NAME
FIRST-NAME
SALARY (1:1)
BONUS (1:1,1:1)

END-READ
END

The above program contains no parameter settings and produces the following output:

Page 1 04-11-11 11:11:11

NAME FIRST-NAME ANNUAL BONUS
SALARY

-------------------- -------------------- ---------- ----------

JONES VIRGINIA 46000 9000
JONES MARSHA 50000 0
JONES ROBERT 31000 0
JONES LILLY 24000 0
JONES EDWARD 37600 0

Example Program with Parameters AL, NL, LC, IC and TC

In this example, the session parameters AL, NL, LC, IC and TC are used.

** Example 'FORMAX04': FORMAT (with parameters AL, NL, LC, TC, IC and
** compare with FORMAX03)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 SALARY (1:1)
2 BONUS (1:1,1:1)

END-DEFINE
*
FORMAT AL=10 NL=6
*
READ (5) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY NAME (LC=*)
FIRST-NAME (TC=*)
SALARY (1:1) (IC=$)
BONUS (1:1,1:1) (LC=>)

END-READ
END

The above program produces the following output. Compare the layout of this output with that
of the previous program to see the effect of the individual parameters:

Page 1 04-11-11 11:11:11

NAME FIRST-NAME ANNUAL BONUS
SALARY

----------- ----------- -------- --------

*JONES VIRGINIA * $46000 > 9000
*JONES MARSHA * $50000 > 0
*JONES ROBERT * $31000 > 0
*JONES LILLY * $24000 > 0
*JONES EDWARD * $37600 > 0

As you can see in the above example, any output length you specify with the AL or NL parameter
does not include any characters specified with the LC, IC and TC parameters: the width of the NAME
column, for example, is 11 characters - 10 for the field value (AL=10) plus 1 leading character.

The width of the SALARY and BONUS columns is 8 characters - 6 for the field value (NL=6), plus 1
leading/inserted character, plus 1 sign position (because SG=ON applies).

Identical Suppress - IS Parameter

With the session parameter IS, you can suppress the display of identical information in successive
lines created by a WRITE or DISPLAY statement.

■ By default, IS=OFF applies, which means that identical field values will be displayed.
■ If IS=ON is specified, a value which is identical to the previous value of that field will not be
displayed.

The IS parameter can be specified

■ with a FORMAT statement to apply to the whole report, or
■ in a DISPLAY or WRITE statement at both statement level and element level.

The effect of the parameter IS=ON can be suspended for one record by using the statement SUSPEND
IDENTICAL SUPPRESS; see the Statements documentation for details.

Compare the output of the following two example programs to see the effect of the IS parameter.
In the second one, the display of identical values in the NAME field is suppressed.

Example Program without IS Parameter

** Example 'FORMAX05': FORMAT (without parameter IS
** and compare with FORMAX06)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME

END-DEFINE
*
READ (3) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY NAME FIRST-NAME
END-READ
END

The above program produces the following output:

  Page      1                                                  04-11-11  11:11:11  
                                                                                 
          NAME              FIRST-NAME     
  -------------------- --------------------
                                           
  JONES                VIRGINIA            
  JONES                MARSHA              
  JONES                ROBERT             ↩

Example Program with IS Parameter

** Example 'FORMAX06': FORMAT (with parameter IS
** and compare with FORMAX05)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME

END-DEFINE
*
FORMAT IS=ON
*
READ (3) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY NAME FIRST-NAME
END-READ
END

The above program produces the following output:

Page 1 04-11-11 11:54:01

NAME FIRST-NAME
-------------------- --------------------

JONES VIRGINIA
MARSHA
ROBERT

Zero Printing - ZP Parameter

With the profile and session parameter ZP, you determine how a field value of zero is to be dis-
played.

■ By default, ZP=ON applies, which means that one 0 (for numeric fields) or all zeros (for time
fields) will be displayed for each field value that is zero.

■ If you specify ZP=OFF, the display of each field value which is zero will be suppressed.

The ZP parameter can be specified

■ with a FORMAT statement to apply to the whole report, or
■ in a DISPLAY or WRITE statement at both statement level and element level.

Compare the output of the following two example programs to see the effect of the parameters
ZP and ES.

Empty Line Suppression - ES Parameter

With the session parameter ES, you can suppress the output of empty lines created by a DISPLAY
or WRITE statement.

■ By default, ES=OFF applies, whichmeans that lines containing all blank valueswill be displayed.
■ If ES=ON is specified, a line resulting from a DISPLAY or WRITE statement which contains all blank
values will not be displayed. This is particularly useful when displaying multiple-value fields
or fields which are part of a periodic group if a large number of empty lines are likely to be
produced.

The ES parameter can be specified

■ with a FORMAT statement to apply to the whole report, or
■ in a DISPLAY or WRITE statement at statement level.

315Programming Guide

Parameters to Influence the Output of Fields



Note: To achieve empty suppression for numeric values, in addition to ES=ON the parameter
ZP=OFFmust also be set for the fields concerned in order to have null values turned into
blanks and thus not output either.

Compare the output of the following two example programs to see the effect of the parameters
ZP and ES.

Example Program without Parameters ZP and ES

** Example 'FORMAX07': FORMAT (without parameter ES and ZP
** and compare with FORMAX08)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 BONUS (1:2,1:1)

END-DEFINE
*
READ (4) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY NAME FIRST-NAME BONUS (1:2,1:1)
END-READ
END

The above program produces the following output:

Page 1 04-11-11 11:58:23

NAME FIRST-NAME BONUS
-------------------- -------------------- ----------

JONES VIRGINIA 9000
6750

JONES MARSHA 0
0

JONES ROBERT 0
0

JONES LILLY 0
0

Example Program with Parameters ZP and ES

** Example 'FORMAX08': FORMAT (with parameters ES and ZP
** and compare with FORMAX07)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 BONUS (1:2,1:1)

END-DEFINE
*
FORMAT ES=ON
*
READ (4) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY NAME FIRST-NAME BONUS (1:2,1:1)(ZP=OFF)
END-READ
END

The above program produces the following output:

Page 1 04-11-11 11:59:09

NAME FIRST-NAME BONUS
-------------------- -------------------- ----------

JONES VIRGINIA 9000
6750

JONES MARSHA
JONES ROBERT
JONES LILLY

Further Examples of Field-Output-Relevant Parameters

For further examples of the parameters LC, IC, TC, AL, NL, IS, ZP and ES, and the SUSPEND IDENTICAL
SUPPRESS statement, see the following example programs:

■ DISPLX17 - DISPLAY (with NL, AL, IC, LC, TC)
■ DISPLX18 - DISPLAY (using default settings for SF, AL, UC, LC, IC, TC and compare with
DISPLX19)

■ DISPLX19 - DISPLAY (with SF, AL, LC, IC, TC and compare with DISPLX18)
■ SUSPEX01 - SUSPEND IDENTICAL SUPPRESS (in conjunction with parameters IS, ES, ZP
in DISPLAY)

■ SUSPEX02 - SUSPEND IDENTICAL SUPPRESS (in conjunction with parameters IS, ES, ZP
in DISPLAY). Identical to SUSPEX01, but with IS=OFF.

■ COMPRX03 - COMPRESS

This chapter describes how you can specify an edit mask for an alphanumeric or numeric field.

Use of EM Parameter

With the session parameter EM you can specify an edit mask for an alphanumeric or numeric field,
that is, determine character by character the format in which the field values are to be output.
Using the session parameter EMU, you can define edit masks with Unicode characters in the same
way as described below for the EM session parameter.

Example:

DISPLAY NAME (EM=X^X^X^X^X^X^X^X^X^X)

In this example, each X represents one character of an alphanumeric field value to be displayed,
and each ^ represents a blank. If displayed via the DISPLAY statement, the name JOHNSONwould
appear as follows:

J O H N S O N

You can specify the session parameter EM

■ at report level (in a FORMAT statement),
■ at statement level (in a DISPLAY, WRITE, INPUT, MOVE EDITED or PRINT statement) or
■ at element level (in a DISPLAY, WRITE or INPUT statement).

An edit mask specified with the session parameter EMwill override a default edit mask specified
for a field in the DDM.

If EM=OFF is specified, no edit mask at all will be used.

An edit mask specified at statement level will override an edit mask specified at report level.

An edit mask specified at element level will override an edit mask specified at statement level.

Edit Masks for Numeric Fields

An edit mask specified for a field of format N, P, I, or F must contain at least one 9 or Z. If more
nines or Zs exist, the number of positions contained in the field value, the number of print positions
in the edit mask will be adjusted to the number of digits defined for the field value. If fewer nines
or Zs exist, the high-order digits before the decimal point and/or low-order digits after the
decimal point will be truncated.

For further information, see session parameter EM, Edit Masks for Numeric Fields in the Parameter
Reference documentation.

Edit Masks for Alphanumeric Fields

Edit masks for alphanumeric fields must include an X for each alphanumeric character that is to
be output.

With a few exceptions, you may add leading, trailing and insertion characters (with or without
enclosing them in apostrophes).

The circumflex character (^) is used to insert blanks in edit mask for both numeric and alphanu-
meric fields.

For further information, see session parameter EM,EditMasks for Alphanumeric Fields in theParameter
Reference documentation.

Length of Fields

It is important to be aware of the length of the field to which you assign an edit mask.

■ If the edit mask is longer than the field, this will yield unexpected results.
■ If the edit mask is shorter than the field, the field output will be truncated to just those positions
specified in the edit mask.

Examples:

Assuming an alphanumeric field that is 12 characters long and the field value to be output is
JOHNSON, the following edit masks will yield the following results:

OutputEdit Mask

J.O.H.N.SEM=X.X.X.X.X

****JOHNSO**EM=****XXXXXX****

Edit Masks for Date and Time Fields

Edit masks for date fields can include the characters D (day), M (month) and Y (year) in various
combinations.

Edit masks for time fields can include the characters H (hour), I (minute), S (second) and T (tenth
of a second) in various combinations.

In conjunctionwith editmasks for date and time fields, see also the date and time system variables.

Customizing Separator Character Displays

Natural programs are used in business applications all over the world. Depending on the local
conventions, it is usual to present numeric data fields and those with a date or time content in a
special output style, when displayed in I/O statements. The different appearance should not be
realized by alternate program coding that is processed selectively as a function of the locale where
the program is being executed, but should be carried outwith the sameprogram image in conjunc-
tion with a set of runtime parameters to specify the decimal point character and the “thousands
separator character”.

The following topics are covered below:

■ Decimal Separator
■ Dynamic Thousands Separator
■ Examples

Decimal Separator

The Natural parameter DC is available to specify the character to be inserted in place of any char-
acters used to represent the decimal separator (also called “radix” character) in edit masks. This
parameter enables the users of a Natural program or application to choose any (special) character
to separate the integer positions from the decimal positions of a numeric data item and enables,
for example, U.S. shops to use the decimal point (.) and European shops to use the comma (,).

Dynamic Thousands Separator

To structure the output of a large integer values, it is commonpractice to insert separators between
every three digits of an integer to separate groups of thousands. This separator is called a “thou-
sands separator”. For example, shops in the United States generally use a comma for this purpose
(1,000,000), whereas shops in Germany use the period (1.000.000), in France a space (1 000 000),
etc.

In a Natural edit mask, a “dynamic thousands separator” is a comma (or period) indicating the
position where thousands separator characters (definedwith the THSEPCH parameter) are inserted
at runtime. At compile time, the Natural profile parameter THSEP or the option THSEP of system
command COMPOPT enables or disables the interpretation of the comma (or period) as dynamic
thousands separator.

If THSEP is set to OFF (default), any character used as thousands separator in the edit mask is treated
as literal and displayed unchanged at runtime. This setting retains downwards compatibility.

If THSEP is set to ON, any comma (or period) in the edit mask is interpreted as dynamic thousands
separators. In general, the dynamic thousands separator is a comma, but if the comma is already
in use as decimal character (DC), the period is used as dynamic thousands separator.

At runtime the dynamic thousands separators are replaced by the current value of the THSEPCH
parameter (thousands separator character).

Examples

A Natural program that is cataloged with parameter settings DC='.' and THSEP=ON uses the edit
mask (EM=ZZ,ZZZ,ZZ9.99).

Displays asParameter Settings at Runtime

1,234,567.89DC='.' and THSEPCH=','

1.234.567,89DC=',' and THSEPCH='.'

1/234/567,89DC=',' and THSEPCH='/'

1 234 567,89DC=',' and THSEPCH=' '

1'234'567,89DC=',' and THSEPCH=''''

Examples of Edit Masks

Some examples of edit masks, along with possible output they produce, are provided below.

In addition, the abbreviated notation for each editmask is given. You can use either the abbreviated
or the long notation.

Output BOutput AAbbreviationEdit Mask

005.40367.32EM=9(3).9(2)EM=999.99

5790EM=Z(5)9(1)EM=ZZZZZ9

A 19379B LUEEM=X(1)^X(5)EM=X^XXXXX

AAB...01BLU...EEM=X(3)...X(2)EM=XXX...XX

12.22.8601.05.87*EM=MM.DD.YY

14.32.54.308.54.12.7**EM=HH.II.SS.T

* Use a date system variable.

** Use a time system variable.

For further information about edit masks, see the session parameter EM in the Parameter Reference.

Example Program without EM Parameters

** Example 'EDITMX01': Edit mask (using default edit masks)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 JOB-TITLE
2 SALARY (1:3)
2 CITY

END-DEFINE
*
READ (3) VIEWEMP BY NAME STARTING FROM 'JONES'

DISPLAY 'N A M E' NAME /
'OCCUPATION' JOB-TITLE
'SALARY' SALARY (1:3)
'LOCATION' CITY

SKIP 1
END-READ
END

Output of Program EDITMX01:

The output of this program shows the default edit masks available.

Page 1 04-11-11 14:15:54

N A M E SALARY LOCATION
OCCUPATION

------------------------- ---------- --------------------

JONES 46000 TULSA
MANAGER 42300

39300

JONES 50000 MOBILE
DIRECTOR 46000

42700

JONES 31000 MILWAUKEE
PROGRAMMER 29400

27600

Example Program with EM Parameters

** Example 'EDITMX02': Edit mask (using EM)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 JOB-TITLE
2 SALARY (1:3)

END-DEFINE
*
READ (3) VIEWEMP BY NAME STARTING FROM 'JONES'
DISPLAY 'N A M E' NAME (EM=X^X^X^X^X^X^X^X^X^X^X^X^X^X^X) /

FIRST-NAME (EM=...X(10)...)
'OCCUPATION' JOB-TITLE (EM=' ___ 'X(12))
'SALARY' SALARY (1:3) (EM=' USD 'ZZZ,999)

SKIP 1
END-READ
END

Output of Program EDITMX02:

Compare the outputwith that of the previous program (Example Programwithout EMParameters)
to see how the EM specifications affect the way the fields are displayed.

Page 1 04-11-11 14:15:54

N A M E OCCUPATION SALARY
FIRST-NAME

----------------------------- ---------------- -----------

J O N E S ___ MANAGER USD 46,000
..VIRGINIA ... USD 42,300

USD 39,300

J O N E S ___ DIRECTOR USD 50,000
..MARSHA ... USD 46,000

USD 42,700

J O N E S ___ PROGRAMMER USD 31,000
..ROBERT ... USD 29,400

USD 27,600

Further Examples of Edit Masks

See the following example programs:

■ EDITMX03 - Edit mask (different EM for alpha-numeric fields)
■ EDITMX04 - Edit mask (different EM for numeric fields)
■ EDITMX05 - Edit mask (EM for date and time system variables)

34 Unicode Edit Masks - EMU Parameter

Unicode edit masks can be used similar to code page edit masks. The difference is that the edit
mask is always stored in Unicode format.

This allows you to specify editmaskswithmixed characters fromdifferent code pages and assures
that always the correct character is displayed, independent of the installed system code page.

For the general usage of edit masks, see Edit Masks - EM Parameter.

For information on the session parameter EMU, see EMU - Unicode Edit Mask (in the Parameter Ref-
erence).

This chapter describes how you can combine the features of the statements DISPLAY and WRITE to
produce vertical displays of field values.

Creating Vertical Displays

There are two ways of creating vertical displays:

■ You can use a combination of the statements DISPLAY and WRITE.
■ You can use the VERT option of the DISPLAY statement.

Combining DISPLAY and WRITE

As described in Statements DISPLAY and WRITE, the DISPLAY statement normally presents the
data in columnswith default headers, while the WRITE statement presents data horizontallywithout
headers.

You can combine the features of the two statements to produce vertical displays of field values.

The DISPLAY statement produces the values of different fields for the same record across the page
with a column for each field. The field values for each record are displayed below the values for
the previous record.

By using a WRITE statement after a DISPLAY statement, you can insert text and/or field values spe-
cified in the WRITE statement between records displayed via the DISPLAY statement.

The following program illustrates the combination of DISPLAY and WRITE:

** Example 'WRITEX04': WRITE (in combination with DISPLAY)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 JOB-TITLE
2 CITY
2 DEPT

END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'SAN FRANCISCO'

DISPLAY NAME JOB-TITLE
WRITE 22T 'DEPT:' DEPT
SKIP 1

END-READ
END

Output of Program WRITEX04:

Page 1 04-11-11 14:15:55

NAME CURRENT
POSITION

-------------------- -------------------------

KOLENCE MANAGER
DEPT: TECH05

GOSDEN ANALYST
DEPT: TECH10

WALLACE SALES PERSON
DEPT: SALE20

Tab Notation - T*field

In the previous example, the position of the field DEPT is determined by the tab notation nT (in this
case 20T, which means that the display begins in column 20 on the screen).

Field values specified in a WRITE statement can be lined up automaticallywith field values specified
in the first DISPLAY statement of the program by using the tab notation T*field (where field is
the name of the field to which the field is to be aligned).

In the following program, the output produced by the WRITE statement is aligned to the field
JOB-TITLE by using the notation T*JOB-TITLE:

** Example 'WRITEX05': WRITE (in combination with DISPLAY)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 JOB-TITLE
2 DEPT
2 CITY

END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'SAN FRANCISCO'

DISPLAY NAME JOB-TITLE
WRITE T*JOB-TITLE 'DEPT:' DEPT
SKIP 1

END-READ
END

Output of Program WRITEX05:

Page 1 04-11-11 14:15:55

NAME CURRENT
POSITION

-------------------- -------------------------

KOLENCE MANAGER
DEPT: TECH05

GOSDEN ANALYST
DEPT: TECH10

WALLACE SALES PERSON
DEPT: SALE20

Positioning Notation x/y

When you use the DISPLAY and WRITE statements in sequence andmultiple lines are to be produced
by the WRITE statement, you can use the notation x/y (number-slash-number) to determine in
which row/column something is to be displayed. The positioning notation causes the next element
in the DISPLAY or WRITE statement to be placed x lines below the last output, beginning in column
y of the output.

The following program illustrates the use of this notation:

** Example 'WRITEX06': WRITE (with n/n)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 MIDDLE-I
2 ADDRESS-LINE (1:1)
2 CITY
2 ZIP

END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'NEW YORK'

DISPLAY 'NAME AND ADDRESS' NAME
WRITE 1/5 FIRST-NAME

1/30 MIDDLE-I
2/5 ADDRESS-LINE (1:1)
3/5 CITY
3/30 ZIP /

END-READ
END

Output of Program WRITEX06:

Page 1 04-11-11 14:15:55

NAME AND ADDRESS
--------------------

RUBIN
SYLVIA L
2003 SARAZEN PLACE
NEW YORK 10036

WALLACE
MARY P
12248 LAUREL GLADE C
NEW YORK 10036

KELLOGG
HENRIETTA S
1001 JEFF RYAN DR.
NEWARK 19711

DISPLAY VERT Statement

The standard display mode in Natural is horizontal.

With the VERT clause option of the DISPLAY statement, you can override the standard display and
produce a vertical field display.

The HORIZ clause option, which can be used in the same DISPLAY statement, re-activates the
standard horizontal display mode.

Columnheadings in verticalmode are controlledwith various forms of the AS clause. The following
example programs illustrate the use of the DISPLAY VERT statement:

■ DISPLAY VERT without AS Clause
■ DISPLAY with VERT AS CAPTIONED and HORIZ Clause
■ DISPLAY with VERT AS 'text' Clause
■ DISPLAY with VERT AS 'text' CAPTIONED Clause

■ Tab Notation P*field

DISPLAY VERT without AS Clause

The following program has no AS clause, which means that no column headings are output.

** Example 'DISPLX09': DISPLAY (without column title)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY

END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'NEW YORK'

DISPLAY VERT NAME FIRST-NAME / CITY
SKIP 2

END-READ
END

Output of Program DISPLX09:

Note that all field values are displayed vertically underneath one another.

Page 1 04-11-11 14:15:54

RUBIN
SYLVIA

NEW YORK

WALLACE
MARY

NEW YORK

KELLOGG
HENRIETTA

NEWARK

DISPLAY with VERT AS CAPTIONED and HORIZ Clause

The following program contains a VERT and a HORIZ clause, which causes some column values to
be output vertically and others horizontally; moreover AS CAPTIONED causes the default column
headers to be displayed.

** Example 'DISPLX10': DISPLAY (with VERT as CAPTIONED and HORIZ clause)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY
2 JOB-TITLE
2 SALARY (1:1)

END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'NEW YORK'

DISPLAY VERT AS CAPTIONED NAME FIRST-NAME
HORIZ JOB-TITLE SALARY (1:1)

SKIP 1
END-READ
END

Output of Program DISPLX10:

Page 1 04-11-11 14:15:54

NAME CURRENT ANNUAL
FIRST-NAME POSITION SALARY

-------------------- ------------------------- ----------

RUBIN SECRETARY 17000
SYLVIA

WALLACE ANALYST 38000
MARY

KELLOGG DIRECTOR 52000
HENRIETTA

DISPLAY with VERT AS 'text' Clause

The following program contains an AS 'text' clause, which displays the specified 'text' as
column header.

Note: A slash (/) within the text element in a DISPLAY statement causes a line advance.

** Example 'DISPLX11': DISPLAY (with VERT AS 'text' clause)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY
2 JOB-TITLE
2 SALARY (1:1)

END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'NEW YORK'

DISPLAY VERT AS 'EMPLOYEES' NAME FIRST-NAME
HORIZ JOB-TITLE SALARY (1:1)

SKIP 1
END-READ
END

Output of Program DISPLX11:

Page 1 04-11-11 14:15:54

EMPLOYEES CURRENT ANNUAL
POSITION SALARY

-------------------- ------------------------- ----------

RUBIN SECRETARY 17000
SYLVIA

WALLACE ANALYST 38000
MARY

KELLOGG DIRECTOR 52000
HENRIETTA

DISPLAY with VERT AS 'text' CAPTIONED Clause

The AS 'text' CAPTIONED clause causes the specified text to be displayed as column heading,
and the default column headings to be displayed immediately before the field value in each line
that is output.

The following program contains an AS 'text' CAPTIONED clause.

** Example 'DISPLX12': DISPLAY (with VERT AS 'text' CAPTIONED clause)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY
2 JOB-TITLE
2 SALARY (1:1)

END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'NEW YORK'

DISPLAY VERT AS 'EMPLOYEES' CAPTIONED NAME FIRST-NAME
HORIZ JOB-TITLE SALARY (1:1)

SKIP 1
END-READ
END

Output of Program DISPLX12:

This clause causes the default column headers (NAME and FIRST-NAME) to be placed before the field
values:

Page 1 04-11-11 14:15:54

EMPLOYEES CURRENT ANNUAL
POSITION SALARY

------------------------------- ------------------------- ----------

NAME RUBIN SECRETARY 17000
FIRST-NAME SYLVIA

NAME WALLACE ANALYST 38000
FIRST-NAME MARY

NAME KELLOGG DIRECTOR 52000
FIRST-NAME HENRIETTA

Tab Notation P*field

If you use a combination of DISPLAY VERT statement and subsequent WRITE statement, you can
use the tab notation P*field-name in the WRITE statement to align the position of a field to the
column and line position of a particular field specified in the DISPLAY VERT statement.

In the following program, the fields SALARY and BONUS are displayed in the same column, SALARY
in every first line, BONUS in every second line. The text ***SALARY PLUS BONUS*** is aligned to
SALARY, which means that it is displayed in the same column as SALARY and in the first line,
whereas the text (IN US DOLLARS) is aligned to BONUS and therefore displayed in the same column
as BONUS and in the second line.

** Example 'WRITEX07': WRITE (with P*field)
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 CITY
2 NAME
2 JOB-TITLE
2 SALARY (1:1)
2 BONUS (1:1,1:1)

END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'LOS ANGELES'

DISPLAY NAME JOB-TITLE
VERT AS 'INCOME' SALARY (1) BONUS (1,1)

WRITE P*SALARY '***SALARY PLUS BONUS***'
P*BONUS '(IN US DOLLARS)'

SKIP 1
END-READ
END

Output of Program WRITEX07:

Page 1 04-11-11 14:15:55

NAME CURRENT INCOME
POSITION

-------------------- ------------------------- ----------

SMITH 0
0

***SALARY PLUS BONUS***
(IN US DOLLARS)

POORE JR SECRETARY 25000
0

***SALARY PLUS BONUS***
(IN US DOLLARS)

PREPARATA MANAGER 46000
9000

***SALARY PLUS BONUS***
(IN US DOLLARS)

Further Example of DISPLAY VERT with WRITE Statement

See the following example program:

■ WRITEX10 - WRITE (with nT, T*field and P*field)

VII Further Programming Aspects

End of Statement, Program or Application

Processing of Application Errors

Conditional Processing - IF Statement

Loop Processing

Control Breaks

Data Computation

System Variables and System Functions

Stack

Processing of Date Information

Text Notation

User Comments

Logical Condition Criteria

Rules for Arithmetic Assignment

Invoking Natural Subprograms from 3GL Programs

Issuing Operating System Commands from within a Natural Program

Statements for Internet and XML Access

End of Statement

To explicitly mark the end of a statement, you can place a semicolon (;) between the statement
and the next statement. This can be used tomake the program structure clearer, but is not required.

End of Program

The END statement is used to mark the end of a Natural program, function, subprogram, external
subroutine or helproutine.

Every one of these objects must contain an END statement as the last statement.

Every object may contain only one END statement.

End of Application

Ending the Execution of an Application by a STOP Statement

The STOP statement is used to terminate the execution of a Natural application. A STOP statement
executed anywherewithin an application immediately stops the execution of the entire application.

Ending the Execution of an Application by a TERMINATE Statement

The TERMINATE statement stops the execution of the Natural application and also ends the Natural
session.

Interrupting a Running Natural Application

During the development of a Natural application and in test situations, the user should be able to
interrupt a running Natural application that does not respond anymore, for example, due to an
endless loop. As the Natural session should not need to be killed, the running Natural application
can be interrupted via the typical system interrupt key combination (for example, CTRL+BREAK for
Windows, CTRL+C for UNIX andOpenVMS). The Natural error NAT1016 is raised and the runtime
error processing is activated. The error can be handled by an ON ERROR processing.

In a production environment, this featurewill typically need to be disabled, because the application
may not be able to recover from a user interrupt at an arbitrary program location.

TheNatural profile parameter RTINTdetermineswhether interrupts are allowed. By default, inter-
rupts are not allowed.

If this parameter is set to ON, a running Natural application may be interrupted with the interrupt
key combination of the operating system (for example, forWindows: CTRL+BREAK; forUNIX: typically
CTRL+C, but can be reconfigured using the stty command; for OpenVMS: CTRL+C).

Natural catches this interrupt request and then offers the user the following possibilities:

■ Perform standard error processing by raising a NAT1016 error.
■ Continue application processing (cancel interrupt).

The choice is shown in a window that is opened after catching the interrupt signal.

37 Processing of Application Errors

This section discusses the two basic methods Natural offers for the handling of application errors:
default processing and application-specific processing. Furthermore, it describes the options you
have to enable the application specific error processing: coding an ON ERROR statement block
within a programming object or using a separate error transaction program.

Finally, this section gives an overview of the features that are provided to configureNatural's error
processing behavior, to retrieve information on an error, to process or debug an application error.

This document covers the following topics:

For information on error handling in a Natural RPC environment, see Handling Errors in the Nat-
ural Remote Procedure Call documentation.

Natural's Default Error Processing

When an error occurs in a Natural application, Natural will by default proceed in the following
way:

1. Natural terminates the execution of the currently running application object;

2. Natural issues an error message;

3. Natural returns to command input mode.

“Command inputmode”means that, depending on yourNatural configuration, theNatural main
menu, the NEXT command prompt, or a user-defined startup menu may appear.

The displayed error message contains the Natural error number, the corresponding message text
and the affected Natural object and line number where the error has occurred.

Because after the occurrence of an error the execution of the affected application object is terminated,
the status of any pending database transactions may be affected by actions required by the setting
of the profile parameter ETEOP. Unless Natural has issued an END TRANSACTION statement as a
result of the settings of these parameters, a BACKOUT TRANSACTION statement is issued when Nat-
ural returns to command input mode.

Application Specific Error Processing

Natural enables you to adapt the error processing if the default error processing does not meet
your application’s requirements. Possible reasons to establish an application specific error processing
may be:

■ The information on the error is to be stored for further analysis by the application developer.
■ The application execution is to be continued after error recovery, if possible.

■ A specific transaction handling is necessary.

Because the execution of the affected Natural application object is terminated after an application
error has occurred, the status of the pending database transactions may be influenced by actions
which are triggered by the settings of the profile parameter ETEOP. Therefore, further transaction
handling (END TRANSACTION or BACKOUT TRANSACTION statement) has to be performed by the ap-
plication’s error processing.

To enable the application specific error processing, you have the following options:

■ You may code an ON ERROR statement block within a programming object.
■ You may use a separate error transaction program.

These options are described in the following sections.

Using an ON ERROR Statement Block

You may use the ON ERROR statement to intercept execution time errors within the application
where an error occurs.

From within an ON ERROR statement block, it is possible to resume application execution on the
current level or on a superior level.

Moreover, you may specify an ON ERROR statement in multiple objects of an application in order
to process any errors that have occurred on subordinate levels. Thus, application specific error
processing may exactly be tailored to the application’s needs.

Exiting from an ON ERROR Statement Block

You may exit from an ON ERROR statement block by specifying one of the following statements:

■ RETRY

Application execution is resumed on the current level.
■ ESCAPE ROUTINE

Error processing is assumed to be complete and application execution is resumed on the super-
ior level.

■ FETCH

Error processing is assumed to be complete and the “fetched” program is executed.

STOP

Natural stops the execution of the affectedprogram, ends the application and returns to command
input mode.

■ TERMINATE

The execution of the Natural application is stopped and also the Natural session is terminated.

Error Processing Rules

■ If the execution of the ON ERROR statement block is not terminated by one of these statements,
the error is percolated to the Natural object on the superior level for processing by an ON ERROR
statement block that exists there.

■ If none of theNatural objects on any of the superior levels contains an ON ERROR statement block,
but if an error transaction program has been specified (as described in the next section), this
error transaction program will receive control.

■ If none of theNatural objects on any of the superior levels contains an ON ERROR statement block
and no error transaction program has been specified there, Natural's default error processing
will be performed as described above.

Using an Error Transaction Program

You may specify an error transaction program in the following places:

■ In the profile parameter ETA.
■ If Natural Security is installed, within the Natural Security library profile; see Components of a
Library Profile in the Natural Security documentation.

■ Within a Natural object by assigning the name of the programwhich is to receive control in the
event of an error condition as a value to the systemvariable *ERROR-TA, using an ASSIGN, COMPUTE
or MOVE statement.

If you assign the name of an error transaction program to the system variable *ERROR-TA during
the Natural session, this assignment supersedes an error transaction program specified using the
profile parameter ETA. Regardless of whether you use the ETA profile parameter or assign a value
to the system variable *ERROR-TA, the error transaction program names are not saved and restored
by Natural for different levels of the call hierarchy. Therefore, if you assign a name to the system
variable *ERROR-TA in a Natural object, the specified programwill be invoked to process any error
that occurs in the current Natural session after the assignment.

On the one hand, if you specify an error transaction program by using the profile parameter ETA,
an error transaction is defined for the complete Natural session without having the need for indi-
vidual assignments in Natural objects. On the other hand, the method of assigning a program to
the system variable *ERROR-TA provides more flexibility and, for example, allows you to have
different error transaction programs in different application branches.

If the system variable *ERROR-TA is reset to blank, Natural's default error processing will be per-
formed as described above.

If an error transaction program is specified and an application error occurs, execution of the ap-
plication is terminated, and the specified error transaction program receives control to perform
the following actions:

■ Analyze the error;
■ Log the error information;
■ Terminate the Natural session;
■ Continue the application execution by calling a program using the FETCH statement.

Because the error transaction program receives control in the same way as if it had been called
from the command prompt, it is not possible to resume application execution in one of theNatural
objects that were active at the time when the error occurred.

If a syntax error occurs and the Natural profile parameter SYNERR is set to ON, the error transaction
program will also receive control.

An error transaction programmust be located in the library to which you are currently logged on
or in a current steplib library.

When an error occurs, Natural executes a STACK TOP DATA statement and places the following in-
formation at the top of the stack:

DescriptionFormat/LengthStack Data

Natural error number.

Note: If session parameter SG is set to ON, the format/length will be N5.

N4Error number

Number of the line where the error has occurred.

If the status is C or L, the line number will be zero.

N4Line number

Status code:A1Status

Command processing errorC

Logon processing errorL

Object (execution) time errorO

Error on remote server (in conjunction
with Natural RPC)

R

Syntax errorS

Name of the Natural object where the error has occurred.A8Object name

Level number of the Natural object where the error has occurred.

If a Natural syntax error occurs at compile time and profile parameter
SYNERR is set to ON, the level number will be zero.

N2Level number

DescriptionFormat/LengthStack Data

If a Natural runtime error occurs and the level number of the Natural
object is greater than 99, the value 99will be stacked, and the current
value will be stacked in the additional stack data “Level number
enhanced”.

If a Natural runtime error occurs and the level number of the Natural object is greater than 99:

Current level number (512 at maximum).I4Level number
enhanced

If a Natural syntax error occurs at compile time and profile parameter SYNERR is set to ON:

Position of the offending item in the source line.N3Error position

Length of the offending item.N3Item length

This information can be retrieved in the error transaction program, using an INPUT statement.

Example:

DEFINE DATA LOCAL
1 #ERROR-NR (N5)
1 #LINE (N4)
1 #STATUS-CODE (A1)
1 #PROGRAM (A8)
1 #LEVEL (N2)
1 #LEVELI4 (I4)
1 #POSITION-IN-LINE (N3)
1 #LENGTH-OF-ITEM (N3)
END-DEFINE
IF *DATA > 6 THEN /* SYNERR = ON and a syntax error occurred

INPUT
#ERROR-NR
#LINE
#STATUS-CODE
#PROGRAM
#LEVEL
#POSITION-IN-LINE
#LENGTH-OF-ITEM

ELSE
INPUT /* other error

#ERROR-NR
#LINE
#STATUS-CODE
#PROGRAM
#LEVEL
#LEVELI4

END-IF
WRITE #STATUS-CODE
* DECIDE ON FIRST VALUE OF STATUS-CODE
* ... /* process error
* END-DECIDE
END

Some of the information placed on top of the stack is equivalent to the contents of several system
variables that are available in an ON ERROR statement block:

Equivalent System Variable in ON ERROR Statement BlockStack Data

*ERROR-NRError number

*ERROR-LINELine number

*PROGRAMObject name

*LEVELLevel number

Rules under Natural Security

If Natural Security is installed, the additional rules for the processing of logon errors apply. For
further information, see Transactions in the Natural Security documentation.

Error Processing Related Features

Natural provides a variety of error processing related features that

■ Enable you to configure Natural’s error processing behavior;
■ Help you in retrieving information about errors that have occurred;
■ Support you in processing these errors;
■ Support you in debugging application errors.

These features can be grouped as follows:

■ Profile parameters
■ System variables
■ Terminal Commands
■ Application programming interfaces

Profile Parameters

The following profile parameters have an influence on the behavior of Natural in the event of an
error:

PurposeProfile Parameter

Conversion ErrorCPCVERR

Error Transaction ProgramETA

Issue END TRANSACTION at End of ProgramETEOP

Handling of Response Code 113 for FIND StatementRCFIND

Handling of Response Code 113 for GET StatementRCGET

Control of Syntax ErrorsSYNERR

System Variables

The following application related systemvariables can be used to locate an error or to obtain/specify
the name of the program which is to receive control in the event of an error condition:

ContentSystem Variable

Source-code line number of the statement that caused an error.

See Example 1.

*ERROR-LINE

Error number of the error which caused an ON ERROR condition to be entered.*ERROR-NR

Name of the program which is to receive control in the event of an error condition.

See Example 2.

*ERROR-TA

Level number of the Natural object where the error has occurred.*LEVEL

Name of the library to which the user is currently logged on.*LIBRARY-ID

Name of the Natural object that is currently being executed.

See Example 1.

*PROGRAM

Example 1:

...
/*
ON ERROR
IF *ERROR-NR = 3009 THEN

WRITE 'LAST TRANSACTION NOT SUCCESSFUL'
/ 'HIT ENTER TO RESTART PROGRAM'

FETCH 'ONEEX1'
END-IF
WRITE 'ERROR' *ERROR-NR 'OCCURRED IN PROGRAM' *PROGRAM

'AT LINE' *ERROR-LINE
FETCH 'MENU'

END-ERROR
/*

...

Example 2:

...
*ERROR-TA := 'ERRORTA1'
/* from now on, program ERRORTA1 will be invoked
/* to process application errors

...
MOVE 'ERRORTA2' TO *ERROR-TA
/* change error transaction program to ERRORTA2

...

For further information on these system variables, see the corresponding sections in the System
Variables documentation.

Terminal Commands

The following terminal command has an influence on the behavior of Natural in the event of an
error:

PurposeTerminal Command

Activate/Deactivate Error Processing%E=

Application Programming Interfaces

The following application programming interfaces (APIs) are generally available for getting addi-
tional information on an error situation or to install an error transaction.

PurposeAPI

Get type of last errorUSR0040N

Get error level for error in nested copycodesUSR1016N

Get information on last errorUSR2001N

Get information from error message collectorUSR2006N

Get or set data for RPC default serverUSR2007N

Get error information on last database callUSR2010N

Get TECH informationUSR2026N

Get dynamic error message parts from the last errorUSR2030N

Find user short error message (including steplibs search)USR3320N

Get program level informationUSR4214N

With the IF statement, you define a logical condition, and the execution of the statement attached
to the IF statement then depends on that condition.

Structure of IF Statement

The IF statement contains three components:

In the IF clause, you specify the logical condition which is to be met.IF

In the THEN clause you specify the statement(s) to be executed if this condition is met.THEN

In the (optional) ELSE clause, you can specify the statement(s) to be executed if this condition is not
met.

ELSE

So, an IF statement takes the following general form:

IF condition
THEN execute statement(s)
ELSE execute other statement(s)

END-IF

Note: If you wish a certain processing to be performed only if the IF condition is notmet,
you can specify the clause THEN IGNORE. The IGNORE statement causes the IF condition to
be ignored if it is met.

Example 1:

** Example 'IFX01': IF
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 BIRTH
2 CITY
2 SALARY (1:1)

END-DEFINE
*
LIMIT 7
READ MYVIEW BY CITY STARTING FROM 'C'
IF SALARY (1) LT 40000 THEN

WRITE NOTITLE '*****' NAME 30X 'SALARY LT 40000'
ELSE

DISPLAY NAME BIRTH (EM=YYYY-MM-DD) SALARY (1)
END-IF

END-READ
END

The IF statement block in the above program causes the following conditional processing to be
performed:

■ IF the salary is less than 40000, THEN the WRITE statement is to be executed;
■ otherwise (ELSE), that is, if the salary is 40000 or more, the DISPLAY statement is to be executed.

Output of Program IFX01:

NAME DATE ANNUAL
OF SALARY

BIRTH
-------------------- ---------- ----------

***** KEEN SALARY LT 40000
***** FORRESTER SALARY LT 40000
***** JONES SALARY LT 40000
***** MELKANOFF SALARY LT 40000
DAVENPORT 1948-12-25 42000
GEORGES 1949-10-26 182800
***** FULLERTON SALARY LT 40000

Example 2:

** Example 'IFX03': IF
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 BONUS (1,1)
2 SALARY (1)

*
1 #INCOME (N9)
1 #TEXT (A26)
END-DEFINE
*
WRITE TITLE '-- DISTRIBUTION OF CATALOGS I AND II --' /
*
READ (3) EMPLOY-VIEW BY CITY = 'SAN FRANSISCO'

COMPUTE #INCOME = BONUS(1,1) + SALARY(1)
/*

IF #INCOME > 40000
MOVE 'CATALOGS I AND II' TO #TEXT

ELSE
MOVE 'CATALOG I' TO #TEXT

END-IF
/*
DISPLAY NAME 5X 'SALARY' SALARY(1) / BONUS(1,1)
WRITE T*SALARY '-'(10) /

16X 'INCOME:' T*SALARY #INCOME 3X #TEXT /

359Programming Guide

Conditional Processing - IF Statement



16X '='(19)
SKIP 1

END-READ
END

Output of Program IFX03:

-- DISTRIBUTION OF CATALOGS I AND II --
NAME SALARY

BONUS
-------------------- ----------

COLVILLE JR 56000
0

----------
INCOME: 56000 CATALOGS I AND II
===================

RICHMOND 9150
0

----------
INCOME: 9150 CATALOG I
===================

MONKTON 13500
600

----------
INCOME: 14100 CATALOG I
===================

Nested IF Statements

It is possible to use various nested IF statements; for example, you can make the execution of a
THEN clause dependent on another IF statement which you specify in the THEN clause.

Example:

** Example 'IFX02': IF (two IF statements nested)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 SALARY (1:1)
2 BIRTH
2 PERSONNEL-ID

1 MYVIEW2 VIEW OF VEHICLES
2 PERSONNEL-ID

2 MAKE
*
1 #BIRTH (D)
END-DEFINE
*
MOVE EDITED '19450101' TO #BIRTH (EM=YYYYMMDD)
*
LIMIT 20
FND1. FIND MYVIEW WITH CITY = 'BOSTON'

SORTED BY NAME
IF SALARY (1) LESS THAN 20000

WRITE NOTITLE '*****' NAME 30X 'SALARY LT 20000'
ELSE

IF BIRTH GT #BIRTH
FIND MYVIEW2 WITH PERSONNEL-ID = PERSONNEL-ID (FND1.)

DISPLAY (IS=ON) NAME BIRTH (EM=YYYY-MM-DD)
SALARY (1) MAKE (AL=8 IS=OFF)

END-FIND
END-IF

END-IF
SKIP 1

END-FIND
END

Output of Program IFX02:

NAME DATE ANNUAL MAKE
OF SALARY

BIRTH
-------------------- ---------- ---------- --------

***** COHEN SALARY LT 20000

CREMER 1972-12-14 20000 FORD

***** FLEMING SALARY LT 20000

PERREAULT 1950-05-12 30500 CHRYSLER

***** SHAW SALARY LT 20000

STANWOOD 1946-09-08 31000 CHRYSLER
FORD

A processing loop is a group of statements which are executed repeatedly until a stated condition
has been satisfied, or as long as a certain condition prevails.

Use of Processing Loops

Processing loops can be subdivided into database loops and non-database loops:

■ Database processing loops
are those created automatically by Natural to process data selected from a database as a result
of a READ, FIND or HISTOGRAM statement. These statements are described in the sectionDatabase
Access.

■ Non-database processing loops
are initiated by the statements REPEAT, FOR, CALL FILE, CALL LOOP, SORT, and READ WORK FILE.

More than one processing loopmay be active at the same time. Loopsmay be embedded or nested
within other loops which remain active (open).

A processing loopmust be explicitly closedwith a corresponding END-... statement (for example,
END-REPEAT, END-FOR)

The SORT statement, which invokes the sort program of the operating system, closes all active
processing loops and initiates a new processing loop.

Limiting Database Loops

The following topics are covered below:

■ Possible Ways of Limiting Database Loops
■ LT Session Parameter
■ LIMIT Statement
■ Limit Notation

■ Priority of Limit Settings

Possible Ways of Limiting Database Loops

With the statements READ, FIND or HISTOGRAM, you have three ways of limiting the number of repe-
titions of the processing loops initiated with these statements:

■ using the session parameter LT,
■ using a LIMIT statement,
■ or using a limit notation in a READ/FIND/HISTOGRAM statement itself.

LT Session Parameter

At session level, you can specify the session parameter LT, which limits the number of records
which may be read in a database processing loop.

Example:

GLOBALS LT=100

This limit applies to all READ, FIND and HISTOGRAM statements in the entire session.

LIMIT Statement

In a program, you can use the LIMIT statement to limit the number of records which may be read
in a database processing loop.

Example:

LIMIT 100

The LIMIT statement applies to the remainder of the program unless it is overridden by another
LIMIT statement or limit notation.

Limit Notation

With a READ, FIND or HISTOGRAM statement itself, you can specify the number of records to be read
in parentheses immediately after the statement name.

Example:

READ (10) VIEWXYZ BY NAME

This limit notation overrides any other limit in effect, but applies only to the statement in which
it is specified.

Priority of Limit Settings

If the limit set with the LT parameter is smaller than a limit specified with a LIMIT statement or a
limit notation, the LT limit has priority over any of these other limits.

Limiting Non-Database Loops - REPEAT Statement

Non-database processing loops begin and end based on logical condition criteria or some other
specified limiting condition.

The REPEAT statement is discussed here as representative of a non-database loop statement.

With the REPEAT statement, you specify one ormore statementswhich are to be executed repeatedly.
Moreover, you can specify a logical condition, so that the statements are only executed either until
or as long as that condition is met. For this purpose you use an UNTIL or WHILE clause.

If you specify the logical condition

■ in an UNTIL clause, the REPEAT loop will continue until the logical condition is met;
■ in a WHILE clause, the REPEAT loop will continue as long as the logical condition remains true.

If you specify no logical condition, the REPEAT loop must be exited with one of the following
statements:

■ ESCAPE terminates the execution of the processing loop and continues processing outside the
loop (see below).

■ STOP stops the execution of the entire Natural application.
■ TERMINATE stops the execution of the Natural application and also ends the Natural session.

Example of REPEAT Statement

** Example 'REPEAX01': REPEAT
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 SALARY (1:1)

*
1 #PAY1 (N8)
END-DEFINE
*
READ (5) MYVIEW BY NAME WHERE SALARY (1) = 30000 THRU 39999

MOVE SALARY (1) TO #PAY1
/*
REPEAT WHILE #PAY1 LT 40000

MULTIPLY #PAY1 BY 1.1
DISPLAY NAME (IS=ON) SALARY (1)(IS=ON) #PAY1

END-REPEAT
/*
SKIP 1

END-READ
END

Output of Program REPEAX01:

Page 1 04-11-11 14:15:54

NAME ANNUAL #PAY1
SALARY

-------------------- ---------- ---------

ADKINSON 34500 37950
41745

33500 36850
40535

36000 39600
43560

AFANASSIEV 37000 40700

ALEXANDER 34500 37950
41745

Terminating a Processing Loop - ESCAPE Statement

The ESCAPE statement is used to terminate the execution of a processing loop based on a logical
condition.

You can place an ESCAPE statement within loops in conditional IF statement groups, in break
processing statement groups (AT END OF DATA, AT END OF PAGE, AT BREAK), or as a stand-alone
statement implementing the basic logical conditions of a non-database loop.

The ESCAPE statement offers the options TOP and BOTTOM, which determine where processing is to
continue after the processing loop has been left via the ESCAPE statement:

■ ESCAPE TOP is used to continue processing at the top of the processing loop.
■ ESCAPE BOTTOM is used to continue processing with the first statement following the processing
loop.

You can specify several ESCAPE statements within the same processing loop.

For further details and examples of the ESCAPE statement, see the Statements documentation.

Loops Within Loops

Adatabase statement can be placedwithin a database processing loop initiated by another database
statement. When database loop-initiating statements are embedded in this way, a “hierarchy” of
loops is created, each of which is processed for each record which meets the selection criteria.

Multiple levels of loops can be embedded. For example, non-database loops can be nested one
inside the other. Database loops can be nested inside non-database loops. Database and non-
database loops can be nested within conditional statement groups.

Example of Nested FIND Statements

The following program illustrates a hierarchy of two loops,with one FIND loop nested or embedded
within another FIND loop.

Programming Guide368

Loop Processing



** Example 'FINDX06': FIND (two FIND statements nested)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 PERSONNEL-ID

1 VEH-VIEW VIEW OF VEHICLES
2 MAKE
2 PERSONNEL-ID

END-DEFINE
*
FND1. FIND EMPLOY-VIEW WITH CITY = 'NEW YORK' OR = 'BEVERLEY HILLS'

FIND (1) VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (FND1.)
DISPLAY NOTITLE NAME CITY MAKE

END-FIND
END-FIND
END

The above program selects data frommultiple files. The outer FIND loop selects from the EMPLOYEES
file all persons who live in New York or Beverley Hills. For each record selected in the outer loop,
the inner FIND loop is entered, selecting the car data of those persons from the VEHICLES file.

Output of Program FINDX06:

NAME CITY MAKE

-------------------- -------------------- --------------------

RUBIN NEW YORK FORD
OLLE BEVERLEY HILLS GENERAL MOTORS
WALLACE NEW YORK MAZDA
JONES BEVERLEY HILLS FORD
SPEISER BEVERLEY HILLS GENERAL MOTORS

Referencing Statements within a Program

Statement reference notation is used for the following purposes:

■ Referring to previous statements in a program in order to specify processing over a particular
range of data.

■ Overriding Natural's default referencing.
■ Documenting.

AnyNatural statement which causes a processing loop to be initiated and/or causes data elements
in a database to be accessed can be referenced, for example:

■ READ

■ FIND

■ HISTOGRAM

■ SORT

■ REPEAT

■ FOR

When multiple processing loops are used in a program, reference notation is used to uniquely
identify the particular database field to be processed by referring back to the statement that origin-
ally accessed that field in the database.

If a field can be referenced in such a way, this is indicated in the Referencing Permitted column
of theOperand Definition Table in the corresponding statement description (in the Statements docu-
mentation). See also User-Defined Variables, Referencing of Database Fields Using (r) Notation.

In addition, reference notation can be specified in some statements. For example:

■ AT START OF DATA

■ AT END OF DATA

■ AT BREAK

■ ESCAPE BOTTOM

Without reference notation, an AT START OF DATA, AT END OF DATA or AT BREAK statement will
be related to the outermost active READ, FIND, HISTOGRAM, SORT or READ WORK FILE loop. With refer-
ence notation, you can relate it to another active processing loop.

If reference notation is specified with an ESCAPE BOTTOM statement, processing will continue with
the first statement following the processing loop identified by the reference notation.

Statement reference notation may be specified in the form of a statement reference label or a source-
code line number.

■ Statement reference label
A statement reference label consists of several characters, the last of which must be a period (.).
The period serves to identify the entry as a label.

A statement that is to be referenced is marked with a label by placing the label at the beginning
of the line that contains the statement. For example:

0030 ...
0040 READ1. READ VIEWXYZ BY NAME
0050 ...

In the statement that references the marked statement, the label is placed in parentheses at the
location indicated in the statement's syntax diagram (as described in the Statements documenta-
tion). For example:

AT BREAK (READ1.) OF NAME

■ Source-code line number
If source-code line numbers are used for referencing, they must be specified as 4-digit numbers
(leading zeros must not be omitted) and in parentheses. For example:

AT BREAK (0040) OF NAME

In a statement where the label/line number relates a particular field to a previous statement, the
label/line number is placed in parentheses after the field name. For example:

DISPLAY NAME (READ1.) JOB-TITLE (READ1.) MAKE MODEL

Line numbers and labels can be used interchangeably.

See also User-Defined Variables, Referencing of Database Fields Using (r) Notation.

Example of Referencing with Line Numbers

The following program uses source code line numbers (4-digit numbers in parentheses) for refer-
encing.

In this particular example, the line numbers refer to the statements that would be referenced in
any case by default.

0010 ** Example 'LABELX01': Labels for READ and FIND loops (line numbers)
0020 ************************************************************************
0030 DEFINE DATA LOCAL
0040 1 MYVIEW1 VIEW OF EMPLOYEES
0050 2 NAME
0060 2 FIRST-NAME
0070 2 PERSONNEL-ID
0080 1 MYVIEW2 VIEW OF VEHICLES
0090 2 PERSONNEL-ID
0100 2 MAKE
0110 END-DEFINE
0120 *

0130 LIMIT 15
0140 READ MYVIEW1 BY NAME STARTING FROM 'JONES'
0150 FIND MYVIEW2 WITH PERSONNEL-ID = PERSONNEL-ID (0140)
0160 IF NO RECORDS FOUND
0170 MOVE '***NO CAR***' TO MAKE
0180 END-NOREC
0190 DISPLAY NOTITLE NAME (0140) (IS=ON)
0200 FIRST-NAME (0140) (IS=ON)
0210 MAKE (0150)
0220 END-FIND /* (0150)
0230 END-READ /* (0140)
0240 END

Example with Statement Reference Labels

The following example illustrates the use of statement reference labels.

It is identical to the previous program, except that labels are used for referencing instead of line
numbers.

** Example 'LABELX02': Labels for READ and FIND loops (user labels)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW1 VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 PERSONNEL-ID

1 MYVIEW2 VIEW OF VEHICLES
2 PERSONNEL-ID
2 MAKE

END-DEFINE
*
LIMIT 15
RD. READ MYVIEW1 BY NAME STARTING FROM 'JONES'

FD. FIND MYVIEW2 WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
IF NO RECORDS FOUND
MOVE '***NO CAR***' TO MAKE

END-NOREC
DISPLAY NOTITLE NAME (RD.) (IS=ON)

FIRST-NAME (RD.) (IS=ON)
MAKE (FD.)

END-FIND /* (FD.)
END-READ /* (RD.)                                                     
END ↩

Both programs produce the following output:

NAME FIRST-NAME MAKE
-------------------- -------------------- --------------------

JONES VIRGINIA CHRYSLER
MARSHA CHRYSLER

CHRYSLER
ROBERT GENERAL MOTORS
LILLY FORD

MG
EDWARD GENERAL MOTORS
LAUREL GENERAL MOTORS
KEVIN DATSUN
GREGORY FORD

JOPER MANFRED ***NO CAR***
JOUSSELIN DANIEL RENAULT
JUBE GABRIEL ***NO CAR***
JUNG ERNST ***NO CAR***
JUNKIN JEREMY ***NO CAR***
KAISER REINER ***NO CAR***
KANT HEIKE ***NO CAR***

This chapter describes how the execution of a statement can bemade dependent on a control break,
and how control breaks can be used for the evaluation of Natural system functions.

Use of Control Breaks

A control break occurs when the value of a control field changes.

The execution of statements can be made dependent on a control break.

A control break can also be used for the evaluation of Natural system functions.

System functions are discussed in SystemVariables and SystemFunctions. For detailed descriptions
of the system functions available, refer to the System Functions documentation.

AT BREAK Statement

With the statement AT BREAK, you specify the processing which is to be performed whenever a
control break occurs, that is, whenever the value of a control field which you specify with the AT
BREAK statement changes. As a control field, you can use a database field or a user-defined variable.

The following topics are covered below:

■ Control Break Based on a Database Field
■ Control Break Based on a User-Defined Variable
■ Multiple Control Break Levels

Control Break Based on a Database Field

The field specified as control field in an AT BREAK statement is usually a database field.

Example:

...
AT BREAK OF DEPT

statements
END-BREAK
...

In this example, the control field is the database field DEPT; if the value of the field changes, for
example, FROM SALE01 to SALE02, the statements specified in the AT BREAK statement would be
executed.

Instead of an entire field, you can also use only part of a field as a control field.With the slash-n-slash
notation /n/, you can determine that only the first n positions of a field are to be checked for a
change in value.

Example:

...
AT BREAK OF DEPT /4/

statements
END-BREAK
...

In this example, the specified statementswould only be executed if the value of the first 4 positions
of the field DEPT changes, for example, FROM SALE to TECH; if, however, the field value changes
from SALE01 to SALE02, this would be ignored and no AT BREAK processing performed.

Example:

** Example 'ATBREX01': AT BREAK OF (with database field)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 COUNTRY
2 JOB-TITLE
2 SALARY (1:1)

END-DEFINE
*
READ (5) MYVIEW BY CITY WHERE COUNTRY = 'USA'

DISPLAY CITY (AL=9) NAME 'POSITION' JOB-TITLE 'SALARY' SALARY(1)
/*
AT BREAK OF CITY

WRITE / OLD(CITY) (EM=X^X^X^X^X^X^X^X^X^X^X^)
5X 'AVERAGE:' T*SALARY AVER(SALARY(1)) //

COUNT(SALARY(1)) 'RECORDS FOUND' /
END-BREAK
/*
AT END OF DATA

WRITE 'TOTAL (ALL RECORDS):' T*SALARY(1) TOTAL(SALARY(1))
END-ENDDATA

END-READ
END

In the above program, the first WRITE statement is executed whenever the value of the field CITY
changes.

In the AT BREAK statement, the Natural system functions OLD, AVER and COUNT are evaluated (and
output in the WRITE statement).

In the AT END OF DATA statement, the Natural system function TOTAL is evaluated.

Output of Program ATBREX01:

Page      1                                                  04-12-14  14:07:26
                                                                               
  CITY            NAME                 POSITION            SALARY              
--------- -------------------- ------------------------- ----------            
                                                                               
AIKEN     SENKO                PROGRAMMER                     31500            
                                                                               
A I K E N                  AVERAGE:                           31500            
                                                                               
       1 RECORDS FOUND                                                         
                                                                               
ALBUQUERQ HAMMOND              SECRETARY                      22000            
ALBUQUERQ ROLLING              MANAGER                        34000            
ALBUQUERQ FREEMAN              MANAGER                        34000            
ALBUQUERQ LINCOLN              ANALYST                        41000            
                                                                               
A L B U Q U E R Q U E      AVERAGE:                           32750            
                                                                               
       4 RECORDS FOUND                                                         
                                                                               
TOTAL (ALL RECORDS):                                         162500           ↩

Control Break Based on a User-Defined Variable

A user-defined variable can also be used as control field in an AT BREAK statement.

In the following program, the user-defined variable #LOCATION is used as control field.

** Example 'ATBREX02': AT BREAK OF (with user-defined variable and
** in conjunction with BEFORE BREAK PROCESSING)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 CITY
2 COUNTRY
2 JOB-TITLE
2 SALARY (1:1)

*
1 #LOCATION (A20)
END-DEFINE
*
READ (5) MYVIEW BY CITY WHERE COUNTRY = 'USA'

BEFORE BREAK PROCESSING
COMPRESS CITY 'USA' INTO #LOCATION

END-BEFORE
DISPLAY #LOCATION 'POSITION' JOB-TITLE 'SALARY' SALARY (1)

/*
AT BREAK OF #LOCATION

SKIP 1
END-BREAK

END-READ
END

Output of Program ATBREX02:

Page      1                                                  04-12-14  14:08:36
                                                                               
     #LOCATION               POSITION            SALARY                        
-------------------- ------------------------- ----------                      
                                                                               
AIKEN USA            PROGRAMMER                     31500                      
                                                                               
ALBUQUERQUE USA      SECRETARY                      22000                      
ALBUQUERQUE USA      MANAGER                        34000                      
ALBUQUERQUE USA      MANAGER                        34000                      
ALBUQUERQUE USA      ANALYST                        41000                     ↩

Multiple Control Break Levels

As explained above, the notation /n/ allows some portion of a field to be checked for a control
break. It is possible to combine several AT BREAK statements, using an entire field as control field
for one break and part of the same field as control field for another break.

In such a case, the break at the lower level (entire field) must be specified before the break at the
higher level (part of field); that is, in the first AT BREAK statement the entire field must be specified
as control field, and in the second one part of the field.

The following example program illustrates this, using the field DEPT as well as the first 4 positions
of that field (DEPT /4/).

** Example 'ATBREX03': AT BREAK OF (two statements in combination)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 JOB-TITLE
2 DEPT
2 SALARY (1:1)
2 CURR-CODE (1:1)

END-DEFINE
*
READ MYVIEW BY DEPT STARTING FROM 'SALE40' ENDING AT 'TECH10'

WHERE SALARY(1) GT 47000 AND CURR-CODE(1) = 'USD'
/*
AT BREAK OF DEPT

WRITE '*** LOWEST BREAK LEVEL ***' /

379Programming Guide

Control Breaks



END-BREAK
AT BREAK OF DEPT /4/

WRITE '*** HIGHEST BREAK LEVEL ***'
END-BREAK
/*
DISPLAY DEPT NAME 'POSITION' JOB-TITLE

END-READ
END

Output of Program ATBREX03:

Page      1                                                  04-12-14  14:09:20
                                                                               
DEPARTMENT         NAME                 POSITION                               
   CODE                                                                        
---------- -------------------- -------------------------                      
                                                                               
TECH05     HERZOG               MANAGER                                        
TECH05     LAWLER               MANAGER                                        
TECH05     MEYER                MANAGER                                        
*** LOWEST BREAK LEVEL ***                                                     
                                                                               
TECH10     DEKKER               DBA                                            
*** LOWEST BREAK LEVEL ***                                                     
                                                                               
*** HIGHEST BREAK LEVEL ***    ↩

In the following program, one blank line is output whenever the value of the field DEPT changes;
and whenever the value in the first 4 positions of DEPT changes, a record count is carried out by
evaluating the system function COUNT.

** Example 'ATBREX04': AT BREAK OF (two statements in combination)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 DEPT
2 REDEFINE DEPT

3 #GENDEP (A4)
2 NAME
2 SALARY (1)

END-DEFINE
*
WRITE TITLE '** PERSONS WITH SALARY > 30000, SORTED BY DEPARTMENT **' /
LIMIT 9
READ MYVIEW BY DEPT FROM 'A' WHERE SALARY(1) > 30000

DISPLAY 'DEPT' DEPT NAME 'SALARY' SALARY(1)
/*
AT BREAK OF DEPT

SKIP 1
END-BREAK
AT BREAK OF DEPT /4/

WRITE COUNT(SALARY(1)) 'RECORDS FOUND IN:' OLD(#GENDEP) /
END-BREAK

END-READ
END

Output of Program ATBREX04:

            ** PERSONS WITH SALARY > 30000, SORTED BY DEPARTMENT **            
                                                                               
 DEPT          NAME           SALARY                                           
------ -------------------- ----------                                         
                                                                               
ADMA01 JENSEN                   180000                                         
ADMA01 PETERSEN                 105000                                         
ADMA01 MORTENSEN                320000                                         
ADMA01 MADSEN                   149000                                         
ADMA01 BUHL                     642000                                         
                                                                               
ADMA02 HERMANSEN                391500                                         
ADMA02 PLOUG                    162900                                         
ADMA02 HANSEN                   234000                                         
                                                                               
       8 RECORDS FOUND IN: ADMA                                                
                                                                               
COMP01 HEURTEBISE               168800                                         
                                                                               
       1 RECORDS FOUND IN: COMP                                               ↩

Automatic Break Processing

Automatic break processing is in effect for a processing loopwhich contains an AT BREAK statement.
This applies to the following statements:

■ FIND

■ READ

■ HISTOGRAM

■ SORT

■ READ WORK FILE

The value of the control field specified with the AT BREAK statement is checked only for records
which satisfy the selection criteria of both the WITH clause and the WHERE clause.

Natural system functions (AVER, MAX, MIN, etc.) are evaluated for each record after all statements
within the processing loop have been executed. System functions are not evaluated for any record
which is rejected by WHERE criteria.

The figure below illustrates the flow logic of automatic break processing.

Example of System Functions with AT BREAK Statement

The following example shows the use of the Natural system functions OLD, MIN, AVER, MAX, SUM and
COUNT in an AT BREAK statement (and of the system function TOTAL in an AT END OF DATA statement).

** Example 'ATBREX05': AT BREAK OF (with system functions)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 SALARY (1:1)
2 CURR-CODE (1:1)

END-DEFINE
*
LIMIT 3
READ MYVIEW BY CITY = 'SALT LAKE CITY'

DISPLAY NOTITLE CITY NAME 'SALARY' SALARY(1) 'CURRENCY' CURR-CODE(1)
/*
AT BREAK OF CITY

WRITE / OLD(CITY) (EM=X^X^X^X^X^X^X^X^X^X^X^X^X^X^X)
31T ' - MINIMUM:' MIN(SALARY(1)) CURR-CODE(1) /
31T ' - AVERAGE:' AVER(SALARY(1)) CURR-CODE(1) /
31T ' - MAXIMUM:' MAX(SALARY(1)) CURR-CODE(1) /
31T ' - SUM:' SUM(SALARY(1)) CURR-CODE(1) /
33T COUNT(SALARY(1)) 'RECORDS FOUND' /

END-BREAK
/*
AT END OF DATA

WRITE 22T 'TOTAL (ALL RECORDS):'
T*SALARY TOTAL(SALARY(1)) CURR-CODE(1)

END-ENDDATA
END-READ
END

Output of Program ATBREX05:

        CITY                 NAME           SALARY   CURRENCY
-------------------- -------------------- ---------- --------
                                                             
SALT LAKE CITY       ANDERSON                  50000 USD     
SALT LAKE CITY       SAMUELSON                 24000 USD     
                                                             
S A L T   L A K E   C I T Y    - MINIMUM:      24000 USD     
                               - AVERAGE:      37000 USD     
                               - MAXIMUM:      50000 USD     
                               -     SUM:      74000 USD     
                                       2 RECORDS FOUND       
                                                             
SAN DIEGO            GEE                       60000 USD     
                                                             
S A N   D I E G O              - MINIMUM:      60000 USD     
                               - AVERAGE:      60000 USD     
                               - MAXIMUM:      60000 USD     
                               -     SUM:      60000 USD     
                                       1 RECORDS FOUND       

                     TOTAL (ALL RECORDS):     134000 USD    ↩

Further Example of AT BREAK Statement

See the following example program:

■ ATBREX06 - AT BREAKOF (comparing NMIN, NAVER, NCOUNTwithMIN, AVER, COUNT)

BEFORE BREAK PROCESSING Statement

With the BEFORE BREAK PROCESSING statement, you can specify statements that are to be executed
immediately before a control break; that is, before the value of the control field is checked, before
the statements specified in the AT BREAK block are executed, and before any Natural system func-
tions are evaluated.

Example of BEFORE BREAK PROCESSING Statement

** Example 'BEFORX01': BEFORE BREAK PROCESSING
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 SALARY (1:1)
2 BONUS (1:1,1:1)

*
1 #INCOME (P11)
END-DEFINE
*
LIMIT 5
READ MYVIEW BY NAME FROM 'B'
BEFORE BREAK PROCESSING

COMPUTE #INCOME = SALARY(1) + BONUS(1,1)
END-BEFORE
/*
DISPLAY NOTITLE NAME FIRST-NAME (AL=10)

'ANNUAL/INCOME' #INCOME 'SALARY' SALARY(1) (LC==) /
'+ BONUS' BONUS(1,1) (IC=+)

AT BREAK OF #INCOME
WRITE T*#INCOME '-'(24)

END-BREAK

END-READ
END

Output of Program BEFORX01:

NAME FIRST-NAME ANNUAL SALARY
INCOME + BONUS

-------------------- ---------- ------------ -----------

BACHMANN HANS 56800 = 52800
+4000

------------------------
BAECKER JOHANNES 81000 = 74400

+6600
------------------------

BAECKER KARL 52650 = 48600
+4050

------------------------
BAGAZJA MARJAN 152700 = 129700

+23000
------------------------

BAILLET PATRICK 198500 = 188000
+10500

------------------------

User-InitiatedBreak Processing - PERFORMBREAKPROCESSINGStatement

With automatic break processing, the statements specified in an AT BREAK block are executed
whenever the value of the specified control field changes - regardless of the position of the AT
BREAK statement in the processing loop.

With a PERFORM BREAK PROCESSING statement, you can perform break processing at a specified
position in a processing loop: the PERFORM BREAK PROCESSING statement is executed when it is
encountered in the processing flow of the program.

Immediately after the PERFORM BREAK PROCESSING, you specify one or more AT BREAK statement
blocks:

...
PERFORM BREAK PROCESSING

AT BREAK OF field1
statements

END-BREAK
AT BREAK OF field2

statements
END-BREAK

...

When a PERFORM BREAK PROCESSING is executed, Natural checks if a break has occurred; that is,
if the value of the specified control field has changed; and if it has, the specified statements are
executed.

With PERFORM BREAK PROCESSING, system functions are evaluated beforeNatural checks if a break
has occurred.

The following figure illustrates the flow logic of user-initiated break processing:

Example of PERFORM BREAK PROCESSING Statement

** Example 'PERFBX01': PERFORM BREAK PROCESSING (with BREAK option
** in IF statement)
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 DEPT
2 SALARY (1:1)

*
1 #CNTL (N2)
END-DEFINE
*
LIMIT 7
READ MYVIEW BY DEPT

AT BREAK OF DEPT /* <- automatic break processing
SKIP 1
WRITE 'SUMMARY FOR ALL SALARIES '

'SUM:' SUM(SALARY(1))
'TOTAL:' TOTAL(SALARY(1))

ADD 1 TO #CNTL
END-BREAK
/*
IF SALARY (1) GREATER THAN 100000 OR BREAK #CNTL

PERFORM BREAK PROCESSING /* <- user-initiated break processing
AT BREAK OF #CNTL
WRITE 'SUMMARY FOR SALARY GREATER 100000'

'SUM:' SUM(SALARY(1))
'TOTAL:' TOTAL(SALARY(1))

END-BREAK
END-IF
/*
IF SALARY (1) GREATER THAN 150000 OR BREAK #CNTL

PERFORM BREAK PROCESSING /* <- user-initiated break processing
AT BREAK OF #CNTL
WRITE 'SUMMARY FOR SALARY GREATER 150000'

'SUM:' SUM(SALARY(1))
'TOTAL:' TOTAL(SALARY(1))

END-BREAK
END-IF
DISPLAY NAME DEPT SALARY(1)

END-READ
END

Output of Program PERFBX01:

Page      1                                                  04-12-14  14:13:35
                                                                               
        NAME         DEPARTMENT   ANNUAL                                       
                        CODE      SALARY                                       
-------------------- ---------- ----------                                     
                                                                               
JENSEN               ADMA01         180000                                     
PETERSEN             ADMA01         105000                                     
MORTENSEN            ADMA01         320000                                     
MADSEN               ADMA01         149000                                     
BUHL                 ADMA01         642000                                     
                                                                               
SUMMARY FOR ALL SALARIES          SUM:    1396000 TOTAL:    1396000            
SUMMARY FOR SALARY GREATER 100000 SUM:    1396000 TOTAL:    1396000            
SUMMARY FOR SALARY GREATER 150000 SUM:    1142000 TOTAL:    1142000            
HERMANSEN            ADMA02         391500                                     
PLOUG                ADMA02         162900                                     
                                                                               
SUMMARY FOR ALL SALARIES          SUM:     554400 TOTAL:    1950400            
SUMMARY FOR SALARY GREATER 100000 SUM:     554400 TOTAL:    1950400            
SUMMARY FOR SALARY GREATER 150000 SUM:     554400 TOTAL:    1696400           ↩

This chapter discusses arithmetic statements that are used for computing data:

■ COMPUTE

■ ADD

■ SUBTRACT

■ MULTIPLY

■ DIVIDE

In addition, the following statements are discussed which are used to transfer the value of an op-
erand into one or more fields:

■ MOVE

■ COMPRESS

Important: For optimum processing, user-defined variables used in arithmetic statements
should be defined with format P (packed numeric).

COMPUTE Statement

The COMPUTE statement is used to perform arithmetic operations. The following connecting oper-
ators are available:

Exponentiation**

Multiplication*

Division/

Addition+

Subtraction-

Parentheses may be used to indicate logical grouping.( )

Example 1:

COMPUTE LEAVE-DUE = LEAVE-DUE * 1.1

In this example, the value of the field LEAVE-DUE is multiplied by 1.1, and the result is placed in
the field LEAVE-DUE.

Example 2:

COMPUTE #A = SQRT (#B)

In this example, the square root of the value of the field #B is evaluated, and the result is assigned
to the field #A.

SQRT is a mathematical function supported in the following arithmetic statements:

■ COMPUTE

■ ADD

■ SUBTRACT

■ MULTIPLY

■ DIVIDE

For an overview of mathematical functions, seeMathematical Functions below.

Example 3:

COMPUTE #INCOME = BONUS (1,1) + SALARY (1)

In this example, the first bonus of the current year and the current salary amount are added and
assigned to the field #INCOME.

Statements MOVE and COMPUTE

The statements MOVE and COMPUTE can be used to transfer the value of an operand into one or more
fields. The operand may be a constant such as a text item or a number, a database field, a user-
defined variable, a system variable, or, in certain cases, a system function.

The difference between the two statements is that in the MOVE statement the value to be moved is
specified on the left; in the COMPUTE statement the value to be assigned is specified on the right, as
shown in the following examples.

Examples:

MOVE NAME TO #LAST-NAME
COMPUTE #LAST-NAME = NAME

Statements ADD, SUBTRACT, MULTIPLY and DIVIDE

The ADD, SUBTRACT, MULTIPLY and DIVIDE statements are used to perform arithmetic operations.

Examples:

ADD +5 -2 -1 GIVING #A
SUBTRACT 6 FROM 11 GIVING #B
MULTIPLY 3 BY 4 GIVING #C
DIVIDE 3 INTO #D GIVING #E

All four statements have a ROUNDED option,which you can use if youwish the result of the operation
to be rounded.

For rules on rounding, see Rules for Arithmetic Assignment.

The Statements documentation provides more detailed information on these statements.

Example of MOVE, SUBTRACT and COMPUTE Statements

The following program demonstrates the use of user-defined variables in arithmetic statements.
It calculates the ages and wages of three employees and outputs these.

** Example 'COMPUX01': COMPUTE
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 BIRTH
2 JOB-TITLE
2 SALARY (1:1)
2 BONUS (1:1,1:1)

*
1 #DATE (N8)
1 REDEFINE #DATE

2 #YEAR (N4)
2 #MONTH (N2)
2 #DAY (N2)

1 #BIRTH-YEAR (A4)
1 REDEFINE #BIRTH-YEAR

2 #BIRTH-YEAR-N (N4)
1 #AGE (N3)
1 #INCOME (P9)
END-DEFINE
*
MOVE *DATN TO #DATE
*
READ (3) MYVIEW BY NAME STARTING FROM 'JONES'

MOVE EDITED BIRTH (EM=YYYY) TO #BIRTH-YEAR
SUBTRACT #BIRTH-YEAR-N FROM #YEAR GIVING #AGE
/*
COMPUTE #INCOME = BONUS (1:1,1:1) + SALARY (1:1)
/*
DISPLAY NAME 'POSITION' JOB-TITLE #AGE #INCOME

END-READ
END

Output of Program COMPUX01:

Page 1 04-11-11 14:15:54

NAME POSITION #AGE #INCOME
-------------------- ------------------------- ---- ----------

JONES MANAGER 63 55000
JONES DIRECTOR 58 50000
JONES PROGRAMMER 48 31000

COMPRESS Statement

The COMPRESS statement is used to transfer (combine) the contents of two or more operands into
a single alphanumeric field.

Leading zeros in a numeric field and trailing blanks in an alphanumeric field are suppressed before
the field value is moved to the receiving field.

By default, the transferred values are separated fromone another by a single blank in the receiving
field. For other separating possibilities, see the COMPRESS statement option LEAVING NO SPACE (in
the Statements documentation).

Example:

COMPRESS 'NAME:' FIRST-NAME #LAST-NAME INTO #FULLNAME

In this example, a COMPRESS statement is used to combine a text constant ('NAME:'), a database
field (FIRST-NAME) and a user-defined variable (#LAST-NAME) into one user-defined variable
(#FULLNAME).

For further information on the COMPRESS statement, please refer to the COMPRESS statement descrip-
tion (in the Statements documentation).

Example of COMPRESS and MOVE Statements

The following program illustrates the use of the statements MOVE and COMPRESS.

** Example 'COMPRX01': COMPRESS
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 MIDDLE-I

*
1 #LAST-NAME (A15)
1 #FULL-NAME (A30)
END-DEFINE
*
READ (3) MYVIEW BY NAME STARTING FROM 'JONES'

MOVE NAME TO #LAST-NAME
/*
COMPRESS 'NAME:' FIRST-NAME MIDDLE-I #LAST-NAME INTO #FULL-NAME
/*
DISPLAY #FULL-NAME (UC==) FIRST-NAME 'I' MIDDLE-I (AL=1) NAME

END-READ
END

Output of Program COMPRX01:

Notice the output format of the compressed field.

Page 1 04-11-11 14:15:54

#FULL-NAME FIRST-NAME I NAME
============================== -------------------- - --------------------

NAME: VIRGINIA J JONES VIRGINIA J JONES
NAME: MARSHA JONES MARSHA JONES
NAME: ROBERT B JONES ROBERT B JONES

In multiple-line displays, it may be useful to combine fields/text in a user-defined variables by
using a COMPRESS statement.

Example of COMPRESS Statement

In the following program, three user-defined variables are used: #FULL-SALARY, #FULL-NAME, and
#FULL-CITY. #FULL-SALARY, for example, contains the text 'SALARY:' and the database fields SALARY
and CURR-CODE. The WRITE statement then references only the compressed variables.

** Example 'COMPRX02': COMPRESS
************************************************************************
DEFINE DATA LOCAL
1 VIEWEMP VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 SALARY (1:1)
2 CURR-CODE (1:1)
2 CITY
2 ADDRESS-LINE (1:1)
2 ZIP

*
1 #FULL-SALARY (A25)
1 #FULL-NAME (A25)
1 #FULL-CITY (A25)
END-DEFINE
*
READ (3) VIEWEMP BY CITY STARTING FROM 'NEW YORK'

COMPRESS 'SALARY:' CURR-CODE(1) SALARY(1) INTO #FULL-SALARY
COMPRESS FIRST-NAME NAME INTO #FULL-NAME
COMPRESS ZIP CITY INTO #FULL-CITY
/*
DISPLAY 'NAME AND ADDRESS' NAME (EM=X^X^X^X^X^X^X^X^X^X^X^X)
WRITE 1/5 #FULL-NAME

1/37 #FULL-SALARY
2/5 ADDRESS-LINE (1)
3/5 #FULL-CITY

SKIP 1
END-READ
END

Output of Program COMPRX02:

Page 1 04-11-11 14:15:54

NAME AND ADDRESS
-----------------------

R U B I N
SYLVIA RUBIN SALARY: USD 17000
2003 SARAZEN PLACE
10036 NEW YORK

W A L L A C E
MARY WALLACE SALARY: USD 38000
12248 LAUREL GLADE C
10036 NEW YORK

K E L L O G G
HENRIETTA KELLOGG SALARY: USD 52000
1001 JEFF RYAN DR.
19711 NEWARK

Mathematical Functions

The following Natural mathematical functions are supported in arithmetic processing statements
(ADD, COMPUTE, DIVIDE, SUBTRACT, MULTIPLY).

Natural System FunctionMathematical Function

ABS(field)Absolute value of field.

ATN(field)Arc tangent of field.

COS(field)Cosine of field.

EXP(field)Exponential of field.

FRAC(field)Fractional part of field.

INT(field)Integer part of field.

LOG(field)Natural logarithm of field.

SGN(field)Sign of field.

SIN(field)Sine of field.

SQRT(field)Square root of field.

TAN(field)Tangent of field.

VAL(field)Numeric value of an alphanumeric field.

See also the System Functions documentation for a detailed explanation of each mathematical
function.

Further Examples of COMPUTE, MOVE and COMPRESS Statements

See the following example programs:

■ WRITEX11 - WRITE (with nX, n/n and COMPRESS)
■ IFX03 - IF statement
■ COMPRX03 - COMPRESS (using parameters LC and TC)

This chapter describes the purpose of Natural system variables and Natural system functions and
how they are used in Natural programs.

System Variables

The following topics are covered below:

■ Purpose
■ Characteristics of System Variables
■ System Variables Grouped by Function

Purpose

System variables are used to display system information. They may be referenced at any point
within a Natural program.

Natural system variables provide variable information, for example, about the current Natural
session:

■ the current library;
■ the user and terminal identification;
■ the current status of a loop processing;
■ the current report processing status;
■ the current date and time.

The typical use of system variables is illustrated in the Example of System Variables and System
Functions below and in the examples contained in library SYSEXPG.

The information contained in a system variable may be used in Natural programs by specifying
the appropriate system variables. For example, date and time system variables may be specified
in a DISPLAY, WRITE, PRINT, MOVE or COMPUTE statement.

Characteristics of System Variables

The names of all system variables begin with an asterisk (*).

Format/Length

Information on format and length is given in the detailed descriptions in the System Variables
documentation. The following abbreviations are used:

Format

AlphanumericA

BinaryB

DateD

IntegerI

LogicalL

Numeric (unpacked)N

Packed numericP

TimeT

Content Modifiable

In the individual descriptions, this indicateswhether in aNatural programyou can assign another
value to the system variable, that is, overwrite its content as generated by Natural.

System Variables Grouped by Function

The Natural system variables are grouped as follows:

■ Application Related System Variables
■ Date and Time System Variables
■ Input/Ouput Related System Variables
■ Natural Environment Related System Variables
■ System Environment Related System Variables
■ XML Related System Variables

For detailed descriptions of all system variables, see the System Variables documentation.

System Functions

Natural system functions comprise a set of statistical andmathematical functions that can be applied
to the data after a record has been processed, but before break processing occurs.

System functions may be specified in a DISPLAY, WRITE, PRINT, MOVE or COMPUTE statement that is
used in conjunction with an AT END OF PAGE, AT END OF DATA or AT BREAK statement.

In the case of an AT END OF PAGE statement, the corresponding DISPLAY statement must include
the GIVE SYSTEM FUNCTIONS clause (as shown in the example below).

The following functional groups of system functions exist:

■ System Functions for Use in Processing Loops
■ Mathematical Functions
■ Miscellaneous Functions

For detailed information on all system functions available, see the System Functionsdocumentation.

See also Using System Functions in Processing Loops (in the System Functions documentation).

The typical use of system functions is explained in the example programs given below and in the
examples contained in library SYSEXPG.

Example of System Variables and System Functions

The following example program illustrates the use of system variables and system functions:

** Example 'SYSVAX01': System variables and system functions
************************************************************************
DEFINE DATA LOCAL
1 MYVIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 JOB-TITLE
2 INCOME (1:1)

3 CURR-CODE
3 SALARY
3 BONUS (1:1)

END-DEFINE
*
WRITE TITLE LEFT JUSTIFIED 'EMPLOYEE SALARY REPORT AS OF' *DAT4E /
*
READ (3) MYVIEW BY CITY STARTING FROM 'E'

DISPLAY GIVE SYSTEM FUNCTIONS
NAME (AL=15) JOB-TITLE (AL=15) INCOME (1:1)

AT START OF DATA
WRITE 'REPORT CREATED AT:' *TIME 'HOURS' /

END-START
AT END OF DATA

WRITE / 'LAST PERSON SELECTED:' OLD (NAME) /
END-ENDDATA

END-READ
*
AT END OF PAGE

WRITE 'AVERAGE SALARY:' AVER (SALARY(1))
END-ENDPAGE
END

Explanation:

■ The system variable *DATE is output with the WRITE TITLE statement.
■ The system variable *TIME is output with the AT START OF DATA statement.
■ The system function OLD is used in the AT END OF DATA statement.
■ The system function AVER is used in the AT END OF PAGE statement.

Output of Program SYSVAX01:

Note how the system variables and system function are displayed.

EMPLOYEE SALARY REPORT AS OF 11/11/2004

NAME CURRENT INCOME
POSITION

CURRENCY ANNUAL BONUS
CODE SALARY

--------------- --------------- -------- ---------- ----------

REPORT CREATED AT: 14:15:55.0 HOURS

DUYVERMAN PROGRAMMER USD 34000 0
PRATT SALES PERSON USD 38000 9000
MARKUSH TRAINEE USD 22000 0

LAST PERSON SELECTED: MARKUSH

AVERAGE SALARY: 31333

Further Examples of System Variables

See the following example programs:

■ EDITMX05 - Edit mask (EM for date and time system variables)
■ READX04 - READ (in combination with FIND and the system variables *NUMBER and
*COUNTER)

■ WTITLX01 - WRITE TITLE (with *PAGE-NUMBER)

Further Examples of System Functions

See the following example programs:

■ ATBREX06 - AT BREAKOF (comparing NMIN, NAVER, NCOUNTwithMIN, AVER, COUNT)
■ ATENPX01 -ATENDOFPAGE (with system function available viaGIVESYSTEMFUNCTIONS
in DISPLAY)

The Natural stack is a kind of “intermediate storage” in which you can store Natural commands,
user-defined commands, and input data to be used by an INPUT statement.

Use of Natural Stack

In the stack you can store a series of functions which are frequently executed one after the other,
such as a series of logon commands.

The data/commands stored in the stack are “stacked” on top of one another. You can decide
whether to put them on top or at the bottom of the stack. The data/command in the stack can only
be processed in the order in which they are stacked, beginning from the top of the stack.

In a program, you may reference the system variable *DATA to determine the content of the stack
(see the System Variables documentation for further information).

Stack Processing

The processing of the commands/data stored in the stack differs depending on the function being
performed.

If a command is expected, that is, the NEXT prompt is about to be displayed, Natural first checks
if a command is on the top of the stack. If there is, the NEXT prompt is suppressed and the command
is read and deleted from the stack; the command is then executed as if it had been enteredmanually
in response to the NEXT prompt.

If an INPUT statement containing input fields is being executed, Natural first checks if there are
any input data on the top of the stack. If there are, these data are passed to the INPUT statement
(in delimiter mode); the data read from the stack must be format-compatible with the variables in
the INPUT statement; the data are then deleted from the stack. See also Processing Data from the
Natural Stack in the INPUT statement description.

If an INPUT statement was executed using data from the stack, and this INPUT statement is re-ex-
ecuted via a REINPUT statement, the INPUT statement screen will be re-executed displaying the
same data from the stack aswhen itwas executed originally.With the REINPUT statement, no further
data are read from the stack.

When a Natural program terminates normally, the stack is flushed beginning from the top until
either a command is on the top of the stack or the stack is cleared. When a Natural program is
terminated via the terminal command %% or with an error, the stack is cleared entirely.

Placing Data on the Stack

The following methods can be used to place data/commands on the stack:

■ STACK Parameter
■ STACK Statement
■ FETCH and RUN Statements

STACK Parameter

TheNatural profile parameter STACKmay be used to place data/commands on the stack. The STACK
parameter (described in the Parameter Reference) can be specified by the Natural administrator in
the Natural parameter module at the installation of Natural; or you can specify it as a dynamic
parameter when you invoke Natural.

When data/commands are to be placed on the stack via the STACK parameter, multiple commands
must be separated from one another by a semicolon (;). If a command is to be passed within a se-
quence of data or command elements, it must be preceded by a semicolon.

Data for multiple INPUT statements must be separated from one another by a colon (:). Data that
are to be read by a separate INPUT statement must be preceded by a colon. If a command is to be
stacked which requires parameters, no colon is to be placed between the command and the para-
meters.

Semicolon and colonmust not be usedwithin the input data themselves as theywill be interpreted
as separation characters.

STACK Statement

The STACK statement can be usedwithin a program to place data/commands in the stack. The data
elements specified in one STACK statement will be used for one INPUT statement, whichmeans that
if data formultiple INPUT statements are to be placed on the stack, multiple STACK statementsmust
be used.

Data may be placed on the stack either unformatted or formatted:

■ If unformatted data are read from the stack, the data string is interpreted in delimiter mode and
the characters specified with the session parameters IA (Input Assignment character) and ID
(Input Delimiter character) are processed as control characters for keyword assignment and
data separation.

■ If formatted data are placed on the stack, each content of a field will be separated and passed
to one input field in the corresponding INPUT statement. If the data to be placed on the stack
contains delimiter, control or DBCS characters, it should be placed formatted on the stack to
avoid unintentional interpretation of these characters.

See the Statements documentation for further information on the STACK statement.

FETCH and RUN Statements

The execution of a FETCH or RUN statement that contains parameters to be passed to the invoked
program will result in these parameters being placed on top of the stack.

Clearing the Stack

The contents of the stack can be deletedwith the RELEASE statement. See the Statements document-
ation for details on the RELEASE statement.

Note: When aNatural program is terminated via the terminal command %% orwith an error,
the stack is cleared entirely.

Programming Guide408

Stack



44 Processing of Date Information

■ Edit Masks for Date Fields and Date System Variables .......................................................................... 410
■ Default Edit Mask for Date - DTFORM Parameter ................................................................................ 410
■ Date Format for Alphanumeric Representation - DF Parameter ............................................................... 411
■ Date Format for Output - DFOUT Parameter ....................................................................................... 413
■ Date Format for Stack - DFSTACK Parameter ..................................................................................... 414
■ Year Sliding Window - YSLW Parameter ............................................................................................. 415
■ Combinations of DFSTACK and YSLW ............................................................................................... 417
■ Year Fixed Window ........................................................................................................................ 419
■ Date Format for Default Page Title - DFTITLE Parameter ...................................................................... 419

409



This chapter covers various aspects concerning the handling of date information inNatural applic-
ations.

Edit Masks for Date Fields and Date System Variables

If you wish the value of a date field to be output in a specific representation, you usually specify
an edit mask for the field. With an edit mask, you determine character by character what the
output is to look like.

If you wish to use the current date in a specific representation, you need not define a date field
and specify an edit mask for it; instead you can simply use a date system variable. Natural provides
various date system variables, which contain the current date in different representations. Some
of these representations contain a 2-digit year component, some a 4-digit year component.

Formore information and a list of all date system variables, see the SystemVariablesdocumentation.

Default Edit Mask for Date - DTFORM Parameter

The profile parameter DTFORM determines the default format used for dates as part of the default
title on Natural reports, for date constants and for date input.

This date format determines the sequence of the day, month and year components of a date, as
well as the delimiter characters to be used between these components.

Possible DTFORM settings are:

ExampleDate Format*Setting

2005-12-31yyyy-mm-ddDTFORM=I

31.12.2005dd.mm.yyyyDTFORM=G

31/12/2005dd/mm/yyyyDTFORM=E

12/31/2005mm/dd/yyyyDTFORM=U

* dd = day, mm = month, yyyy = year.

The DTFORM parameter can be set in the Natural parameter module/file or dynamically when
Natural is invoked. By default, DTFORM=I applies.

Programming Guide410

Processing of Date Information



Date Format for Alphanumeric Representation - DF Parameter

If an edit mask is specified, the representation of the field value is determined by the edit mask.
If no edit mask is specified, the representation of the field value is determined by the session
parameter DF in combination with the profile parameter DTFORM.

With the DF parameter, you can choose one of the following date representations:

8-byte representation with 2-digit year component and delimiters (yy-mm-dd).DF=S

8-byte representation with 4-digit year component without delimiters (yyyymmdd).DF=I

10-byte representation with 4-digit year component and delimiters (yyyy-mm-dd).DF=L

For each representation, the sequence of the day, month and year components, and the delimiter
characters used, are determined by the DTFORM parameter.

By default, DF=S applies (except for INPUT statements; see below).

The session parameter DF is evaluated at compilation.

It can be specified with the following statements:

■ FORMAT,
■ INPUT, DISPLAY, WRITE and PRINT at statement and element (field) level,
■ MOVE, COMPRESS, STACK, RUN and FETCH at element (field) level.

When specified in one of these statements, the DF parameter applies to the following:

Effect of DF parameterStatement

When the value of a date variable is output with one of these statements, the value
is converted to an alphanumeric representation before it is output. The DF parameter
determines which representation is used.

DISPLAY, WRITE,
PRINT

When the value of a date variable is transferred to an alphanumeric field with a MOVE
or COMPRESS statement, the value is converted to an alphanumeric representation
before it is transferred. The DF parameter determines which representation is used.

MOVE, COMPRESS

When the value of a date variable is placed on the stack, it is converted to alphanumeric
representation before it is placed on the stack. The DF parameter determines which
representation is used.

The same applies when a date variable is specified as a parameter in a FETCH or RUN
statement (as these parameters are also passed via the stack).

STACK, RUN, FETCH

Effect of DF parameterStatement

When a data variable is used in an INPUT statement, the DF parameter determines
how a value must be entered in the field.

However, when a date variable for which no DF parameter is specified is used in an
INPUT statement, the date can be entered either with a 2-digit year component and

INPUT

delimiters or with a 4-digit year component and no delimiters. In this case, too, the
sequence of the day, month and year components, and the delimiter characters to be
used, are determined by the DTFORM parameter.

Note: With DF=S, only 2 digits are provided for the year information; this means that if a
date value contained the century, this information would be lost during the conversion. To
retain the century information, you set DF=I or DF=L.

Examples of DF Parameter with WRITE Statements

These examples assume that DTFORM=G applies.

/* DF=S (default)
WRITE *DATX /* Output has this format: dd.mm.yy
END

FORMAT DF=I
WRITE *DATX /* Output has this format: ddmmyyyy
END

FORMAT DF=L
WRITE *DATX /* Output has this format: dd.mm.yyyy
END

Example of DF Parameter with MOVE Statement

This example assumes that DTFORM=E applies.

DEFINE DATA LOCAL
1 #DATE (D) INIT <D'31/12/2005'>
1 #ALPHA (A10)

END-DEFINE
...
MOVE #DATE TO #ALPHA /* Result: #ALPHA contains 31/12/05
MOVE #DATE (DF=I) TO #ALPHA /* Result: #ALPHA contains 31122005
MOVE #DATE (DF=L) TO #ALPHA /* Result: #ALPHA contains 31/12/2005
...

Example of DF Parameter with STACK Statement

This example assumes that DTFORM=I applies.

DEFINE DATA LOCAL
1 #DATE (D) INIT <D'2005-12-31'>
1 #ALPHA1(A10)
1 #ALPHA2(A10)
1 #ALPHA3(A10)

END-DEFINE
...
STACK TOP DATA #DATE (DF=S) #DATE (DF=I) #DATE (DF=L)
...
INPUT #ALPHA1 #ALPHA2 #ALPHA3
...
/* Result: #ALPHA1 contains 05-12-31
/* #ALPHA2 contains 20051231
/* #ALPHA3 contains 2005-12-31
...

Example of DF Parameter with INPUT Statement

This example assumes that DTFORM=I applies.

DEFINE DATA LOCAL
1 #DATE1 (D)
1 #DATE2 (D)
1 #DATE3 (D)
1 #DATE4 (D)

END-DEFINE
...
INPUT #DATE1 (DF=S) /* Input must have this format: yy-mm-dd

#DATE2 (DF=I) /* Input must have this format: yyyymmdd
#DATE3 (DF=L) /* Input must have this format: yyyy-mm-dd
#DATE4 /* Input must have this format: yy-mm-dd or yyyymmdd

...

Date Format for Output - DFOUT Parameter

The session/profile parameter DFOUT only applies to date fields in INPUT, DISPLAY, WRITE and PRINT
statements for which no edit mask is specified, and for which no DF parameter applies.

For date fields which are displayed by INPUT, DISPLAY, PRINT and WRITE statements and for which
neither an edit mask is specified nor a DF parameter applies, the profile/session parameter DFOUT
determines the format in which the field values are displayed.

Possible DFOUT settings are:

Date variables are displayed with a 2-digit year component, and delimiters as determined by
the DTFORM parameter (yy-mm-dd).

DFOUT=S

Date variables are displayed with a 4-digit year component and no delimiters (yyyymmdd).DFOUT=I

By default, DFOUT=S applies. For either DFOUT setting, the sequence of the day, month and year
components in the date values is determined by the DTFORM parameter.

The lengths of the date fields are not affected by the DFOUT setting, as either date value represent-
ation fits into an 8-byte field.

The DFOUT parameter can be set in the Natural parameter module/file, dynamically when Natural
is invoked, or at session level. It is evaluated at runtime.

Example:

This example assumes that DTFORM=I applies.

DEFINE DATA LOCAL
1 #DATE (D) INIT <D'2005-12-31'>
END-DEFINE
...
WRITE #DATE /* Output if DFOUT=S is set ...: 05-12-31

/* Output if DFOUT=I is set ...: 20051231
WRITE #DATE (DF=L) /* Output (regardless of DFOUT): 2005-12-31
...

Date Format for Stack - DFSTACK Parameter

The session/profile parameter DFSTACK only applies to date fields used in STACK, FETCH and RUN
statements for which no DF parameter has been specified.

The DFSTACK parameter determines the format in which the values of date variables are placed on
the stack via a STACK, RUN or FETCH statement.

Possible DFSTACK settings are:

Date variables are placed on the stack with a 2-digit year component, and delimiters as
determined by the profile parameter DTFORM (yy-mm-dd).

DFSTACK=S

Same as DFSTACK=S. However, a change in the century will be intercepted at runtime.DFSTACK=C

Date variables are placed on the stack with a 4-digit year component and no delimiters
(yyyymmdd).

DFSTACK=I

By default, DFSTACK=S applies. DFSTACK=Smeans that when a date value is placed on the stack, it
is placed there without the century information (which is lost). When the value is then read from
the stack and placed into another date variable, the century is either assumed to be the current

one or determined by the setting of the YSLW parameter (see below). Thismight lead to the century
being different from that of the original date value; however, Natural would not issue any error
in this case.

DFSTACK=Cworks the same as DFSTACK=S in that a date value is placed on the stack without the
century information. However, if the value is read from the stack and the resulting century is
different from that of the original date value (either because of the YSLW parameter, or the original
century not being the current one), Natural issues a runtime error.

Note: This runtime error is already issued at the timewhen the value is placed on the stack.

DFSTACK=I allows you to place a date value on the stack in a length of 8 bytes without losing the
century information.

The DFSTACK parameter can be set in the Natural parameter module/file, dynamically when Nat-
ural is invoked, or at session level . It is evaluated at runtime.

Example:

This example assumes that DTFORM=I and YSLW=0 apply.

DEFINE DATA LOCAL
1 #DATE (D) INIT <D'2005-12-31'>
1 #ALPHA1(A8)
1 #ALPHA2(A10)

END-DEFINE
...
STACK TOP DATA #DATE #DATE (DF=L)
...
INPUT #ALPHA1 #ALPHA2
...
/* Result if DFSTACK=S or =C is set: #ALPHA1 contains 05-12-31
/* Result if DFSTACK=I is set .....: #ALPHA1 contains 20051231
/* Result (regardless of DFSTACK) .: #ALPHA2 contains 2005-12-31
...

Year Sliding Window - YSLW Parameter

The profile parameter YSLW allows you determine the century of a 2-digit year value.

The YSLW parameter can be set in theNatural parametermodule/file or dynamicallywhenNatural
is invoked. It is evaluated at runtime when an alphanumeric date value with a 2-digit year com-
ponent is moved into a date variable. This applies to data values which are:

■ used with themathematical function VAL(field),
■ used with the IS(D) option in a logical condition,

■ read from the stack as input data, or
■ entered in an input field as input data.

The YSLW parameter determines the range of years covered by a so-called “year sliding window”.
The sliding-window mechanism assumes a date with a 2-digit year to be within a “window” of
100 years. Within these 100 years, every 2-digit year value can be uniquely related to a specific
century.

With the YSLW parameter, you determine howmany years in the past that 100-year range is to begin:
The YSLW value is subtracted from the current year to determine the first year of thewindow range.

Possible values of the YSLW parameter are 0 to 99. The default value is YSLW=0, which means that
no sliding-window mechanism is used; that is, a date with a 2-digit year is assumed to be in the
current century.

Example 1:

If the current year is 2005 and you specify YSLW=40, the sliding window will cover the years 1965
to 2064. A 2-digit year value nn from 65 to 99 is interpreted accordingly as 19nn, while a 2-digit
year value nn from 00 to 64 is interpreted as 20nn.

Example 2:

If the current year is 2005 and you specify YSLW=20, the sliding window will cover the years 1985
to 2084. A 2-digit year value nn from 85 to 99 is interpreted accordingly as 19nn, while a 2-digit
year value nn from 00 to 84 is interpreted as 20nn.

Combinations of DFSTACK and YSLW

The following examples illustrate the effects of using various combinations of the parameters
DFSTACK and YSLW.

Note: All these examples assume that DTFORM=I applies.

Example 1:

This example assumes the current year to be 2005, and that the parameter settings DFSTACK=S
(default) and YSLW=20 apply.

DEFINE DATA LOCAL
1 #DATE1 (D) INIT <D'1956-12-31'>
1 #DATE2 (D)

END-DEFINE
...
STACK TOP DATA #DATE1 /* century information is lost (year 56 is stacked)
...
INPUT #DATE2 /* year sliding window determines 56 to be 2056
...
/* Result: #DATE2 contains 2056-12-31

even if #DATE1 is set to <D'2156-12-31'>

In this case, the year sliding window is not set appropriately, so that the century information is
(inadvertently) changed.

Example 2:

This example assumes the current year to be 2005, and that the parameter settings DFSTACK=S
(default) and YSLW=60 apply.

DEFINE DATA LOCAL
1 #DATE1 (D) INIT <D'1956-12-31'>
1 #DATE2 (D)

END-DEFINE
...
STACK TOP DATA #DATE1 /* century information is lost (year 56 is stacked)
...
INPUT #DATE2 /* year sliding window determines 56 to be 1956
...
/* Result: #DATE2 contains 1956-12-31

even if #DATE1 is set to <D'2056-12-31'>

In this case, the year sliding window is set appropriately, so that the original century information
is correctly restored.

Example 3:

This example assumes the current year to be 2005, and that the parameter settings DFSTACK=C and
YSLW=0 (default) apply.

DEFINE DATA LOCAL
1 #DATE1 (D) INIT <D'1956-12-31'>
1 #DATE2 (D)

END-DEFINE
...
STACK TOP DATA #DATE1 /* century information is lost (year 56 is stacked)
...
INPUT #DATE2 /* 56 is assumed to be in current century -> 1956

...
/* Result: RUNTIME ERROR (UNINTENDED CENTURY CHANGE)

In this case, the century information is (inadvertently) changed.However, this change is intercepted
by the DFSTACK=C setting.

Example 4:

This example assumes the current year to be 2005, and that the parameter settings DFSTACK=C and
YSLW=60 (default) apply.

DEFINE DATA LOCAL
1 #DATE1 (D) INIT <D'2056-12-31'>
1 #DATE2 (D)

END-DEFINE
...
STACK TOP DATA #DATE1 /* century information is lost (year 56 is stacked)
...
INPUT #DATE2 /* year sliding window determines 56 to be 1956
...
/* Result: RUNTIME ERROR (UNINTENDED CENTURY CHANGE)

In this case, the century information is changed due to the year sliding window. However, this
change is intercepted by the DFSTACK=C setting.

Year Fixed Window

For information on this topic, see the description of the profile parameter YSLW.

Date Format for Default Page Title - DFTITLE Parameter

The session/profile parameter DFTITLE determines the format of the date in a default page title
(as output with a DISPLAY, WRITE or PRINT statement).

The date is output with a 2-digit year component and delimiters (yy-mm-dd).DFTITLE=S

The date is output with a 4-digit year component and delimiters (yyyy-mm-dd).DFTITLE=L

The date is output with a 4-digit year component and no delimiters (yyyymmdd).DFTITLE=I

For each of these output formats, the sequence of the day, month and year components, and the
delimiter characters used, are determined by the DTFORM parameter.

The DFTITLE parameter can be set in the Natural parameter module/file, dynamically when Nat-
ural is invoked, or at session level. It is evaluated at runtime.

Example:

This example assumes that DTFORM=I applies.

WRITE 'HELLO'
END
/*
/* Date in page title if DFTITLE=S is set ...: 05-10-31
/* Date in page title if DFTITLE=L is set ...: 2005-10-31
/* Date in page title if DFTITLE=I is set ...: 20051031

Note: The DFTITLE parameter has no effect on a user-defined page title as specified with a
WRITE TITLE statement.

In an INPUT, DISPLAY, WRITE, WRITE TITLE or WRITE TRAILER statement, you can use text notation
to define a text to be used in conjunction with such a statement.

Defining a Text to Be Used with a Statement - the 'text' Notation

The text to be used with the statement (for example, a prompting message) must be enclosed in
either apostrophes (') or quotationmarks ("). Do not confuse double apostrophes ('') with a quotation
mark (").

Text enclosed in quotation marks can be converted automatically from lower-case letters to upper
case. To switch off automatic conversion, change the settings in the editor profile.

The text itself may be 1 to 72 characters and must not be continued from one line to the next.

Text elements may be concatenated by using a hyphen.

Examples:

DEFINE DATA LOCAL
1 #A(A10)
END-DEFINE

INPUT 'Input XYZ' (CD=BL) #A
WRITE '=' #A
WRITE 'Write1 ' - 'Write2 ' - 'Write3' (CD=RE)
END

Using Apostrophes as Part of a Text String

The following applies, if Natural profile parameter TQMARK (Translate Quotation Marks) is set to
ON. This is the default setting.

If you want an apostrophe to be part of a text string that is enclosed in apostrophes, you must
write this as double apostrophes ('') or as a quotation mark ("). Either notation will be output as a
single apostrophe.

If you want an apostrophe to be part of a text string that is enclosed in quotation marks, you write
this as a single apostrophe.

Examples of Apostrophe:

#FIELDA = 'O''CONNOR'
#FIELDA = 'O"CONNOR'
#FIELDA = "O'CONNOR"

In all three cases, the result will be:

O'CONNOR

Using Quotation Marks as Part of a Text String

The following applies, if the Natural profile parameter TQ (Translate Quotation Marks) is set to
OFF. The default setting is TQ=ON.

If youwant a quotationmark to be part of a text string that is enclosed in single apostrophes, write
a quotation mark.

If you want a quotation mark to be part of a text string that is enclosed in quotation marks, write
double quotation marks ("").

Example of Quotation Mark:

#FIELDA = 'O"CONNOR'
#FIELDA = "O""CONNOR"

In both cases, the result will be:

O"CONNOR

Defining a Character to Be Displayed n Times before a Field Value - the 'c'(n)
Notation

If a single character is to be output several times as text, you use the following notation:

'c'(n)

As c you specify the character, and as n the number of times the character is to be generated. The
maximum value for n is 249.

Example:

WRITE '*'(3)

Instead of apostrophes before and after the character c you can also use quotation marks.

User comments are descriptions or explanatory notes added to or interspersed among the statements
of the source code. Such informationmay be particularly helpful in understanding andmaintenaing
source code that was written or edited by another programmer. Also, the characters marking the
beginning of a comment can be used to temporarily disable the function of a statement or several
source code lines for test purposes.

Using an Entire Source Code Line for Comments

If you wish to use an entire source-code line for a user comment, you enter one of the following
at the beginning of the line:

■ an asterisk and a blank (* ),
■ two asterisks (**), or
■ a slash and an asterisk (/*).

* USER COMMENT
** USER COMMENT
/* USER COMMENT

Example:

As can be seen from the following example, comment lines may also be used to provide for a clear
source code structure.

** Example 'LOGICX03': BREAK option in logical condition
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 BIRTH

*
1 #BIRTH (A8)
END-DEFINE
*
LIMIT 10
READ EMPLOY-VIEW BY BIRTH

MOVE EDITED BIRTH (EM=YYYYMMDD) TO #BIRTH
/*
IF BREAK OF #BIRTH /6/

NEWPAGE IF LESS THAN 5 LINES LEFT
WRITE / '-' (50) /

END-IF
/*
DISPLAY NOTITLE BIRTH (EM=YYYY-MM-DD) NAME FIRST-NAME

END-READ
END

Using the Latter Part of a Source Code Line for Comments

If you wish to use only the latter part of a source-code line for a user comment, you enter a blank,
a slash and an asterisk ( /*); the remainder of the line after this notation is thusmarked as a comment:

ADD 5 TO #A /* USER COMMENT

Example:

** Example 'LOGICX04': IS option as format/length check
************************************************************************
DEFINE DATA LOCAL
1 #FIELDA (A10) /* INPUT FIELD TO BE CHECKED
1 #FIELDB (N5) /* RECEIVING FIELD OF VAL FUNCTION
1 #DATE (A10) /* INPUT FIELD FOR DATE
END-DEFINE
*
INPUT #DATE #FIELDA
IF #DATE IS(D)

IF #FIELDA IS (N5)
COMPUTE #FIELDB = VAL(#FIELDA)
WRITE NOTITLE 'VAL FUNCTION OK' // '=' #FIELDA '=' #FIELDB

ELSE
REINPUT 'FIELD DOES NOT FIT INTO N5 FORMAT'

MARK *#FIELDA
END-IF

ELSE
REINPUT 'INPUT IS NOT IN DATE FORMAT (YY-MM-DD) '

MARK *#DATE
END-IF
*
END

This chapter describes purpose and use of logical condition criteria that can be used in the state-
ments FIND, READ, HISTOGRAM, ACCEPT/REJECT, IF, DECIDE FOR, REPEAT.

Introduction

The basic criterion is a relational expression. Multiple relational expressions may be combined
with logical operators (AND, OR) to form complex criteria.

Arithmetic expressions may also be used to form a relational expression.

Logical condition criteria can be used in the following statements:

UsageStatement

A WHERE clause containing logical condition criteria may be used to indicate criteria in
addition to the basic selection criteria as specified in the WITH clause. The logical condition

FIND

criteria specified with the WHERE clause are evaluated after the record has been selected
and read.

In a WITH clause, “basic search criteria” (as describedwith the FIND statement) are used,
but not logical condition criteria.

A WHERE clause containing logical condition criteria may be used to specify whether a
record that has just been read is to be processed. The logical condition criteria are
evaluated after the record has been read.

READ

A WHERE clause containing logical condition criteria may be used to specify whether the
value that has just been read is to be processed. The logical condition criteria are evaluated
after the value has been read.

HISTOGRAM

An IF clause may be used with an ACCEPT or REJECT statement to specify logical
condition criteria in addition to that specified when the record was selected/read with

ACCEPT/REJECT

a FIND, READ, or HISTOGRAM statement. The logical condition criteria are evaluated after
the record has been read and after record processing has started.

Logical condition criteria are used to control statement execution.IF

Logical condition criteria are used to control statement execution.DECIDE FOR

The UNTIL or WHILE clause of a REPEAT statement contain logical condition criteriawhich
determine when a processing loop is to be terminated.

REPEAT

Relational Expression

Syntax:

operand2operand1

EQ
=
EQUAL
EQUAL TO
NE
^=
<>
NOT =
NOT EQ
NOTEQUAL
NOT EQUAL
NOT EQUAL TO
LT
LESS THAN
<
GE
GREATER EQUAL
>=
NOT <
NOT LT
GT
GREATER THAN
>
LE
LESS EQUAL
<=
NOT >
NOT GT

Operand Definition Table:

Dynamic
Definition

Referencing
Permitted

Possible FormatsPossible StructureOperand

yesyesOGLTDBFIPNUAENASCoperand1

noyesOGLTDBFIPNUAENASCoperand2

For an explanation of the Operand Definition Table shown above, see Syntax Symbols and Operand
Definition Tables in the Statements documentation.

In the “Possible Structure” column of the table above, “E” stands for arithmetic expressions; that
is, any arithmetic expression may be specified as an operand within the relational expression. For

further information on arithmetic expressions, see arithmetic-expression in the COMPUTE statement
description.

Explanation of the comparison operators:

ExplanationComparison Operator

equal toEQ
=
EQUAL
EQUAL TO

not equal toNE
^=
<>
NOT =
NOT EQ
NOTEQUAL
NOT EQUAL
NOT EQUAL TO

less thanLT
LESS THAN
<

greater than or equal toGE
GREATER EQUAL
>=

not less thanNOT <
NOT LT

greater thanGT
GREATER THAN
>

less than or equal toLE
LESS EQUAL
<=

not greater thanNOT >
NOT GT

Examples of Relational Expressions:

IF NAME = 'SMITH'
IF LEAVE-DUE GT 40
IF NAME = #NAME

For information on comparing arrays in a relational expression, see Processing of Arrays.

Note: If a floating-point operand is used, comparison is performed in floating point.
Floating-point numbers as suchhave only a limitedprecision; therefore, rounding/truncation

errors cannot be precluded when numbers are converted to/from floating-point represent-
ation.

Arithmetic Expressions in Logical Conditions

The following example shows how arithmetic expressions can be used in logical conditions:

IF #A + 3 GT #B - 5 AND #C * 3 LE #A + #B

Handles in Logical Conditions

If the operands in a relation expression are handles, only EQUAL and NOT EQUAL operators may be
used.

SUBSTRING Option in Relational Expression

Syntax:

operand2

=

SUBSTRING
(operand1,operand3,operand4) SUBSTRING

(operand2,operand5,operand6)

EQ

operand1

EQUAL [TO]
<>
NE
NOT =
NOT EQ
NOT EQUAL
NOT EQUAL TO
<
LT
LESS THAN
<=
LE
LESS EQUAL
>
GT
GREATER THAN
>=
GE
GREATER EQUAL

Operand Definition Table:

Dynamic
Definition

Referencing
Permitted

Possible FormatsPossible StructureOperand

yesyesBUANASCoperand1

noyesBUANASCoperand2

noyesBIPNSCoperand3

noyesIPNSCoperand4

noyesIPNSCoperand5

noyesIPNSCoperand6

With the SUBSTRING option, you can compare a part of an alphanumeric, a binary or a Unicode
field. After the field name (operand1) you specify first the starting position (operand3) and then
the length (operand4) of the field portion to be compared.

Also, you can compare a field valuewith part of another field value. After the field name (operand2)
you specify first the starting position (operand5) and then the length (operand6) of the field portion
operand1 is to be compared with.

You can also combine both forms, that is, you can specify a SUBSTRING for both operand1 and
operand2.

Examples:

The following expression compares the 5th to 12th position inclusive of the value in field #Awith
the value of field #B:

SUBSTRING(#A,5,8) = #B

- where 5 is the starting position and 8 is the length.

The following expression compares the value of field #Awith the 3rd to 6th position inclusive of
the value in field #B:

#A = SUBSTRING(#B,3,4)

Note: If you omit operand3/operand5, the starting position is assumed to be 1. If you omit
operand4/operand6, the length is assumed to be from the starting position to the end of the
field.

Extended Relational Expression

Syntax:

operand2operand1
=
EQ
EQUAL [TO]

operand3OR
=
EQ
EQUAL [TO]

THRU operand4 [BUT NOT operand5 [THRU operand6]]

Operand Definition Table:

Dynamic
Definition

Referencing
Permitted

Possible FormatsPossible StructureOperand

noyesOGTDBFIPNUAEN*ASCoperand1

noyesOGTDBFIPNUAEN*ASCoperand2

noyesOGTDBFIPNUAEN*ASCoperand3

noyesOGTDBFIPNUAEN*ASCoperand4

noyesOGTDBFIPNUAEN*ASCoperand5

noyesOGTDBFIPNUAEN*ASCoperand6

* Mathematical functions and system variables are permitted. Break functions are not permitted.

operand3 can also be specified using a MASK or SCAN option; that is, it can be specified as:

MASK (mask-definition) [operand]

MASK operand

SCAN operand

For details on these options, see the sectionsMASK Option and SCAN Option.

Examples:

IF #A = 2 OR = 4 OR = 7
IF #A = 5 THRU 11 BUT NOT 7 THRU 8

Evaluation of a Logical Variable

Syntax:

operand1

This option is used in conjunction with a logical variable (format L). A logical variable may take
the value TRUE or FALSE. As operand1 you specify the name of the logical variable to be used.

Operand Definition Table:

Dynamic DefinitionReferencing PermittedPossible FormatsPossible StructureOperand

nonoLASCoperand1

Example of Logical Variable:

** Example 'LOGICX05': Logical variable in logical condition
************************************************************************
DEFINE DATA LOCAL
1 #SWITCH (L) INIT <true>
1 #INDEX (I1)
END-DEFINE
*
FOR #INDEX 1 5

WRITE NOTITLE #SWITCH (EM=FALSE/TRUE) 5X 'INDEX =' #INDEX
WRITE NOTITLE #SWITCH (EM=OFF/ON) 7X 'INDEX =' #INDEX
IF #SWITCH

MOVE FALSE TO #SWITCH
ELSE

MOVE TRUE TO #SWITCH
END-IF
/*
SKIP 1

END-FOR
END

Output of Program LOGICX05:

TRUE INDEX = 1
ON INDEX = 1

FALSE INDEX = 2
OFF INDEX = 2

TRUE INDEX = 3
ON INDEX = 3

FALSE INDEX = 4
OFF INDEX = 4

TRUE INDEX = 5
ON INDEX = 5

Fields Used within Logical Condition Criteria

Database fields and user-defined variables may be used to construct logical condition criteria. A
database field which is a multiple-value field or is contained in a periodic group can also be used.
If a range of values for a multiple-value field or a range of occurrences for a periodic group is
specified, the condition is true if the search value is found in any value/occurrence within the
specified range.

Each value used must be compatible with the field used on the opposite side of the expression.
Decimal notation may be specified only for values used with numeric fields, and the number of
decimal positions of the value must agree with the number of decimal positions defined for the
field.

If the operands are not of the same format, the second operand is converted to the format of the
first operand.

Note: A numeric constant without decimal point notation is stored with format I for the
values -2147483648 to +2147483647, seeNumeric Constants. Consequently the comparison
with such an integer constant as operand1 is performed by converting operand2 to a integer
value. This means that the digits after the decimal point of operand2 are not considered
due to truncation.

Example:

 IF 0 = 0.5    /* is true because 0.5 (operand2) is converted to 0 (format I of ↩
operand1)
 IF 0.0 = 0.5  /* is false 
 IF 0.5 = 0    /* is false 
 IF 0.5 = 0.0  /* is false ↩

The following table shows which operand formats can be used together in a logical condition:

operand2operand1

OHGHPNLFITDBn (n>=5)Bn (n=<4)UA

YYYYA

[2][2]YYU

YYYYYYYYYYBn (n=<4)

YYYYBn (n>=5)

YYYYYYYD

YYYYYYYT

YYYYYYYI

YYYYYYYF
L

YYYYYYYN

YYYYYYYP

YGH [1]

YOH [1]

Notes:

1. [1] where GH = GUI handle, OH = object handle.

2. [2] The binary value will be assumed to contain Unicode code points, and the comparison is
performed as for a comparison of two Unicode values. The length of the binary field must be
even.

If two values are compared as alphanumeric values, the shorter value is assumed to be extended
with trailing blanks in order to get the same length as the longer value.

If two values are compared as binary values, the shorter value is assumed to be extended with
leading binary zeroes in order to get the same length as the longer value.

If two values are compared asUnicode values, trailing blanks are removed from both values before
the ICU collation algorithm is used to compare the two resulting values. See also Logical Condition
Criteria in the Unicode and Code Page Support documentation.

Comparison Examples:

A1(A1) := 'A'            
A5(A5) := 'A    '        
B1(B1) := H'FF'          
B5(B5) := H'00000000FF'  
U1(U1) := UH'00E4'       
U2(U2) := UH'00610308'   
IF A1 = A5 THEN ...                 /* TRUE 
IF B1 = B5 THEN ...                 /* TRUE
IF U1 = U2 THEN ...                 /* TRUE ↩

If an array is compared with a scalar value, each element of the array will be compared with the
scalar value. The condition will be true if at least one of the array elements meets the condition
(OR operation).

If an array is comparedwith an array, each element in the array is comparedwith the corresponding
element of the other array. The result is true only if all element comparisons meet the condition
(AND operation).

See also Processing of Arrays.

Note: An Adabas phonetic descriptor cannot be used within a logical condition.

Examples of Logical Condition Criteria:

FIND EMPLOYEES-VIEW WITH CITY = 'BOSTON' WHERE SEX = 'M'
READ EMPLOYEES-VIEW BY NAME WHERE SEX = 'M'
ACCEPT IF LEAVE-DUE GT 45
IF #A GT #B THEN COMPUTE #C = #A + #B
REPEAT UNTIL #X = 500

Logical Operators in Complex Logical Expressions

Logical condition criteria may be combined using the Boolean operators AND, OR, and NOT. Paren-
theses may also be used to indicate logical grouping.

The operators are evaluated in the following order:

MeaningOperatorPriority

Parentheses( )1

NegationNOT2

AND operationAND3

OR operationOR4

The following logical-condition-criteriamay be combined by logical operators to form a
complex logical-expression:

■ Relational expressions
■ Extended relational expressions
■ MASK option
■ SCAN option
■ BREAK option

The syntax for a logical-expression is as follows:

logical-expression
OR

[NOT]
logical-condition-criterion
(logical-expression) AND

Examples of Logical Expressions:

FIND STAFF-VIEW WITH CITY = 'TOKYO'
WHERE BIRTH GT 19610101 AND SEX = 'F'

IF NOT (#CITY = 'A' THRU 'E')

For information on comparing arrays in a logical expression, see Processing of Arrays.

Note: If multiple logical-condition-criteria are connectedwith AND, the evaluation terminates
as soon as the first of these criteria is not true.

BREAK Option - Compare Current Value with Value of Previous Loop Pass

The BREAK option allows the current value or a portion of a value of a field to be compared with
the value contained in the same field in the previous pass through the processing loop.

Syntax:

BREAK [OF] operand1 [/n/]

Operand Definition Table:

Dynamic DefinitionReferencing
Permitted

Possible FormatsPossible StructureOperand

noyesLTDBFIPNUASoperand1

Syntax Element Description:

Specifies the control field which is to be checked. A specific occurrence of an array can also be
used as a control field.

operand1

The notation /n/may be used to indicate that only the first n positions (counting from left to
right) of the control field are to be checked for a change in value. This notation can only be
used with operands of format A, B, N, or P.

/n/

The result of the BREAK operation is true when a change in the specified positions of the field
occurs. The result of the BREAK operation is not true if an AT END OF DATA condition occurs.

Example:

In this example, a check is made for a different value in the first position of the field
FIRST-NAME.

BREAK FIRST-NAME /1/

Natural system functions (which are available with the AT BREAK statement) are not available
with this option.

Example of BREAK Option:

** Example 'LOGICX03': BREAK option in logical condition
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 BIRTH

*
1 #BIRTH (A8)
END-DEFINE
*
LIMIT 10
READ EMPLOY-VIEW BY BIRTH

MOVE EDITED BIRTH (EM=YYYYMMDD) TO #BIRTH
/*
IF BREAK OF #BIRTH /6/

NEWPAGE IF LESS THAN 5 LINES LEFT
WRITE / '-' (50) /

END-IF
/*
DISPLAY NOTITLE BIRTH (EM=YYYY-MM-DD) NAME FIRST-NAME

END-READ
END

Output of Program LOGICX03:

DATE NAME FIRST-NAME
OF

BIRTH
---------- -------------------- --------------------

1940-01-01 GARRET WILLIAM
1940-01-09 TAILOR ROBERT
1940-01-09 PIETSCH VENUS
1940-01-31 LYTTLETON BETTY

--------------------------------------------------

1940-02-02 WINTRICH MARIA
1940-02-13 KUNEY MARY
1940-02-14 KOLENCE MARSHA
1940-02-24 DILWORTH TOM

--------------------------------------------------

1940-03-03 DEKKER SYLVIA
1940-03-06 STEFFERUD BILL

IS Option - Check whether Content of Alphanumeric or Unicode Field can be
Converted

Syntax:

operand1 IS (format)

This option is used to check whether the content of an alphanumeric or Unicode field (operand1)
can be converted to a specific other format.

Operand Definition Table:

Dynamic DefinitionReferencing PermittedPossible FormatsPossible StructureOperand

noyesUANASCoperand1

The format for which the check is performed can be:

Numeric with length ll.ll.Nll.ll

Floating point with length ll.Fll

Date. The following date formats are possible: dd-mm-yy, dd-mm-yyyy, ddmmyyyy (dd = day,
mm = month, yy or yyyy = year). The sequence of the day, month and year components as well

D

as the characters between the components are determined by the profile parameter DTFORM
(which is described in the Parameter Reference).

Time (according to the default time display format).T

Packed numeric with length ll.ll.Pll.ll

Integer with length ll.Ill

When the check is performed, leading and trailing blanks in operand1will be ignored.

The IS option may, for example, be used to check the content of a field before the mathematical
function VAL (extract numeric value from an alphanumeric field) is used to ensure that it will not
result in a runtime error.

Note: The IS option cannot be used to check if the value of an alphanumeric field is in the
specified “format”, but if it can be converted to that “format”. To check if a value is in a
specific format, you can use the MASK option. For further information, seeMASK Option
Compared with IS Option and Checking Packed or Unpacked Numeric Data.

Example of IS Option:

** Example 'LOGICX04': IS option as format/length check
************************************************************************
DEFINE DATA LOCAL
1 #FIELDA (A10) /* INPUT FIELD TO BE CHECKED
1 #FIELDB (N5) /* RECEIVING FIELD OF VAL FUNCTION
1 #DATE (A10) /* INPUT FIELD FOR DATE
END-DEFINE
*
INPUT #DATE #FIELDA
IF #DATE IS(D)

IF #FIELDA IS (N5)
COMPUTE #FIELDB = VAL(#FIELDA)
WRITE NOTITLE 'VAL FUNCTION OK' // '=' #FIELDA '=' #FIELDB

ELSE
REINPUT 'FIELD DOES NOT FIT INTO N5 FORMAT'

MARK *#FIELDA
END-IF

ELSE
REINPUT 'INPUT IS NOT IN DATE FORMAT (YY-MM-DD) '

MARK *#DATE
END-IF
*
END

Output of Program LOGICX04:

#DATE 150487 #FIELDA

INPUT IS NOT IN DATE FORMAT (YY-MM-DD)

MASK Option - Check Selected Positions of a Field for Specific Content

With the MASK option, you can check selected positions of a field for specific content.

The following topics are covered below:

■ Constant Mask
■ Variable Mask
■ Characters in a Mask
■ Mask Length
■ Checking Dates
■ Checking Against the Content of Constants or Variables
■ Range Checks
■ Checking Packed or Unpacked Numeric Data

Constant Mask

Syntax:

MASK (mask-definition) [operand2]operand1

=
EQ
EQUAL TO
NE
NOT EQUAL

Operand Definition Table:

Dynamic DefinitionReferencing
Permitted

Possible FormatsPossible StructureOperand

noyesPNUANASCoperand1

noyesBPNUASCoperand2

operand2 can only be used if the mask-definition contains at least one X. operand1 and operand2
must be format-compatible:

■ If operand1 is of format A, operand2must be of format A, B, N or U.
■ If operand1 is of format U, operand2must be of format A, B, N or U.
■ If operand1 is of format N or P, operand2must be of format N or P.

An X in the mask-definition selects the corresponding positions of the content of operand1 and
operand2 for comparison.

Variable Mask

Apart from a constant mask-definition (see above), you may also specify a variable mask defin-
ition.

Syntax:

MASK operand2operand1

=
EQ
EQUAL TO
NE
NOT EQUAL

Operand Definition Table:

Dynamic DefinitionReferencing
Permitted

Possible FormatsPossible StructureOperand

noyesPNUANASCoperand1

noyesUASoperand2

The content of operand2will be taken as the mask definition. Trailing blanks in operand2will be
ignored.

■ If operand1 is of format A, N or P, operand2must be of format A.
■ If operand1 is of format U, operand2must be of format U.

Characters in a Mask

The following characters may be used within a mask definition (the mask definition is contained
in mask-definition for a constant mask and operand2 for a variable mask):

MeaningCharacter

Aperiod, questionmark or underscore indicates a single position that is not to be checked.. or ? or _

An asterisk or percentmark is used to indicate any number of positions not to be checked.* or %

A slash (/) is used to check if a value endswith a specific character (or string of characters).

For example, the following condition will be true if there is either an E in the last position
of the field, or the last E in the field is followed by nothing but blanks:

/

IF #FIELD = MASK (*'E'/)

The position is to be checked for an alphabetical character (upper or lower case).A

One or more positions are to be checked for the characters bounded by apostrophes (a
double apostrophe indicates that a single apostrophe is the character to be checked for).
If operand1 is in Unicode format, 'c'must contain Unicode characters.

'c'

The position is to be checked for an alphabetical character (upper or lower case), a numeric
character, or a blank.

C

The two positions are to be checked for a valid day notation (01 - 31; dependent on the
values of MM and YY/YYYY, if specified; see also Checking Dates).

DD

The position is to be checked for hexadecimal content (A - F, 0 - 9).H

The positions are to be checked for a valid Julian Day; that is, the day number in the year
(001-366, dependent on the value of YY/YYYY, if specified. See also Checking Dates.)

JJJ

The position is to be checked for a lower-case alphabetical character (a - z).L

The positions are to be checked for a valid month (01 - 12); see also Checking Dates.MM

The position is to be checked for a numeric digit.N

One (or more) positions are to be checked for a numeric value in the range 0 - n.n...

The positions are checked for a numeric value in the range n1-n2.

n1 and n2 must be of the same length.

n1-n2 or n1:n2

The position is to be checked for a displayable character (U, L, N or S).P

The position is to be checked for special characters. See also Support of Different Character
Sets with NATCONV.INI in the Operations documentation.

S

The position is to be checked for an upper-case alphabetical character (A - Z).U

The position is to be checked against the equivalent position in the value (operand2)
following the mask-definition.

X is not allowed in a variable mask definition, as it makes no sense.

X

The two positions are to be checked for a valid year (00 - 99). See also Checking Dates.YY

The four positions are checked for a valid year (0000 - 2699).YYYY

MeaningCharacter

The position is to be checked for a character whose left half-byte is hexadecimally 3 or 7,
and whose right half-byte is hexadecimally 0 - 9.

Thismay be used to correctly check for numeric digits in negative numbers.With N (which
indicates a position to be checked for a numeric digit), a check for numeric digits in negative

Z

numbers leads to incorrect results, because the sign of the number is stored in the last
digit of the number, causing that digit to be hexadecimally represented as non-numeric.

Within a mask, use only one Z for each sequence of numeric digits that is checked.

Mask Length

The length of the mask determines how many positions are to be checked.

Example:

DEFINE DATA LOCAL
1 #CODE (A15)
END-DEFINE
...
IF #CODE = MASK (NN'ABC'....NN)
...

In the above example, the first two positions of #CODE are to be checked for numeric content. The
three following positions are checked for the contents ABC. The next four positions are not to be
checked. Positions ten and eleven are to be checked for numeric content. Positions twelve to fifteen
are not to be checked.

Checking Dates

Only one date may be checked within a given mask. When the same date component (JJJ, DD, MM,
YY or YYYY) is specifiedmore than once in themask, only the value of the last occurrence is checked
for consistency with other date components.

When dates are checked for a day (DD) and no month (MM) is specified in the mask, the current
month will be assumed.

When dates are checked for a day (DD) or a Julian day (JJJ) and no year (YY or YYYY) is specified
in the mask, the current year will be assumed.

When dates are checked for a 2-digit year (YY), the current century will be assumed if no Sliding
or FixedWindow is set. Formore details about Sliding or FixedWindows, refer to profile parameter
YSLW in the Parameter Reference.

Example 1:

MOVE 1131 TO #DATE (N4)
IF #DATE = MASK (MMDD)

In this example,month andday are checked for validity. The value formonth (11)will be considered
valid, whereas the value for day (31) will be invalid since the 11th month has only 30 days.

Example 2:

IF #DATE(A8) = MASK (MM'/'DD'/'YY)

In this example, the content of the field #DATE is be checked for a valid date with the format
MM/DD/YY (month/day/year).

Example 3:

IF #DATE (A8) = MASK (1950-2020MMDD)

In this example, the content of field #DATE is checked for a four-digit number in the range 1950 to
2020 followed by a valid month and day in the current year.

Note: Although apparent, the above mask does not allow to check for a valid date in the
years 1950 through 2020, because the numeric value range 1950-2020 is checked independent
of the validation of month and day. The check will deliver the intended results except for
February, 29, where the result depends on whether the current year is a leap year or not.
To check for a specific year range in addition to the date validation, code one check for the
date validation and another for the range validation:

IF #DATE (A8) = MASK (YYYYMMDD) AND #DATE = MASK (1950-2020)

Example 4:

IF #DATE (A4) = MASK (19-20YY)

In this example, the content of field #DATE is checked for a two-digit number in the range 19 to 20
followed by a valid two-digit year (00 through 99). The century is supplied byNatural as described
above.

Note: Although apparent, the above mask does not allow to check for a valid year in the
range 1900 through 2099, because the numeric value range 19-20 is checked independent
of the year validation. To check for year ranges, code one check for the date validation and
another for the range validation:

IF #DATE (A10) = MASK (YYYY'-'MM'-'DD) AND #DATE = MASK (19-20)

Checking Against the Content of Constants or Variables

If the value for the mask check is to be taken from either a constant or a variable, this value
(operand2) must be specified immediately following the mask-definition.

operand2must be at least as long as the mask.

In themask, you indicate each position to be checkedwith an X, and each position not to be checked
with a period (.) or a question mark (?) or an underscore (_).

Example:

DEFINE DATA LOCAL
1 #NAME (A15)
END-DEFINE
...
IF #NAME = MASK (..XX) 'ABCD'
...

In the above example, it is checked whether the field #NAME contains CD in the third and fourth
positions. Positions one and two are not checked.

The length of themask determines howmany positions are to be checked. Themask is left-justified
against any field or constant used in the mask operation. The format of the field (or constant) on
the right side of the expression must be the same as the format of the field on the left side of the
expression.

If the field to be checked (operand1) is of format A, any constant used (operand2) must be enclosed
in apostrophes. If the field is numeric, the value used must be a numeric constant or the content
of a numeric database field or user-defined variable.

In either case, any characters/digits within the value specified whose positions do not match the
X indicator within the mask are ignored.

The result of the MASK operation is true when the indicated positions in both values are identical.

Example:

** Example 'LOGICX01': MASK option in logical condition
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
END-DEFINE
*
HISTOGRAM EMPLOY-VIEW CITY

IF CITY =

MASK (....XX) '....NN'

DISPLAY NOTITLE CITY *NUMBER
END-IF

END-HISTOGRAM
*
END

In the above example, the record will be accepted if the fifth and sixth positions of the field CITY
each contain the character N.

Range Checks

When performing range checks, the number of positions verified in the supplied variable is defined
by the precision of the value supplied in themask specification. For example, amask of (...193...)
will verify positions 4 to 6 for a three-digit number in the range 000 to 193.

Additional Examples of Mask Definitions:

■ In this example, each character of #NAME is checked for an alphabetical character:

IF #NAME (A10) = MASK (AAAAAAAAAA)

■ In this example, positions 4 to 6 of #NUMBER are checked for a numeric value:

IF #NUMBER (A6) = MASK (...NNN)

■ In this example, positions 4 to 6 of #VALUE are to be checked for the value 123:

IF #VALUE(A10) = MASK (...'123')

■ This examplewill check if #LICENSE contains a license numberwhich beginswith NY- andwhose
last five characters are identical to the last five positions of #VALUE:

DEFINE DATA LOCAL
1 #VALUE(A8)
1 #LICENSE(A8)
END-DEFINE
INPUT 'ENTER KNOWN POSITIONS OF LICENSE PLATE:' #VALUE
IF #LICENSE = MASK ('NY-'XXXXX) #VALUE

■ The following conditionwould bemet by any valuewhich contains NAT and AL nomatter which
and howmany other characters are between NAT and AL (this would include the values NATURAL
and NATIONALITY as well as NATAL):

MASK('NAT'*'AL')

Checking Packed or Unpacked Numeric Data

Legacy applications often have packed or unpacked numeric variables redefined with alphanu-
meric or binary fields. Such redefinitions are not recommended, because using the packed or un-
packed variable in an assignment or computation may lead to errors or unpredictable results. To
validate the contents of such a redefined variable before the variable is used, use the N option (see
Characters in a Mask) as many as number of digits - 1 times followed by a single Z option.

Examples :

IF #P1 (P1) = MASK (Z)
IF #N4 (N4) = MASK (NNNZ)
IF #P5 (P5) = MASK (NNNNZ)

For further information about checking field contents, seeMASKOptionComparedwith ISOption.

MASK Option Compared with IS Option

This section points out the difference between MASK option and IS option and contains a sample
program to illustrate the difference.

The IS option can be used to check whether the content of an alphanumeric or Unicode field can
be converted to a specific other format, but it cannot be used to check if the value of an alphanu-
meric field is in the specified format.

The MASK option can be used to validate the contents of a redefined packed or unpacked numeric
variable.

Example Illustrating the Difference:

** Example 'LOGICX09': MASK versus IS option in logical condition       
************************************************************************
DEFINE DATA LOCAL                                                       
1 #A2   (A2)                                                            
1 REDEFINE #A2                                                          
  2 #N2 (N2)                                                            
1 REDEFINE #A2                                                          
  2 #P3 (P3)                                                            
1 #CONV-N2 (N2)                                                         
1 #CONV-P3 (P3)                                                         
END-DEFINE                                                              
*                                                                       
#A2 := '12'                                         
WRITE NOTITLE   'Assignment #A2 := "12" results in:'

PERFORM SUBTEST                                     
#A2 := '-1'                                         
WRITE NOTITLE / 'Assignment #A2 := "-1" results in:'
PERFORM SUBTEST                                     
#N2 :=  12                                          
WRITE NOTITLE / 'Assignment #N2 :=  12  results in:'
PERFORM SUBTEST                                     
#N2 :=  -1                                          
WRITE NOTITLE / 'Assignment #N2 :=  -1  results in:'
PERFORM SUBTEST                                     
#P3 :=  12                                          
WRITE NOTITLE / 'Assignment #P3 :=  12  results in:'
PERFORM SUBTEST                                     
#P3 :=  -1                                          
WRITE NOTITLE / 'Assignment #P3 :=  -1  results in:'
PERFORM SUBTEST                                     
*                                                                                 ↩
                                               
DEFINE SUBROUTINE SUBTEST                                       
IF #A2 IS (N2) THEN                                             
  #CONV-N2 := VAL(#A2)                                          
  WRITE NOTITLE 12T '#A2 can be converted to' #CONV-N2 '(N2)'   
END-IF                                                          
IF #A2 IS (P3) THEN                                             
  #CONV-P3 := VAL(#A2)                                          
  WRITE NOTITLE 12T '#A2 can be converted to' #CONV-P3 '(P3)'   
END-IF                                                          
IF #N2 = MASK(NZ) THEN                                          
  WRITE NOTITLE 12T '#N2 contains the valid unpacked number' #N2
END-IF                                                          
IF #P3 = MASK(NNZ) THEN                                         
  WRITE NOTITLE 12T '#P3 contains the valid packed number' #P3  
END-IF                                                          
END-SUBROUTINE                                                  
*                                                               
END                                                             ↩

Output of Program LOGICX09:

Assignment #A2 := '12' results in:
#A2 can be converted to 12 (N2)
#A2 can be converted to 12 (P3)
#N2 contains the valid unpacked number 12

Assignment #A2 := '-1' results in:
#A2 can be converted to -1 (N2)
#A2 can be converted to -1 (P3)

Assignment #N2 := 12 results in:
#A2 can be converted to 12 (N2)
#A2 can be converted to 12 (P3)
#N2 contains the valid unpacked number 12

Assignment #N2 := -1 results in:
#N2 contains the valid unpacked number -1

Assignment #P3 := 12 results in:
#P3 contains the valid packed number 12

Assignment #P3 := -1 results in:
#P3 contains the valid packed number -1

MODIFIED Option - Check whether Field Content has been Modified

Syntax:

operand1 [NOT] MODIFIED

This option is used to determine whether the content of a field has been modified during the exe-
cution of an INPUT or PROCESS PAGE statement. As a precondition, a control variable must have
been assigned using the parameter CV.

Operand Definition Table:

Dynamic DefinitionReferencing PermittedPossible FormatsPossible StructureOperand

nonoCASoperand1

Attribute control variables referenced in an INPUT or PROCESS PAGE statement are always assigned
the status “not modified” when the map is transmitted to the terminal.

Whenever the content of a field referencing an attribute control variable is modified, the attribute
control variable has been assigned the status “modified”.Whenmultiple fields reference the same
attribute control variable, the variable is marked “modified” if any of these fields is modified.

If operand1 is an array, the result will be true if at least one of the array elements has been assigned
the status “modified” (OR operation).

Example of MODIFIED Option:

** Example 'LOGICX06': MODIFIED option in logical condition
************************************************************************
DEFINE DATA LOCAL
1 #ATTR (C)
1 #A (A1)
1 #B (A1)
END-DEFINE
*
MOVE (AD=I) TO #ATTR

*
INPUT (CV=#ATTR) #A #B
IF #ATTR NOT MODIFIED

WRITE NOTITLE 'FIELD #A OR #B HAS NOT BEEN MODIFIED'
END-IF
*
IF #ATTR MODIFIED

WRITE NOTITLE 'FIELD #A OR #B HAS BEEN MODIFIED'
END-IF
*
END

Output of Program LOGICX06:

#A #B

After entering any value and pressing ENTER, the following output is displayed:

FIELD #A OR #B HAS BEEN MODIFIED

SCAN Option - Scan for a Value within a Field

Syntax:

SCANoperand1
operand2

=

(operand2)

EQ
EQUAL TO
NE
NOT EQUAL

Operand Definition Table:

Dynamic DefinitionReferencing
Permitted

Possible FormatsPossible StructureOperand

noyesPNUANASCoperand1

noyesB*UASCoperand2

* operand2may only be binary if operand1 is of format A or U. If operand1 is of format U and op-
erand2 is of format B, then the length of operand2must be even.

The SCAN option is used to scan for a specific value within a field.

The characters used in the SCAN option (operand2) may be specified as an alphanumeric or Unicode
constant (a character string bounded by apostrophes) or the contents of an alphanumeric orUnicode
database field or user-defined variable.

Caution: Trailing blanks are automatically eliminated from operand1 and operand2. Therefore,
the SCAN option cannot be used to scan for values containing trailing blanks. operand1 and
operand2may contain leading or embedded blanks. If operand2 consists of blanks only,
scanning will be assumed to be successful, regardless of the value of operand1; confer
EXAMINE FULL statement if trailing blanks are not to be ignored in the scan operation.

The field to be scanned (operand1) may be of format A, N, P or U. The SCAN operation may be
specified with the equal (EQ) or not equal (NE) operators.

The length of the character string for the SCAN operation should be less than the length of the field
to be scanned. If the length of the character string specified is identical to the length of the field
to be scanned, then an EQUAL operator should be used instead of SCAN.

Example of SCAN Option:

** Example 'LOGICX02': SCAN option in logical condition
************************************************************************
DEFINE DATA
LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
*
1 #VALUE (A4)
1 #COMMENT (A10)
END-DEFINE
*
INPUT 'ENTER SCAN VALUE:' #VALUE
LIMIT 15
*
HISTOGRAM EMPLOY-VIEW FOR NAME

RESET #COMMENT
IF NAME = SCAN #VALUE

MOVE 'MATCH' TO #COMMENT
END-IF
DISPLAY NOTITLE NAME *NUMBER #COMMENT

END-HISTOGRAM
*
END

Output of Program LOGICX02:

ENTER SCAN VALUE:  ↩

A scan for example for LL delivers three matches in 15 names:

NAME NMBR #COMMENT
-------------------- --------- ----------

ABELLAN 1 MATCH
ACHIESON 1
ADAM 1
ADKINSON 8
AECKERLE 1
AFANASSIEV 2
AHL 1
AKROYD 1
ALEMAN 1
ALESTIA 1
ALEXANDER 5
ALLEGRE 1 MATCH
ALLSOP 1 MATCH
ALTINOK 1
ALVAREZ 1

SPECIFIED Option - Check whether a Value is Passed for an Optional Para-
meter

Syntax:

parameter-name [NOT] SPECIFIED

This option is used to check whether an optional parameter in an invoked object (subprogram,
external subroutine or ActiveX control) has received a value from the invoking object or not.

An optional parameter is a field definedwith the keyword OPTIONAL in the DEFINE DATA PARAMETER
statement of the invoked object. If a field is defined as OPTIONAL, a value can - but need not - be
passed from an invoking object to this field.

In the invoking statement, the notation nX is used to indicate parameters for which no values are
passed.

If you process an optional parameter which has not received a value, this will cause a runtime
error. To avoid such an error, you use the SPECIFIED option in the invoked object to check
whether an optional parameter has received a value or not, and then only process it if it has.

parameter-name is the name of the parameter as specified in the DEFINE DATA PARAMETER statement
of the invoked object.

For a field not defined as OPTIONAL, the SPECIFIED condition is always TRUE.

Example of SPECIFIED Option:

Calling Programming:

** Example 'LOGICX07': SPECIFIED option in logical condition
************************************************************************
DEFINE DATA LOCAL
1 #PARM1 (A3)
1 #PARM3 (N2)
END-DEFINE
*
#PARM1 := 'ABC'
#PARM3 := 20
*
CALLNAT 'LOGICX08' #PARM1 1X #PARM3                                     
*                                                                       
END  ↩

Subprogram Called:

** Example 'LOGICX08': SPECIFIED option in logical condition
************************************************************************
DEFINE DATA PARAMETER
1 #PARM1 (A3)
1 #PARM2 (N2) OPTIONAL
1 #PARM3 (N2) OPTIONAL
END-DEFINE
*
WRITE '=' #PARM1
*
IF #PARM2 SPECIFIED

WRITE '#PARM2 is specified'
WRITE '=' #PARM2

ELSE
WRITE '#PARM2 is not specified'

* WRITE '=' #PARM2 /* would cause runtime error NAT1322
END-IF
*
IF #PARM3 NOT SPECIFIED
  WRITE '#PARM3 is not specified'
ELSE                             
  WRITE '#PARM3 is specified'    
  WRITE '=' #PARM3               
END-IF                           
END ↩

Output of Program LOGICX07:

Page 1 04-12-15 11:25:41

#PARM1: ABC
#PARM2 is not specified
#PARM3 is specified
#PARM3: 20

Field Initialization

Afield (user-defined variable or database field) which is to be used as an operand in an arithmetic
operation must be defined with one of the following formats:

Format

Numeric unpackedN

Packed numericP

IntegerI

Floating pointF

DateD

TimeT

Note: For reporting mode: A field which is to be used as an operand in an arithmetic oper-
ation must have been previously defined. A user-defined variable or database field used
as a result field in an arithmetic operation need not have been previously defined.

All user-defined variables and all database fields defined in a DEFINE DATA statement are initialized
to the appropriate zero or blank value when the program is invoked for execution.

Data Transfer

Data transfer is performedwith a MOVE or COMPUTE statement. The following table summarizes the
data transfer compatibility of the formats an operand may take.

Receiving Field FormatSending Field Format
OGFTDCLIBn (n>4)Bn (n<5)UAN or P

--YY---Y-[ 3 ][ 14 ][ 2 ]YN or P

--------[ 1 ][ 1 ][ 13 ]Y-A

--------[ 12 ][ 12 ]Y[ 11 ]-U

--YY---Y[ 5 ][ 5 ][ 14 ][ 2 ][ 4 ]Bn (n<5)

--------[ 5 ][ 5 ][ 15 ][ 6 ]-Bn (n>4)

--YY---Y-[ 3 ][ 14 ][ 2 ]YI

------Y---[ 16 ][ 9 ]-L

-----Y-------C

--Y[7]Y--Y-Y[ 16 ][ 9 ]YD

--YY[ 8 ]--Y-Y[ 16 ][ 9 ]YT

--YY---Y-[ 3 ][ 10 ] [ 16 ][ 9 ] [ 10 ]YF

-Y-----------G

Y------------O

Where:

Indicates data transfer compatibility.Y

Indicates data transfer incompatibility.-

Numbers in brackets [ ] refer to the corresponding rule for data transfer given below.[ ]

Data Conversion

The following rules apply to converting data values:

1. Alphanumeric to binary:
The value will be moved byte by byte from left to right. The result may be truncated or padded
with trailing blank characters depending on the length defined and the number of bytes specified.

2. (N,P,I) and binary (length 1-4) to alphanumeric:
The value will be converted to unpacked form and moved into the alphanumeric field left jus-
tified, that is, leading zeros will be suppressed and the field will be filled with trailing blank
characters. For negative numeric values, the sign will be converted to the hexadecimal notation
Dx. Any decimal point in the numeric value will be ignored. All digits before and after the
decimal point will be treated as one integer value.

3. (N,P,I,F) to binary (1-4 bytes):
The numeric value will be converted to binary (4 bytes). Any decimal point in the numeric
value will be ignored (the digits of the value before and after the decimal point will be treated
as an integer value). The resulting binary number will be positive or a two's complement of the
number depending on the sign of the value.

4. Binary (1-4 bytes) to numeric:
The value will be converted and assigned to the numeric value right justified, that is, with
leading zeros. (Binary values of the length 1-3 bytes are always assumed to have a positive sign.
For binary values of 4 bytes, the leftmost bit determines the sign of the number: 1=negative,
0=positive.) Any decimal point in the receiving numeric value will be ignored. All digits before
and after the decimal point will be treated as one integer value.

5. Binary to binary:
The value will be moved from right to left byte by byte. Leading binary zeros will be inserted
into the receiving field.

6. Binary (>4 bytes) to alphanumeric:
The value will be moved byte by byte from left to right. The result may be truncated or padded
with trailing blanks depending on the length defined and the number of bytes specified.

7. Date (D) to time (T):
If date is moved to time, it is converted to time assuming time 00:00:00:0.

8. Time (T) to date (D):
If time is moved to date, the time information is truncated, leaving only the date information.

9. L,D,T,F to A:
The values are converted to display form and are assigned left justified.

10. F:
If F is assigned to an alphanumeric or Unicode field which is too short, the mantissa is reduced
accordingly.

11.Unicode to alphanumeric:
The Unicode value will be converted to alphanumeric character codes according to the default
code page (value of the system variable *CODEPAGE) using the International Components for
Unicode (ICU) library. The result may be truncated or padded with trailing blank characters,
depending on the length defined and the number of bytes specified.

12.Unicode to binary:
The value will be moved code unit by code unit from left to right. The result may be truncated
or padded with trailing blank characters, depending on the length defined and the number of
bytes specified. The length of the receiving binary field must be even.

13.Alphanumeric to Unicode:
The alphanumeric value will be converted from the default code page to a Unicode value using
the International Components for Unicode (ICU) library. The resultmay be truncated or padded
with trailing blank characters, depending on the length defined and the number of code units
specified.

14. (N,P,I) and binary (length 1-4) to Unicode:
The value will be converted to unpacked form from which an alphanumeric value will be ob-
tained by suppression of leading zeros. For negative numeric values, the sign will be converted
to the hexadecimal notation Dx. Any decimal point in the numeric value will be ignored. All
digits before and after the decimal pointwill be treated as one integer value. The resulting value
will be converted from alphanumeric to Unicode. The result may be truncated or padded with
trailing blank characters, depending on the length defined and the number of codeunits specified.

15. Binary (>4 bytes) to Unicode:
The value will be moved byte by byte from left to right. The result may be truncated or padded
with trailing blanks, depending on the length defined and the number of bytes specified. The
length of the sending binary field must be even.

16. L,D,T,F to U:
The values are converted to an alphanumeric display form. The resulting valuewill be converted
from alphanumeric to Unicode and assigned left justified.

If source and target format are identical, the result may be truncated or paddedwith trailing blank
characters (format A and U) or leading binary zeros (format B) depending on the length defined
and the number of bytes (format A and B) or code units (format U) specified.

See also Using Dynamic Variables.

Field Truncation and Field Rounding

The following rules apply to field truncation and rounding:

■ High-order numeric field truncation is allowed only when the digits to be truncated are leading
zeros. Digits following an expressed or implied decimal point may be truncated.

■ Trailing positions of an alphanumeric field may be truncated.
■ If the option ROUNDED is specified, the last position of the result will be rounded up if the first
truncated decimal position of the value being assigned contains a value greater than or equal
to 5. For the result precision of a division, see alsoPrecision of Results of ArithmeticOperations.

Result Format and Length in Arithmetic Operations

The following table shows the format and length of the result of an arithmetic operation:

F8F4N or PI4I2I1

F8F4P*I4I2I1I1

F8F4P*I4I2I2I2

F8F4P*I4I4I4I4

F8F4P*P*P*P*N or P

F8F4F4F4F4F4F4

F8F8F8F8F8F8F8

On a mainframe computer, format/length F8 is used instead of F4 for improved precision of the
results of an arithmetic operation.

P* is determined from the integer length and precision of the operands individually for each oper-
ation, as shown under Precision of Results of Arithmetic Operations.

The following decimal integer lengths and possible values are applicable for format I:

Possible ValuesDecimal Integer LengthFormat/Length

-128 to 1273I1

-32768 to 327675I2

-2147483648 to 214748364710I4

Arithmetic Operations with Floating-Point Numbers

The following topics are covered below:

■ General Considerations
■ Precision of Floating-Point Numbers
■ Conversion to Floating-Point Representation
■ Platform Dependency

General Considerations

Floating-point numbers (format F) are represented as a sum of powers of two (as are integer
numbers (format I)), whereas unpacked and packed numbers (formats N and P) are represented
as a sum of powers of ten.

In unpacked or packed numbers, the position of the decimal point is fixed. In floating-point
numbers, however, the position of the decimal point (as the name indicates) is “floating”, that is,
its position is not fixed, but depends on the actual value.

Floating-point numbers are essential for the computing of trigonometric functions ormathematical
functions such as sinus or logarithm.

Precision of Floating-Point Numbers

Due to the nature of floating-point numbers, their precision is limited:

■ For a variable of format/length F4, the precision is limited to approximately 7 digits.
■ For a variable of format/length F8, the precision is limited to approximately 15 digits.

Valueswhich havemore significant digits cannot be represented exactly as a floating-point number.
Nomatter howmany additional digits there are before or after the decimal point, a floating-point
number can cover only the leading 7 or 15 digits respectively.

An integer value can only be represented exactly in a variable of format/length F4 if its absolute
value does not exceed 2 23 -1.

Conversion to Floating-Point Representation

When an alphanumeric, unpackednumeric or packednumeric value is converted to floating-point
format (for example, in an assignment operation), the representation has to be changed, that is, a
sum of powers of ten has to be converted to a sum of powers of two.

Consequently, only numbers that are representable as a finite sum of powers of two can be repres-
ented exactly; all other numbers can only be represented approximately.

Examples:

This number has an exact floating-point representation:

1.25 = 20 + 2-2

This number is a periodic floating-point number without an exact representation:

1.2 = 20 + 2-3 + 2-4 + 2-7 + 2-8 + 2-11 + 2-12 + ...

Thus, the conversion of alphanumeric, unpacked numeric or packed numeric values to floating-
point values, and vice versa, can introduce small errors.

Platform Dependency

Because of different hardware architecture, the representation of floating-point numbers varies
according to platforms. This explains why the same application, when run on different platforms,
may return slightly different results when floating-point arithmetics are involved. The respective
representation also determines the range of possible values for floating-point variables, which is
(approximately)

■ ±1.17 * 10-38 to ±3.40 * 1038 for F4 variables,
■ ±2.22 * 10-308 to ±1.79 * 10308 for F8 variables.

Note: The representation used by your pocket calculator may also be different from the
one used by your computer - which explains why results for the same computation may
differ.

Arithmetic Operations with Date and Time

With formats D (date) and T (time), only addition, subtraction, multiplication and division are
allowed.Multiplication and division are allowed on intermediate results of additions and subtrac-
tions only.

Date/time values can be added to/subtracted fromone another; or integer values (no decimal digits)
can be added to/subtracted from date/time values. Such integer values can be contained in fields
of formats N, P, I, D, or T.

The intermediate results of such an addition or subtraction may be used as a multiplicand or di-
vidend in a subsequent operation.

An integer value added to/subtracted from a date value is assumed to be in days. An integer value
added to/subtracted from a time value is assumed to be in tenths of seconds.

For arithmetic operations with date and time, certain restrictions apply, which are due to the
Natural's internal handling of arithmetic operations with date and time, as explained below.

Internally, Natural handles an arithmetic operation with date/time variables as follows:

COMPUTE result-field = operand1 +/- operand2

The above statement is resolved as:

1. intermediate-result = operand1 +/- operand2

2. result-field = intermediate-result

That is, in a first stepNatural computes the result of the addition/subtraction, and in a second step
assigns this result to the result field.

More complex arithmetic operations are resolved following the same pattern:

COMPUTE result-field = operand1 +/- operand2 +/- operand3 +/- operand4

The above statement is resolved as:

1. intermediate-result1 = operand1 +/- operand2

2. intermediate-result2 = intermediate-result1 +/- operand3

3. intermediate-result3 = intermediate-result2 +/- operand4

4. result-field = intermediate-result3

The resolution of multiplication and division operations is similar to the resolution for addition
and subtraction.

The internal format of such an intermediate result depends on the formats of the operands, as
shown in the tables below.

Addition

The following table shows the format of the intermediate result of an addition
(intermediate-result = operand1 + operand2):

Format of intermediate-resultFormat of operand2Format of operand1

DiDD

TTD

DDi, Ti, N, P, ID

TD, T, Di, Ti, N, P, IT

DDDi, Ti, N, P, I

TTDi, Ti, N, P, I

DiDiDi, N, P, I

TiTiTi, N, P, I

DiTi, N, P, IDi

TiDi, N, P, ITi

Subtraction

The following table shows the format of the intermediate result of a subtraction
(intermediate-result = operand1 - operand2):

Format of intermediate-resultFormat of operand2Format of operand1

DiDD

TiTD

DDi, Ti, N, P, ID

TiD, TT

TDi, Ti, N, P, IT

DiDDi, N, P, I

TiTDi, N, P, I

DiDi, Ti, N, P, IDi

TiD, T, Di, Ti, N, P, ITi

P12Di, TiN, P, I

Multiplication or Division

The following table shows the format of the intermediate result of a multiplication
(intermediate-result = operand1 * operand2) or division (intermediate-result = operand1
/ operand2):

Format of intermediate-resultFormat of operand2Format of operand1

DiD, Di, Ti, N, P, ID

TiTD

TiD, T, Di, Ti, N, P, IT

TiTDi

DiD, Di, Ti, N, P, IDi

DiDTi

TiDi, T, Ti, N, P, ITi

DiD, DiN, P, I

TiT, TiN, P, I

Internal Assignments

Di is a value in internal date format; Ti is a value in internal time format; such values can be used
in further arithmetic date/time operations, but they cannot be assigned to a result field of format
D (see the assignment table below).

In complex arithmetic operations in which an intermediate result of internal format Di or Ti is
used as operand in a further addition/subtraction/multiplication/division, its format is assumed
to be D or T respectively.

The following table shows which intermediate results can internally be assigned to which result
fields (result-field = intermediate-result).

Assignment possibleFormat of intermediate-resultFormat of result-field

yesD, TD

noDi, Ti, N, P, ID

yesD, T, Di, Ti, N, P, IT

yesD, T, Di, Ti, N, P, IN, P, I

A result field of format D or T must not contain a negative value.

Examples 1 and 2 (invalid):

COMPUTE DATE1 (D) = DATE2 (D) + DATE3 (D)
COMPUTE DATE1 (D) = DATE2 (D) - DATE3 (D)

These operations are not possible, because the intermediate result of the addition/subtraction
would be format Di, and a value of format Di cannot be assigned to a result field of format D.

Examples 3 and 4 (invalid):

COMPUTE DATE1 (D) = TIME2 (T) - TIME3 (T)
COMPUTE DATE1 (D) = DATE2 (D) - TIME3 (T)

These operations are not possible, because the intermediate result of the addition/subtraction
would be format Ti, and a value of format Ti cannot be assigned to a result field of format D.

Example 5 (valid):

COMPUTE DATE1 (D) = DATE2 (D) - DATE3 (D) + TIME3 (T)

This operation is possible. First, DATE3 is subtracted from DATE2, giving an intermediate result of
format Di; then, this intermediate result is added to TIME3, giving an intermediate result of format
T; finally, this second intermediate result is assigned to the result field DATE1.

Examples 6 and 7 (invalid):

COMPUTE DATE1 (D) = DATE2 (D) + DATE3 (D) * 2
COMPUTE TIME1 (T) = TIME2 (T) - TIME3 (T) / 3

These operations are not possible, because the attempted multiplication/division is performed
with date/time fields and not with intermediate results.

Example 8 (valid):

COMPUTE DATE1 (D) = DATE2 (D) + (DATE3(D) - DATE4 (D)) * 2

This operation is possible. First, DATE4 is subtracted from DATE3 giving an intermediate result of
format Di; then, this intermediate result is multiplied by two giving an intermediate result of
format Di; this intermediate result is added to DATE2 giving an intermediate result of format D;
finally, this third intermediate result is assigned to the result field DATE1.

If a format T value is assigned to a format D field, you must ensure that the time value contains a
valid date component.

Performance Considerations for Mixed Format Expressions

When doing arithmetic operations, the choice of field formats has considerable impact on perform-
ance:

For business arithmetic, only fields of format I (integer) should be used, if possible.

For scientific arithmetic, fields of format F (floating point) should be used, if possible.

In expressionswhere formats aremixed between numeric (N, P) and floating point (F), a conversion
to floating point format is performed. This conversion results in considerable CPU load. Therefore
it is recommended to avoid mixed format expressions in arithmetic operations.

Precision of Results of Arithmetic Operations

Digits After Decimal PointDigits Before Decimal PointOperation

Fd or Sd (whichever is greater)Fi + 1 or Si + 1 (whichever is greater)Addition/Subtraction

Fd + Sd (maximum 7)Fi + Si + 2Multiplication

(see below)Fi + SdDivision

Fd15 - Fd (See Exception below)Exponentiation

FdFiSquare Root

- where:

First operandF

Second operandS

ResultR

Digits before decimal pointi

Digits after decimal pointd

Exception:

If the exponent has one or more digits after the decimal point, the exponentiation is internally
carried out in floating point format and the result will also have floating point format. See Arith-
metic Operations with Floating-Point Numbers for further information.

Digits after Decimal Point for Division Results

The precision of the result of a division depends whether a result field is available or not:

■ If a result field is available, the precision is: Fd or Rd (whichever is greater) *.
■ If no result field is available, the precision is: Fd or Sd (whichever is greater) *.

* If the ROUNDED option is used, the precision of the result is internally increased by one digit before
the result is actually rounded.

A result field is available (or assumed to be available) in a COMPUTE and DIVIDE statement, and in
a logical condition in which the division is placed after the comparison operator (for example: IF
#A = #B / #C THEN ...).

A result field is not (or assumed to be not) available in a logical condition in which the division is
placed before the comparison operator (for example: IF #B / #C = #A THEN ...).

Exception:

If both dividend and divisor are of integer format and at least one of them is a variable, the division
result is always of integer format (regardless of the precision of the result field and of whether the
ROUNDED option is used or not).

Precision of Results for Arithmetic Expressions

The precision of arithmetic expressions, for example: #A / (#B * #C) + #D * (#E - #F + #G),
is derived by evaluating the results of the arithmetic operations in their processing order. For
further information on arithmetic expressions, see arithmetic-expression in the COMPUTE statement
description.

Error Conditions in Arithmetic Operations

In an addition, subtraction, multiplication or division, an error occurs if the total number of digits
(before and after the decimal point) of the result is greater than 31.

In an exponentiation, an error occurs in any of the following situations:

■ if the base is of packed format and either the result has over 16 digits or any intermediate result
has over 15 digits;

■ if the base is of floating-point format and the result is greater than approximately 7 * 1075.

Processing of Arrays

Generally, the following rules apply:

■ All scalar operations may be applied to array elements which consist of a single occurrence.
■ If a variable is defined with a constant value (for example, #FIELD (I2) CONSTANT <8>), the
valuewill be assigned to the variable at compilation, and the variablewill be treated as a constant.
This means that if such a variable is used in an array index, the dimension concerned has a def-
inite number of occurrences.

■ If an assignment/comparison operation involves two arrays with a different number of dimen-
sions, the “missing” dimension in the array with fewer dimensions is assumed to be (1:1).

Example: If #ARRAY1 (1:2) is assigned to #ARRAY2 (1:2,1:2), #ARRAY1 is assumed to be #ARRAY1
(1:1,1:2).

The following topics are covered below:

■ Definitions of Array Dimensions
■ Assignment Operations with Arrays
■ Comparison Operations with Arrays
■ Arithmetic Operations with Arrays

Definitions of Array Dimensions

The first, second and third dimensions of an array are defined as follows:

PropertiesNumber of Dimensions

#a3(3rd dim., 2nd dim., 1st dim.)3

#a2(2nd dim., 1st dim.)2

#a1(1st dim.)1

Assignment Operations with Arrays

If an array range is assigned to another array range, the assignment is performed element by ele-
ment.

Example:

DEFINE DATA LOCAL
1 #ARRAY(I4/1:5) INIT <10,20,30,40,50>
END-DEFINE
*
MOVE #ARRAY(2:4) TO #ARRAY(3:5)
/* is identical to
/* MOVE #ARRAY(2) TO #ARRAY(3)
/* MOVE #ARRAY(3) TO #ARRAY(4)
/* MOVE #ARRAY(4) TO #ARRAY(5)
/*
/* #ARRAY contains 10,20,20,20,20

If a single occurrence is assigned to an array range, each element of the range is filled with the
value of the single occurrence. (For a mathematical function, each element of the range is filled
with the result of the function.)

Before an assignment operation is executed, the individual dimensions of the arrays involved are
comparedwith one another to check if theymeet one of the conditions listed below. The dimensions
are compared independently of one another; that is, the 1st dimension of the one array is compared
with the 1st dimension of the other array, the 2nd dimension of the one array is compared with
the 2nd dimension of the other array, and the 3rd dimension of the one array is compared with
the 3rd dimension of the other array.

The assignment of values from one array to another is only allowed under one of the following
conditions:

■ The number of occurrences is the same for both dimensions compared.
■ The number of occurrences is indefinite for both dimensions compared.
■ The dimension that is assigned to another dimension consists of a single occurrence.

Example - Array Assignments:

The following program shows which array assignment operations are possible.

 DEFINE DATA LOCAL
 1 A1   (N1/1:8)
 1 B1   (N1/1:8)
 1 A2   (N1/1:8,1:8)
 1 B2   (N1/1:8,1:8)
 1 A3   (N1/1:8,1:8,1:8)
 1 I    (I2)          INIT <4>
 1 J    (I2)          INIT <8>
 1 K    (I2)          CONST <8>
 END-DEFINE
 *
 COMPUTE A1(1:3) = B1(6:8)                                      /* allowed
 COMPUTE A1(1:I) = B1(1:I)                                      /* allowed
 COMPUTE A1(*)   = B1(1:8)                                      /* allowed
 COMPUTE A1(2:3) = B1(I:I+1)                                    /* allowed

 COMPUTE A1(1)   = B1(I)                                        /* allowed
 COMPUTE A1(1:I) = B1(3)                                        /* allowed
 COMPUTE A1(I:J) = B1(I+2)                                      /* allowed
 COMPUTE A1(1:I) = B1(5:J)                                      /* allowed
 COMPUTE A1(1:I) = B1(2)                                        /* allowed
 COMPUTE A1(1:2) = B1(1:J)                                      /* NOT ALLOWED ↩
(NAT0631)
 COMPUTE A1(*)   = B1(1:J)                                      /* NOT ALLOWED ↩
(NAT0631)
 COMPUTE A1(*)   = B1(1:K)                                      /* allowed
 COMPUTE A1(1:J) = B1(1:K)                                      /* NOT ALLOWED ↩
(NAT0631)
 *
 COMPUTE A1(*)       = B2(1,*)                                  /* allowed
 COMPUTE A1(1:3)     = B2(1,I:I+2)                              /* allowed
 COMPUTE A1(1:3)     = B2(1:3,1)                                /* NOT ALLOWED ↩
(NAT0631)
 *
 COMPUTE A2(1,1:3)   = B1(6:8)                                  /* allowed
 COMPUTE A2(*,1:I)   = B1(5:J)                                  /* allowed
 COMPUTE A2(*,1)     = B1(*)                                    /* NOT ALLOWED ↩
(NAT0631)
 COMPUTE A2(1:I,1)   = B1(1:J)                                  /* NOT ALLOWED ↩
(NAT0631)
 COMPUTE A2(1:I,1:J) = B1(1:J)                                  /* allowed
 *
 COMPUTE A2(1,I)     = B2(1,1)                                  /* allowed
 COMPUTE A2(1:I,1)   = B2(1:I,2)                                /* allowed
 COMPUTE A2(1:2,1:8) = B2(I:I+1,*)                              /* allowed
 *
 COMPUTE A3(1,1,1:I)   = B1(1)                                  /* allowed
 COMPUTE A3(1,1,1:J)   = B1(*)                                  /* NOT ALLOWED ↩
(NAT0631)
  COMPUTE A3(1,1,1:I)   = B1(1:I)                               /* allowed
 COMPUTE A3(1,1:2,1:I) = B2(1,1:I)                              /* allowed
 COMPUTE A3(1,1,1:I)   = B2(1:2,1:I)                            /* NOT ALLOWED ↩
(NAT0631)
 END

Comparison Operations with Arrays

Generally, the following applies: if arrays with multiple dimensions are compared, the individual
dimensions are handled independently of one another; that is, the 1st dimension of the one array
is compared with the 1st dimension of the other array, the 2nd dimension of the one array is
compared with the 2nd dimension of the other array, and the 3rd dimension of the one array is
compared with the 3rd dimension of the other array.

The comparison of two array dimensions is only allowed under one of the following conditions:

■ The array dimensions compared with one another have the same number of occurrences.

■ The array dimensions compared with one another have an indefinite number of occurrences.
■ All array dimensions of one of the arrays involved are single occurrences.

Example - Array Comparisons:

The following program shows which array comparison operations are possible:

DEFINE DATA LOCAL
1 A3 (N1/1:8,1:8,1:8)
1 A2 (N1/1:8,1:8)

1 A1 (N1/1:8)
1 I (I2) INIT <4>
1 J (I2) INIT <8>
1 K (I2) CONST <8>
END-DEFINE
*
IF A2(1,1) = A1(1) THEN IGNORE END-IF /* allowed
IF A2(1,1) = A1(I) THEN IGNORE END-IF /* allowed
IF A2(1,*) = A1(1) THEN IGNORE END-IF /* allowed
IF A2(1,*) = A1(I) THEN IGNORE END-IF /* allowed
IF A2(1,*) = A1(*) THEN IGNORE END-IF /* allowed
IF A2(1,*) = A1(I -3:I+4) THEN IGNORE END-IF /* allowed
IF A2(1,5:J) = A1(1:I) THEN IGNORE END-IF /* allowed
IF A2(1,*) = A1(1:I) THEN IGNORE END-IF /* NOT ALLOWED(NAT0629)
IF A2(1,*) = A1(1:K) THEN IGNORE END-IF /* allowed
*
IF A2(1,1) = A2(1,1) THEN IGNORE END-IF /* allowed
IF A2(1,1) = A2(1,I) THEN IGNORE END-IF /* allowed
IF A2(1,*) = A2(1,1:8) THEN IGNORE END-IF /* allowed
IF A2(1,*) = A2(I,I -3:I+4) THEN IGNORE END-IF /* allowed
IF A2(1,1:I) = A2(1,I+1:J) THEN IGNORE END-IF /* allowed
IF A2(1,1:I) = A2(1,I:I+1) THEN IGNORE END-IF /* NOT ALLOWED(NAT0629)
IF A2(*,1) = A2(1,*) THEN IGNORE END-IF /* NOT ALLOWED(NAT0629)
IF A2(1,1:I) = A1(2,1:K) THEN IGNORE END-IF /* NOT ALLOWED(NAT0629)
*
IF A3(1,1,*) = A2(1,*) THEN IGNORE END-IF /* allowed
IF A3(1,1,*) = A2(1,I -3:I+4) THEN IGNORE END-IF /* allowed
IF A3(1,*,I:J) = A2(*,1:I+1) THEN IGNORE END-IF /* allowed
IF A3(1,*,I:J) = A2(*,I:J) THEN IGNORE END-IF /* allowed
END

When you compare two array ranges, note that the following two expressions lead to different
results:

#ARRAY1(*) NOT EQUAL #ARRAY2(*)
NOT #ARRAY1(*) = #ARRAY2(*)

Example:

■ Condition A:

IF #ARRAY1(1:2) NOT EQUAL #ARRAY2(1:2)

This is equivalent to:

IF (#ARRAY1(1) NOT EQUAL #ARRAY2(1)) AND (#ARRAY1(2) NOT EQUAL #ARRAY2(2))

ConditionA is therefore true if the first occurrence of #ARRAY1 does not equal the first occurrence
of #ARRAY2 and the second occurrence of #ARRAY1 does not equal the second occurrence of
#ARRAY2.

■ Condition B:

IF NOT #ARRAY1(1:2) = #ARRAY2(1:2)

This is equivalent to:

IF NOT (#ARRAY1(1)= #ARRAY2(1) AND #ARRAY1(2) = #ARRAY2(2))

This in turn is equivalent to:

IF (#ARRAY1(1) NOT EQUAL #ARRAY2(1)) OR (#ARRAY1(2) NOT EQUAL #ARRAY2(2))

Condition B is therefore true if either the first occurrence of #ARRAY1 does not equal the first oc-
currence of #ARRAY2 or the second occurrence of #ARRAY1 does not equal the second occurrence
of #ARRAY2.

Arithmetic Operations with Arrays

A general rule about arithmetic operations with arrays is that the number of occurrences of the
corresponding dimensions must be equal.

The following illustrates this rule:

#c(2:3,2:4) := #a(3:4,1:3) + #b(3:5)

In other words:

Programming Guide476

Rules for Arithmetic Assignment



RangeNumber of OccurrencesDimension NumberArray

2:322nd#c

2:431st#c

3:422nd#a

1:331st#a

3:531st#b

The operation is performed element by element.

Note: An arithmetic operation of a different number of dimensions is allowed.

For the example above, the following operations are executed:

#c(2,2) := #a(3,1) + #b(3)

#c(2,3) := #a(3,2) + #b(4)

#c(2,4) := #a(3,3) + #b(5)

#c(3,2) := #a(4,1) + #b(3)

#c(3,3) := #a(4,2) + #b(4)

#c(3,4) := #a(4,3) + #b(5)

Below is a list of examples of how array ranges may be used in the following ways in arithmetic
operations (in COMPUTE, ADD or MULTIPLY statements). In examples 1-4, the number of occurrences
of the corresponding dimensions must be equal.

1. range + range = range.

The addition is performed element by element.

2. range * range = range.

The multiplication is performed element by element.

3. scalar + range = range.

The scalar is added to each element of the range.

4. range * scalar = range.

Each element of the range is multiplied by the scalar.

5. range + scalar = scalar.

Each element of the range is added to the scalar and the result is assigned to the scalar.

6. scalar * range = scalar2.

The scalar is multiplied by each element of the array and the result is assigned to scalar2.

Since intermediate results will be generated for arithmetic operations as shown in the above ex-
amples, the result of overlapping index ranges is computed element by element in an intermediate
result array and finally the intermediate result array is assigned to the result field.

Example:

DEFINE DATA LOCAL
1 #ARRAY(I4/1:5) INIT <10,20,30,40,50>
END-DEFINE

#ARRAY(3:5) := #ARRAY(2:4) + 1

/* A temporary array for the
/* intermediate result values is
/* generated implicitly: #temp(1:3).
/* The following operations are
/* performed internally:
/* #temp(1) := #ARRAY(2) + 1
/* #temp(2) := #ARRAY(3) + 1
/* #temp(3) := #ARRAY(4) + 1
/* #ARRAY(3:5) := #temp(1:3)
/*
/* #ARRAY contains 10,20,21,31,41

Natural subprograms can be invoked from a programming object written in a 3rd generation
programming language (3GL). The invoking program can bewritten in anyprogramming language
that supports a standard CALL interface.

For this purpose, Natural provides the interface ncxr_callnat. The 3GL program invokes this
interface with a specification of the name of the desired subprogram.

Note: Natural must have been activated beforehand; that is, the invoking 3GL program
must in turn have been invoked by a Natural object with a CALL statement.

The subprogram is executed as if it had been invoked from another Natural object with a CALLNAT
statement.

When the processing of the subprogram stops (either with the END statement or with an ESCAPE
ROUTINE statement), control is returned to the 3GL program.

Passing Parameters from the 3GL Program to the Subprogram

Parameters can be passed from the invoking 3GLprogram to theNatural subprogram. For passing
parameters, the same rules apply as for passing parameters with a CALL statement.

The 3GL program invokes the Natural interface ncxr_callnatwith four parameters:

■ The 1st parameter is the name of the Natural subprogram to be invoked.
■ The 2nd parameter contains the number of parameters to be passed to the subprogram.
■ The 3rd parameter contains the address of the table that contains the addresses of the parameters
to be passed to the subprogram.

■ The 4th parameter contains the address of the table that contains the format/length specifications
of the parameters to be passed to the subprogram.

The sequence, format and length of the parameters in the invoking program must match exactly
the sequence, format and length of the fields in the DEFINE DATA PARAMETER statement of the
subprogram. The names of the fields in the invoking program and the invoked subprogram can
be different.

Example of Invoking a Natural Subprogram from a 3GL Program

For an example of how to invoke aNatural subprogram from a 3GLprogram, refer to the following
samples in NATDIR:[vnnnn.samples] (where nnnn is the current version number).

■ MY3GL.NSP (for the main program),
■ MY3GLSUB.NSN (for the subprogram),
■ MYC3GL.C (for the C function).

The Natural user exit SHCMD can be used to issue an operating system command, call a DCL com-
mand procedure or execute an executable program onOpenVMS fromwithin a Natural program.

Syntax

CALL 'SHCMD' 'command' ['option']

Parameters

Command to be executed under control of the operating system command interpreter. Natural
waits until the command is completed and then Natural returns control back to the Natural
program.

command

For more information, see Examples below.

option describes how the command should be executed. This parameter is optional. The
following options are available:

option

■ SCREENIO

■ NOSCREENIO

See Parameter Options below.

Parameter Options

The following parameter options are available:

DescriptionOption

This option is used to hide the output generated by the command. The hidden output is
redirected to the null device.

NOSCREENIO

This option is used to refresh the Natural screen output after the command is completed.SCREENIO

Note: The options SCREENIO and NOSCREENIOmay be not set at the same time.

Return Codes

Natural manipulates return codes from the control interpreter in the following way:

■ Even numbers are interpreted as error codes and are retained unchanged.
■ Odd numbers are interpreted as success codes and are converted to 0.

The following return code values are available:

DescriptionReturn Code

Command successfully executed.0

Illegal SHCMD parameter specified.4

Command has returned an error.All other codes

Examples

Execute a DCL command procedure from within Natural:

CALL 'SHCMD' '@MYDCL.COM'

Execute an executable program from within Natural:

CALL 'SHCMD' 'MYPROGRAM.EXE'

After executing the DIRECTORY command, you will recognize that the output generated by this
command has changed the last Natural screen output. You have to press the refresh-screen key
to clear the screen. To do this automatically, you can specify the SCREENIO option:

CALL 'SHCMD' 'DIRECTORY' 'SCREENIO'

Retrieve the return code by using the RET function:

DEFINE DATA LOCAL
1 rc (I4)

END-DEFINE
CALL 'SHCMD' '' 'SCREENIO'
ASSIGN rc = RET( 'SHCMD' ) /* retrieve return code
IF rc <> 0 THEN

IF rc = 4 THEN
WRITE NOTITLE 'Illegal option specified'

ELSE
WRITE NOTITLE 'Command not executed successfully (rc=' rc ')'

END-IF
ELSE

WRITE NOTITLE 'Command executed successfully'
END-IF
END

Execute theDCLcommandprocedureMYSUCCESS.COM.Naturalwill recognize that the command
procedure returns an odd return number and assumes success by mapping the exit code 15 to 0.

DCL command procedureMYSUCCESS.COM:

$ WRITE SYS$OUTPUT "DCL returns SUCCESS"
$ EXIT 15

Executing the DCL command procedureMYERROR.COM, Natural would recognize that the
command procedure returns an even return number, assumes an error and leaves the number
unchanged.

DCL command procedureMYERROR.COM:

$ WRITE SYS$OUTPUT "DCL returns ERROR"
$ EXIT 14

This chapter gives an overview of theNatural statements for internet and XML access and contains
a list of further references. To take full advantage of these statements, a thorough knowledge of
the underlying communication standards is required.

Statements Available

The following statements are available for internet and XML access:

■ REQUEST DOCUMENT
■ PARSE XML

REQUEST DOCUMENT

This statement enables you to use the HTTP/HTTPS protocol.

The following is an example of how this statement can be used to access an externally-located
document:

REQUEST DOCUMENT FROM
"http://bolsap1:5555/invoke/sap.demo/handle_RFC_XML_POST" 
WITH 
USER #User PASSWORD #Password
DATA
NAME 'XMLData' VALUE #Queryxml
NAME 'repServerName' VALUE 'NT2'
RETURN 
PAGE #Resultxml
RESPONSE #rc ↩

For further information, see REQUEST DOCUMENT in the Statements documentation.

PARSE XML

The PARSE XML statement allows you to parse XML documents from a Natural program.

For further information, see PARSE XML in the Statements documentation.

Further References

Below is a list of resources that you may find useful.

■ Sample Programs
■ Training Courses
■ Useful Links

Sample Programs

In addition to the sample programs provided at the end of the description of each statement, some
sample programs are included in the Natural library SYSEXV.

Training Courses

SoftwareAG's CorporateUniversity offers special training courses on this subject. See theCorporate
University offerings on Empower at https://empower.softwareag.com/.

Or, ask your local Software AG representative for the availability of special on-site training courses
at your location.

Useful Links

Below is a collection of links that may be of interest.

■ World Wide Web Consortium (W3C): https://www.w3.org/
■ Extensible Markup Language (XML): https://www.w3.org/XML/
■ HyperText Markup Language (HTML) Home Page: https://www.w3.org/MarkUp/
■ W3 Schools: https://www.w3schools.com/

As of Natural Version 5, Natural generated programs (GPs) are portable across UNIX, OpenVMS
and Windows platforms.

Compatibility

As ofNatural Version 5, a sourcewhichwas cataloged on anyNatural-supportedUNIX,OpenVMS
andWindows platform is executable with all of these Open Systems platforms without recompil-
ation. This feature simplifies the deployment of applications across Open Systems platforms.

Natural applications generated with Natural Version 4 or Natural Version 3 can be executed with
Natural Version 5 or above without cataloging the applications again (upward compatibility). In
this case, the portable GP functionality is not available. To make use of the portable GP and other
improvements, cataloging with Natural Version 5 or above is required.

Command processor GPs are not portable. The portable GP feature is not available for mainframe
platforms. This means that Natural GPs which are generated on mainframe computers are not
executable onUNIX, OpenVMS andWindows platformswithout recompilation of the application
and vice versa.

Endian Mode Considerations

AsofNatural Version 5,Natural acts as follows:Depending onwhichUNIX,OpenVMSorWindows
platform it is running, Natural will consider the byte order inwhichmulti-byte numbers are stored
in the GP. The two byte order modes are called “Little Endian” and “Big Endian”.

■ “Little Endian” means that the low-order byte of the number is stored in memory at the lowest
address, and the high-order byte at the highest address (the little end comes first).

■ “Big Endian” means that the high-order byte of the number is stored in memory at the lowest
address, and the low-order byte at the highest address (the big end comes first).

The UNIX, OpenVMS andWindows platforms use both endian modes: Intel processors and AXP
computers have “Little Endian” byte order, other processors such asHP-UX, Sun Solaris, or RS6000
use “Big Endian” mode.

Natural converts a portable GP automatically into the endian mode of the execution platform, if
necessary. This endian conversion is not performed if the GP has been generated in the endian
mode of the platform.

ENDIAN Parameter

In order to increase execution performance of portable GPs, the profile parameter ENDIAN has been
introduced. ENDIAN determines the endian mode in which a GP is generated during compilation:

The endian mode of the machine on which the GP is generated.DEFAULT

Big endian mode (high order byte first).BIG

Little endian mode (low order byte first).LITTLE

The values DEFAULT, BIG and LITTLE are alternatives whereby the default value is DEFAULT.

The ENDIANmode parameter may be set

■ as a profile parameter,
■ as a start-up parameter,
■ as a session parameter.

Portable FILEDIR.SAG and Error Message Files

As ofNatural Version 6.2, the file FILEDIR.SAG and the errormessage files are platform independ-
ent. Hence, it is possible to share common FUSER system files among different Open Systems
platforms. For example, it is possible to copy sets of Natural libraries from one Open Systems
platform to another with operating system copy procedures. However, it is not recommended to
share FNAT system files. For more information about the portable FILEDIR.SAG, refer to Portable
Natural System Files in the Operations documentation.

IX Designing Application User Interfaces

The user interface of an application, that is, the way an application presents itself to the user, is a
key consideration when writing an application.

This part provides information on the various possibilities Natural offers for designing character-
based user interfaces that are uniform in presentation and provide powerful mechanisms for user
guidance and interaction.

When designing user interfaces, standards and standardization are key factors.

Using Natural, you can offer the end user common functionality across various hardware and
operating systems.

This includes the general screen layout (information, data andmessage areas), function-key assign-
ment and the layout of windows.

This part covers the following topics:

Screen Design

Dialog Design

Control of Function-Key Lines - Terminal Command %Y

With the terminal command %Y you can define how and where the Natural function-key lines are
to be displayed.

Below is information on:

■ Format of Function-Key Lines
■ Positioning of Function-Key Lines
■ Cursor-Sensitivity

Format of Function-Key Lines

The following terminal commands are available for defining the format of function-key lines:

%YN
The function-key lines are displayed in tabular Software AG format:

Command ===>
Enter-PF1---PF2---PF3---PF4---PF5---PF6---PF7---PF8---PF9---PF10--PF11--PF12---

Help Exit Canc

%YS
The function-key lines display the keys sequentially and only show those keys towhich names
have been assigned (PF1=value, PF2=value, etc.):

  Command ===>                                                                 ↩
  
  PF1=Help,PF3=Exit,PF12=Canc

%YP
The function-key lines are displayed in PC-like format, that is, sequentially and only showing
those keys to which names have been assigned (F1=value, F2=value, etc.):

  Command ===>                                                                 ↩
  
  F1=Help,F3=Exit,F12=Canc

Other Display Options

Various other command options are available for function-key lines, such as:

■ single- and double-line display,
■ intensified display,
■ reverse video display,
■ color display.

For details on these options, see%Y - Control of PF-Key Lines in the Terminal Commands document-
ation.

Positioning of Function-Key Lines

%YB
The function-key lines are displayed at the bottom of the screen.

%YT
The function-key lines are displayed at the top of the screen.

%Ynn
The function-key lines are displayed on line nn of the screen.

Cursor-Sensitivity

%YC
This command makes the function-key lines cursor-sensitive. This means that they act like an
action bar on a PC screen: you just move the cursor to the desired function-key number or
name and press Enter, andNatural reacts as if the corresponding function key had been pressed.

To switch cursor-sensitivity off, you enter %YC again (toggle switch).

By using %YC in conjunction with tabular display format (%YN) and having only the function-
key names displayed (%YH), you can equip your applications with very comfortable action bar
processing: the user merely has to select a function name with the cursor and press Enter, and
the function is executed.

Control of the Message Line - Terminal Command %M

Various options of the terminal command %M are available for defining how andwhere theNatural
message line is to be displayed.

Below is information on:

■ Positioning the Message Line
■ Message Line Color

Positioning the Message Line

%MB
The message line is displayed at the bottom of the screen.

%MT
The message line is displayed at the top of the screen.

Other options for the positioning of the message line are described in%M - Control of Message Line
in the Terminal Commands documentation.

Message Line Color

%M=color-code
The message line is displayed in the specified color (for an explanation of color codes, see the
session parameter CD as described in the Parameter Reference).

Assigning Colors to Fields - Terminal Command %=

You can use the terminal command %= to assign colors to field attributes for programs that were
originally not written for color support. The command causes all fields/text defined with the spe-
cified attributes to be displayed in the specified color.

If predefined color assignments are not suitable for your terminal type, you can use this command
to override the original assignments with new ones.

You can also use the %= terminal command within Natural editors, for example to define color
assignments dynamically during map creation.

DescriptionCodes

Clear color translate table.blank

Newly defined colors are to override colors assigned by the program.F

Color attributes assigned by program are not to be modified.N

Output field.O

Modifiable field (output and input).M

Text constant.T

BlinkingB

ItalicC

DefaultD

IntensifiedI

UnderlinedU

Reverse videoV

BackgroundBG

BlueBL

GreenGR

NeutralNE

PinkPI

RedRE

TurquoiseTU

YellowYE

Example:

%=TI=RE,OB=YE

This example assigns the color red to all intensified text fields and yellow to all blinking output
fields.

Infoline - Terminal Command %X

The terminal command %X controls the display of the Natural infoline.

For further information, see the description of the terminal command %X in the Terminal Commands
documentation.

Windows

Below is information on:

■ What is a Window?
■ DEFINE WINDOW Statement
■ INPUT WINDOW Statement

What is a Window?

Awindow is that segment of a logical page, built by a program, which is displayed on the terminal
screen.

A logical page is the output area for Natural; in other words the logical page contains the current
report/map produced by the Natural program for display. This logical page may be larger than
the physical screen.

There is always awindowpresent, although youmay not be aware of its existence. Unless specified
differently (by a DEFINE WINDOW statement), the size of the window is identical to the physical size
of your terminal screen.

You can manipulate a window in two ways:

■ You can control the size and position of the window on the physical screen.
■ You can control the position of the window on the logical page.

Positioning on the Physical Screen

The figure below illustrates the positioning of a window on the physical screen. Note that the
same section of the logical page is displayed in both cases, only the position of the window on the
screen has changed.

Positioning on the Logical Page

The figure below illustrates the positioning of a window on the logical page.

When you change the position of thewindowon the logical page, the size andposition of thewindow
on the physical screenwill remain unchanged. In other words, the window is not moved over the
page, but the page is moved “underneath” the window.

DEFINE WINDOW Statement

You use the DEFINE WINDOW statement to specify the size, position and attributes of a window on
the physical screen.

A DEFINE WINDOW statement does not activate a window; this is donewith a SET WINDOW statement
or with the WINDOW clause of an INPUT statement.

Various options are available with the DEFINE WINDOW statement. These are described below in
the context of the following example.

The following program defines a window on the physical screen.

** Example 'WINDX01': DEFINE WINDOW
************************************************************************
DEFINE DATA LOCAL
1 COMMAND (A10)
END-DEFINE
*
DEFINE WINDOW TEST

SIZE 5*25
BASE 5/40
TITLE 'Sample Window'
CONTROL WINDOW
FRAMED POSITION SYMBOL BOTTOM LEFT

*
INPUT WINDOW='TEST' WITH TEXT 'message line'

COMMAND (AD=I'_') /
'dataline 1' /
'dataline 2' /
'dataline 3' 'long data line'

*
IF COMMAND = 'TEST2'

FETCH 'WINDX02'
ELSE

IF COMMAND = '.'
STOP

ELSE
REINPUT 'invalid command'

END-IF
END-IF
END

Thewindow-name identifies thewindow. The namemay be up to 32 characters long. For awindow
name, the same naming conventions apply as for user-defined variables. Here the name is TEST.

The window size is set with the SIZE option. Here the window is 5 lines high and 25 columns
(positions) wide.

The position of the window is set by the BASE option. Here the top left-hand corner of the window
is positioned on line 5, column 40.

With the TITLE option, you can define a title that is to be displayed in thewindow frame (of course,
only if you have defined a frame for the window).

With the CONTROL clause, you determine whether the PF-key lines, the message line and the stat-
istics line are displayed in the window or on the full physical screen. Here CONTROL WINDOW causes
the message line to be displayed inside the window. CONTROL SCREENwould cause the lines to be
displayed on the full physical screen outside the window. If you omit the CONTROL clause, CONTROL
WINDOW applies by default.

With the FRAMED option, you define that the window is to be framed. This frame is then cursor-
sensitive. Where applicable, you can page forward, backward, left or right within the window by
simply placing the cursor over the appropriate symbol (<, -, +, or >; see POSITION clause) and then
pressing Enter. In other words, you are moving the logical page underneath the window on the
physical screen. If no symbols are displayed, you can page backward and forward within the
window by placing the cursor in the top frame line (for backward positioning) or bottom frame
line (for forward positioning) and then pressing Enter.

With the POSITION clause of the FRAMED option, you define that information on the position of the
window on the logical page is to be displayed in the frame of the window. This applies only if the
logical page is larger than thewindow; if it is not, the POSITION clausewill be ignored. The position
information indicates in which directions the logical page extends above, below, to the left and to
the right of the current window.

507Programming Guide

Screen Design



If the POSITION clause is omitted, POSITION SYMBOL TOP RIGHT applies by default.

POSITION SYMBOL causes the position information to be displayed in form of symbols: “More: < -
+ >”. The information is displayed in the top and/or bottom frame line.

TOP/BOTTOM determines whether the position information is displayed in the top or bottom frame
line.

LEFT/RIGHT determines whether the position information is displayed in the left or right part of
the frame line.

You can define which characters are to be used for the frame with the terminal command %F=chv.

The first character will be used for the four corners of the window frame.c

The second character will be used for the horizontal frame lines.h

The third character will be used for the vertical frame lines.v

Example:

%F=+-!

The above command makes the window frame look like this:

+------------------------+
! !
! !
! !
! !
+------------------------+

INPUT WINDOW Statement

The INPUT WINDOW statement activates the window defined in the DEFINE WINDOW statement. In
the example, the window TEST is activated. Note that if you wish to output data in a window (for
example, with a WRITE statement), you use the SET WINDOW statement.

When the above program is run, thewindow is displayedwith one input field COMMAND. The session
parameter AD is used to define that the value of the field is displayed intensified and an underscore
is used as filler character.

Output of ProgramWINDX01:

> r > + Program WINDX01 Lib SYSEXPG
Top ....+....1....+....2....+....3....+....4....+....5....+....6....+....7..

0010 ** Example 'WINDX01': DEFINE WINDOW
0020 ******************************* +----Sample Window-----+ ***************
0030 DEFINE DATA LOCAL ! message line !
0040 1 COMMAND (A10) ! COMMAND __________ !
0050 END-DEFINE ! dataline 1 !
0060 * +More: + >---------+
0070 DEFINE WINDOW TEST
0080 SIZE 5*25
0090 BASE 5/40
0100 TITLE 'Sample Window'
0110 CONTROL WINDOW
0120 FRAMED POSITION SYMBOL BOTTOM LEFT
0130 *
0140 INPUT WINDOW='TEST' WITH TEXT 'message line'
0150 COMMAND (AD=I'_') /
0160 'dataline 1' /
0170 'dataline 2' /
0180 'dataline 3' 'long data line'
0190 *
0200 IF COMMAND = 'TEST2'

....+....1....+....2....+....3....+....4....+....5....+... S 29 L 1

In the bottom frame line, the position information More: + > indicates that there is more inform-
ation on the logical page than is displayed in the window.

To see the information that is further down on the logical page, you place the cursor in the bottom
frame line on the plus (+) sign and press Enter.

The window is now moved downwards. Note that the text long data line does not fit in the
window and is consequently not fully visible.

> r > + Program WINDX01 Lib SYSEXPG
Top ....+....1....+....2....+....3....+....4....+....5....+....6....+....7..

0010 ** Example 'WINDX01': DEFINE WINDOW
0020 ******************************* +----Sample Window-----+ ***************
0030 DEFINE DATA LOCAL ! message line !
0040 1 COMMAND (A10) ! dataline 3 long data !
0050 END-DEFINE ! dataline 2 !
0060 * +More: - >---------+
0070 DEFINE WINDOW TEST
0080 SIZE 5*25
0090 BASE 5/40
0100 TITLE 'Sample Window'
0110 CONTROL WINDOW
0120 FRAMED POSITION SYMBOL BOTTOM LEFT
0130 *
0140 INPUT WINDOW='TEST' WITH TEXT 'message line'
0150 COMMAND (AD=I'_') /
0160 'dataline 1' /
0170 'dataline 2' /
0180 'dataline 3' 'long data line'
0190 *
0200 IF COMMAND = 'TEST2'

....+....1....+....2....+....3....+....4....+....5....+... S 29 L 1

To see this hidden information to the right, you place the cursor in the bottom frame line on the
less-than symbol (>) and press Enter. The window is now moved to the right on the logical page
and displays the previously invisible word line:

> r > + Program WINDX01 Lib SYSEXPG
Top ....+....1....+....2....+....3....+....4....+....5....+....6....+....7..

0010 ** Example 'WINDX01': DEFINE WINDOW
0020 ******************************* +----Sample Window-----+ ***************
0030 DEFINE DATA LOCAL ! message line !
0040 1 COMMAND (A10) ! line ! <==
0050 END-DEFINE ! !
0060 * +More: < - ---------+
0070 DEFINE WINDOW TEST
0080 SIZE 5*25
0090 BASE 5/40
0100 TITLE 'Sample Window'
0110 CONTROL WINDOW
0120 FRAMED POSITION SYMBOL BOTTOM LEFT
0130 *
0140 INPUT WINDOW='TEST' WITH TEXT 'message line'
0150 COMMAND (AD=I'_') /
0160 'dataline 1' /
0170 'dataline 2' /
0180 'dataline 3' 'long data line'
0190 *
0200 IF COMMAND = 'TEST2'

....+....1....+....2....+....3....+....4....+....5....+... S 29 L 1

Multiple Windows

You can, of course, open multiple windows. However, only one Natural window is active at any
one time, that is, themost recentwindow.Any previouswindowsmay still be visible on the screen,
but are no longer active and are ignored by Natural. You may enter input only in the most recent
window. If there is not enough space to enter input, the window size must be adjusted first.

When TEST2 is entered in the COMMAND field, the program WINDX02 is executed.

** Example 'WINDX02': DEFINE WINDOW
************************************************************************
DEFINE DATA LOCAL
1 COMMAND (A10)
END-DEFINE
*
DEFINE WINDOW TEST2

SIZE 5*30
BASE 15/40
TITLE 'Another Window'
CONTROL SCREEN
FRAMED POSITION SYMBOL BOTTOM LEFT

*
INPUT WINDOW='TEST2' WITH TEXT 'message line'

COMMAND (AD=I'_') /
'dataline 1' /

'dataline 2' /
'dataline 3' 'long data line'

*
IF COMMAND = 'TEST'

FETCH 'WINDX01'
ELSE

IF COMMAND = '.'
STOP

ELSE
REINPUT 'invalid command'

END-IF
END-IF
END

A second window is opened. The other window is still visible, but it is inactive.

message line
> r > + Program WINDX01 Lib SYSEXPG
Top ....+....1....+....2....+....3....+....4....+....5....+....6....+....7..

0010 ** Example 'WINDX01': DEFINE WINDOW
0020 ******************************* +----Sample Window-----+ ***************
0030 DEFINE DATA LOCAL ! message line ! Inactive
0040 1 COMMAND (A10) ! COMMAND TEST2_____ ! Window
0050 END-DEFINE ! dataline 1 ! <==
0060 * +More: + >---------+
0070 DEFINE WINDOW TEST
0080 SIZE 5*25
0090 BASE 5/40
0100 TITLE 'Sample Window'
0110 CONTROL WINDOW
0120 FRAMED POSITION SYMBOL B +------Another Window-------+ Currently   ↩

      
  0130 *                               ! COMMAND __________ ! Active

0140 INPUT WINDOW='TEST' WITH TEXT ' ! dataline 1 ! Window
0150 COMMAND (AD=I'_') / ! dataline 2 ! <==
0160 'dataline 1' / +More: +------------------+           

  0170       'dataline 2' /                                                    
  0180       'dataline 3' 'long data line'                                     
  0190 *                                                                       
  0200 IF COMMAND = 'TEST2'                                                       ↩
     

Note that for the new window the message line is now displayed on the full physical screen (at
the top) and not in the window. This was defined by the CONTROL SCREEN clause in the WINDX02
program.

For further details on the statements DEFINE WINDOW, INPUT WINDOW and SET WINDOW, see the cor-
responding descriptions in the Statements documentation.

Standard/Dynamic Layout Maps

Standard Layout Maps

A standard layout can be defined in the map editor. This layout guarantees a uniform appearance
for all maps that reference it throughout the application.

When a map that references a standard layout is initialized, the standard layout becomes a fixed
part of the map. This means that if this standard layout is modified, all affected maps must be re-
cataloged before the changes take effect.

Dynamic Layout Maps

In contrast to a standard layout, a dynamic layout does not become a fixed part of a map that refer-
ences it, rather it is executed at runtime.

This means that if you define the layout map as “dynamic” on the Define Map Settings For MAP
screen in the map editor, any modifications to the layout map are also carried out on all maps that
reference it. The maps need not be re-cataloged.

For further details on layout maps, seeMap Editor in the Editors documentation.

Multilingual User Interfaces

Using Natural, you can create multilingual applications for international use.

Maps, helproutines, error messages, programs, functions, subprograms and copycodes can be
defined in up to 60 different languages (including languages with double-byte character sets).

Below is information on:

■ Language Codes
■ Defining the Language of a Natural Object
■ Defining the User Language
■ Referencing Multilingual Objects
■ Programs

■ Edit Masks for Date and Time Fields

Language Codes

In Natural, each language has a language code (from 1 to 60). The table below is an extract from the
full table of language codes. For a complete overview, refer to the description of the system variable
*LANGUAGE in the System Variables documentation.

Map Code in Object NamesLanguageLanguage Code

1English1

2German2

3French3

4Spanish4

5Italian5

6Dutch6

7Turkish7

8Danish8

9Norwegian9

AAlbanian10

BPortuguese11

The language code (left column) is the code that is contained in the system variable *LANGUAGE.
This code is used by Natural internally. It is the code you use to define the user language (see
Defining the User Language below). The code you use to identify the language of a Natural object
is the map code in the right-hand column of the table.

Example:

The language code for Portuguese is “11”. The code you usewhen cataloging a PortugueseNatural
object is “B”.

For the full table of language codes, see the system variable *LANGUAGE as described in the System
Variables documentation.

Defining the Language of a Natural Object

To define the language of a Natural object (map, helproutine, program, function, subprogram or
copycode), you add the corresponding map code to the object name. Apart from the map code,
the name of the object must be identical for all languages.

In the example below, amap has been created in English and in German. To identify the languages
of the maps, the map code that corresponds to the respective language has been included in the
map name.

Example of Map Names for a Multilingual Application

DEMO1 = English map (map code 1)

DEMO2 = German map (map code 2)

Defining Languages with Alphabetical Map Codes

Map codes are in the range 1-9, A-Z or a-y. The alphabetical map codes require special handling.

Normally, it is not possible to catalog an object with a lower-case letter in the name - all characters
are automatically converted into capitals.

This is however necessary, if for example you wish to define an object for Kanji (Japanese) which
has the language code 59 and the map code x.

To catalog such an object, you first set the correct language code (here 59) using the terminal
command %L=nn, where nn is the language code.

You then catalog the object using the ampersand (&) character instead of the actual map code in
the object name. So to have a Japanese version of the map DEMO, you stow themap under the name
DEMO&.

If you now look at the list of Natural objects, you will see that the map is correctly listed as DEMOx.

Objects with language codes 1-9 and upper case A-Z can be cataloged directly without the use of
the ampersand (&) notation.

Defining the User Language

You define the language to be used per user - as defined in the system variable *LANGUAGE - with
the profile parameter ULANG (which is described in the Parameter Reference) or with the terminal
command %L=nn (where nn is the language code).

Referencing Multilingual Objects

To reference multilingual objects in a program, you use the ampersand (&) character in the name
of the object.

The program below uses the maps DEMO1 and DEMO2. The ampersand (&) character at the end of
the map name stands for the map code and indicates that the map with the current language as
defined in the *LANGUAGE system variable is to be used.

DEFINE DATA LOCAL
1 PERSONNEL VIEW OF EMPLOYEES

2 NAME (A20)
2 PERSONNEL-ID (A8)

1 CAR VIEW OF VEHICLES
2 REG-NUM (A15)

1 #CODE (N1)
END-DEFINE
*
INPUT USING MAP 'DEMO&' /* <--- INVOKE MAP WITH CURRENT LANGUAGE CODE
...

When this program is run, the English map (DEMO1) is displayed. This is because the current value
of *LANGUAGE is 1 = English.

MAP DEMO1

SAMPLE MAP

Please select a function!

1.) Employee information

2.) Vehicle information

Enter code here: _

In the example below, the language code has been switched to 2 = German with the terminal
command %L=2.

When the program is now run, the German map (DEMO2) is displayed.

BEISPIEL-MAP

Bitte wählen Sie eine Funktion!

1.) Mitarbeiterdaten

2.) Fahrzeugdaten

Code hier eingeben: _

Programs

For some applications it may be useful to define multilingual programs. For example, a standard
invoicing program, might use different subprograms to handle various tax aspects, depending on
the country where the invoice is to be written.

Multilingual programs are defined with the same technique as described above for maps.

Edit Masks for Date and Time Fields

The language used for date and time fields defined with edit masks also depends on the system
variable *LANGUAGE.

For details on edit masks, see the session parameter EM as described in the Parameter Reference.

Skill-Sensitive User Interfaces

Users with varying levels of skill may wish to have different maps (of varying detail) while using
the same application.

If your application is not for international use by users speaking different languages, you can use
the techniques for multilingual maps to define maps of varying detail.

For example, you could define language code 1 as corresponding to the skill of the beginner, and
language code 2 as corresponding to the skill of the advanced user. This simple but effective
technique is illustrated below.

The followingmap (PERS1) includes instructions for the end user on how to select a function from
the menu. The information is very detailed. The name of the map contains the map code 1:

MAP PERS1

SAMPLE MAP

Please select a function

1.) Employee information _

2.) Vehicle information _

Enter code: _

To select a function, do one of the following:

- place the cursor on the input field next to desired function and press Enter
- mark the input field next to desired function with an X and press Enter
- enter the desired function code (1 or 2) in the 'Enter code' field and press

Enter

The samemap, but without the detailed instructions is saved under the same name, but with map
code 2.

MAP PERS2

SAMPLE MAP

Please select a function

1.) Employee information _

2.) Vehicle information _

Enter code: _

In the example above, the map with the detailed instructions is output, if the ULANG profile para-
meter has the value 1, the map without the instructions if the value is 2. See also the description
of the profile parameter ULANG (in the Parameter Reference).

Programming Guide518

Screen Design



54 Dialog Design

■ Field-Sensitive Processing .............................................................................................................. 520
■ Simplifying Programming ................................................................................................................ 522
■ Line-Sensitive Processing ............................................................................................................... 523
■ Column-Sensitive Processing ........................................................................................................... 524
■ Processing Based on Function Keys .................................................................................................. 524
■ Processing Based on Function-Key Names ......................................................................................... 525
■ Processing Data Outside an Active Window ........................................................................................ 526
■ Copying Data from a Screen ............................................................................................................ 529
■ Statements REINPUT/REINPUT FULL .............................................................................................. 532

519



This chapter tells you howyou can design character-based user interfaces thatmake user interaction
with the application simple and flexible.

Field-Sensitive Processing

*CURS-FIELD and POS(field-name)

Using the system variable *CURS-FIELD together with the system function POS(field-name), you
can define processing based on the fieldwhere the cursor is positioned at the time the user presses
Enter.

*CURS-FIELD contains the internal identification of the fieldwhere the cursor is currently positioned;
it cannot be used by itself, but only in conjunction with POS(field-name).

You can use *CURS-FIELD and POS(field-name), for example, to enable a user to select a function
simply by placing the cursor on a specific field and pressing Enter.

The example below illustrates such an application:

DEFINE DATA LOCAL
1 #EMP (A1)
1 #CAR (A1)
1 #CODE (N1)
END-DEFINE
*
INPUT USING MAP 'CURS'
*
DECIDE FOR FIRST CONDITION

WHEN *CURS-FIELD = POS(#EMP) OR #EMP = 'X' OR #CODE = 1
FETCH 'LISTEMP'

WHEN *CURS-FIELD = POS(#CAR) OR #CAR = 'X' OR #CODE = 2
FETCH 'LISTCAR'

WHEN NONE
REINPUT 'PLEASE MAKE A VALID SELECTION'

END-DECIDE
END

And the result:

SAMPLE MAP

Please select a function

1.) Employee information _
2.) Vehicle information _ <== Cursor positioned

on field

Enter code: _

To select a function, do one of the following:

- place the cursor on the input field next to desired function and press Enter
- mark the input field next to desired function with an X and press Enter
- enter the desired function code (1 or 2) in the 'Enter code' field and press

Enter

If the user places the cursor on the input field (#EMP) next to Employee information, and presses
Enter, the program LISTEMP displays a list of employee names:

Page 1 2001-01-22 09:39:32

NAME
--------------------

ABELLAN
ACHIESON
ADAM
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
AECKERLE
AFANASSIEV
AFANASSIEV
AHL
AKROYD

Notes:

1. In Natural for Ajax applications, *CURS-FIELD identifies the operand that represents the value
of the control that has the input focus. You may use *CURS-FIELD in conjunction with the POS
function to check for the control that has the input focus and perform processing depending
on that condition.

2. The values of *CURS-FIELD and POS(field-name) serve for internal identification of the fields
only. They cannot be used for arithmetical operations.

Simplifying Programming

System Function POS

The Natural system function POS(field-name) contains the internal identification of the field
whose name is specified with the system function.

POS(field-name)may be used to identify a specific field, regardless of its position in a map. This
means that the sequence and number of fields in a map may be changed, but POS(field-name)
will still uniquely identify the same field. With this, for example, you need only a single REINPUT
statement to make the field to be MARKed dependent on the program logic.

Note: The value POS(field-name) serves for internal identification of the fields only. It
cannot be used for arithmetical operations.

Example:

...
DECIDE ON FIRST VALUE OF ...

VALUE ...
COMPUTE #FIELDX = POS(FIELD1)

VALUE ...
COMPUTE #FIELDX = POS(FIELD2)

...
END-DECIDE
...
REINPUT ... MARK #FIELDX
...

Full details on *CURS-FIELD and POS(field-name) are described in the System Variables and System
Functions documention.

Line-Sensitive Processing

System Variable *CURS-LINE

Using the system variable *CURS-LINE, you can make processing dependent on the line where the
cursor is positioned at the time the user presses Enter.

Using this variable, you can make user-friendly menus. With the appropriate programming, the
usermerely has to place the cursor on the line of the desiredmenu option and press Enter to execute
the option.

The cursor position is definedwithin the current activewindow, regardless of its physical placement
on the screen.

Note: The message line, function-key lines and statistics line/infoline are not counted as
data lines on the screen.

The example below demonstrates line-sensitive processing using the *CURS-LINE system variable.
When the user presses Enter on the map, the program checks if the cursor is positioned on line 8
of the screen which contains the option Employee information. If this is the case, the program
that lists the names of employees LISTEMP is executed.

DEFINE DATA LOCAL
1 #EMP (A1)
1 #CAR (A1)
1 #CODE (N1)
END-DEFINE
*
INPUT USING MAP 'CURS'
*
DECIDE FOR FIRST CONDITION

WHEN *CURS-LINE = 8
FETCH 'LISTEMP'

WHEN NONE
REINPUT 'PLACE CURSOR ON LINE OF OPTION YOU WISH TO SELECT'

END-DECIDE
END

Output:

Company Information

Please select a function

[] 1.) Employee information

2.) Vehicle information

Place the cursor on the line of the option you wish to select and press
Enter

The user places the cursor indicated by square brackets [] on the line of the desired option and
presses Enter and the corresponding program is executed.

Column-Sensitive Processing

System Variable *CURS-COL

The system variable *CURS-COL can be used in a similar way to *CURS-LINE described above. With
*CURS-COL you can make processing dependent on the column where the cursor is positioned on
the screen.

Processing Based on Function Keys

System Variable *PF-KEY

Frequently you may wish to make processing dependent on the function key a user presses.

This is achieved with the statement SET KEY, the system variable *PF-KEY and a modification of
the default map settings (Standard Keys = Y).

The SET KEY statement assigns functions to function keys during program execution. The system
variable *PF-KEY contains the identification of the last function key the user pressed.

The example below illustrates the use of SET KEY in combination with *PF-KEY.

...
SET KEY PF1
*
INPUT USING MAP 'DEMO&'
IF *PF-KEY = 'PF1'

WRITE 'Help is currently not active'
END-IF
...

The SET KEY statement activates PF1 as a function key.

The IF statement defines what action is to be taken when the user presses PF1. The system variable
*PF-KEY is checked for its current content; if it contains PF1, the corresponding action is taken.

Further details regarding the statement SET KEY and the system variable *PF-KEY are described
in the Statements and the System Variables documentation respectively.

Processing Based on Function-Key Names

System Variable *PF-NAME

When defining processing based on function keys, further comfort can be added by using the
system variable *PF-NAME. With this variable you can make processing dependent on the name of
a function, not on a specific key.

The variable *PF-NAME contains the name of the last function key the user pressed (that is, the
name as assigned to the key with the NAMED clause of the SET KEY statement).

For example, if you wish to allow users to invoke help by pressing either PF3 or PF12, you assign
the same name (in the example below: INFO) to both keys. When the user presses either one of the
keys, the processing defined in the IF statement is performed.

...
SET KEY PF3 NAMED 'INFO'

PF12 NAMED 'INFO'
INPUT USING MAP 'DEMO&'
IF *PF-NAME = 'INFO'

WRITE 'Help is currently not active'
END-IF
...

The function names defined with NAMED appear in the function-key lines:

Processing Data Outside an Active Window

Below is information on:

■ System Variable *COM
■ Example Usage of *COM
■ Positioning the Cursor to *COM - the %T* Terminal Command

System Variable *COM

As stated in the section Screen Design -Windows, only onewindow is active at any one time. This
normally means that input is only possible within that particular window.

Using the *COM system variable, which can be regarded as a communication area, it is possible to
enter data outside the current window.

The prerequisite is that a map contains *COM as a modifiable field. This field is then available for
the user to enter data when a window is currently on the screen. Further processing can then be
made dependent on the content of *COM.

This allows you to implement user interfaces as already used, for example, by Con-nect, Software
AG's office system,where a user can always enter data in the command line, evenwhen awindow
with its own input fields is active.

Note that *COM is only cleared when the Natural session is ended.

Example Usage of *COM

In the example below, the program ADD performs a simple addition using the input data from a
map. In this map, *COM has been defined as a modifiable field (at the bottom of the map) with the
length specified in the AL field of the Extended Field Editing. The result of the calculation is dis-
played in a window. Although this window offers no possibility for input, the user can still use
the *COM field in the map outside the window.

Program ADD:

DEFINE DATA LOCAL
1 #VALUE1 (N4)
1 #VALUE2 (N4)
1 #SUM3 (N8)
END-DEFINE
*
DEFINE WINDOW EMP

SIZE 8*17
BASE 10/2

TITLE 'Total of Add'
CONTROL SCREEN
FRAMED POSITION SYMBOL BOT LEFT

*
INPUT USING MAP 'WINDOW'
*
COMPUTE #SUM3 = #VALUE1 + #VALUE2
*
SET WINDOW 'EMP'
INPUT (AD=O) / 'Value 1 +' /

'Value 2 =' //
' ' #SUM3

*
IF *COM = 'M'

FETCH 'MULTIPLY' #VALUE1 #VALUE2
END-IF
END

Output of Program ADD:

Map to Demonstrate Windows with *COM

CALCULATOR

Enter values you wish to calculate

Value 1: 12__
Value 2: 12__

+-Total of Add-+
! !
! Value 1 + !
! Value 2 = !
! !
! 24 !
! !
+--------------+

Next line is input field (*COM) for input outside the window:

In this example, by entering the value M, the user initiates amultiplication function; the two values
from the input map are multiplied and the result is displayed in a second window:

Map to Demonstrate Windows with *COM

CALCULATOR

Enter values you wish to calculate

Value 1: 12__
Value 2: 12__

+-Total of Add-+ +--------------+
! ! ! !
! Value 1 + ! ! Value 1 x !
! Value 2 = ! ! Value 2 = !
! ! ! !
! 24 ! ! 144 !
! ! ! !
+--------------+ +--------------+

Next line is input field (*COM) for input outside the window:
M

Positioning the Cursor to *COM - the %T* Terminal Command

Normally, when a window is active and the window contains no input fields (AD=M or AD=A), the
cursor is placed in the top left corner of the window.

With the terminal command %T*, you can position the cursor to a *COM system variable outside
the window when the active window contains no input fields.

By using %T* again, you can switch back to standard cursor placement.

Example:

...
INPUT USING MAP 'WINDOW'
*
COMPUTE #SUM3 = #VALUE1 + #VALUE2
*
SET CONTROL 'T*'
SET WINDOW 'EMP'
INPUT (AD=O) / 'Value 1 +' /

'Value 2 =' //
' ' #SUM3

...

Copying Data from a Screen

Below is information on:

■ Terminal Commands %CS and %CC
■ Selecting a Line from Report Output for Further Processing

Terminal Commands %CS and %CC

With these terminal commands, you can copy parts of a screen into the Natural stack (%CS) or into
the system variable *COM (%CC). The protected data from a specific screen line are copied field by
field.

The full options of these terminal commands are described in the Terminal Commands documenta-
tion.

Once copied to the stack or *COM, the data are available for further processing. Using these com-
mands, you can make user-friendly interfaces as in the example below.

Selecting a Line from Report Output for Further Processing

In the following example, the program COM1 lists all employee names from Abellan to Alestia.

Program COM1:

DEFINE DATA LOCAL
1 EMP VIEW OF EMPLOYEES

2 NAME(A20)
2 MIDDLE-NAME (A20)
2 PERSONNEL-ID (A8)

END-DEFINE
*
READ EMP BY NAME STARTING FROM 'ABELLAN' THRU 'ALESTIA'

DISPLAY NAME
END-READ
FETCH 'COM2'
END

Output of Program COM1:

Page 1 2006-08-12 09:41:21

NAME
--------------------

ABELLAN
ACHIESON
ADAM
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
AECKERLE
AFANASSIEV
AFANASSIEV
AHL
AKROYD
ALEMAN
ALESTIA
MORE

Control is now passed to the program COM2.

Program COM2:

DEFINE DATA LOCAL
1 EMP VIEW OF EMPLOYEES

2 NAME(A20)
2 MIDDLE-NAME (A20)
2 PERSONNEL-ID (A8)

1 SELECTNAME (A20)
END-DEFINE
*
SET KEY PF5 = '%CCC'
*
INPUT NO ERASE 'SELECT FIELD WITH CURSOR AND PRESS PF5'
*
MOVE *COM TO SELECTNAME
FIND EMP WITH NAME = SELECTNAME

DISPLAY NAME PERSONNEL-ID
END-FIND
END

In this program, the terminal command %CCC is assigned to PF5. The terminal command copies all
protected data from the line where the cursor is positioned to the system variable *COM. This in-

formation is then available for further processing. This further processing is defined in the program
lines shown in boldface.

The user can now position the cursor on the name that interests him; when he/she now presses
PF5, further employee information is supplied.

SELECT FIELD WITH CURSOR AND PRESS PF5 2006-08-12 09:44:25

NAME
--------------------

ABELLAN
ACHIESON
ADAM <== Cursor positioned on name for which more information is required
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
ADKINSON
AECKERLE
AFANASSIEV
AFANASSIEV
AHL
AKROYD
ALEMAN
ALESTIA

In this case, the personnel ID of the selected employee is displayed:

Page 1 2006-08-12 09:44:52

NAME PERSONNEL
ID

-------------------- ---------

ADAM 50005800

Statements REINPUT/REINPUT FULL

If you wish to return to and re-execute an INPUT statement, you use the REINPUT statement. It is
generally used to display amessage indicating that the data input as a result of the previous INPUT
statement were invalid.

If you specify the FULL option in a REINPUT statement, the corresponding INPUT statement will be
re-executed fully:

■ With an ordinary REINPUT statement (without FULL option), the contents of variables that were
changed between the INPUT and REINPUT statement will not be displayed; that is, all variables
on the screenwill show the contents they hadwhen the INPUT statementwas originally executed.

■ With a REINPUT FULL statement, all changes that have been made after the initial execution of
the INPUT statement will be applied to the INPUT statement when it is re-executed; that is, all
variables on the screen contain the values they had when the REINPUT statement was executed.

■ If youwish to position the cursor to a specified field, you can use the MARK option, and to position
to a particular position within a specified field, you use the MARK POSITION option.

The example below illustrates the use of REINPUT FULLwith MARK POSITION.

DEFINE DATA LOCAL
1 #A (A10)
1 #B (N4)
1 #C (N4)
END-DEFINE
*
INPUT (AD=M) #A #B #C
IF #A = ' '

COMPUTE #B = #B + #C
RESET #C
REINPUT FULL 'Enter a value' MARK POSITION 5 IN *#A

END-IF
END

The user enters 3 in field #B and 3 in field #C and presses Enter.

#A #B 3 #C 3

The program requires field #A to be non-blank. The REINPUT FULL statement with MARK POSITION
5 IN *#A returns the input screen; the now modified variable #B contains the value 6 (after the
COMPUTE calculation has been performed). The cursor is positioned to the 5th position in field #A
ready for new input.

Enter name of field
#A _ #B 6 #C 0

Enter a value

This is the screen that would be returned by the same statement, without the FULL option. Note
that the variables #B and #C have been reset to their status at the time of execution of the INPUT
statement (each field contains the value 3).

#A _ #B 3 #C 3

This chapter contains a list of all keywords that are reserved in theNatural programming language.

Important: To avoid any naming conflicts, you are strongly recommended not to useNatural
reserved keywords as names for variables.

Alphabetical List of Natural Reserved Keywords

The following list is an overview of Natural reserved keywords and is for general information
only. In case of doubt, use the keyword check function of the compiler.

[ A | B | C | D | E | F | G | H | I | J | K | L |M | N | O | P | Q | R | S | T | U | V |W | X | Y
| Z ]

- A -

ABS
ABSOLUTE
ACCEPT
ACTION
ACTIVATION
AD
ADD
AFTER
AL
ALARM
ALL
ALPHA
ALPHABETICALLY
AND
ANY
APPL
APPLICATION
ARRAY
AS
ASC
ASCENDING
ASSIGN
ASSIGNING
ASYNC
AT
ATN
ATT
ATTRIBUTES
AUTH

AUTHORIZATION
AUTO
AVER
AVG

- B -

BACKOUT
BACKWARD
BASE
BEFORE
BETWEEN
BLOCK
BOT
BOTTOM
BREAK
BROWSE
BUT
BX
BY

- C -

CABINET
CALL
CALLDBPROC
CALLING
CALLNAT
CAP
CAPTIONED
CASE
CC
CD
CDID
CF
CHAR
CHARLENGTH
CHARPOSITION
CHILD
CIPH
CIPHER
CLASS
CLOSE
CLR
COALESCE
CODEPAGE

COMMAND
COMMIT
COMPOSE
COMPRESS
COMPUTE
CONCAT
CONDITION
CONST
CONSTANT
CONTEXT
CONTROL
CONVERSATION
COPIES
COPY
COS
COUNT
COUPLED
CS
CURRENT
CURSOR
CV

- D -

DATA
DATAAREA
DATE
DAY
DAYS
DC
DECIDE
DECIMAL
DEFINE
DEFINITION
DELETE
DELIMITED
DELIMITER
DELIMITERS
DESC
DESCENDING
DIALOG
DIALOG-ID
DIGITS
DIRECTION
DISABLED

DISP
DISPLAY
DISTINCT
DIVIDE
DL
DLOGOFF
DLOGON
DNATIVE
DNRET
DO
DOCUMENT
DOEND
DOWNLOAD
DU
DY
DYNAMIC

- E -

EDITED
EJ
EJECT
ELSE
EM
ENCODED
END
END-ALL
END-BEFORE
END-BREAK
END-BROWSE
END-CLASS
END-DECIDE
END-DEFINE
END-ENDDATA
END-ENDFILE
END-ENDPAGE
END-ERROR
END-FILE
END-FIND
END-FOR
END-FUNCTION
END-HISTOGRAM
ENDHOC
END-IF
END-INTERFACE

END-LOOP
END-METHOD
END-NOREC
END-PARAMETERS
END-PARSE
END-PROCESS
END-PROPERTY
END-PROTOTYPE
END-READ
END-REPEAT
END-RESULT
END-SELECT
END-SORT
END-START
END-SUBROUTINE
END-TOPPAGE
END-WORK
ENDING
ENTER
ENTIRE
ENTR
EQ
EQUAL
ERASE
ERROR
ERRORS
ES
ESCAPE
EVEN
EVENT
EVERY
EXAMINE
EXCEPT
EXISTS
EXIT
EXP
EXPAND
EXPORT
EXTERNAL
EXTRACTING

- F -

FALSE
FC

FETCH
FIELD
FIELDS
FILE
FILL
FILLER
FINAL
FIND
FIRST
FL
FLOAT
FOR
FORM
FORMAT
FORMATTED
FORMATTING
FORMS
FORWARD
FOUND
FRAC
FRAMED
FROM
FS
FULL
FUNCTION
FUNCTIONS

- G -

GC
GE
GEN
GENERATED
GET
GFID
GIVE
GIVING
GLOBAL
GLOBALS
GREATER
GT
GUI

- H -

HANDLE

HAVING
HC
HD
HE
HEADER
HEX
HISTOGRAM
HOLD
HORIZ
HORIZONTALLY
HOUR
HOURS
HW

- I -

IA
IC
ID
IDENTICAL
IF
IGNORE
IM
IMMEDIATE
IMPORT
IN
INC
INCCONT
INCDIC
INCDIR
INCLUDE
INCLUDED
INCLUDING
INCMAC
INDEPENDENT
INDEX
INDEXED
INDICATOR
INIT
INITIAL
INNER
INPUT
INSENSITIVE
INSERT
INT

INTEGER
INTERCEPTED
INTERFACE
INTERFACE4
INTERMEDIATE
INTERSECT
INTO
INVERTED
INVESTIGATE
IP
IS
ISN

- J -

JOIN
JUST
JUSTIFIED

- K -

KD
KEEP
KEY
KEYS

- L -

LANGUAGE
LAST
LC
LE
LEAVE
LEAVING
LEFT
LENGTH
LESS
LEVEL
LIB
LIBPW
LIBRARY
LIBRARY-PASSWORD
LIKE
LIMIT
LINDICATOR
LINES

LISTED
LOCAL
LOCKS
LOG
LOG-LS
LOG-PS
LOGICAL
LOOP
LOWER
LS
LT

- M -

MACROAREA
MAP
MARK
MASK
MAX
MC
MCG
MESSAGES
METHOD
MGID
MICROSECOND
MIN
MINUTE
MODAL
MODIFIED
MODULE
MONTH
MORE
MOVE
MOVING
MP
MS
MT
MULTI-FETCH
MULTIPLY

- N -

NAME
NAMED
NAMESPACE
NATIVE

NAVER
NC
NCOUNT
NE
NEWPAGE
NL
NMIN
NO
NODE
NOHDR
NONE
NORMALIZE
NORMALIZED
NOT
NOTIT
NOTITLE
NULL
NULL-HANDLE
NUMBER
NUMERIC

- O -

OBJECT
OBTAIN
OCCURRENCES
OF
OFF
OFFSET
OLD
ON
ONCE
ONLY
OPEN
OPTIMIZE
OPTIONAL
OPTIONS
OR
ORDER
OUTER
OUTPUT

- P -

PACKAGESET
PAGE

PARAMETER
PARAMETERS
PARENT
PARSE
PASS
PASSW
PASSWORD
PATH
PATTERN
PA1
PA2
PA3
PC
PD
PEN
PERFORM
PFn (n = 1 to 9)
PFnn(nn = 10 to 99)
PGDN
PGUP
PGM
PHYSICAL
PM
POLICY
POS
POSITION
PREFIX
PRINT
PRINTER
PROCESS
PROCESSING
PROFILE
PROGRAM
PROPERTY
PROTOTYPE
PRTY
PS
PT
PW

- Q -

QUARTER
QUERYNO

- R -

RD
READ
READONLY
REC
RECORD
RECORDS
RECURSIVELY
REDEFINE
REDUCE
REFERENCED
REFERENCING
REINPUT
REJECT
REL
RELATION
RELATIONSHIP
RELEASE
REMAINDER
REPEAT
REPLACE
REPORT
REPORTER
REPOSITION
REQUEST
REQUIRED
RESET
RESETTING
RESIZE
RESPONSE
RESTORE
RESULT
RET
RETAIN
RETAINED
RETRY
RETURN
RETURNS
REVERSED
RG
RIGHT
ROLLBACK
ROUNDED
ROUTINE

ROW
ROWS
RR
RS
RULEVAR
RUN

- S -

SA
SAME
SCAN
SCREEN
SCROLL
SECOND
SELECT
SELECTION
SEND
SENSITIVE
SEPARATE
SEQUENCE
SERVER
SET
SETS
SETTIME
SF
SG
SGN
SHORT
SHOW
SIN
SINGLE
SIZE
SKIP
SL
SM
SOME
SORT
SORTED
SORTKEY
SOUND
SPACE
SPECIFIED
SQL
SQLID

SQRT
STACK
START
STARTING
STATEMENT
STATIC
STATUS
STEP
STOP
STORE
SUBPROGRAM
SUBPROGRAMS
SUBROUTINE
SUBSTR
SUBSTRING
SUBTRACT
SUM
SUPPRESS
SUPPRESSED
SUSPEND
SYMBOL
SYNC
SYSTEM

- T -

TAN
TC
TERMINATE
TEXT
TEXTAREA
TEXTVARIABLE
THAN
THEM
THEN
THRU
TIME
TIMESTAMP
TIMEZONE
TITLE
TO
TOP
TOTAL
TP
TR

TRAILER
TRANSACTION
TRANSFER
TRANSLATE
TREQ
TRUE
TS
TYPE
TYPES

- U -

UC
UNDERLINED
UNION
UNIQUE
UNKNOWN
UNTIL
UPDATE
UPLOAD
UPPER
UR
USED
USER
USING

- V -

VAL
VALUE
VALUES
VARGRAPHIC
VARIABLE
VARIABLES
VERT
VERTICALLY
VIA
VIEW

- W -

WH
WHEN
WHERE
WHILE
WINDOW

WITH
WORK
WRITE
WITH_CTE

- X -

XML

- Y -

YEAR

- Z -

ZD
ZP

Performing a Check for Natural Reserved Keywords

There is a subset ofNatural keywordswhich,whenused as names for variables,would be ambigu-
ous. These are in particular keywordswhich identifyNatural statements (ADD, FIND, etc.) or system
functions (ABS, SUM, etc.). If you use such a keyword as the name of a variable, you cannot use this
variable in the context of optional operands (with CALLNAT, WRITE, etc.).

Example:

DEFINE DATA LOCAL
1 ADD (A10)
END-DEFINE
CALLNAT 'MYSUB' ADD 4 /* ADD is regarded as ADD statement
END

To check variable names in a programming object against such Natural reserved keywords, you
can use the Natural profile parameter KCHECK

The following table contains a list of Natural reserved keywords that are checked by KC or KCHECK.

T - WR - SG - PE - FA - D

TANREADGETEJECTA-AVER
TERMINATEREDEFINEHISTOGRAMELSEABS
TOPREDUCEIFENDACCEPT
TOTALREINPUTIGNOREEND-ALLADD
TRANSFERREJECTIMPORTEND-BEFOREALL
TRUERELEASEINCCONTEND-BREAKA-MAX
UNTILREPEATINCDICEND-BROWSEA-MIN
UPDATEREQUESTINCDIREND-DECIDEA-NAVER
UPLOADRESETINCLUDEEND-ENDDATAA-NCOUNT
VALRESIZEINCMACEND-DECIDEA-NMIN
VALUERESTOREINPUTEND-ENDDATAANY
VALUESRETINSERTEND-ENDFILEASSIGN
WASTERETRYINTEND-ENDPAGEA-SUM
WHENRETURNINVESTIGATEEND-ERRORAT
WHILEROLLBACKLIMITEND-FILEATN
WITH_CTERULEVARLOGEND-FINDAVER
WRITERUNLOOPEND-FORBACKOUT

SELECTMAPEND-HISTOGRAMBEFORE
SENDMAXENDHOCBREAK
SEPARATEMINEND-IFBROWSE
SETMOVEEND-LOOPCALL
SETTIMEMULTIPLYEND-NORECCALLDBPROC
SGNNAVEREND-PARSECALLNAT
SHOWNCOUNTEND-PROCESSCLOSE
SINNEWPAGEEND-READCOMMIT
SKIPNMINEND-REPEATCOMPOSE
SORTNONEEND-RESULTCOMPRESS
SORTKEYNULL-HANDLEEND-SELECTCOMPUTE
SQRTOBTAINEND-SORTCOPY
STACKOLDEND-STARTCOS
STARTONEND-SUBROUTINECOUNT
STOPOPENEND-TOPPAGECREATE
STOREOPTIONSEND-WORKDECIDE
SUBSTRPARSEENTIREDEFINE
SUBSTRINGPASSWESCAPEDELETE
SUBTRACTPERFORMEXAMINEDISPLAY
SUMPOSEXPDIVIDE
SUSPENDPRINTEXPANDDLOGOFF

PROCESSEXPORTDLOGON
FALSEDNATIVE
FETCHDO
FINDDOEND
FORDOWNLOAD
FORMAT
FRAC

By default, no keyword check is performed.

This chapter contains some additional example programs that are referenced in the Programming
Guide.

READ Statement

The following example is referenced in the section Statements for Database Access.

READX03 - READ statement (with LOGICAL clause)

** Example 'READX03': READ (with LOGICAL clause)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 PERSONNEL-ID
2 JOB-TITLE

END-DEFINE
*
LIMIT 8
READ EMPLOY-VIEW LOGICAL BY PERSONNEL-ID

DISPLAY NOTITLE *ISN NAME
'PERS-NO' PERSONNEL-ID
'POSITION' JOB-TITLE

END-READ
END

Output of Program READX03:

ISN NAME PERS-NO POSITION
----------- -------------------- -------- -------------------------

204 SCHINDLER 11100102 PROGRAMMIERER
205 SCHIRM 11100105 SYSTEMPROGRAMMIERER
206 SCHMITT 11100106 OPERATOR
207 SCHMIDT 11100107 SEKRETAERIN
208 SCHNEIDER 11100108 SACHBEARBEITER
209 SCHNEIDER 11100109 SEKRETAERIN
210 BUNGERT 11100110 SYSTEMPROGRAMMIERER
211 THIELE 11100111 SEKRETAERIN

FIND Statement

The following examples are referenced in the section Statements for Database Access.

FINDX07 - FIND statement (with several clauses)

** Example 'FINDX07': FIND (with several clauses)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY
2 SALARY (1)
2 CURR-CODE (1)

END-DEFINE
*
FIND EMPLOY-VIEW WITH PHONETIC-NAME = 'JONES' OR = 'BECKR'

AND CITY = 'BOSTON' THRU 'NEW YORK'
BUT NOT 'CHAPEL HILL'
SORTED BY NAME
WHERE SALARY (1) < 28000

DISPLAY NOTITLE NAME FIRST-NAME CITY SALARY (1)
END-FIND
END

Output of Program FINDX07:

NAME FIRST-NAME CITY ANNUAL
SALARY

-------------------- -------------------- -------------------- ----------

BAKER PAULINE DERBY 4450
JONES MARTHA KALAMAZOO 21000
JONES KEVIN DERBY 7000

FINDX08 - FIND statement (with LIMIT)

** Example 'FINDX08': FIND (with LIMIT)
** Demonstrates FIND statement with LIMIT option to
** terminate program with an error message if the
** number of records selected exceeds a specified
** limit (no output).
************************************************************************
DEFINE DATA LOCAL

1 EMPLOY-VIEW VIEW OF EMPLOYEES
2 NAME
2 JOB-TITLE

END-DEFINE
*
FIND EMPLOY-VIEW WITH LIMIT (5) JOB-TITLE = 'SALES PERSON'

DISPLAY NAME JOB-TITLE
END-FIND
END

Runtime Error Caused by Program FINDX08:

NAT1005 More records found than specified in search limit.

FINDX09 - FIND statement (using *NUMBER, *COUNTER, *ISN)

** Example 'FINDX09': FIND (using *NUMBER, *COUNTER, *ISN)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 DEPT
2 NAME

END-DEFINE
*
FIND EMPLOY-VIEW WITH CITY = 'BOSTON'

WHERE DEPT = 'TECH00' THRU 'TECH10'
DISPLAY NOTITLE

'COUNTER' *COUNTER NAME DEPT 'ISN' *ISN
AT START OF DATA

WRITE '(TOTAL NUMBER IN BOSTON:' *NUMBER ')' /
END-START

END-FIND
END

Output of Program FINDX09:

COUNTER NAME DEPARTMENT ISN
CODE

----------- -------------------- ---------- -----------

(TOTAL NUMBER IN BOSTON: 7 )

1 STANWOOD TECH10 782
2 PERREAULT TECH10 842

FINDX10 - FIND statement (combined with READ)

** Example 'FINDX10': FIND (combined with READ)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 FIRST-NAME

1 VEHIC-VIEW VIEW OF VEHICLES
2 PERSONNEL-ID
2 MAKE

END-DEFINE
*
LIMIT 15
*
EMP. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'

VEH. FIND VEHIC-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (EMP.)
IF NO RECORDS FOUND
MOVE '*** NO CAR ***' TO MAKE

END-NOREC
DISPLAY NOTITLE

NAME (EMP.) (IS=ON)
FIRST-NAME (EMP.) (IS=ON)
MAKE (VEH.)

END-FIND
END-READ
END

Output of Program FINDX10:

NAME FIRST-NAME MAKE
-------------------- -------------------- --------------------

JONES VIRGINIA CHRYSLER
MARSHA CHRYSLER

CHRYSLER
ROBERT GENERAL MOTORS
LILLY FORD

MG
EDWARD GENERAL MOTORS
MARTHA GENERAL MOTORS
LAUREL GENERAL MOTORS
KEVIN DATSUN
GREGORY FORD

JOPER MANFRED *** NO CAR ***
JOUSSELIN DANIEL RENAULT
JUBE GABRIEL *** NO CAR ***
JUNG ERNST *** NO CAR ***

JUNKIN JEREMY *** NO CAR ***
KAISER REINER *** NO CAR ***

FINDX11 - FIND NUMBER statement (with *NUMBER)

** Example 'FINDX11': FIND NUMBER (with *NUMBER)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 FIRST-NAME
2 NAME
2 CITY
2 JOB-TITLE
2 SALARY (1)

*
1 #PERCENT (N.2)
1 REDEFINE #PERCENT

2 #WHOLE-NBR (N2)
1 #ALL-BOST (N3.2)
1 #SECR-ONLY (N3.2)
1 #BOST-NBR (N3)
1 #SECR-NBR (N3)
END-DEFINE
*
F1. FIND NUMBER EMPLOY-VIEW WITH CITY = 'BOSTON'
F2. FIND NUMBER EMPLOY-VIEW WITH CITY = 'BOSTON'

AND JOB-TITLE = 'SECRETARY'
*
MOVE *NUMBER(F1.) TO #ALL-BOST #BOST-NBR
MOVE *NUMBER(F2.) TO #SECR-ONLY #SECR-NBR
DIVIDE #ALL-BOST INTO #SECR-ONLY GIVING #PERCENT
*
WRITE TITLE LEFT JUSTIFIED UNDERLINED

'There are' #BOST-NBR 'employees in the Boston offices.' /
#SECR-NBR '(=' #WHOLE-NBR (EM=99%')') 'are secretaries.'

*
SKIP 1
FIND EMPLOY-VIEW WITH CITY = 'BOSTON'

AND JOB-TITLE = 'SECRETARY'
DISPLAY NAME FIRST-NAME JOB-TITLE SALARY (1)

END-FIND
END

Output of Program FINDX11:

Programming Guide560

Referenced Example Programs



There are 7 employees in the Boston offices.
3 (= 42%) are secretaries.

-------------------------------------------------------------------------------

NAME FIRST-NAME CURRENT ANNUAL
POSITION SALARY

-------------------- -------------------- ------------------------- ----------

SHAW LESLIE SECRETARY 18000
CREMER WALT SECRETARY 20000
COHEN JOHN SECRETARY 16000

Nested READ and FIND Statements

The following examples are referenced in the section Database Processing Loops.

READX04 - READ statement (in combination with FIND and the system variables *NUMBER
and *COUNTER)

** Example 'READX04': READ (in combination with FIND and the system
** variables *NUMBER and *COUNTER)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 FIRST-NAME

1 VEHIC-VIEW VIEW OF VEHICLES
2 PERSONNEL-ID
2 MAKE

END-DEFINE
*
LIMIT 10
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'

FD. FIND VEHIC-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
IF NO RECORDS FOUND
ENTER

END-NOREC
/*
DISPLAY NOTITLE

*COUNTER (RD.)(NL=8) NAME (AL=15) FIRST-NAME (AL=10)
*NUMBER (FD.)(NL=8) *COUNTER (FD.)(NL=8) MAKE

END-FIND
END-READ
END

Output of Program READX04:

CNT NAME FIRST-NAME NMBR CNT MAKE
--------- --------------- ---------- --------- --------- --------------------

1 JONES VIRGINIA 1 1 CHRYSLER
2 JONES MARSHA 2 1 CHRYSLER
2 JONES MARSHA 2 2 CHRYSLER
3 JONES ROBERT 1 1 GENERAL MOTORS
4 JONES LILLY 2 1 FORD
4 JONES LILLY 2 2 MG
5 JONES EDWARD 1 1 GENERAL MOTORS
6 JONES MARTHA 1 1 GENERAL MOTORS
7 JONES LAUREL 1 1 GENERAL MOTORS
8 JONES KEVIN 1 1 DATSUN
9 JONES GREGORY 1 1 FORD
10 JOPER MANFRED 0 0

LIMITX01 - LIMIT statement (for READ, FIND loop processing)

** Example 'LIMITX01': LIMIT (for READ, FIND loop processing)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 FIRST-NAME
2 NAME

1 VEH-VIEW VIEW OF VEHICLES
2 PERSONNEL-ID
2 MAKE

END-DEFINE
*
LIMIT 4
*
READ EMPLOY-VIEW BY NAME STARTING FROM 'A'

FIND VEH-VIEW WITH PERSONNEL-ID = EMPLOY-VIEW.PERSONNEL-ID
IF NO RECORDS FOUND
MOVE 'NO CAR' TO MAKE

END-NOREC
DISPLAY PERSONNEL-ID NAME FIRST-NAME MAKE

END-FIND
END-READ
END

Output of Program LIMITX01:

Page 1 04-12-13 14:01:57

PERSONNEL-ID NAME FIRST-NAME MAKE
------------ -------------------- -------------------- --------------------

ABELLAN KEPA NO CAR
30000231 ACHIESON ROBERT FORD

ADAM SIMONE NO CAR
20008800 ADKINSON JEFF GENERAL MOTORS

ACCEPT and REJECT Statements

The following examples are referenced in the section Selecting Records Using ACCEPT/REJECT.

ACCEPX04 - ACCEPT IF ... LESS THAN ... statement

** Example 'ACCEPX04': ACCEPT IF ... LES THAN ...
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 JOB-TITLE
2 SALARY (1)

END-DEFINE
*
LIMIT 20
READ EMPLOY-VIEW BY PERSONNEL-ID = '20017000'

ACCEPT IF SALARY (1) LESS THAN 38000
DISPLAY NOTITLE PERSONNEL-ID NAME JOB-TITLE SALARY (1)

END-READ
END

Output of Program ACCEPX04:

PERSONNEL NAME CURRENT ANNUAL
ID POSITION SALARY

--------- -------------------- ------------------------- ----------

20017000 CREMER ANALYST 34000
20017100 MARKUSH TRAINEE 22000
20017400 NEEDHAM PROGRAMMER 32500
20017500 JACKSON PROGRAMMER 33000
20017600 PIETSCH SECRETARY 22000
20017700 PAUL SECRETARY 23000

20018000 FARRIS PROGRAMMER 30500
20018100 EVANS PROGRAMMER 31000
20018200 HERZOG PROGRAMMER 31500
20018300 LORIE SALES PERSON 28000
20018400 HALL SALES PERSON 30000
20018500 JACKSON MANAGER 36000
20018800 SMITH SECRETARY 24000
20018900 LOWRY SECRETARY 25000

ACCEPX05 - ACCEPT IF ... AND ... statement

** Example 'ACCEPX05': ACCEPT IF ... AND ...
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 JOB-TITLE
2 SALARY (1:2)

END-DEFINE
*
LIMIT 6
READ EMPLOY-VIEW PHYSICAL WHERE SALARY(2) > 0

ACCEPT IF SALARY(1) > 10000
AND SALARY(1) < 50000

DISPLAY (AL=15) 'SALARY I' SALARY (1) 'SALARY II' SALARY (2)
NAME JOB-TITLE CITY

END-READ
END

Output of Program ACCEPX05:

Page 1 04-12-13 14:05:28

SALARY I SALARY II NAME CURRENT CITY
POSITION

---------- ---------- --------------- --------------- ---------------

48000 46000 SPENGLER SACHBEARBEITER DARMSTADT
45000 40000 SPECK SACHBEARBEITER DARMSTADT
48000 46000 SCHINDLER PROGRAMMIERER HEPPENHEIM
36000 32000 SCHMIDT SEKRETAERIN HEPPENHEIM

ACCEPX06 - REJECT IF ... OR ... statement

** Example 'ACCEPX06': REJECT IF ... OR ...
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 SALARY (1)
2 JOB-TITLE
2 CITY
2 NAME

END-DEFINE
*
LIMIT 20
READ EMPLOY-VIEW LOGICAL BY PERSONNEL-ID = '20017000'

REJECT IF SALARY (1) < 20000
OR SALARY (1) > 26000

DISPLAY NOTITLE SALARY (1) NAME JOB-TITLE CITY
END-READ
END

Output of Program ACCEPX06:

ANNUAL NAME CURRENT CITY
SALARY POSITION

---------- -------------------- ------------------------- --------------------

22000 MARKUSH TRAINEE LOS ANGELES
22000 PIETSCH SECRETARY VISTA
23000 PAUL SECRETARY NORFOLK
24000 SMITH SECRETARY SILVER SPRING
25000 LOWRY SECRETARY LEXINGTON

AT START OF DATA and AT END OF DATA Statements

The following examples are referenced in the section AT START/END OF DATA Statements.

ATENDX01 - AT END OF DATA statement

** Example 'ATENDX01': AT END OF DATA
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 JOB-TITLE

END-DEFINE
*
READ (6) EMPLOY-VIEW BY PERSONNEL-ID FROM '20017000'

DISPLAY NOTITLE NAME JOB-TITLE
AT END OF DATA

WRITE / 'LAST PERSON SELECTED:' OLD(NAME)
END-ENDDATA

END-READ
END

Output of Program ATENDX01:

NAME CURRENT
POSITION

-------------------- -------------------------

CREMER ANALYST
MARKUSH TRAINEE
GEE MANAGER
KUNEY DBA
NEEDHAM PROGRAMMER
JACKSON PROGRAMMER

LAST PERSON SELECTED: JACKSON

ATSTAX02 - AT START OF DATA statement

** Example 'ATSTAX02': AT START OF DATA
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 FIRST-NAME
2 NAME
2 SALARY (1)
2 CURR-CODE (1)
2 BONUS (1,1)

END-DEFINE
*

LIMIT 3
FIND EMPLOY-VIEW WITH CITY = 'MADRID'

DISPLAY NAME FIRST-NAME SALARY(1) BONUS(1,1) CURR-CODE (1)
/*
AT START OF DATA

WRITE NOTITLE *DAT4E /
END-START

END-FIND
END

Output of Program ATSTAX02:

NAME FIRST-NAME ANNUAL BONUS CURRENCY
SALARY CODE

-------------------- -------------------- ---------- ---------- --------

13/12/2004

DE JUAN JAVIER 1988000 0 PTA
DE LA MADRID ANSELMO 3120000 0 PTA
PINERO PAULA 1756000 0 PTA

WRITEX09 - WRITE statement (in combination with AT END OF DATA)

** Example 'WRITEX09': WRITE (in combination with AT END OF DATA )
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 BIRTH
2 JOB-TITLE
2 DEPT

END-DEFINE
*
READ (3) EMPLOY-VIEW BY CITY

DISPLAY NOTITLE NAME BIRTH (EM=YYYY-MM-DD) JOB-TITLE
WRITE 38T 'DEPT CODE:' DEPT
/*
AT END OF DATA

WRITE / 'LAST PERSON SELECTED:' OLD(NAME)
END-ENDDATA
SKIP 1

END-READ
END

Output of ProgramWRITEX09:

NAME DATE CURRENT
OF POSITION

BIRTH
-------------------- ---------- -------------------------

SENKO 1971-09-11 PROGRAMMER
DEPT CODE: TECH10

GODEFROY 1949-01-09 COMPTABLE
DEPT CODE: COMP02

CANALE 1942-01-01 CONSULTANT
DEPT CODE: TECH03

LAST PERSON SELECTED: CANALE

DISPLAY and WRITE Statements

The following examples are referenced in the section Statements DISPLAY and WRITE.

DISPLX13 - DISPLAY statement (compare with WRITEX08 using WRITE)

** Example 'DISPLX13': DISPLAY (compare with WRITEX08 using WRITE)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 FIRST-NAME
2 NAME
2 SALARY (2)
2 BONUS (1,1)
2 CITY

END-DEFINE
*
LIMIT 2
READ EMPLOY-VIEW WITH CITY = 'CHAPEL HILL' WHERE BONUS(1,1) NE 0

/*
DISPLAY 'PERS/ID' PERSONNEL-ID NAME / FIRST-NAME

'**' '=' SALARY(1:2) 'BONUS' BONUS(1,1) CITY (AL=15)
/*
SKIP 1

END-READ
END

Output of Program DISPLX13:

Page 1 04-12-13 14:11:28

PERS NAME ANNUAL BONUS CITY
ID FIRST-NAME SALARY

-------- -------------------- ------------- ---------- ---------------

20027000 CUMMINGS ** 41000 1500 CHAPEL HILL
PUALA 38900

20000200 WOOLSEY ** 26000 3000 CHAPEL HILL
LOUISE 24700

WRITEX08 - WRITE statement (compare with DISPLX13 using DISPLAY)

** Example 'WRITEX08': WRITE (compare with DISPLX13 using DISPLAY)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 FIRST-NAME
2 NAME
2 SALARY (2)
2 BONUS (1,1)
2 CITY

END-DEFINE
*
LIMIT 2
READ EMPLOY-VIEW WITH CITY = 'CHAPEL HILL' WHERE BONUS(1,1) NE 0

/*
WRITE 'PERS/ID' PERSONNEL-ID NAME / FIRST-NAME

'**' '=' SALARY(1:2) 'BONUS' BONUS(1,1) CITY (AL=15)
/*
SKIP 1

END-READ
END

Output of ProgramWRITEX08:

Page 1 04-12-13 14:12:43

PERS/ID 20027000 CUMMINGS
PUALA ** ANNUAL SALARY: 41000 38900 BONUS 1500
CHAPEL HILL

PERS/ID 20000200 WOOLSEY
LOUISE ** ANNUAL SALARY: 26000 24700 BONUS 3000
CHAPEL HILL

DISPLX14 - DISPLAY statement (with AL, SF and nX)

** Example 'DISPLX14': DISPLAY (with AL, SF and nX)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 FIRST-NAME
2 NAME
2 ADDRESS-LINE (1)
2 TELEPHONE

3 AREA-CODE
3 PHONE

2 CITY
END-DEFINE
*
READ (3) EMPLOY-VIEW BY NAME STARTING FROM 'W'

DISPLAY (AL=15 SF=5) NAME CITY / ADDRESS-LINE(1) 2X TELEPHONE
SKIP 1

END-READ
END

Output of Program DISPLX14:

Page 1 04-12-13 14:14:00

NAME CITY TELEPHONE
ADDRESS

AREA TELEPHONE
CODE

--------------- --------------- --------------- ---------------

WABER HEIDELBERG 06221 456452
ERBACHERSTR. 78

WADSWORTH DERBY 0332 515365
56 PINECROFT CO

WAGENBACH FRANKFURT 069 983218
BECKERSTR. 4

WRITEX09 - WRITE statement (in combination with AT END OF DATA)

** Example 'WRITEX09': WRITE (in combination with AT END OF DATA )
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 BIRTH
2 JOB-TITLE
2 DEPT

END-DEFINE
*
READ (3) EMPLOY-VIEW BY CITY

DISPLAY NOTITLE NAME BIRTH (EM=YYYY-MM-DD) JOB-TITLE
WRITE 38T 'DEPT CODE:' DEPT
/*
AT END OF DATA

WRITE / 'LAST PERSON SELECTED:' OLD(NAME)
END-ENDDATA
SKIP 1

END-READ
END

Output of ProgramWRITEX09:

NAME DATE CURRENT
OF POSITION

BIRTH
-------------------- ---------- -------------------------

SENKO 1971-09-11 PROGRAMMER
DEPT CODE: TECH10

GODEFROY 1949-01-09 COMPTABLE
DEPT CODE: COMP02

CANALE 1942-01-01 CONSULTANT
DEPT CODE: TECH03

LAST PERSON SELECTED: CANALE

DISPLAY Statement

The following example is referenced in the section Page Titles, Page Breaks, Blank Lines.

DISPLX21 DISPLAY statement (with slash '/' and compare with WRITE)

** Example 'DISPLX21': DISPLAY (usage of slash '/' in DISPLAY and WRITE)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 NAME
2 FIRST-NAME
2 ADDRESS-LINE (1)

END-DEFINE
*
WRITE TITLE LEFT JUSTIFIED UNDERLINED

*TIME
5X 'PEOPLE LIVING IN SALT LAKE CITY'
21X 'PAGE:' *PAGE-NUMBER /
15X 'AS OF' *DAT4E //

*
WRITE TRAILER UNDERLINED 'REGISTER OF' / 'SALT LAKE CITY'
*
READ (2) EMPLOY-VIEW WITH CITY = 'SALT LAKE CITY'

DISPLAY NAME /
FIRST-NAME
'HOME/CITY' CITY
'STREET/OR BOX NO.' ADDRESS-LINE (1)

SKIP 1
END-READ
END

Output of Program DISPLX21:

14:15:50.1 PEOPLE LIVING IN SALT LAKE CITY PAGE: 1
AS OF 13/12/2004

-------------------------------------------------------------------------------
NAME HOME STREET

FIRST-NAME CITY OR BOX NO.
-------------------- -------------------- --------------------

ANDERSON SALT LAKE CITY 3701 S. GEORGE MASON
JENNY

SAMUELSON SALT LAKE CITY 7610 W. 86TH STREET
MARTIN

REGISTER OF
SALT LAKE CITY

-------------------------------------------------------------------------------

Column Headers

The following example is referenced in the section Column Headers.

DISPLX15 - DISPLAY statement (with FC, UC)

** Example 'DISPLX15': DISPLAY (with FC, UC)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 FIRST-NAME
2 NAME
2 ADDRESS-LINE (1)
2 CITY
2 TELEPHONE

3 AREA-CODE
3 PHONE

END-DEFINE
*
FORMAT AL=12 GC== UC=%
*
READ (3) EMPLOY-VIEW BY NAME STARTING FROM 'R'

DISPLAY NOTITLE (FC=*)
NAME FIRST-NAME CITY (FC=- UC=-) /
ADDRESS-LINE(1) TELEPHONE

SKIP 1
END-READ
END

Output of Program DISPLX15:

****NAME**** *FIRST-NAME* ----CITY---- ========TELEPHONE========
**ADDRESS***

****AREA**** *TELEPHONE**
****CODE****

%%%%%%%%%%%% %%%%%%%%%%%% ------------ %%%%%%%%%%%% %%%%%%%%%%%%

RACKMANN MARIAN FRANKFURT 069 375849
FINKENSTR. 1

RAMAMOORTHY TY SEPULVEDA 209 175-1885
12018 BROOKS

RAMAMOORTHY TIMMIE SEATTLE 206 151-4673
921-178TH PL

DISPLX16 - DISPLAY statement (with '/', 'text', 'text/text')

** Example 'DISPLX16': DISPLAY (with '/', 'text', 'text/text')
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 FIRST-NAME
2 NAME
2 ADDRESS-LINE (1)
2 CITY
2 TELEPHONE

3 AREA-CODE
3 PHONE

END-DEFINE
*
READ (5) EMPLOY-VIEW BY NAME STARTING FROM 'E'

DISPLAY NOTITLE
'/' NAME (AL=12) /* suppressed header
'FIRST/NAME' FIRST-NAME (AL=10) /* two-line user-defined header
'ADDRESS' CITY / /* user-defined header
' ' ADDRESS-LINE(1) /* 'blank' header

TELEPHONE (HC=L) /* default header
SKIP 1

END-READ
END

Output of Program DISPLX16:

FIRST ADDRESS TELEPHONE
NAME

AREA TELEPHONE
CODE

---------- -------------------- ------ ---------------

EAVES TREVOR DERBY 0332 657623
17 HARTON ROAD

ECKERT KARL OBERRAMSTADT 06154 99722
FORSTWEG 22

ECKHARDT RICHARD DARMSTADT
BRESLAUERPL. 4

EDMUNDSON LES TULSA 918 945-4916
2415 ALSOP CT.

EGGERT HERMANN STUTTGART 0711 981237
RABENGASSE 8

Field-Output-Relevant Parameters

The following examples are referenced in the sectionParameters to Influence theOutput of Fields.

They are provided to demonstrate the use of the parameters LC, IC, TC, AL, NL, IS, ZP and ES, and
the SUSPEND IDENTICAL SUPPRESS statement:

DISPLX17 - DISPLAY statement (with NL, AL, IC, LC, TC)

** Example 'DISPLX17': DISPLAY (with NL, AL, IC, LC, TC)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 FIRST-NAME
2 NAME
2 SALARY (1)
2 BONUS (1,1)

END-DEFINE
*
READ (3) EMPLOY-VIEW BY NAME STARTING FROM 'JONES'

DISPLAY NOTITLE (IS=ON NL=15)
NAME

'-' '=' FIRST-NAME (AL=12)
'ANNUAL SALARY' SALARY(1) (LC=USD TC=.00) /
'+ BONUSES' BONUS(1,1) (IC='+ ' TC=.00)

SKIP 1
END-READ
END

Output of Program DISPLX17:

NAME FIRST-NAME ANNUAL SALARY
+ BONUSES

-------------------- -------------- ----------------------

JONES - VIRGINIA USD 46000.00
+ 9000.00

- MARSHA USD 50000.00

+ 0.00

- ROBERT USD 31000.00
+ 0.00

DISPLX18 - DISPLAY statement (using default settings for SF, AL, UC, LC, IC, TC and compare
with DISPLX19)

** Example 'DISPLX18': DISPLAY (using default settings for SF, AL, UC,
** LC, IC, TC and compare with DISPLX19)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY
2 SALARY (1)
2 BONUS (1,1)

END-DEFINE
*
FIND (6) EMPLOY-VIEW WITH CITY = 'CHAPEL HILL'

DISPLAY NAME FIRST-NAME SALARY(1) BONUS(1,1)
END-FIND
END

Output of Program DISPLX18:

Page 1 04-12-13 14:20:48

NAME FIRST-NAME ANNUAL BONUS
SALARY

-------------------- -------------------- ---------- ----------

KESSLER CLARE 41000 0
ADKINSON DAVID 24000 0
GEE TOMMIE 39500 0
HERZOG JOHN 31500 0
QUILLION TIMOTHY 30500 0
CUMMINGS PUALA 41000 1500

DISPLX19 - DISPLAY statement (with SF, AL, LC, IC, TC and compare with DISPLX18)

** Example 'DISPLX19': DISPLAY (with SF, AL, LC, IC, TC and compare
** with DISPLX19)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 FIRST-NAME
2 CITY
2 SALARY (1)
2 BONUS (1,1)

END-DEFINE
*
FORMAT SF=3 AL=15 UC==
*
FIND (6) EMPLOY-VIEW WITH CITY = 'CHAPEL HILL'

DISPLAY (NL=10)
NAME
FIRST-NAME (LC='- ' UC=-)
SALARY (1) (LC=USD)
BONUS (1,1) (IC='*** ' TC=' ***')

END-FIND
END

Output of Program DISPLX19:

Page 1 04-12-13 14:21:57

NAME FIRST-NAME ANNUAL BONUS
SALARY

=============== ----------------- ============== ===================

KESSLER - CLARE USD 41000 *** 0 ***
ADKINSON - DAVID USD 24000 *** 0 ***
GEE - TOMMIE USD 39500 *** 0 ***
HERZOG - JOHN USD 31500 *** 0 ***
QUILLION - TIMOTHY USD 30500 *** 0 ***
CUMMINGS - PUALA USD 41000 *** 1500 ***

SUSPEX01 - SUSPEND IDENTICALSUPPRESS statement (in conjunctionwith parameters IS,
ES, ZP in DISPLAY)

** Example 'SUSPEX01': SUSPEND IDENTICAL SUPPRESS (in conjunction with
** parameters IS, ES, ZP in DISPLAY)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 FIRST-NAME
2 NAME
2 CITY

1 VEH-VIEW VIEW OF VEHICLES
2 PERSONNEL-ID
2 MAKE

END-DEFINE
*
LIMIT 15
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'

SUSPEND IDENTICAL SUPPRESS
FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)

IF NO RECORDS FOUND
MOVE '******' TO MAKE

END-NOREC
DISPLAY NOTITLE (ES=OFF IS=ON ZP=ON AL=15)

NAME (RD.)
FIRST-NAME (RD.)
MAKE (FD.) (IS=OFF)

END-FIND
END-READ
END

Output of Program SUSPEX01:

NAME FIRST-NAME MAKE
--------------- --------------- ---------------

JONES VIRGINIA CHRYSLER
JONES MARSHA CHRYSLER

CHRYSLER
JONES ROBERT GENERAL MOTORS
JONES LILLY FORD

MG
JONES EDWARD GENERAL MOTORS
JONES MARTHA GENERAL MOTORS
JONES LAUREL GENERAL MOTORS
JONES KEVIN DATSUN
JONES GREGORY FORD
JOPER MANFRED ******

JOUSSELIN DANIEL RENAULT
JUBE GABRIEL ******
JUNG ERNST ******
JUNKIN JEREMY ******
KAISER REINER ******

SUSPEX02 - SUSPEND IDENTICALSUPPRESS statement (in conjunctionwith parameters IS,
ES, ZP in DISPLAY) Identical to SUSPEX01, but with IS=OFF.

** Example 'SUSPEX02': SUSPEND IDENTICAL SUPPRESS (in conjunction with
** parameters IS, ES, ZP in DISPLAY)
** Identical to SUSPEX01, but with IS=OFF.
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 FIRST-NAME
2 NAME
2 CITY

1 VEH-VIEW VIEW OF VEHICLES
2 PERSONNEL-ID
2 MAKE

END-DEFINE
*
LIMIT 15
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'

SUSPEND IDENTICAL SUPPRESS
FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)

IF NO RECORDS FOUND
MOVE '******' TO MAKE

END-NOREC
DISPLAY NOTITLE (ES=OFF IS=OFF ZP=ON AL=15)

NAME (RD.)
FIRST-NAME (RD.)
MAKE (FD.) (IS=OFF)

END-FIND
END-READ
END

Output of Program SUSPEX02:

NAME FIRST-NAME MAKE
--------------- --------------- ---------------

JONES VIRGINIA CHRYSLER
JONES MARSHA CHRYSLER
JONES MARSHA CHRYSLER
JONES ROBERT GENERAL MOTORS
JONES LILLY FORD

JONES LILLY MG
JONES EDWARD GENERAL MOTORS
JONES MARTHA GENERAL MOTORS
JONES LAUREL GENERAL MOTORS
JONES KEVIN DATSUN
JONES GREGORY FORD
JOPER MANFRED ******
JOUSSELIN DANIEL RENAULT
JUBE GABRIEL ******
JUNG ERNST ******
JUNKIN JEREMY ******
KAISER REINER ******

COMPRX03 - COMPRESS statement

** Example 'COMPRX03': COMPRESS (using parameters LC and TC)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 SALARY (1)
2 CURR-CODE (1)
2 LEAVE-DUE
2 NAME
2 FIRST-NAME
2 JOB-TITLE

*
1 #SALARY (N9)
1 #FULL-SALARY (A25)
1 #VACATION (A11)
END-DEFINE
*
READ (3) EMPLOY-VIEW WITH CITY = 'BOSTON'

MOVE SALARY(1) TO #SALARY
COMPRESS 'SALARY :' CURR-CODE(1) #SALARY INTO #FULL-SALARY
COMPRESS 'VACATION:' LEAVE-DUE INTO #VACATION
/*
DISPLAY NOTITLE NAME FIRST-NAME

'JOB DESCRIPTION' JOB-TITLE (LC='JOB : ') /
'/' #FULL-SALARY /
'/' #VACATION (TC='DAYS')

SKIP 1
END-READ
END

Output of Program COMPRX03:

NAME FIRST-NAME JOB DESCRIPTION
-------------------- -------------------- -----------------------------------

SHAW LESLIE JOB : SECRETARY
SALARY : USD 18000
VACATION: 2DAYS

STANWOOD VERNON JOB : PROGRAMMER
SALARY : USD 31000
VACATION: 1DAYS

CREMER WALT JOB : SECRETARY
SALARY : USD 20000
VACATION: 3DAYS

Edit Masks

The following examples are referenced in the section Edit Masks - EM Parameter.

EDITMX03 - Edit mask (different EM for alpha-numeric fields)

** Example 'EDITMX03': Edit mask (different EM for alpha-numeric fields)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 FIRST-NAME
2 NAME
2 CITY
2 SALARY(1)

END-DEFINE
*
LIMIT 3
READ EMPLOY-VIEW BY PERSONNEL-ID FROM '20018000'

WHERE SALARY(1) = 28000 THRU 30000
DISPLAY 'N A M E' NAME (EM=X^^X^^X^^X^^X^^X^^X^^X^^X^^X^^X) /

'NAME HEX' NAME (EM=H^H^H^H^H^H^H^H^H^H^H)
FIRST-NAME (EM=' - 'X(15)*)
CITY (EM=X..X(10))

SKIP 1
END-READ
END

Output of Program EDITMX03:

Page 1 04-12-13 14:26:57

N A M E FIRST-NAME CITY
NAME HEX

-------------------------------- ------------------ -------------

L O R I E - JEAN-PAUL * C..LEVELAND
D3 D6 D9 C9 C5 40 40 40 40 40 40

H A L L - ARTHUR * A..NN ARBER
C8 C1 D3 D3 40 40 40 40 40 40 40

V A S W A N I - TOMMIE * M..ONTERREY
E5 C1 E2 E6 C1 D5 C9 40 40 40 40

EDITMX04 - Edit mask (different EM for numeric fields)

** Example 'EDITMX04': Edit mask (different EM for numeric fields)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 FIRST-NAME
2 NAME
2 SALARY (1)
2 BONUS (1,1)
2 LEAVE-DUE

END-DEFINE
*
LIMIT 2
READ EMPLOY-VIEW BY PERSONNEL-ID = '20018000'

WHERE SALARY(1) = 28000 THRU 30000
DISPLAY (SF=4)

'N A M E' NAME
'SALARY' SALARY(1) (EM=*USD^ZZZ,999)
'BONUS (ZZ)' BONUS(1,1) (EM=S*ZZZ,999) /
'BONUS (Z9)' BONUS(1,1) (EM=SZ99,999+) /
'->' '=' BONUS(1,1) (EM=-999,999)
'VAC/DUE' LEAVE-DUE (EM=+999)

SKIP 1
END-READ
END

Output of Program EDITMX04:

Page 1 04-12-13 14:27:43

N A M E SALARY BONUS (ZZ) VAC
BONUS (Z9) DUE

BONUS
-------------------- ----------- ----------- ---

LORIE USD *28,000 +**4,000 +13
+ 04,000+
-> 004,000

HALL USD *30,000 +**5,000 +14
+ 05,000+
-> 005,000

EDITMX05 - Edit mask (EM for date and time system variables)

** Example 'EDITMX05': Edit mask (EM for date and time system variables)
************************************************************************
WRITE NOTITLE //

'DATE INTERNAL :' *DATX (DF=L) /
' :' *DATX (EM=N(9)' 'ZW.'WEEK 'YYYY) /
' :' *DATX (EM=ZZJ'.DAY 'YYYY) /
' ROMAN :' *DATX (EM=R) /
' AMERICAN :' *DATX (EM=MM/DD/YYYY) 12X 'OR ' *DAT4U /
' JULIAN :' *DATX (EM=YYYYJJJ) 15X 'OR ' *DAT4J /
' GREGORIAN:' *DATX (EM=ZD.''L(10)''YYYY) 5X 'OR ' *DATG ///
'TIME INTERNAL :' *TIMX 14X 'OR ' *TIME /
' :' *TIMX (EM=HH.II.SS.T) /
' :' *TIMX (EM=HH.II.SS' 'AP) /
' :' *TIMX (EM=HH)

END

Output of Program EDITMX05:

DATE INTERNAL : 2004-12-13
: Monday 51.WEEK 2004
: 348.DAY 2004

ROMAN : MMIV
AMERICAN : 12/13/2004 OR 12/13/2004
JULIAN : 2004348 OR 2004348
GREGORIAN: 13.December2004 OR 13December 2004

TIME INTERNAL : 14:28:49 OR 14:28:49.1
: 14.28.49.1

583Programming Guide

Referenced Example Programs



: 02.28.49 PM
: 14

DISPLAY VERT with WRITE Statement

WRITEX10 - WRITE statement (with nT, T*field and P*field)

** Example 'WRITEX10': WRITE (with nT, T*field and P*field)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 JOB-TITLE
2 NAME
2 SALARY (1)
2 BONUS (1,1)

END-DEFINE
*
READ (3) EMPLOY-VIEW WITH JOB-TITLE FROM 'SALES PERSON'

DISPLAY NOTITLE NAME 30T JOB-TITLE
VERT AS 'SALARY/BONUS' SALARY(1) BONUS(1,1)

AT BREAK OF JOB-TITLE
WRITE 20T 'AVERAGE' T*JOB-TITLE OLD(JOB-TITLE) (AL=15)

'(SAL)' P*SALARY AVER(SALARY(1)) /
46T '(BON)' P*BONUS AVER(BONUS(1,1)) /

END-BREAK
SKIP 1

END-READ
END

Output of ProgramWRITEX10:

NAME CURRENT SALARY
POSITION BONUS

-------------------- ------------------------- ----------

SAMUELSON SALES PERSON 32000
6000

PAPAYANOPOULOS SALES PERSON 34000
7000

HELL SALES PERSON 38000
9000

AVERAGE SALES PERSON (SAL) 34666
(BON) 7333

AT BREAK Statement

The following example is referenced in the section Control Breaks.

ATBREX06 -ATBREAKOFstatement (comparingNMIN,NAVER,NCOUNTwithMIN,AVER,
COUNT)

** Example 'ATBREX06': AT BREAK OF (comparing NMIN, NAVER, NCOUNT with
** MIN, AVER, COUNT)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 SALARY (1:2)

END-DEFINE
*
WRITE TITLE '-- SALARY STATISTICS BY CITY --' /
*
READ (2) EMPLOY-VIEW WITH CITY = 'NEW YORK'

DISPLAY CITY 'SALARY (1)' SALARY(1) 15X 'SALARY (2)' SALARY(2)
AT BREAK OF CITY

WRITE /
14T 'S A L A R Y (1)' 39T 'S A L A R Y (2)' /
13T '- MIN:' MIN(SALARY(1)) 38T '- MIN:' MIN(SALARY(2)) /
13T '- AVER:' AVER(SALARY(1)) 38T '- AVER:' AVER(SALARY(2)) /
16T COUNT(SALARY(1)) 'RECORDS' 41T COUNT(SALARY(2)) 'RECORDS' //
13T '- NMIN:' NMIN(SALARY(1)) 38T '- NMIN:' NMIN(SALARY(2)) /
13T '- NAVER:' NAVER(SALARY(1)) 38T '- NAVER:' NAVER(SALARY(2)) /
16T NCOUNT(SALARY(1)) 'RECORDS' 41T NCOUNT(SALARY(2)) 'RECORDS'

END-BREAK
END-READ
END

Output of Program ATBREX06:

-- SALARY STATISTICS BY CITY --

CITY SALARY (1) SALARY (2)
-------------------- ---------- ----------

NEW YORK 17000 16100
NEW YORK 38000 34900

S A L A R Y (1) S A L A R Y (2)
- MIN: 17000 - MIN: 16100
- AVER: 27500 - AVER: 25500

2 RECORDS 2 RECORDS

- NMIN: 17000 - NMIN: 16100
- NAVER: 27500 - NAVER: 25500

2 RECORDS 2 RECORDS

COMPUTE, MOVE and COMPRESS Statements

The following examples are referenced in the section Data Computation.

WRITEX11 - WRITE statement (with nX, n/n and COMPRESS)

** Example 'WRITEX11': WRITE (with nX, n/n and COMPRESS)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 SALARY (1)
2 FIRST-NAME
2 NAME
2 CITY
2 ZIP
2 CURR-CODE (1)
2 JOB-TITLE
2 LEAVE-DUE
2 ADDRESS-LINE (1)

*
1 #SALARY (A8)
1 #FULL-NAME (A25)
1 #FULL-CITY (A25)
1 #FULL-SALARY (A25)
1 #VACATION (A16)
END-DEFINE
*
READ (3) EMPLOY-VIEW LOGICAL BY PERSONNEL-ID = '2001800'

MOVE SALARY(1) TO #SALARY
COMPRESS FIRST-NAME NAME INTO #FULL-NAME
COMPRESS ZIP CITY INTO #FULL-CITY
COMPRESS 'SALARY :' CURR-CODE(1) #SALARY INTO #FULL-SALARY
COMPRESS 'VACATION:' LEAVE-DUE 'DAYS' INTO #VACATION
/*
DISPLAY NOTITLE 'NAME AND ADDRESS' NAME

5X 'PERS-NO.' PERSONNEL-ID
3X 'JOB TITLE' JOB-TITLE (LC='JOB : ')

WRITE 1/5 #FULL-NAME 1/37 #FULL-SALARY
2/5 ADDRESS-LINE(1) 2/37 #VACATION
3/5 #FULL-CITY

SKIP 1

END-READ
END

Output of ProgramWRITEX11:

NAME AND ADDRESS PERS-NO. JOB TITLE
-------------------- -------- -----------------------------------

FARRIS 20018000 JOB : PROGRAMMER
JACKIE FARRIS SALARY : USD 30500
918 ELM STREET VACATION: 10 DAY
32306 TALLAHASSEE

EVANS 20018100 JOB : PROGRAMMER
JO EVANS SALARY : USD 31000
1058 REDSTONE LANE VACATION: 11 DAY
68508 LINCOLN

HERZOG 20018200 JOB : PROGRAMMER
JOHN HERZOG SALARY : USD 31500
255 ZANG STREET #253 VACATION: 12 DAY
27514 CHAPEL HILL

IFX03 - IF statement

** Example 'IFX03': IF
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 NAME
2 CITY
2 BONUS (1,1)
2 SALARY (1)

*
1 #INCOME (N9)
1 #TEXT (A26)
END-DEFINE
*
WRITE TITLE '-- DISTRIBUTION OF CATALOGS I AND II --' /
*
READ (3) EMPLOY-VIEW BY CITY = 'SAN FRANSISCO'

COMPUTE #INCOME = BONUS(1,1) + SALARY(1)
/*
IF #INCOME > 40000

MOVE 'CATALOGS I AND II' TO #TEXT
ELSE

MOVE 'CATALOG I' TO #TEXT
END-IF
/*

DISPLAY NAME 5X 'SALARY' SALARY(1) / BONUS(1,1)
WRITE T*SALARY '-'(10) /

16X 'INCOME:' T*SALARY #INCOME 3X #TEXT /
16X '='(19)

SKIP 1
END-READ
END

Output of Program IFX03:

-- DISTRIBUTION OF CATALOGS I AND II --

NAME SALARY
BONUS

-------------------- ----------

COLVILLE JR 56000
0

----------
INCOME: 56000 CATALOGS I AND II
===================

RICHMOND 9150
0

----------
INCOME: 9150 CATALOG I
===================

MONKTON 13500
600

----------
INCOME: 14100 CATALOG I
===================

COMPRX03 - COMPRESS statement (using parameters LC and TC)

** Example 'COMPRX03': COMPRESS (using parameters LC and TC)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 SALARY (1)
2 CURR-CODE (1)
2 LEAVE-DUE
2 NAME
2 FIRST-NAME
2 JOB-TITLE

*
1 #SALARY (N9)

1 #FULL-SALARY (A25)
1 #VACATION (A11)
END-DEFINE
*
READ (3) EMPLOY-VIEW WITH CITY = 'BOSTON'

MOVE SALARY(1) TO #SALARY
COMPRESS 'SALARY :' CURR-CODE(1) #SALARY INTO #FULL-SALARY
COMPRESS 'VACATION:' LEAVE-DUE INTO #VACATION
/*
DISPLAY NOTITLE NAME FIRST-NAME

'JOB DESCRIPTION' JOB-TITLE (LC='JOB : ') /
'/' #FULL-SALARY /
'/' #VACATION (TC='DAYS')

SKIP 1
END-READ
END

Output of Program COMPRX03:

NAME FIRST-NAME JOB DESCRIPTION
-------------------- -------------------- -----------------------------------

SHAW LESLIE JOB : SECRETARY
SALARY : USD 18000
VACATION: 2DAYS

STANWOOD VERNON JOB : PROGRAMMER
SALARY : USD 31000
VACATION: 1DAYS

CREMER WALT JOB : SECRETARY
SALARY : USD 20000
VACATION: 3DAYS

System Variables

The following examples are referenced in the section System Variables and System Functions.

EDITMX05 - Edit mask (EM for date and time system variables)

** Example 'EDITMX05': Edit mask (EM for date and time system variables)
************************************************************************
WRITE NOTITLE //

'DATE INTERNAL :' *DATX (DF=L) /
' :' *DATX (EM=N(9)' 'ZW.'WEEK 'YYYY) /
' :' *DATX (EM=ZZJ'.DAY 'YYYY) /
' ROMAN :' *DATX (EM=R) /
' AMERICAN :' *DATX (EM=MM/DD/YYYY) 12X 'OR ' *DAT4U /
' JULIAN :' *DATX (EM=YYYYJJJ) 15X 'OR ' *DAT4J /
' GREGORIAN:' *DATX (EM=ZD.''L(10)''YYYY) 5X 'OR ' *DATG ///
'TIME INTERNAL :' *TIMX 14X 'OR ' *TIME /
' :' *TIMX (EM=HH.II.SS.T) /
' :' *TIMX (EM=HH.II.SS' 'AP) /
' :' *TIMX (EM=HH)

END

Output of Program EDITMX05:

DATE INTERNAL : 2004-12-13
: Monday 51.WEEK 2004
: 348.DAY 2004

ROMAN : MMIV
AMERICAN : 12/13/2004 OR 12/13/2004
JULIAN : 2004348 OR 2004348
GREGORIAN: 13.December2004 OR 13December 2004

TIME INTERNAL : 14:36:58 OR 14:36:58.8
: 14.36.58.8
: 02.36.58 PM
: 14

READX04 - READ statement (in combination with FIND and the system variables *NUMBER
and *COUNTER)

** Example 'READX04': READ (in combination with FIND and the system
** variables *NUMBER and *COUNTER)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 FIRST-NAME

1 VEHIC-VIEW VIEW OF VEHICLES

2 PERSONNEL-ID
2 MAKE

END-DEFINE
*
LIMIT 10
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'

FD. FIND VEHIC-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
IF NO RECORDS FOUND
ENTER

END-NOREC
/*
DISPLAY NOTITLE

*COUNTER (RD.)(NL=8) NAME (AL=15) FIRST-NAME (AL=10)
*NUMBER (FD.)(NL=8) *COUNTER (FD.)(NL=8) MAKE

END-FIND
END-READ
END

Output of Program READX04:

CNT NAME FIRST-NAME NMBR CNT MAKE
--------- --------------- ---------- --------- --------- --------------------

1 JONES VIRGINIA 1 1 CHRYSLER
2 JONES MARSHA 2 1 CHRYSLER
2 JONES MARSHA 2 2 CHRYSLER
3 JONES ROBERT 1 1 GENERAL MOTORS
4 JONES LILLY 2 1 FORD
4 JONES LILLY 2 2 MG
5 JONES EDWARD 1 1 GENERAL MOTORS
6 JONES MARTHA 1 1 GENERAL MOTORS
7 JONES LAUREL 1 1 GENERAL MOTORS
8 JONES KEVIN 1 1 DATSUN
9 JONES GREGORY 1 1 FORD

10 JOPER MANFRED 0 0

WTITLX01 - WRITE TITLE statement (with *PAGE-NUMBER)

** Example 'WTITLX01': WRITE TITLE (with *PAGE-NUMBER)
************************************************************************
DEFINE DATA LOCAL
1 VEHIC-VIEW VIEW OF VEHICLES

2 MAKE
2 YEAR
2 MAINT-COST (1)

END-DEFINE
*
LIMIT 5
*

READ VEHIC-VIEW
END-ALL
SORT BY YEAR USING MAKE MAINT-COST (1)

DISPLAY NOTITLE YEAR MAKE MAINT-COST (1)
AT BREAK OF YEAR

MOVE 1 TO *PAGE-NUMBER
NEWPAGE

END-BREAK
/*
WRITE TITLE LEFT JUSTIFIED

'YEAR:' YEAR 15X 'PAGE' *PAGE-NUMBER
END-SORT
END

Output of ProgramWTITLX01:

YEAR: 1980 PAGE 1
YEAR MAKE MAINT-COST
----- -------------------- ----------

1980 RENAULT 20000
1980 RENAULT 20000
1980 PEUGEOT 20000

System Functions

The following examples are referenced in the section System Variables and System Functions.

ATBREX06 -ATBREAKOFstatement (comparingNMIN,NAVER,NCOUNTwithMIN,AVER,
COUNT)

** Example 'ATBREX06': AT BREAK OF (comparing NMIN, NAVER, NCOUNT with
** MIN, AVER, COUNT)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 CITY
2 SALARY (1:2)

END-DEFINE
*
WRITE TITLE '-- SALARY STATISTICS BY CITY --' /
*
READ (2) EMPLOY-VIEW WITH CITY = 'NEW YORK'

DISPLAY CITY 'SALARY (1)' SALARY(1) 15X 'SALARY (2)' SALARY(2)
AT BREAK OF CITY

WRITE /

14T 'S A L A R Y (1)' 39T 'S A L A R Y (2)' /
13T '- MIN:' MIN(SALARY(1)) 38T '- MIN:' MIN(SALARY(2)) /
13T '- AVER:' AVER(SALARY(1)) 38T '- AVER:' AVER(SALARY(2)) /
16T COUNT(SALARY(1)) 'RECORDS' 41T COUNT(SALARY(2)) 'RECORDS' //
13T '- NMIN:' NMIN(SALARY(1)) 38T '- NMIN:' NMIN(SALARY(2)) /
13T '- NAVER:' NAVER(SALARY(1)) 38T '- NAVER:' NAVER(SALARY(2)) /
16T NCOUNT(SALARY(1)) 'RECORDS' 41T NCOUNT(SALARY(2)) 'RECORDS'

END-BREAK
END-READ
END

Output of Program ATBREX06:

-- SALARY STATISTICS BY CITY --

CITY SALARY (1) SALARY (2)
-------------------- ---------- ----------

NEW YORK 17000 16100
NEW YORK 38000 34900

S A L A R Y (1) S A L A R Y (2)
- MIN: 17000 - MIN: 16100
- AVER: 27500 - AVER: 25500

2 RECORDS 2 RECORDS

- NMIN: 17000 - NMIN: 16100
- NAVER: 27500 - NAVER: 25500

2 RECORDS 2 RECORDS

ATENPX01 - AT ENDOF PAGE statement (with system function available via GIVE SYSTEM
FUNCTIONS in DISPLAY)

** Example 'ATENPX01': AT END OF PAGE (with system function available
** via GIVE SYSTEM FUNCTIONS in DISPLAY)
************************************************************************
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES

2 PERSONNEL-ID
2 NAME
2 JOB-TITLE
2 SALARY (1)

END-DEFINE
*
READ (10) EMPLOY-VIEW BY PERSONNEL-ID = '20017000'

DISPLAY NOTITLE GIVE SYSTEM FUNCTIONS
NAME JOB-TITLE 'SALARY' SALARY(1)

/*
AT END OF PAGE

WRITE / 24T 'AVERAGE SALARY: ...' AVER(SALARY(1))
END-ENDPAGE

END-READ
END

Output of Program ATENPX01:

NAME CURRENT SALARY
POSITION

-------------------- ------------------------- ----------

CREMER ANALYST 34000
MARKUSH TRAINEE 22000
GEE MANAGER 39500
KUNEY DBA 40200
NEEDHAM PROGRAMMER 32500
JACKSON PROGRAMMER 33000
PIETSCH SECRETARY 22000
PAUL SECRETARY 23000
HERZOG MANAGER 48500
DEKKER DBA 48000

AVERAGE SALARY: ... 34270

